Chapter 1 PRECALCULUS REVIEW

1.1 Real Numbers, Functions, and Graphs

1.1.1 Independent variables, dependent variables, and constants

When you use MAPLE, a name usually needs to be assigned to each variable or constant for reference. A name of a variable can be a letter followed by a zero or by more letters, digits, and underscores, with lowercase and uppercase letters distinct. Any name (a character or a string) is assumed to be an independent variable before it is assigned to an expression or a number. It then can be assigned to a constant, or it can be assigned as an expression of other variables, becoming a dependent variable. For example, in the following assignment

```
> y:=x^2+1;
```

$$y := x^2 + 1 \qquad (1.1.1)$$

y is a dependent variable, while x is an independent variable. A name can also stand for a constant. For example, we can assign y to the constant π:

```
> y:=Pi;
```

$$y := \pi \qquad (1.1.2)$$

You can get the numerical value of π:

```
> evalf(Pi);
```

$$3.141592654 \qquad (1.1.3)$$

Since y now is assigned to the constant π, you can get the same result using the syntax

```
> evalf(y);
```

$$3.141592654 \qquad (1.1.4)$$

To release y so that it returns to an independent variable, we can use the following:

```
> y:='y';
```

$$y := y \qquad (1.1.5)$$

Please note the difference between ":=" and "=". The former means "is assigned to", the latter is only a symbol here.

1.1.2 The ditto operators

The ditto operator is a way of referring to previously computed results in Maple. For example, percentage signs are used to refer to previously computed expressions. Specifically, the following operators are defined:

```
%   last expression
%%  second last expression
%%% third last expression
```

For example,

```
> 2*4;
```

$$8 \quad (1.2.1)$$

```
> %+2;
```

$$10 \quad (1.2.2)$$

```
> %;
```

$$10 \quad (1.2.3)$$

```
> %%%;
```

$$8 \quad (1.2.4)$$

1.1.3 Display mode and hidden mode in MAPLE

When you make an assignment or write a command, you can choose to display the result or hide the result: ":" is used in hidden mode while ";" is in display mode. For example,

```
> y:=x^2+1:
```

is in the hidden mode. If you want to see the assignment for y, use the display mode:

```
> y;
```

$$x^2+1 \quad (1.3.1)$$

1.1.4 Functions

A function can be defined as a dependent variable (i.e., an expression of an independent variable), or as a rule to produce an output (an expression) from an input (an independent variable). To define a function as a dependent variable, use the following syntax:

Example 1. Define the function $y=x^2+1$ (as a dependent variable) and evaluate it at $x=2$ and $x=a^2+2a$.

```
> y:=x^2+1;
```

$$y:=x^2+1 \quad (1.4.1)$$

To evaluate an expression (i.e., a dependent variable), you can use one of these two ways:

```
> eval(y,x=2);
```

$$5 \quad (1.4.2)$$

Note that here we use **x=2**, not **x:=2**. Another way to evaluate the function is the following:

```
> subs(x=2,y);
```

$$5 \quad (1.4.3)$$

Similarly, to evaluate $y=x^2+1$ at $x=a^2+2a$, you can do one of the followings:

```
> eval(y,x=a^2+2*a);
```

$$\left(a^2+2\,a\right)^2+1 \quad (1.4.4)$$

```
> subs(x=a^2+2*a,y);
```

$$\left(a^2+2\,a\right)^2+1 \quad (1.4.5)$$

Example 2. Define the function $y = x^2 + 1$ (as a mapping from input to output) and evaluate it at $x = 2$ and $x = a^2 + 2a$.

```
> y:=x->x^2+1;
```

$$y := x \to x^2 + 1 \quad (1.4.6)$$

```
> y(2);
```

$$5 \quad (1.4.7)$$

```
> y(a^2+2*a);
```

$$\left(a^2 + 2\,a\right)^2 + 1 \quad (1.4.8)$$

This way is very similar to the notation in the textbook.

1.1.5 Graphs of functions

Example 3. Sketch the graph of $f(x) = x^3 - 2\,x$ and find its roots.

```
> f:=x->x^3-2*x:
> plot(f,-2..2);
```



The another way to sketch the graph is the following:

```
> y:=x^3-2*x;
```

$$y := x^3 - 2\,x \quad (1.5.1)$$

```
> plot(y,x=-2..2);
```



To find its roots, we use the following:

```
> solve(y,x);
```

$$0, \sqrt{2}, -\sqrt{2} \tag{1.5.2}$$

To sketch a graph of a function, we can add options to change the display of the graph. The popular options include "viewing range", "title", and "color", which are shown as the following:

```
> plot(x^3-2*x,x=-2..2, -3..3, title="Graph of x^3-2*x", color=
  green);
```



To find more options of plot, please search for "plot" from the Help menu".

Example 4. Sketch the graph of $f(x) = \dfrac{1}{x^2+1}$.

```
> plot(1/(x^2+1),x=-3..3, title="Graph of 1/(x^2+1)",
  xtickmarks=13);
```



Example 5. Sketch the graphs of $f(x)=\sin(\pi x)$ and its two dilates $f(3x)$ and $3f(x)$.

```
> f:=x->sin(Pi*x);
```

$$f:=x\rightarrow\sin(\pi x) \tag{1.5.3}$$

```
> plot({f(x),f(3*x),3*f(x)},x=0..4);
```



1.1.6 Solve linear systems and function equations

Example 6. Solve the linear system $2x+y=3$ and $3x-y=2$.

```
> x:='x': y:='y':
> eq1:=2*x+y=3; eq2:=3*x-y=2;
```

$$eq1:=2x+y=3 \tag{1.6.1}$$

$$eq2:=3x-y=2$$

```
> solve({eq1,eq2},{x,y});
```

$$\{y=1, x=1\} \tag{1.6.2}$$

Example 7. Solve the linear system $ax+by=n$ and $cx+dy=m$.

```
> eq1:=a*x+b*y=n; eq2:=c*x+d*y=m;
```

$$eq1:=a\,x+b\,y=n \tag{1.6.3}$$

$$eq2:=c\,x+d\,y=m$$

```
> solve({eq1,eq2},{x,y});
```

$$\left\{x=\frac{-b\,m+d\,n}{-b\,c+d\,a}, y=-\frac{-a\,m+n\,c}{-b\,c+d\,a}\right\} \tag{1.6.4}$$

Example 8. Solve the equation $x^2+y^2=1$ for y.

```
> eq:=x^2+y^2=1;
```

$$eq:=x^2+y^2=1 \tag{1.6.5}$$

```
> solve(eq,y);
```

$$\sqrt{-x^2+1}, -\sqrt{-x^2+1} \tag{1.6.6}$$

Exercises

1. Evaluate the function $y=\sqrt{x+1}$ at $x=2$ and $x=a^2+2a$.
2. Sketch the graph of $f(x)=x^2-2x-3$ and find its roots.
3. Sketch the graph of $f(x)=\dfrac{x+1}{\sqrt{x^2+1}}$.
4. Sketch the graphs of $f(x)=\cos(\pi x)$ and its two dilates $f(2x)$ and $-f(x)$.
5. Solve the linear system $3x+2y=8$ and $x-2y=4$.
6. Solve the equation $2x^2+xy=3$ for y .

1.2 Linear and Quadratic Functions

Example 1. Find the quadratic formula for the quadratic function $f(x)=ax^2+bx+c$.

```
> eq:=a*x^2+b*x+c;
```

$$eq := a\,x^2+b\,x+c \tag{2.1}$$

```
> solve(eq,x);
```

$$-\frac{1}{2}\,\frac{b-\sqrt{b^2-4\,a\,c}}{a},\ -\frac{1}{2}\,\frac{b+\sqrt{b^2-4\,a\,c}}{a} \tag{2.2}$$

```
>
```

Example 2. Factor the quadratic form $f(x)=2\,x^2-3\,x+1$, and then find its roots.

```
> f:=2*x^2-3*x+1;
```

$$f := 2\,x^2-3\,x+1 \tag{2.3}$$

```
> factor(f);
```

$$(2\,x-1)\,(x-1) \tag{2.4}$$

Hence, the roots are 1/2 and 1.

Example 3. Complete the square for $f(x)=4x^2-12x+3$.
This quadratic function has a minimum.

```
> minimize(4*x^2-12*x+3,location);
```

$$-6,\ \left\{\left[\left\{x=\frac{3}{2}\right\},\ -6\right]\right\} \tag{2.5}$$

Hence, the complete square form of $f(x)$ is

```
> 4*(x-3/2)^2-6;
```

$$4\left(x-\frac{3}{2}\right)^2-6 \tag{2.6}$$

Example 4. Find the minimum value and its location of $f(x)=x^2-4x+9$.

```
> minimize(x^2-4*x+9,location);
```

$$5,\ \{[\{x=2\},5]\} \tag{2.7}$$

Example 5. Find the maximum value of $f(x)=\dfrac{x}{x^2+1}$ on the interval $[-2, 2]$ and draw the graph to verify the result.

```
> maximize(x/(x^2+1),x=-2..2, location);
```

$$\frac{1}{2}, \left\{\left[\{x=1\}, \frac{1}{2}\right]\right\} \quad (2.8)$$

The maximum is $\frac{1}{2}$ at $x=1$.

```
> plot(x/(x^2+1),x=-2..2);
```



Exercises

1. Find the roots for the quadratic function $f(x)=2x^2+5x+1$.
2. Factor the quadratic form $f(x)=2x^2+5x-3$ and then find its roots.
3. Complete the square for $f(x)=3x^2-5x+2$.
4. Find the minimum value and its location of $f(x)=2x^2-3x+2$.
5. Find the maximum value and minimum value of $f(x)=x^3-3x^2+x-1$ on $[-1, 1]$ and draw its graph to verify the results.

1.3 The Basic Classes of Functions and Trigonometric Functions

MAPLE preserves names for most elementary functions such as natural exponential and logarithmic functions, trigonometric functions and their inverses. You can directly call these functions. It is important that you do not assign names for variables that are preserved for these built-in functions.

Example 1. Find the domain of $f(x)=\sqrt{x^2-2x-3}$.

```
> solve(x^2-2*x-3>=0,x);
```

$$RealRange(-\infty, -1), RealRange(3, \infty) \quad (3.1)$$

The domain of the function is $(-\infty, -1]$ and $[3, \infty)$.
Remark. Note that MAPLE displays a closed interval, say [0, 1} as "(0, 1)", and displays an open interval, say (0, 1), as "(Open(0), Open(1))". Compare **Example 1** to the following.

Example 1*. Find the domain of $f(x)=\frac{1}{\sqrt{x^2-2x-3}}$.

```
> solve(x^2-2*x-3>0,x);
```

$$RealRange(-\infty, Open(-1)), RealRange(Open(3), \infty) \quad (3.2)$$

The domain of the function is $(-\infty, -1)$ and $(3, \infty)$.

Example 2. Compute the composite function $f(g(x))$ and $g(f(x))$, where $f(x) = \sqrt{x}$ and $g(x) = x - 1$.

```
> f:=x->sqrt(x): g:=x->x-1:
> gf=g(f(x));
```

$$gf = \sqrt{x} - 1 \tag{3.3}$$

```
> fg=f(g(x));
```

$$fg = \sqrt{x - 1} \tag{3.4}$$

Example 3. Compute the following values of trigonometric functions, both exact values and numerical values: $\sin\left(\frac{\pi}{4}\right)$, $\tan\left(\frac{\pi}{6}\right)$, $\sec\left(\frac{\pi}{3}\right)$, $\sin(3.2)$.

```
> sin(Pi/4);
```

$$\frac{1}{2}\sqrt{2} \tag{3.5}$$

```
> evalf(%);
```

$$0.7071067810 \tag{3.6}$$

```
> tan(Pi/6);
```

$$\frac{1}{3}\sqrt{3} \tag{3.7}$$

```
> evalf(%);
```

$$0.5773502693 \tag{3.8}$$

```
> sec(Pi/3);
```

$$2 \tag{3.9}$$

```
> sin(3.2);
```

$$-0.05837414343 \tag{3.10}$$

Note that MAPLE uses the radian system for angle measures. If the input angle is in degrees, you have to convert it to radians.

Example 4. Compute the numerical values of trigonometric functions: sin(15 $^\circ$) and tan(23 $^\circ$).

```
> evalf(sin(15*Pi/180));
```

$$0.2588190451 \tag{3.11}$$

```
> evalf(tan(23*Pi/180));
```

$$0.4244748164 \tag{3.12}$$

Example 5. Simplify $\frac{\tan(x)}{\sec(x)}$.

```
> simplify(tan(x)/sec(x));
```

$$\sin(x) \tag{3.13}$$

Example 6. Simplify $\sin(x)^2 + \cos(x)^2$.

```
> simplify(sin(x)^2+cos(x)^2);
```

$$1 \tag{3.14}$$

Exercises

1. Find the domain of $f(x)=\sqrt{x^2-2x-3}$.

2. Compute the composite function $f(g(x))$ and $g(f(x))$, where $f(x)=\sqrt{2x^2}$ and $g(x)=x+3$.

3. Compute the following values of trigonometric functions, both exact values and numerical values: $\sin\left(\frac{\pi}{3}\right)$, $\tan\left(\frac{5\pi}{6}\right)$, $\csc\left(\frac{5\pi}{3}\right)$.

4. Compute the numerical values of trigonometric functions: sin(22.5°), cos(75°).

5. Simplify $(\tan(x)+\cot(x))\cos(x)$.

Review Exercises

1. Find the domain of the function. (1) $f(x)=\sqrt{x+3}$; (2) $f(x)=\frac{2}{3-x}$; (3) $f(x)=\sqrt{x^2-x-5}$.

2. Let $f(x)=\frac{3-x}{x^2+1}$. Evaluate (1) $f(2)$; (2) $f(a)$; (3) $f\left(2a^2+1\right)$.

3. Let $f(x)=\frac{x-1}{x+1}$ and $g(x)=\frac{x+1}{x-1}$. Find the composite functions $f(g(x))$ and $g(f(x))$.

4. Sketch the graph of the function $y=\sin\left(\frac{t}{2}\right)$ on $[-2\pi, 2\pi]$.

5. Sketch the graph of $f(x)=\frac{x+1}{x^2+1}$ on $[-2, 2]$ with blue color, 9 x-ticketmarks, and a title.

6. Sketch $\sin(x)$, $\cos(x)$, $\sin(2x)$, and $\cos\left(x-\frac{\pi}{6}\right)$ on one screen.

7. Solve the equation $x^3+3x^2-x-3=0$.

8. Solve the linear system $2x+3y=7,\ x-2y=1$.

9. Factor x^2-4x-5.

10. Find the minimum and its location for the quadratic function $f(x)=2x^2-3x+1$.

11. Evaluate the following values (both exact values and numerical values): $\sin\left(\frac{11\pi}{12}\right)$, $\cos\left(\frac{7\pi}{8}\right)$, $\cot\left(\frac{3\pi}{12}\right)$, $\csc\left(\frac{3\pi}{4}\right)$.

12. Evaluate the following values (both exact values and numerical values): sin (135°), cos(150°), sec(315°).

13. Simplify the following: $\frac{(\csc(x)-\cot(x))(\csc(x)+\cot(x))}{\sec(x)}$.

14. Simplify the following: $(\tan(x)+\cot(x))\sin(x)$.

15. Verify the following: $\frac{2\sin(x)\cos(x)}{\cos(x)^2-\sin(x)^2}=\tan(2x)$.

Chapter 2 LIMITS

2.1 Limits, Rates of Change, and Tangent Lines

The **average velocity** over a time interval $[t_0, t]$ is given by $\frac{s(t) - s(t_0)}{t - t_0}$. Then the instantaneous velocity at t_0 is the limit of the average velocity as $t \to t_0$, which can be estimated by a list of the average velocities over intervals $[t_0, t_1]$, $[t_0, t_2]$,..., $[t_0, t_n]$, where the lengths of the intervals tend to 0.

2.1.1 Find Limits, and velocities

Example 1. Assume $s(t) = 16t^2$. Estimate the average velocity over [0.5, t], with $t = 0.6$, 0.55, 0.51, 0.505, and 0.5001, and then estimate the instantaneous velocity at $t = 0.5$.

Step 1. Get the initial distance at the initial time:

```
> s0:=16*0.5^2;
```

$$s0 := 4.00 \qquad (1.1.1)$$

Step 2. Define the average speed as a function of t:

```
> Avrspd:=t->(16*t^2-4)/(t-0.5);
```

$$Avrspd := t \to \frac{16\,t^2 - 4}{t - 0.5} \qquad (1.1.2)$$

Step 3. Create a list of the end times for computing the average speeds.

```
> tlist:=[0.6,0.55,0.51,0.505,0.50001];
```

$$tlist := [0.6, 0.55, 0.51, 0.505, 0.50001] \qquad (1.1.3)$$

Step4. Evaluate the average speeds for all end times in the list.

```
> spdlist:=map(Avrspd,tlist);
```

$$spdlist := [17.60000000, 16.80000000, 16.16000000, 16.08000000, 16.00020000] \qquad (1.1.4)$$

Using the speed list above we can guess the instantaneous velocity at t = .5, which is 16.

MAPLE syntax remark: You cannot use "eval" to evaluate an expression (i.e., a dependent variable) in a list. You need to use the syntax "map" to do it. Compare **Example 1** to following two examples.

Example 2. A stone is tossed in the air from ground level with an initial velocity of 15 m/s. Its height at time t is $h(t) = 15t - 4.9t^2$ m. Compute its average velocity over the time interval $[0.5, 25]$.

```
> h0:=15*0.5-4.9*0.5^2;
```

$$h0 := 6.275 \qquad (1.1.5)$$

```
> Avrspd:=t->(15*t-4.9*t^2-6.275)/(t-0.5);
```

$$Avrspd := t \to \frac{15\,t - 4.9\,t^2 - 6.275}{t - 0.5} \qquad (1.1.6)$$

```
> spdv:=Avrspd(2.5);
```

$$spdv := 0.3000000000 \quad (1.1.7)$$

The result is 0.3 m/s.

Example 3. A stone is tossed in the air from ground level with an initial velocity of 15 m/s. Its height at time t is $h(t) = 15t - 4.9t^2$ m. Compute its average velocity over the time intervals $[1, 1.01]$, $[1, 1.001]$, $[1, 1.0001]$, $[0.99, 1]$, $[0.999, 1]$, and $[0.9999, 1]$. Use these values to estimate the instantaneous velocity at $t = 1$.

```
> h0:=15-4.9;
```

$$h0 := 10.1 \quad (1.1.8)$$

```
> Avrspd:=t->(15*t-4.9*t^2-10.1)/(t-1);
```

$$Avrspd := t \rightarrow \frac{15\,t - 4.9\,t^2 - 10.1}{t - 1} \quad (1.1.9)$$

```
> tlist1:=[1.01,1.001,1.0001];
```

$$tlist1 := [1.01, 1.001, 1.0001] \quad (1.1.10)$$

```
> spdlist1:=map(Avrspd,tlist1);
```

$$spdlist1 := [5.151000000, 5.195100000, 5.199500000] \quad (1.1.11)$$

```
> tlist2:=[0.99,0.999,0.9999];
```

$$tlist2 := [0.99, 0.999, 0.9999] \quad (1.1.12)$$

```
> spdlist2:=map(Avrspd,tlist2);
```

$$spdlist2 := [5.249000000, 5.204900000, 5.200500000] \quad (1.1.13)$$

Hence, we can estimate the instantaneous velocity at $t = 1$ by 5.2 m/s.

Example 4. Estimate the instantaneous rate of change of $f(x) = \sqrt{3t + 1}$ at $t = 1$.

```
> sqrt(3*1+1);
```

$$2 \quad (1.1.14)$$

```
> rcf:=t->(sqrt(3*t+1)-2)/(t-1);
```

$$rcf := t \rightarrow \frac{\sqrt{3\,t + 1} - 2}{t - 1} \quad (1.1.15)$$

```
> tlist1:=[1.01,1.001,1.0001];
```

$$tlist1 := [1.01, 1.001, 1.0001] \quad (1.1.16)$$

```
> rclist1:=map(rcf,tlist1);
```

$$rclist1 := [0.7485990000, 0.7498590000, 0.7499900000] \quad (1.1.17)$$

```
> tlist2:=[0.99,0.999,0.9999];
```

$$tlist2 := [0.99, 0.999, 0.9999] \quad (1.1.18)$$

```
> rclist2:=map(rcf,tlist2);
```

$$rclist2 := [0.7514115000, 0.7501410000, 0.7500100000] \quad (1.1.19)$$

The instantaneous rate of change of $f(x) = \sqrt{3t + 1}$ at $t = 1$ is estimated by 0.75.

Exercises

1. Assume $s(t)=4t^2-2t$. Estimate the average velocity over $[0.5, t]$, with $t=0.6, 0.55, 0.51, 0.505$, and 0.5001. Then estimate the instantaneous velocity at $t=0.5$.
2. A stone is tossed in the air from ground level with an initial velocity of 15 m/s. Its height at time t is $h(t)=15t-4.9t^2$ m. Compute its average velocity over the time interval [0.5, 25].
3. A stone is tossed in the air from ground level with an initial velocity of 15 m/s. Its height at time t is $h(t)=2t-4.9t^2+5$ m. Compute its average velocity over the time intervals $[1, 1.01]$, $[1, 1.001]$, $[1, 1.0001]$ and $[0.99, 1]$, $[0.999, 1]$, $[0.9999, 1]$. Use these values to estimate the instantaneous velocity at $t=1$.
4. Estimate the instantaneous rate of change of $f(x)=\dfrac{1}{\sqrt{2t^2+t+1}}$ at $t=1$.
5. Estimate the instantaneous rate of change of $f(x)=\dfrac{2t}{(3t+1)^2}$ at $t=a$.

2.2 Limit: A Numerical and Graphical Approach

2.2.1 A numerical approach

Example 1. Use a numerical approach to estimate $\lim_{x\to 2}\left(\dfrac{x-2}{x^2-4}\right)$.

Step 1. Define the function.

```
> f:=x->(x-2)/(x^2-4);
```

$$f := x \to \frac{x-2}{x^2-4} \qquad (2.1.1)$$

Step 2. Create a right-list of x with values near 2:

```
> rlist:=[2.1,2.01,2.001,2.0001];
```

$$rlist := [2.1, 2.01, 2.001, 2.0001] \qquad (2.1.2)$$

Step 3. Evaluate the function at the list.

```
> map(f,rlist);
```

$$[0.2439024390, 0.2493765586, 0.2499375156, 0.2499937502] \qquad (2.1.3)$$

Step 4. Repeat the evaluation for a left-list.

```
> llist:=[1.9,1.99,1.999,1.9999];
```

$$llist := [1.9, 1.99, 1.999, 1.9999] \qquad (2.1.4)$$

```
> map(f,llist);
```

$$[0.2564102564, 0.2506265664, 0.2500625156, 0.2500062502] \qquad (2.1.5)$$

The estimate of the limit is 0.25.

Example 2. Use a numerical approach to estimate the limit $\lim_{x \to 0} \left(\frac{\sin(x)}{x} \right)$.

```
> f:=x->sin(x)/x;
```

$$f := x \to \frac{\sin(x)}{x} \quad (2.1.6)$$

```
> rlist:=[0.1,0.01,0.001,0.0001];
```

$$rlist := [0.1, 0.01, 0.001, 0.0001] \quad (2.1.7)$$

```
> map(f,rlist);
```

$$[0.9983341665, 0.9999833334, 0.9999998333, 0.9999999983] \quad (2.1.8)$$

```
> llist:=-rlist;
```

$$llist := [-.1, -0.01, -0.001, -0.0001] \quad (2.1.9)$$

```
> map(f,llist);
```

$$[0.9983341665, 0.9999833334, 0.9999998333, 0.9999999983] \quad (2.1.10)$$

The estimate of the limit is 1.

2.2.2 A graphical approach

The graphical approach uses the graph of a function around the given point to estimate the limit of the function at the point.

Example 3. Investigate graphically $\lim_{x \to 2} \left(\frac{x-2}{x^2-4} \right)$.

```
> f :=(x-2)/(x^2-4);
```

$$f := \frac{x-2}{x^2-4} \quad (2.2.1)$$

```
> plot(f(x), x=1.9..2.1);
```



From the graph, you can see that the limit is 0.25.

Example 4. Investigate graphically $\lim_{x\to 0}\left(\frac{\sin(x)}{x}\right)$.

```
> plot(sin(x)/x,x=-0.1..0.1);
```



The graph shows that the limit is 1.

Example 5. Investigate graphically $\lim_{x\to 0}\left(\frac{|x|}{x}\right)$.

```
> plot(abs(x)/x, x=-1..1);
```



The graph shows that the function has a jump at $x = 0$. Hence, the limit does not exist.

Example 6. Investigate graphically $\lim_{x\to 0}\sin\left(\frac{1}{x}\right)$.

```
> plot(sin(1/x),x=-0.1..0.1);
```



```
> plot(sin(1/x),x=-0.05..0.05);
```



These graphs show that the limit does not exist.

Remark. The approaches shown in this section are for understanding the concept of limit. They are not the approach usually used to find the limits of functions. The popular approach to limits is the algebraic one. MAPLE has a built-in command called "limit". You can directly use this command to find the limits of functions, which we will introduce in the following sections.

Exercises

1. Use a numerical approach to estimate $\lim_{x\to\sqrt{2}}\left(\frac{x-2}{\sqrt{x}-\sqrt{2}}\right)$.

2. Use a numerical approach to estimate $\lim_{x\to 0}\left(\frac{\sin(x)^2}{1-\cos(x)}\right)$.

3. Investigate graphically $\lim_{x\to 2}[x]$, where $[x]$ is the integer part of x. (Hint: In MAPLE the integer part of x is defined by floor(x)).

4. Investigate graphically $\lim_{x\to 0}\left(\frac{1-\cos(x)}{x^2}\right)$.

5. Investigate graphically $\lim_{x\to 0}\cos\left(\frac{1}{x}\right)$.

2.3 Find Limits Algebraically

2.3.1 Find two-sided limits

You can define a function first and then find the limit, or directly find the limit.

Example 1. Find $\lim_{x\to 2}\left(\frac{x-2}{x^2-4}\right)$.

```
> f :=(x-2)/(x^2-4):
> limit( f, x=2);
```

$$\frac{1}{4} \tag{3.1.1}$$

Or use a direct way.

```
> limit( (x-2)/(x^2-4), x=2 );
```

$$\frac{1}{4} \tag{3.1.2}$$

You can also define the function in the input-output mode, then take its limit.

```
> f :=x->(x-2)/(x^2-4):
> limit(f(x),x=2);
```

$$\frac{1}{4} \qquad (3.1.3)$$

Example 2. Find $\lim_{x \to 0} \left(\frac{\sin(2x)}{x} \right)$.

```
> limit( sin(2*x)/x, x=0);
```

$$2 \qquad (3.1.4)$$

or, do the following.

```
> f:=x->sin(2*x)/x;
```

$$f := x \to \frac{\sin(2x)}{x} \qquad (3.1.5)$$

```
> limit(f(x),x=0);
```

$$2 \qquad (3.1.6)$$

If the limit of a function does not exist, then the result is shown as "undefined" or a region of values.

Example 3. Determine if $\lim_{x \to 0} \sin\left(\frac{1}{x}\right)$ exists.

```
> limit( sin(1/x), x=0 );
```

$$-1 \,..\, 1 \qquad (3.1.7)$$

The result shows that the limit does not exist. The given region "$-1\,..\,1$" means that when x tends to 0, the values of the function can take any values between -1 and 1, but not outside $[-1, 1]$. Compare the following example with it:

Example 4. Determine if $\lim_{x \to 0} \frac{1}{x} \sin\left(\frac{1}{x}\right)$ exists.

```
> limit( sin(1/x)/x, x=0);
```

$$\mathit{undefined} \qquad (3.1.8)$$

where "*undefined*" means that when x tends to 0, the values of the function can take any values, which are unbounded.

If you get the result that shows that the limit is ∞, then the limit of the function does not exist, but the function values have a certain tendency. See the following example.

Example 5. Determine if $\lim_{x \to 0} \left(\frac{1}{x^2}\right)$ exists.

```
> limit( 1/x^2, x=0);
```

$$\infty \qquad (3.1.9)$$

2.3.2 Find one-sided limits

You can add an option "right" or "left" in the limit expression to specify the one-sided limit.

Example 6. Find the right-hand limit: $\lim_{x \to 0+} \left(\frac{|x|}{x} \right)$.

```
> limit( abs(x)/x, x=0, right);
```

$$1 \tag{3.2.1}$$

Example 7. Find the left-hand limit: $\lim_{x \to 0-} \left(\frac{|x|}{x}\right)$.

```
> limit(abs(x)/x, x=0, left);
```

$$-1 \tag{3.2.2}$$

Example 8. Find the right-hand limit: $\lim_{x \to \frac{\pi}{2}+} \tan(x)$.

```
> limit(tan(x), x=Pi/2, right);
```

$$-\infty \tag{3.2.3}$$

Example 9. Find the left-hand limit: $\lim_{x \to \frac{\pi}{2}-} \tan(x)$.

```
> limit(tan(x), x=Pi/2, left);
```

$$\infty \tag{3.2.4}$$

Remark. Note that in "Pi", the capital "P" cannot be replaced by small "p". See what will happen if you use "p" instead in the following example.

```
> limit(tan(x), x=pi/2, right);
```

$$\tan\left(\frac{1}{2}\pi\right) \tag{3.2.5}$$

In the above expression, the "pi" only means a Greek character, not the ratio of the circumference of a circle to its diameter.
The following example shows how to get the one-sided limit of a piecewise function.

Example 10. Find the right-hand limit of the function at $x = 3$, where $f(x)$ is the piecewise function defined by $f(x) = \begin{cases} x+1 & 3 < x, \\ x^2 - 1 & \textit{otherwise.} \end{cases}$

```
> f:=piecewise( x>3, x+1, x^2-1);
```

$$f := \begin{cases} x+1 & 3 < x \\ x^2 - 1 & \textit{otherwise} \end{cases} \tag{3.2.6}$$

```
> limit(f, x=3, right);
```

$$4 \tag{3.2.7}$$

Example 11. Find the left-hand limit of the piecewise function in **Example 5** at $x = 3$.

```
> limit(f, x=3, left);
```

$$8 \tag{3.2.8}$$

Since its right-hand limit is not equal to the left-hand limit, the piecewise function has no limit at

$x = 3$. Its graph shows this fact clearly.

```
> plot(f,x=1..5);
```



Exercises

1. Find the limit of $\frac{x^2-9}{x^2-4x+3}$ at $x = 3$.
2. Find $\lim_{x\to 0}\left(\frac{\sin(3x)}{2x}\right)$.
3. Find $\lim_{x\to\infty}\left(\frac{2x^2-9}{x^2-4x+3}\right)$.
4. Find $\lim_{x\to\infty}\left(\frac{\sin(x)}{x}\right)$.
5. Find $\lim_{x\to 1-}\tan\left(\frac{\pi x}{2}\right)$.
6. Find the right-hand limit of the function at $x = a$, where $f(x)$ is the piecewise function defined by $f(x) = \begin{cases} x^2+1 & a < x, \\ x^2-1 & \textit{otherwise.} \end{cases}$

2.4 Trigonometric Limits

When you use MAPLE to find the trigonometric limits, you do the same job as for other functions.

Example 1. Find $\lim_{x\to 0}\left(x\sin\left(\frac{1}{x}\right)\right)$.

```
> limit(x*sin(1/x),x=0) ;
```

$$0 \quad (4.1)$$

Example 2. Find $\lim_{x\to 0}\left(\frac{\sqrt{x+2}\,\sin(x)}{x}\right)$.

```
> limit(sqrt(x+2)*sin(x)/x,x=0);
```

(4.2)

$$\sqrt{2} \tag{4.2}$$

Example 3. Find $\lim_{x \to 0} \left(\frac{1 - \cos(x)}{\sin(x)^2} \right)$.

```
> limit((1-cos(x))/sin(x)^2,x=0);
```

$$\frac{1}{2} \tag{4.3}$$

Example 4. Find $\lim_{x \to 0} \left(\frac{x - \sin(x)}{x^3} \right)$.

```
> limit((x-sin(x))/x^3,x=0);
```

$$\frac{1}{6} \tag{4.4}$$

Exercises

1. Find $\lim_{x \to 0} \left(x^2 \cos\left(\frac{1}{x} \right) \right)$.
2. Find $\lim_{x \to 0} \left(\frac{1 - \cos(x)}{2x^2 + x^3} \right)$.
3. Find $\lim_{x \to 0} \left(\frac{x - \sin(x)}{\tan(x^3)} \right)$.

2.5 Limits and Continuity

A function $f(x)$ is continuous at $x = a$ if $\lim_{x \to a} f(x) = f(a)$. A function is continuous on $[a, b]$ if it is continuous everywhere in (a, b) and left-continuous at $x = b$ and right-continuous at $x = a$. An important fact is that **an elementary function is continuous on its domain.**

Example 1. Show that the function $f(x) = \frac{1}{x}$ is continuous at $x = 1$ by the definition of continuity.

```
> f:=x->1/x;
```

$$f := x \to \frac{1}{x} \tag{5.1}$$

```
> limit(f(x),x=1);
```

$$1 \tag{5.2}$$

```
> f(1);
```

$$1 \tag{5.3}$$

Example 2. Determine the points at which the function $f(x) = \frac{1 - 2z}{z^2 - z - 6}$ is discontinuous.

```
> solve(z^2-z-6=0,z);
```

$$3, -2 \tag{5.4}$$

The function is discontinuous at 3 and −2.

Example 3. Determine the points at which the function $f(x) = [x]$ is discontinuous.

Remark: The function [*x*] is the integer part of *x*. In MAPLE its syntax is floor(*x*).

```
> f:=x->floor(x);
```

$$f := x \rightarrow \text{floor}(x) \tag{5.5}$$

```
> plot(f(x),x=-5..5);
```



Hence, each integer number is a discontinuous point of the function.

Example 4. Determine the domain of the function $f(x) = \sqrt{9 - x^2}$.

```
> solve(9-x^2>=0,x);
```

$$RealRange(-3, 3) \tag{5.6}$$

The domain is the closed interval $[-3, 3]$.
Please compare this example to the following one to see how MAPLE displays an open interval.

Example 5. Determine the domain of the function $f(x) = \dfrac{1}{\sqrt{9 - x^2}}$.

```
> solve(9-x^2>0,x);
```

$$RealRange(Open(-3), Open(3)) \tag{5.7}$$

Then the domain is the open interval $(-3, 3)$.

Example 6. Sketch the graph of the piecewise function $f(x) = \begin{cases} 2 - x & 1 < x \\ x^2 & otherwise \end{cases}$ and determine if $x = 1$ is its discontinuous point.

```
> f:=piecewise(x>1,2-x,x^2);
```

$$f := \begin{cases} 2 - x & 1 < x \\ x^2 & otherwise \end{cases} \tag{5.8}$$

```
> plot(f,x=-1..3);
```




```
> limit(f,x=1,right); limit(f,x=1,left); eval(f,x=1);
```

$$1$$
$$1$$
$$1 \tag{5.9}$$

Hence, $f(x)$ is continuous at $x = 1$.

Example 7. Sketch the graph of the piecewise function $f(x) = \begin{cases} x^3 + 1 & x \leq 0 \\ -x^2 + 10x - 15 & 2 \leq x \\ 1 - x & \textit{otherwise} \end{cases}$ and determine if $x = 0$ and $x = 2$ are the discontinuous points.

```
> f:=piecewise(x<=0,x^3+1,x>=2,-x^2+10*x-15,-x+1);
```

$$f := \begin{cases} x^3 + 1 & x \leq 0 \\ -x^2 + 10\,x - 15 & 2 \leq x \\ 1 - x & \textit{otherwise} \end{cases} \tag{5.10}$$

```
> limit(f,x=0,left); limit(f,x=0,right); eval(f,x=0);
```

$$1 \tag{5.11}$$
$$1$$
$$1$$

```
> limit(f,x=2,left); limit(f,x=2,right); eval(f,x=2);
```

$$-1 \tag{5.12}$$
$$1$$
$$1$$

Hence, $f(x)$ is continuous at $x = 0$, but discontinuous at $x = 2$, where it is right-continuous. The following graph shows the results.

```
> plot(f,x=-2..4);
```



Exercises

1. Determine the points at which the function $f(x) = \dfrac{2x}{z^2 - 3x - 4}$ is discontinuous.

2. Determine the points in the interval $(-2, 2)$ at which the function $f(x) = \left[\dfrac{x}{2} - 1\right]$ is discontinuous, where $[x]$ denotes the integer part of x.

3. Sketch the graph of the piecewise function $f(x) = \begin{cases} 4 - x^2 & 2 < x \\ \dfrac{x^2}{2} & otherwise \end{cases}$ and determine where it is discontinuous.

2.6 Intermediate Value Theorem (IVT)

Example 1. Prove that the equation $\sin(x) = 0.3$ has at least one solution.
Recall that sin(x) is a continuous function on the real line. Hence, IVT can be applied for it.

```
> a:=sin(0);b:=sin(Pi/2);
```

$$a := 0 \qquad (6.1)$$
$$b := 1$$

Since $0 < 0.3 < 1$, there is a c in $[0, \pi]$ such that $\sin(c) = 0.3$.

Example 2. Use the Bisection Method to find the zero of $f(x) = \cos(x)^2 - 2\sin\left(\dfrac{x}{4}\right)$ in [0, 2].

```
> f:=x->cos(x)^2-2*sin(x/4);
```

$$f := x \rightarrow \cos(x)^2 - 2\sin\left(\frac{1}{4}x\right) \qquad (6.2)$$

```
> a:=0; b:=2;
```

$$a := 0 \qquad (6.3)$$
$$b := 2$$

```
> evalf({f(a),f(b)});
```

$$\{1., -.7856728877\} \qquad (6.4)$$

```
> c:=(a+b)/2;
```

$$c := 1 \tag{6.5}$$

```
> evalf(f(c));
```

$$-.2028813368 \tag{6.6}$$

```
> b:=c; c:=(a+b)/2;
```

$$b := 1 \tag{6.7}$$

$$c := \frac{1}{2}$$

```
> evalf(f(c));
```

$$0.5208016862 \tag{6.8}$$

```
> a:=c; c:=(a+b)/2;
```

$$a := \frac{1}{2} \tag{6.9}$$

$$c := \frac{3}{4}$$

```
> evalf(f(c));
```

$$0.1625620073 \tag{6.10}$$

```
> a:=c; c:=(a+b)/2;
```

$$a := \frac{3}{4} \tag{6.11}$$

$$c := \frac{7}{8}$$

```
> evalf(f(c));
```

$$-0.0231421900 \tag{6.12}$$

```
> b:=c; c:=(a+b)/2;
```

$$b := \frac{7}{8} \tag{6.13}$$

$$c := \frac{13}{16}$$

```
> evalf(f(c));
```

$$0.0694493049 \tag{6.14}$$

```
> a:=c; c:=(a+b)/2;
```

$$a := \frac{13}{16} \tag{6.15}$$

$$c := \frac{27}{32}$$

```
> evalf(f(c));
```

$$0.0230271052 \tag{6.16}$$

We conclude that $f(x)$ has a zero c satisfying $13/16 < c < 14/16$ $(= 7/8)$.

▼ Exercises

1. Prove that the equation $x^3 - 3 = 0$ has at least one solution in $[0, 2]$.
2. Prove that the equation $\sin(x) - \frac{x}{2} = 0$ has at least one solution in $\left[\frac{\pi}{6}, \pi\right]$.
3. Use the Bisection Method to find the zero of $f(x) = x^3 - 2x - 3$ in $[0, 2]$.
4. Use the Bisection Method to find the zero of $f(x) = \cos(x) - 2\sin(x)$ in $[0, 1]$.

Chapter 3 DIFFERENTIATION

Section 3.1 Definition of the Derivative

The difference quotient of function $f(x)$ at a is $\frac{f(a+h)-f(a)}{h}$.

The derivative of a function f at $x=a$ is defined by

$$f'(a)=\lim_{h\to 0}\left(\frac{f(a+h)-f(a)}{h}\right).$$

The equation of the tangent line to the graph of $y=f(x)$ at $P=(a, f(a))$ is

$$y-f(a)=f'(a)(x-a)$$

or

$$y=f'(a)(x-a)+f(a).$$

3.1.1 Find the difference quotient and derivative of a function at a point

Example 1. Find the difference quotient of $f(x)=4x^2-7x$ at $x=3$.

Method 1. Function is represented as a rule to produce output from input (input-output pair).

```
> f :=x->4*x^2-7*x;
```

$$f := x \to 4x^2-7x \qquad (1.1.1)$$

```
> qf:=x->(f(x+h)-f(x))/h;
```

$$qf := x \to \frac{f(x+h)-f(x)}{h} \qquad (1.1.2)$$

```
> qf3:=qf(3);
```

$$qf3 := \frac{4(3+h)^2-36-7h}{h} \qquad (1.1.3)$$

```
> simplify(qf3);
```

$$17+4h \qquad (1.1.4)$$

Method 2. Function is represented as an expression of an independent variable.

```
> f:=4*x^2-7*x;
```

$$f := 4x^2-7x \qquad (1.1.5)$$

```
> Qf:=( eval(f,x=3+h)-eval(f,x=3))/h;
```

$$Qf := \frac{4(3+h)^2-36-7h}{h} \qquad (1.1.6)$$

```
> simplify(Qf);
```

$$17+4h \qquad (1.1.7)$$

Example 2. Find the derivative of $f(x) = 4x^2 - 7x$ at $a = 3$ by definition.
Method: Repeat the first two steps above, then take the limit for $h \to 0$.

```
> f :=x->4*x^2-7*x;
```

$$f := x \to 4x^2 - 7x \tag{1.1.8}$$

```
> qfh:=(f(3+h)-f(3))/h;
```

$$qfh := \frac{4(3+h)^2 - 36 - 7h}{h} \tag{1.1.9}$$

```
> dfa:=limit(qfh, h=0);
```

$$dfa := 17 \tag{1.1.10}$$

Remark: You can combine the last two steps to one step.

```
> dfat3:= limit((f(3+h)-f(3))/h, h=0 );
```

$$dfat3 := 17 \tag{1.1.11}$$

Example 3. Find the derivative of $f(x) = 4x^2 - 7x$ at a general real number a.

```
> f :=x->4*x^2-7*x;
```

$$f := x \to 4x^2 - 7x \tag{1.1.12}$$

```
> dfata:= limit((f(a+h)-f(a))/h, h=0 );
```

$$dfata := 8a - 7 \tag{1.1.13}$$

Example 4. Find the derivative of $f(x) = \sin(2x)$ at $x = \pi$.

```
> f:=x->sin(2*x);
```

$$f := x \to \sin(2x) \tag{1.1.14}$$

```
> dfPi:=limit( (f(Pi+h)-f(Pi))/h, h=0);
```

$$dfPi := 2 \tag{1.1.15}$$

You can also use the expression representation of the function to do it.

```
> f:=sin(2*x);
```

$$f := \sin(2x) \tag{1.1.16}$$

```
> dfPi:=limit( (eval(f,x=Pi+h)-eval(f,x=Pi))/h, h=0);
```

$$dfPi := 2 \tag{1.1.17}$$

The following is another way to do the same job:

```
> f:=x->sin(2*x);  x:=Pi;
```

$$f := x \to \sin(2x) \tag{1.1.18}$$

$$x := \pi$$

```
> dfPi:=limit( (f(x+h)-f(x))/h, h=0);
```

$$dfPi := 2 \tag{1.1.19}$$

```
> x:='x':
```

Remark. (1) Do not replace the capital "P" in "Pi" above by the small "p". See what happens if you use "p" for "P".
(2) When you finish a job in which a variable is assigned to a constant, you need to release the assigned variable. Releasing the variables, say *x*, *y*, etc., is very important. After the variables are assigned to specific constants, they represent constants. Then when you call them later without releasing them, they will be treated as the assigned constants, not as variables. See the following example.

```
> f:=x->sin(2*x);   x:=Pi;
```

$$f := x \to \sin(2x)$$

$$x := \pi \qquad (1.1.20)$$

```
> dfPi:=limit( (f(x+h)-f(x))/h, h=0);
```

$$dfPi := 2 \qquad (1.1.21)$$

```
> g:= x^2-2;
```

$$g := \pi^2 - 2 \qquad (1.1.22)$$

Here, you obtain a constant instead of an expression of the function $g(x)$. Hence, you need to release 'x' as follows:

```
> x:='x':
```

After '*x*' is released, it represents an independent variable '*x*' again. See the following.

```
> g:= x^2-2;
```

$$g := x^2 - 2 \qquad (1.1.23)$$

This time you get '*g*' as an expression of '*x*'.

The following example shows that the derivative does not exist at a certain number *a*.

Example 5. Determine if the derivative of $f(x) = \begin{cases} x+1 & 3 < x \\ x^2 - 5 & otherwise \end{cases}$ exists at $x = 3$.

```
> f:=x->piecewise( x>3, x+1, x^2-5);
```

$$f := x \to piecewise\left(3 < x, x+1, x^2 - 5\right) \qquad (1.1.24)$$

We first verify the continuity of $f(x)$ at $x = 3$.

```
> fr3:=limit(f(x),x=3,right);
```

$$fr3 := 4 \qquad (1.1.25)$$

```
> fl3:=limit(f(x),x=3,left);
```

$$fl3 := 4 \qquad (1.1.26)$$

```
> f(3);
```

$$4 \qquad (1.1.27)$$

Since $\lim_{x \to 3+} f(x) = \lim_{x \to 3-} f(x) = f(3)$, $f(x)$ is continuous at $x = 3$. Then we determine if the derivative exists.

```
> dfat3:=limit(  (f(3+h)-f(3))/h, h=0);
```

$$dfat3 := undefined \qquad (1.1.28)$$

This means that the derivative does not exist. It can be verified as follows:

```
> limit(  (f(3+h)-f(3))/h, h=0, left);
```

$$6 \quad (1.1.29)$$

```
> limit(  (f(3+h)-f(3))/h, h=0, right);
```

$$1 \quad (1.1.30)$$

Since the right limit of the difference quotient is not equal to its left limit, the derivative does not exist at $x=3$. It can be also shown on the following graph of $f(x)$:

```
> plot(f, 0..5);
```



Remark. If you want to present a piecewise function in a normal way, do the following:

```
> f:=piecewise( x>3, x+1, x^2-5);
```

$$f := \begin{cases} x+1 & 3 < x \\ x^2-5 & \textit{otherwise} \end{cases} \quad (1.1.31)$$

3.1.2 Find a tangent line to the graph of a function at a given point

Example 6. Find an equation of the tangent line to the graph of $f(x)=x^2$ at $x=5$.
Find the derivative of the function at $x=5$.

```
> f:=x->x^2;
```

$$f := x \rightarrow x^2 \quad (1.2.1)$$

```
> m:= limit( (f(5+h) - f(5))/h, h=0);
```

$$m := 10 \quad (1.2.2)$$

The tangent line equation is

```
> y-f(5)=m*(x-5);
```

$$y-25=10\,x-50 \quad (1.2.3)$$

Example 7. Plot the function and the tangent line in **Example 6** in a graph.

```
> L:=x->10*x-50+25;
```

$$L := x \rightarrow 10\,x-25 \quad (1.2.4)$$

```
> plot([f,L],  2..7);
```



If you want to see the detail about the point (5, $f(5)$), you can choose a small domain of x. The graph shows how well the tangent line is close to the function locally.

```
> plot([f,L],4.5..5.5);
```



Exercises

1. Find the difference quotient of the function $f(x)$ at a: $\dfrac{f(a+h)-f(a)}{h}$, where $f(x)=2x^2-3x-5$, (a) $a=2$, (b) $a=3$, (c) $a=x-2$.
2. Find the derivative of the function $f(x)$ at the given number a by the definition, where $f(x)=\sin(3x)$, at $a=\dfrac{\pi}{6}$.
3. Find the derivative $w'(a+1)$ for $w(y)=2y^2-3y$.
4. Determine if the derivative of the piecewise function $f(x)=\begin{cases} 3x+1 & 3<x \\ x^2-3x+7 & \textit{otherwise} \end{cases}$ exists at the joint point $x=3$, and draw the graph of $f(x)$.

In Exercises 5−8, compute the derivative of $f(x)$ at $x=a$ using the limit definition; find the equation of its tangent line at $(a, f(a))$; and then draw the graphs of the function and its tangent line at the given point together.

5. $f(x) = 3x^2 + 2x,\ a = 2$.
6. $f(t) = \sqrt{t+1}$, $a = 0$.
7. $f(x) = \dfrac{x+1}{x-1}$, $a = 3$.
8. $f(t) = \dfrac{2}{1-t}$, $a = -1$.

Section 3.2 The Derivative as a Function

Let x be a variable, then $\dfrac{df(x)}{dx} = \lim_{h\to 0}\left(\dfrac{f(x+h)-f(x)}{h}\right)$ is a function of x, called the derivative function of $f(x)$.

To find the derivative of a given function, we can apply the limit technique as in the definition. However, in practice, we rarely find a derivative function by its definition. Instead, for elementary functions, we find their derivatives using the following procedure: (1) Build a derivative table for "basic" functions. (2) Set rules for functions which are built up from these functions. These rules includes (1) linearity rules; (2) product and quotient rules; and (3) the chain rule.

However, in MAPLE all rules are hidden. You get the derivative of a function without mentioning what rules should be applied.

3.2.1 Find the derivative of an explicit function

In MAPLE, there are two syntaxes for calculating derivatives. We show them in the following example.

Example 1. Find $\dfrac{df(x)}{dx}$, where $f(x) = x^n$.

Syntax 1: represent the function as an expression.

```
> y:=x^n;
```

$$y := x^n \qquad (2.1.1)$$

```
> predf:= diff(y,x);
```

$$predf := \frac{x^n\, n}{x} \qquad (2.1.2)$$

```
> df:=simplify(predf);
```

$$df := x^{(n-1)}\, n \qquad (2.1.3)$$

Syntax 2: represent the function as a pair of input-output.

```
> f:=x->x^n;
```

$$f := x \rightarrow x^n \qquad (2.1.4)$$

```
> predf:= D(f);
```

$$predf := x \rightarrow \frac{x^n\, n}{x} \qquad (2.1.5)$$

```
> df:= simplify(predf(x));
```

$$df := x^{(n-1)}\, n \tag{2.1.6}$$

Example 2. Find $f\,'(x)$, where $f(x) = x^{0.35}$.

```
> f:=x^0.35;
```

$$f := x^{0.35} \tag{2.1.7}$$

```
> df:= diff(f,x);
```

$$df := \frac{0.35}{x^{0.65}} \tag{2.1.8}$$

Example 3. Find $\frac{df(s)}{ds}$, where $f(s) = s^{\frac{1}{4}} + s^{\frac{1}{3}}$.

```
> f:=s^(1/4)+s^(1/3);
```

$$f := s^{(1/4)} + s^{(1/3)} \tag{2.1.9}$$

```
> diff(f, s);
```

$$\frac{1}{4}\,\frac{1}{s^{(3/4)}} + \frac{1}{3}\,\frac{1}{s^{(2/3)}} \tag{2.1.10}$$

Example 4. Use the sum rule to find the derivative of the function in **Example 3.**

```
> f:= s^(1/4); g:=s^(1/3);
```

$$f := s^{(1/4)}$$
$$g := s^{(1/3)} \tag{2.1.11}$$

```
> diff(f,s)+diff(g,s);
```

$$\frac{1}{4}\,\frac{1}{s^{(3/4)}} + \frac{1}{3}\,\frac{1}{s^{(2/3)}} \tag{2.1.12}$$

You get the same result as in **Example 3.**

▶ 3.2.2 Find the derivative of a function at a given value

▼ Exercises

In Exercises 1−2, compute $\frac{df(x)}{dx}$ using the limit definition.

1. $f(x) = \sqrt{x}$.

2. $f(x) = \frac{1}{x}$.

In Exercises 3–4, compute the derivative using MAPLE syntax directly.

3. $\frac{dg(y)}{dy}$, $g(y) = (y+1)^{\frac{1}{3}}$.

4. $\frac{du}{dx}$, $u(x) = x^{0.35}$.

In Exercises 5−6, calculate the indicated derivative.

5. $f'(2)$, $f(x) = \frac{3}{x^4}$.

6. $\frac{dC(100)}{dx}$, $C = 1500 + 120x - 0.01x^2$.

In Exercises 7−8, sketch the graphs of $f(x)$ together with $f'(x)$.

7. $f(x) = x^2$, x in $[-3, 3]$

8. $C(x) = 1500 + 120x - 0.01x^2$, $x \in [1000, 10000]$.

9. Find the point on the curve $y = x^2 + 3x - 7$ at which the slope of the tangent line is equal to 4.

10. Find all values of x where the tangent lines to $y = x^3$ and $y = x^4$ are parallel.

Section 3.3 Product and Quotient Rules

Product rule: $\frac{d(f(x)\,g(x))}{dx} = g(x)\,\frac{df(x)}{dx} + f(x)\,\frac{dg(x)}{dx}$.

Quotient rule: $\frac{d\left(\frac{f(x)}{g(x)}\right)}{dx} = \frac{g(x)\,\frac{df(x)}{dx} - f(x)\,\frac{dg(x)}{dx}}{[g(x)]^2}$.

In MAPLE, all derivative rules are hidden. When you find a derivative of a product or a quotient of functions, you do not need to mention them.

3.3.1 Product rule

Example 1. Find the derivative of $h(x) = 3x^2(5x+1)$.

```
> h:=3*x^2*(5*x+1);
```

$$h := 3\,x^2\,(5\,x+1) \qquad (3.1.1)$$

```
> dh:=diff(h,x);
```

$$dh := 6\,x\,(5\,x+1) + 15\,x^2 \qquad (3.1.2)$$

Example 2. Use the product rule to find the derivative of the function in **Example 1.**

```
> f:=3*x^2; g:=5*x+1;
```

$$f := 3\,x^2$$
$$g := 5\,x+1 \qquad (3.1.3)$$

```
> dh:=g*diff(f,x)+f*diff(g,x);
```

$$dh := 6\,(5\,x+1)\,x + 15\,x^2 \qquad (3.1.4)$$

Example 3. Find $\frac{d}{dx}\left((ax+b)\left(abx^2+1\right)\right)$, where a and b are constants.

```
> y:=(a*x+b)*(a*b*x^2+1);
```

$$y := (a\,x+b)\,\left(a\,b\,x^2+1\right) \qquad (3.1.5)$$

```
> dy:=diff(y,x);
```

(3.1.6)

$$dy := a\left(a\,b\,x^2+1\right)+2\,(a\,x+b)\,a\,b\,x \quad (3.1.6)$$

```
> simplify(dy);
```

$$3\,a^2\,b\,x^2+a+2\,a\,b^2\,x \quad (3.1.7)$$

```
> y:='y':
```

3.3.2 Quotient rule

Example 4. Find the equation of the tangent line to the graph of $f(x)=\dfrac{3x^2-2}{4x^3+1}$ at $x=1$.

```
> f:=x->(3*x^2-2)/(4*x^3+1);
```

$$f := x \rightarrow \frac{3\,x^2-2}{4\,x^3+1} \quad (3.2.1)$$

```
> m:=D(f)(1);
```

$$m := \frac{18}{25} \quad (3.2.2)$$

The equation of the tangent line is

```
> y=m*(x-1)+f(1);
```

$$y=\frac{18}{25}\,x-\frac{13}{25} \quad (3.2.3)$$

```
> m:='m':
```

Example 5. Find the derivative of the function $f(x)=\dfrac{x^4+2x+1}{x+1}$.

```
> f:=(x^4+2*x+1)/(x+1);
```

$$f := \frac{x^4+2\,x+1}{x+1} \quad (3.2.4)$$

```
> df:= diff(f,x);
```

$$df := \frac{4\,x^3+2}{x+1}-\frac{x^4+2\,x+1}{(x+1)^2} \quad (3.2.5)$$

```
> simplify(df);
```

$$3\,x^2-2\,x+1 \quad (3.2.6)$$

Example 6. Find the derivative of the function in **Example 5** by using the quotient rule.

```
> f:=x^4+2*x+1; g:=x+1;
```

$$f := x^4+2\,x+1 \quad (3.2.7)$$
$$g := x+1$$

```
> ndf:=(g*diff(f,x)-f*diff(g,x))/g^2;
```

$$ndf := \frac{(x+1)\left(4\,x^3+2\right)-x^4-2\,x-1}{(x+1)^2} \quad (3.2.8)$$

```
> simplify(ndf);
```

$$3\,x^2 - 2\,x + 1 \tag{3.2.9}$$

Example 7. Find $\frac{d}{dx}\left(\frac{ax+b}{cx+d}\right)$, where a, b, c, d are constants.

```
> f:= (a*x+b)/(c*x+d);
```

$$f := \frac{3\,x + b}{c\,x + d} \tag{3.2.10}$$

```
> diff(f,x);
```

$$\frac{3}{c\,x + d} - \frac{(3\,x + b)\,c}{(c\,x + d)^2} \tag{3.2.11}$$

Exercises

In Exercises 1−2, find the derivative using the product rule.

1. $f(x) = x\left(x^2 + 1\right)$.
2. $y = \left(x^2 + 1\right)\left(x^3 - 3x^2 + 2x\right)$. Find $y\,'(3)$.

In Exercises 3−4, find the derivative using MAPLE syntax directly.

3. $f(x) = x\left(x^2 + 1\right)$.
4. $f(x) = (x + 1)\left(x^2 + x - 3\right)\left(x^3 - x - 1\right)$, find $\frac{df(2)}{dx}$.

In Exercises 5−6, find the derivative using the quotient rule.

5. $f(x) = \frac{3\sqrt{x} - 2}{\sqrt{x} + 3}$.
6. $w = \frac{z^2}{\sqrt{z} + z}$. Find $w\,'(9)$.

In Exercises 7−8, find the derivative using MAPLE syntax directly.

7. $g(z) = \frac{\left(z^2 - 4\right)(z - 1)}{\left(z^2 + 1\right)(z + 2)}$
8. $f(x) = \frac{x^9 + x^8 + 4x^4 - 7x}{x^4 - 3x^2 + 2x + 1}$. Find $f\,'(0)$.

Section 3.4 Rates of Change

Basic formulas.

$$\text{Average ROC} = \frac{f(x_1) - f(x_0)}{x_1 - x_0}.$$

$$\text{Instantaneous ROC} = f\,'(x_0).$$

The effect of a one-unit change $f(x_0 + 1) - f(x_0)$ is approximated by $f\,'(x_0)$.

Remark. We usually refer to the derivative as the ROC omitting the word "instantaneous".

3.4.1 Applications of rates of changes

Example 1. Let $A=\pi r^2$ be the area of a circle of radius r. (a) Calculate the ROC of area with respect to radius. (b) Compute $\frac{dA}{dr}$ for $r=2$ and $r=5$.

```
> A:=r->Pi*r^2;
```

$$A := r \to \pi r^2 \quad (4.1.1)$$

```
> ROC:= D(A);
```

$$ROC := r \to 2\pi r \quad (4.1.2)$$

```
> ROC(2);
```

$$4\pi \quad (4.1.3)$$

```
> evalf(ROC(2));
```

$$12.56637062 \quad (4.1.4)$$

```
> ROC(5);
```

$$10\pi \quad (4.1.5)$$

```
> evalf(ROC(5));
```

$$31.41592654 \quad (4.1.6)$$

Example 2. Company data suggest that the total dollar cost of a 1,200-mile flight is approximately

$$C(x)=0.0005x^3-0.38x^2+120x,$$

where x is the number of passengers.
(a) Estimate the marginal cost of an additional passenger if the flight already has 150 passengers.
(b) Compare your estimate with the actual cost. (c) Is it more expensive to add a passenger when $x=150$ or $x=200$?

```
> C:=x->0.0005*x^3-0.38*x^2 + 120*x;
```

$$C := x \to 0.0005\,x^3 - 0.38\,x^2 + 120\,x \quad (4.1.7)$$

```
> MagCost150:=D(C)(150);
```

$$MagCost150 := 39.7500 \quad (4.1.8)$$

```
> ActuCost:=C(151)-C(150);
```

$$ActuCost := 39.5955 \quad (4.1.9)$$

The marginal cost when $x=150$ is very close to the actual cost.

```
> MAgCost200:=D(C)(200);
```

$$MAgCost200 := 28.0000 \quad (4.1.10)$$

It costs approximately \$28 to add one additional passenger when $x=200$. It is less expensive than to add a passenger when $x=150$ (that costs \$39.75).

Example 3. By Faraday's Law, if a conducting wire of length l meters moves at velocity v m/s in a perpendicular magnetic field of strength B (in teslas), a voltage of size $V=-Blv$ is induced in the wire. Assume that $B=2$ and $l=0.5$. (a) Find the rate of change dV/dv. (b) Find the rate of change of V with respect to time t if $v=4t+9$.

```
> B:=2: l:=0.5:
> V:=-B*l*v;
```
$$V := -1.0\, v \qquad (4.1.11)$$

Find *dV/dv*.

```
> dV:=diff(V,v);
```
$$dV := -1.0 \qquad (4.1.12)$$

Find the rate of change of V with respect to time t if $v = 4t + 9$.

```
> v:=4*t+9;
```
$$v := 4\, t + 9 \qquad (4.1.13)$$

```
> dVt:=diff(V,t);
```
$$dVt := -4.0 \qquad (4.1.14)$$

```
> B:='B': l:='l': v:='v':
```
A better way to do it is the following:

```
> V:=-B*l*v;
```
$$V := -B\, l\, v \qquad (4.1.15)$$

```
> Vt:=eval(V,v=4*t+9);
```
$$Vt := -B\, l\, (4\, t + 9) \qquad (4.1.16)$$

```
> dVt:=diff(Vt,t);
```
$$dVt := -4\, B\, l \qquad (4.1.17)$$

```
> eval(dVt, [B=2,l=0.5]);
```
$$-4.0 \qquad (4.1.18)$$

Exercises

1. Find the ROC of the area of a square with respect to the length of its side s when (1) $s = 3$, and (2) $s = 5$.
2. Find the ROC of the volume of a cube with respect to the length of its side s when (1) $s = 3$, and (2) $s = 5$.
3. Find the ROC of the volume V of a cube with respect to its surface area A.
4. Find the ROC of the volume V of a sphere with respect to its radius r $\left(V = \frac{4\pi r^3}{3}\right)$.
5. The temperature of an object (in degrees Fahrenheit) as a function of time (in minutes) is $T(t) = \frac{3t^2}{4} - 30t + 340$ for $0 \le t$ and $t \le 20$. At what rate does the object cool after 10 min (giving correct units)?
6. The escape velocity at a distance r meters from the center of the earth is $v = 2.82\times10^7 r^{-\frac{1}{2}}$ m/s. Calculate the rate at which v changes with respect to distance at the surface of the earth.
7. The population $P(t)$ of a city (in millions) is given by the formula $P(t) = 0.00005t^2 + 0.01t + 1$, where t denotes the number of years since 1990. (a) How large is the population in 1996 and how fast is it growing in the year? (b) When does the population grow at a rate of 12, 000 people per year?

8. The dollar cost of producing x bagels is $C(x) = 300 + 0.25x - 0.5\left(\frac{x}{1000}\right)^3$. Determine the cost of producing 2000 bagels and estimate the cost of the 2001st bagel. Compare your estimate with the actual cost of the 2001st bagel.
9. Suppose the dollar cost of producing x video cameras is $C(x) = 500x - 0.003x^2 + 10^{-8}x^3$. (a) Estimate the marginal cost at production level $x = 5000$ and compare it with the actual cost $C(5001) - C(5000)$. (b) Compare the marginal cost at $x = 5000$ with the average cost per camera, defined as $\frac{C(x)}{x}$.
10. The cost in dollars of producing alarm clocks is $C(x) = 50x^3 - 750x^2 + 3740x + 3750$, where x is in unit of 1000. (a) Calculate the average cost at $x = 4$, 6, 8, and 10. (b) Find the production level x^* at which the average cost is lowest. (c) Draw the graphs of $C(x)$, the average cost, and the marginal cost. Find the relation between average cost and marginal cost at x^*.

Section 3.5 High Derivatives

If a function $f(x)$ is represented by expression $f := f(x)$ (*Syntax 1*), then its second derivative can be calculated by the MAPLE syntaxes "diff(f, x, x)" or "diff(f, x$2)".
The nth derivative of $f(x)$ is calculated by using the syntax "diff(f, x$n)".

If $f(x)$ is represented by the input-output pair $f := x \to f(x)$ (*Syntax 2*), then its second derivative can be calculated by the syntax "D(D(f))" or the syntax "(D@@2)(f)".
The nth derivative of $f(x)$ is calculated by the syntax "(D@@n)(f)".

3.5.1 Compute high derivatives

Example 1. Find the $f''(x)$ and $f'''(x)$ for $f(x) = x^4 + 2x - \frac{9}{x^2}$ and evaluate $f'''(-1)$.

```
> f:=x->x^4+2*x-9*x^(-2);
```

$$f := x \to x^4 + 2\,x - \frac{9}{x^2} \quad (5.1.1)$$

```
> d2f := D(D(f));
```

$$d2f := x \to 12\,x^2 - \frac{54}{x^4} \quad (5.1.2)$$

```
> d3f:= (D@@3)(f);
```

$$d3f := x \to 24\,x + \frac{216}{x^5} \quad (5.1.3)$$

```
> d3f(-1);
```

$$-240 \quad (5.1.4)$$

Example 2. Find $f^{(6)}(x)$ for $f(x) = 7x^4 + 4x - x^{-1}$.

```
> f:=7*x^4+4*x-x^(-1);
```

$$f := 7\,x^4 + 4\,x - \frac{1}{x} \quad (5.1.5)$$

```
> d6f :=diff(f,x$6);
```

(5.1.6)

$$d6f := -\frac{720}{x^7} \tag{5.1.6}$$

Exercises

In Exercises 1−2, calculate the second and third derivatives.

1. $y = 4t^3 - 9t^2 + 7$.
2. $y = (x^2 + x)(x^3 + 1)$.

In Exercises 3−4, calculate the derivatives indicated.

3. $\frac{d^4}{dx^4} f(x)$, $f(x) = x^4$, at $x = 1$.
4. $h''(1)$, $h(x) = \frac{1}{\sqrt{x} + 1}$.

5. Calculate the first five derivatives of $f(x) = \sqrt{x}$.

In Exercises 6−7, find a general formula for $\frac{d^n}{dx^n} f(x)$.

6. $f(x) = (x + 1)^{-1}$.
7. $f(x) = x^{-\frac{1}{2}}$.

8. Find the acceleration at time $t = 5$ min of a helicopter whose height (in feet) is $h(t) = -3t^3 + 400t$.

9. Use a computer algebra system to compute $\frac{d^k}{dx^k} f(x)$ for $k = 1, 2, 3$ for the functions:

(a) $f(x) = (1 + x^3)^{\frac{5}{3}}$.

(b) $f(x) = \frac{1 - x^4}{1 - 5x - 6x^2}$.

10. Let $f(x) = \frac{x + 20}{x - 1}$. Compute the $\frac{d^k}{dx^k} f(x)$ for $k = 1, 2, 3, 4$. Can you find a general formula for $\frac{d^k}{dx^k} f(x)$?

Section 3.6 Trigonometric Functions

The following is the table of derivatives of trigonometric functions.

```
> diff(sin(x),x);
```

$$\cos(x) \tag{6.1}$$

```
> diff(cos(x),x);
```

$$-\sin(x) \tag{6.2}$$

```
> diff(tan(x),x);
```

$$1+\tan(x)^2 \tag{6.3}$$

```
> diff(cot(x),x);
```

$$-1-\cot(x)^2 \tag{6.4}$$

```
> diff(sec(x),x);
```

$$\sec(x)\tan(x) \tag{6.5}$$

```
> diff(csc(x),x);
```

$$-\csc(x)\cot(x) \tag{6.6}$$

3.6.1 Compute the derivatives of trigonometric functions

Example 1. Find the derivative of each function. (a) $f(x)=\sin(x)^2$; (b) $f(x)=x^3\sin(x)$; (c) $f(x)=\dfrac{\sin(x)}{x}$.

```
> diff(sin(x)^2,x);
```

$$2\sin(x)\cos(x) \tag{6.1.1}$$

```
> diff(x^3*sin(x),x);
```

$$3x^2\sin(x)+x^3\cos(x) \tag{6.1.2}$$

```
> diff(sin(x)/x,x);
```

$$\frac{\cos(x)}{x}-\frac{\sin(x)}{x^2} \tag{6.1.3}$$

Example 2. Find the equation of the tangent line at $x=\dfrac{\pi}{4}$ for $y=\sin(x)$.

```
> f:=x->sin(x);
```

$$f:=x\rightarrow\sin(x) \tag{6.1.4}$$

```
> m:=D(f)(Pi/4);
```

$$m:=\frac{1}{2}\sqrt{2} \tag{6.1.5}$$

The equation is

```
> y=m*(x-Pi/4)+f(Pi/4);
```

$$y=\frac{1}{2}\sqrt{2}\left(x-\frac{1}{4}\pi\right)+\frac{1}{2}\sqrt{2} \tag{6.1.6}$$

Example 3. Find the second derivative of $f(\theta)=\csc(\theta)$.

```
> diff(csc(theta),theta,theta);
```

$$\csc(\theta)\cot(\theta)^2-\csc(\theta)\left(-1-\cot(\theta)^2\right) \tag{6.1.7}$$

Exercises

In Exercises 1−2, find an equation of the tangent line at the point indicated.

1. $y=\sin(x)$, $x=\dfrac{\pi}{4}$.

2. $y = \tan(x)$, $x = \dfrac{\pi}{4}$.

In Exercises 3−5, use the Product and Quotient Rules as necessary to find the derivative of each function.
3. $f(x) = \sin(x)\cos(x)$.
4. $f(x) = \dfrac{x}{\tan(x)}$.
5. $h(x) = \dfrac{\sin(x)}{4 + \cos(x)}$.

In Exercises 6−7, calculate the second derivative.
6. $f(x) = 3\sin(x) + 4\cos(x)$.
7. $h(t) = \csc(t)$.

In Exercises 8−9, find an equation of the tangent line at the point specified.
8. $y = x^2 + \sin(x)$, $x = 0$.
9. $y = \dfrac{\cos(t)}{1 + \sin(t)}$, $t = \dfrac{\pi}{3}$.
10. Find the values of x between 0 and $\sqrt{2}$ where the tangent line to the graph of $y = \sin(x)\cos(x)$ is horizontal.

Section 3.7 The Chain Rule

The Chain rule is

$$(f(g(x)))' = f'(g(x))g'(x),$$

or let

$$h(x) = f(u), \text{ where } u = g(x).$$

Then

$$\frac{d}{dx}(h(x)) = \frac{d}{du}(f(u)) \cdot \frac{d}{dx}(g(x))$$

3.7.1 Find the derivatives by the chain rule

Example 1. Calculate the derivative of $y = \sqrt{x^3 + 1}$.

```
> dy:= diff(sqrt(x^3+1),x);
```

$$dy := \frac{3}{2} \frac{x^2}{\sqrt{x^3 + 1}} \qquad (7.1.1)$$

Example 2. Use the chain rule to calculate the derivative of the function in **Example 1.**

$$u := u \qquad (7.1.2)$$

```
> f:= sqrt(u);
```

$$f := \sqrt{u} \qquad (7.1.3)$$

```
> df:=diff(f,u);
```

$$df := \frac{1}{2} \frac{1}{\sqrt{u}} \qquad (7.1.4)$$

```
> u:=x^3+1;
```

$$u := x^3 + 1 \tag{7.1.5}$$

```
> du:=diff(u,x);
```

$$du := 3\,x^2 \tag{7.1.6}$$

```
> dy:=df*du;
```

$$dy := \frac{3}{2}\,\frac{x^2}{\sqrt{x^3+1}} \tag{7.1.7}$$

```
> u:='u':
```

Example 3. Find the derivative of $y = \left(x^3 + \cos(x)\right)^{-4}$.

```
> y:= (x^3 +cos(x))^(-4);
```

$$y := \frac{1}{\left(x^3 + \cos(x)\right)^4} \tag{7.1.8}$$

```
> diff(y,x);
```

$$-\frac{4\left(3\,x^2 - \sin(x)\right)}{\left(x^3 + \cos(x)\right)^5} \tag{7.1.9}$$

Example 4. Find the second derivative of $y = \dfrac{\cos(1+x)}{1+\cos(x)}$.

```
> y:=cos(1+x)/(1+cos(x));
```

$$y := \frac{\cos(x+1)}{1+\cos(x)} \tag{7.1.10}$$

```
> diff(y,x);
```

$$-\frac{\sin(x+1)}{1+\cos(x)} + \frac{\cos(x+1)\sin(x)}{(1+\cos(x))^2} \tag{7.1.11}$$

```
> y:='y':
```

Exercises

In Exercises 1−2, write the function as a composite $f(g(x))$ and compute its derivative using the Chain Rule.

1. $y = (x + \sin(x))^4$.
2. $y = \cos\left(x^3\right)$.

In Exercises 3−4, use the General Power Rule or the Shifting and Scaling Rule to find the derivative.

3. $y = \left(x^2 + 9\right)^4$.
4. $y = \sin(1 - 4x)$.

In Exercises 5−6, find the derivative of $f(g(x))$ first using the Chain Rule, then applying MAPLE syntax directly.

5. $f(u) = \sin(u),\;\; g(x) = 2\,x + 1$.

6. $f(u) = 2u + 1,\ g(x) = \sin(x)$.

In Exercise 7, find the derivatives of $f(g(x))$ and $g(f(x))$.

7. $f(u) = u^3,\ g(x) = \dfrac{1}{x+1}$.

In Exercises 8−10, find the derivative directly using MAPLE syntax.

8. $y = \sin\left(x^2\right)$.

9. $y = \dfrac{1}{\sqrt{\cos\left(x^2\right) + 1}}$.

10. $y = \dfrac{1}{\sqrt{kt^4 + b}}$, where k and b are nonzero constants.

Section 3.8 Implicit Differentiation

Assume that the function $y = f(x)$ is given by an equation $F(x, y) = c$, where c is a constant. The derivative of the implicit function $y = y(x)$ can be computed by differentiating the equation $F(x, y(x)) = c$.
It will lead to the formula:

$$\frac{dy}{dx} = -\frac{\frac{\partial}{\partial x} F(x, y)}{\frac{\partial}{\partial y} F(x, y)},$$

where $\frac{\partial}{\partial x} F(x, y)$ is the derivative of F with respect to x, and $\frac{\partial}{\partial y} F(x, y)$ is the derivative of F with respect to y.

3.8.1 Find the derivatives of implicit functions

Example 1. Find $y'(x)$, where $y(x)$ is defined by $\dfrac{x^2}{9} + \dfrac{y^2}{16} = 1$.

Method 1. Differentiate the equation $F(x, y) = c$.

```
> F:=x^2/9+y(x)^2/16;
```

$$F := \frac{1}{9} x^2 + \frac{1}{16} y(x)^2 \qquad (8.1.1)$$

```
> dEq:= diff(F,x);
```

$$dEq := \frac{2}{9} x + \frac{1}{8} y(x) \left(\frac{d}{dx} y(x)\right) \qquad (8.1.2)$$

```
> dyx:=solve(dEq, diff(y(x),x));
```

$$dyx := -\frac{16}{9} \frac{x}{y(x)} \qquad (8.1.3)$$

Hence, $\dfrac{dy}{dx} = \dfrac{-16x}{9y}$.

Method 2. Apply the formula.

```
> F:=x^2/9+y^2/16;
```

(8.1.4)

$$F := \frac{1}{9}x^2 + \frac{1}{16}y^2 \quad (8.1.4)$$

```
> dFx:=diff(F,x);
```

$$dFx := \frac{2}{9}x \quad (8.1.5)$$

```
> dFy:=diff(F,y);
```

$$dFy := \frac{1}{8}y \quad (8.1.6)$$

```
> dyx:=-dFx/dFy;
```

$$dyx := -\frac{16}{9}\frac{x}{y} \quad (8.1.7)$$

Example 2. Find $y(x)$, where $y(x)$ is defined by $\sin(xy) = x^2\cos(y)$

```
> F:=sin(x*y)-x^2*cos(y);
```

$$F := \sin(x\,y) - x^2\cos(y) \quad (8.1.8)$$

```
> impdy:=-diff(F,x)/diff(F,y);
```

$$impdy := -\frac{\cos(x\,y)\,y - 2\,x\cos(y)}{\cos(x\,y)\,x + x^2\sin(y)} \quad (8.1.9)$$

If you want to get the value of the derivative of an implicit function at a given point, you may assign a function as an input-output pairs as follows.

Example 3. Find the value of $y'(x)$ at $(2, \frac{\pi}{2})$, where $y(x)$ is defined by $\sin(xy) = x^2\cos(y)$.

```
> F:=(x,y)->sin(x*y)-x^2*cos(y);
```

$$F := (x, y) \rightarrow \sin(x\,y) - x^2\cos(y) \quad (8.1.10)$$

```
> impdy:=-(D[1](F)/D[2](F));
```

$$impdy := -\frac{(x, y) \rightarrow \cos(x\,y)\,y - 2\,x\cos(y)}{(x, y) \rightarrow \cos(x\,y)\,x + x^2\sin(y)} \quad (8.1.11)$$

```
> impdy(2,Pi/2);
```

$$\frac{1}{4}\pi \quad (8.1.12)$$

You can also combine the last two steps above into one:

```
> F:=(x,y)->sin(x*y)-x^2*cos(y);
```

$$F := (x, y) \rightarrow \sin(x\,y) - x^2\cos(y) \quad (8.1.13)$$

```
> impdyValue:=-(D[1](F)/D[2](F))(2, Pi/2);
```

$$impdyValue := \frac{1}{4}\pi \quad (8.1.14)$$

Example 4. Find an equation of the tangent line at the point $P = (1, 1)$ on the curve $y^4 + xy = x^3 - x + 2$.

```
> F:=(x,y)-> y^4+x*y - (x^3-x+2 );
```

$$F := (x, y) \rightarrow y^4 + x\,y - x^3 + x - 2 \quad (8.1.15)$$

```
> m:=-(D[1](F)/D[2](F))(1,1);
```

$$m := \frac{1}{5} \tag{8.1.16}$$

```
> Leq:=(y-1)-m*(x-1);
```

$$Leq := y - \frac{4}{5} - \frac{1}{5}\,x \tag{8.1.17}$$

Remark: When you want to plot the implicit function, you need to call "**implicitplot**", which is in the "**plots**" package. Before you draw the graph you have to activate the "**plots**" package using "**with(plots):**". You will see a warning message that warns you that something is changed. If you want to inactivate an activated package, use "**unwith(package name):**".

Example 5. Draw the graphs of the curve $y^4 + xy = x^3 - x + 2$ and its tangent line at $P = (1, 1)$ on the same screen.

```
> with(plots):
Warning, the name changecoords has been redefined
> implicitplot([F(x,y)=0, Leq=0], x=-4..4, y=-4..4,color=[blue,
  green]);
```



Remark. The "color=[blue,green]" in the command is optional. If you do not add a color option, the default color for the first curve is **red**.

Exercises

In Exercises 1−3, differentiate the expression with respect to x.

1. $x^2 y^3$.
2. $\dfrac{x^{30}}{y^3}$.
3. $5z + z^2 - 3z^3$.

In Exercises 4−10, calculate the derivative of y with respect to x.

4. $3y^3 + x^2 = 5$.

5. $x^2y + 2xy^2 = x + y$.
6. $y^3 = 1$.
7. $y^{-\frac{3}{2}} + x^{\frac{3}{2}} = 1$.
8. $\sqrt{x+y} = \frac{1}{x} + \frac{1}{y}$.
9. $\frac{x}{y} + \frac{y^2}{x+1} = 0$.
10. $\sin(x+y) = x + \cos(y)$.

11. Find $\frac{dy}{dx}$ of $(x+2)^2 - 6\,(2y+3)^2 = 3$ at $(1, -1)$.

12. Show that no point on the graph of $x^2 - 3xy + y^2 = 1$ has a horizontal tangent line.

In Exercises 13−15, find an equation of the tangent line at the given point and plot the graphs of the function and the tangent line.
13. $xy - 2y = 1$, at (3, 1).
14. $x^{\frac{2}{3}} + y^{\frac{2}{3}} = 2$, at (1, 1).
15. $x^{\frac{1}{2}} + y^{-\frac{1}{2}} = 2xy$, at (1, 1).

16. Find the points on the graph of $y^2 = x^3 - 3x + 1$, where the tangent line is horizontal, and draw the graphs of the function and the tangent line to verify your answer.
17. If the derivative $\frac{dx}{dy}$ exists at a point and $\frac{dx}{dy} = 0$, then the tangent line is vertical. Calculate $\frac{dx}{dy}$ for the equation $y^4 + 1 = y^2 + x^2$, and find the points on the graph where the tangent line is vertical.
18. Find all points on the folium $x^3 + y^3 = 3xy$ at which the tangent line is horizontal.
19. Draw the graph of the folium $\left(x^2 + y^2\right)^2 = \frac{25xy^2}{4}$, and find equations of its tangent lines at $x = 1$. [Hint: use x in [0, 2], y in [−2, 2] to draw the graph.]
20. Plot $y^2 = x^3 - 4x$ for x in $[-4, 4]$, y in $[-4, 4]$. Show that if $\frac{dx}{dy} = 0$, then $y = 0$. Plot the function and its vertical tangent line to confirm the conclusion.

Section 3.9 Related Rates

This is an application section. MAPLE can only help you to do the computational part of a problem.

3.9.1 Applications involving related rates

Example 1. Water pours into a conical tank (V-shape) of height 10 ft and radius 4 ft at a rate of 10 ft^3/min. How fast is the water level rising when it is 5 ft high?
***Step 1.* Find the relation of variables.** Let V and h be the volume and height of the water in the tank at time t. Then $V = \frac{\pi h r^2}{3}$.

Step 2. **List the known condition(s) and what is required.** ***Known***: $h = 5$ and $\frac{dV}{dt} = 10$. ***To be found***: $\frac{dh}{dt}$ at $h = 5$.

Step 3. **Solve the problem.**

```
> V:= Pi*h*r^2/3;
```

$$V := \frac{1}{3}\pi h r^2 \tag{9.1.1}$$

```
> r:= solve(r/h=4/10, r);
```

$$r := \frac{2}{5} h \tag{9.1.2}$$

```
> dVh:=diff(V, h);
```

$$dVh := \frac{4}{25}\pi h^2 \tag{9.1.3}$$

```
> dVt:=10: h:=5:
> solution:=solve(dVt=dVh*dht, dht);
```

$$solution := \frac{5}{2}\frac{1}{\pi} \tag{9.1.4}$$

```
> evalf(solution);
```

$$0.7957747152 \tag{9.1.5}$$

```
> h:='h': dVt:='dVt':
```

Answer: When $h = 5$, the water level is rising at the rate 0.80 ft/min.

Exercises

1. At what rate is the diagonal of a cube increasing if its edges are increasing at a rate of 2 cm/s?

In Exercises 2−4, assume that the radius r of a sphere is expanding at a rare of 14 in/min . [Hint: The volume of a sphere is $V = \frac{4\pi r^3}{3}$ and its surface area is $S = 4\,\pi r^2$.]

2. Determine the rate at which the volume is changing with respect to time when $r = 8$ in.
3. Determine the rate at which the volume is changing with respect to time at $t = 2$ min, assuming that $r = 0$ at $t = 0$.
4. Determine the rate at which the surface area is changing when the radius is $r = 8$ in.

5. A conical tank has height 3 m and radius 2 m at the top. Water flows in at a rate of 2 m^3/min. How fast is the water level rising when it is 2 m?
6. Follow the same set up as the previous exercise, but assume that the water level is rising at a rate of 0.3 m/min when it is 2 m. At what rate is the water flowing in?
7. Sonya and Isaac are in motorboats located at the center of a lake. At a time $t = 0$, Sonya begins traveling south at a speed of 32 mph. At the same time, Isaac takes off, heading east at a speed of 27 mph. (a) How far have Sonya and Isaac each traveled after 13 min? (b) At what rate is the distance between them increasing at $t = 12$ min?
8. At a given moment, a plane passes directly above a radar station at an altitude of 6 miles. (a) If the plane's speed is 500 mph, how fast is the distance between the plane and the station changing half an hour later? (b) How fast is the distance between the plane and the station changing when the plane passes directly above the station?

9. A hot-air balloon rising vertically is tracked by an observer located 2 miles from the liftoff point. At a certain moment, the angle between the observer's line of sight and the horizontal line is $\frac{\pi}{5}$, and it is changing at a rate of 0.2 rad/min. How fast is the balloon rising at this moment?

10. As a man walks away from a 1-ft lamppost, the tip of his shadow moves twice as fast as he does. What is the man's height?

Chapter 4 APPLICATIONS OF THE DERIVATIVE

Section 4.1 Linear Approximation and Application

Linear approximation of Δf: If f is differentiable at $x = a$ and Δx is small, then

$$\Delta f \sim f'(a)\Delta x$$

Linearization of a function $f(x)$ **at** $x = a$:

$$f(x) \sim f'(a)(x-a) + f(a)$$

4.1.1 Use Linear Approximation to estimate Δf

Example 1. Use Linear Approximation to estimate $\frac{1}{10.2} - \frac{1}{10}$ and find the approximation error.

```
> f :=x->1/x;
```

$$f := x \rightarrow \frac{1}{x} \qquad (1.1.1)$$

```
> df10:=D(f)(10);
```

$$df10 := \frac{-1}{100} \qquad (1.1.2)$$

```
> Est:=df10*(10.2-10);
```

$$Est := 0.2\ df10 \qquad (1.1.3)$$

```
> Exctv:=1/10.2-1/10;
```

$$Exctv := -0.00196078431 \qquad (1.1.4)$$

```
> err:=abs(Exctv-Est);
```

$$err := |0.00196078431 + 0.2\ df10| \qquad (1.1.5)$$

Example 2. Use Linear Approximation to estimate Δf for $f(x) = \tan(x)$, $a = \frac{\pi}{4}$, $\Delta x = 0.013$.

```
> f:=x->tan(x); a:=Pi/4;dx:=0.013;
```

$$f := x \rightarrow \tan(x) \qquad (1.1.6)$$

$$a := \frac{1}{4}\pi$$

$$dx := 0.013$$

```
> Est:=D(f)(a)*dx;
```

$$Est := 0.026 \qquad (1.1.7)$$

```
> simplify(Qf);
```

$$17 + 4\ h \qquad (1.1.8)$$

Example 3. The position function of an object is $s(t) = t^3 - 20t + 8$ m. Estimate the distance traveled over the time interval [3, 3.025].

```
> s :=t->t^3-20*t+8; a:=3; dt:=3.025-3;
```

$$s := t \to t^3 - 20\,t + 8 \qquad (1.1.9)$$

$$a := 3$$

$$dt := 0.025$$

```
> Estd:=D(s)(a)*dt;
```

$$Estd := 0.175 \qquad (1.1.10)$$

The distance traveled over the time interval [3, 3.025] is approximately 0.175 m.

Example 4. Use Linear Approximation to estimate the value of $\sqrt{82}$.

```
> f :=x->sqrt(x);a:=81; dx:=1;
```

$$f := x \to \sqrt{x} \qquad (1.1.11)$$

$$a := 81$$

$$dx := 1$$

```
> Estv:=sqrt(a)+D(f)(a)*dx;
```

$$Estv := \frac{163}{18} \qquad (1.1.12)$$

The numerical estimate of $\sqrt{82}$ is

```
> evalf(Estv);
```

$$9.055555556 \qquad (1.1.13)$$

The accurate value of $\sqrt{82}$ is

```
> evalf(sqrt(82));
```

$$9.055385138 \qquad (1.1.14)$$

4.1.2 Linearization

Example 5. Compute the linearization of $f(x) = \sqrt{x}$ at $x = 1$.

```
> f:=x->sqrt(x); a:=1;
```

$$f := x \to \sqrt{x} \qquad (1.2.1)$$

$$a := 1$$

```
> Lnzf:=D(f)(a)*(x-a)+f(a);
```

$$Lnzf := \frac{1}{2}\,x + \frac{1}{2} \qquad (1.2.2)$$

The tangent line equation is

```
> y-f(5)=m*(x-5);
```

$$y - 25 = 10\,x - 50 \qquad (1.2.3)$$

Example 6. Plot $f(x) = \tan(x)$ and its linearization $L(x)$ at $a = \frac{\pi}{4}$ on the same set of axes.

```
> f:=x->tan(x); a:=Pi/4;
```

$$f := x \rightarrow \tan(x) \tag{1.2.4}$$

$$a := \frac{1}{4}\pi$$

```
> Lx:=D(f)(a)*(x-a)+f(a);
```

$$Lx := 2x - \frac{1}{2}\pi + 1 \tag{1.2.5}$$

```
> plot({f(x), Lx},x=Pi/6..Pi/3);
```



Exercises

1. Use Linear Approximation to estimate $\sqrt{4.01} - \sqrt{4}$ and find the error of the estimate.
2. Use Linear Approximation to estimate Δf for $f(x) = \frac{1}{1+x^2}$, $a = 3$, $\Delta x = 0.5$. Compute both the error and the percentage error.
3. Use Linear Approximation to estimate $15.8^{\frac{1}{4}}$. Compute both the error and the percentage error.
4. Find Linearization of the function $f(t) = \sqrt{t+1}$ at $a = 0$.
5. Plot $f(x) = \frac{1}{\sqrt{1+x^2}}$ and its linearization $L(x)$ at $a = 3$ on the same set of axes.

Section 4.2 Extreme Values

(1) **A critical point** is a point at which the derivative is zero or does not exist.
(2) A function has a local extremum at a critical point.
(3) For finding the (global) extrema of a function $f(x)$ on a closed interval $[a, b]$, evaluate the function at all critical points in (a, b) and $f(a), f(b)$. The largest one is the maximum and the smallest one is the minimum.

4.2.1 Find local extrema and critical points

Example 1. Find the critical points and local maxima and minima of the function $f(x) = 2x^3 - 15x^2 + 24x + 19$.

Step 1. Find the critical points.

```
> f :=x-> 2*x^3-15*x^2+24*x+19;
```

$$f := x \rightarrow 2\,x^3 - 15\,x^2 + 24\,x + 19 \qquad (2.1.1)$$

```
> D(f);
```

$$x \rightarrow 6\,x^2 - 30\,x + 24 \qquad (2.1.2)$$

```
> solve(D(f)(x), x);
```

$$4, 1 \qquad (2.1.3)$$

Hence, the critical points of $f(x)$ are 4 and 1.

```
> f(4),f(1);
```

$$3, 30 \qquad (2.1.4)$$

Step 2. Draw the graph of $f(x)$ on an interval containing $x = 1, 4$.

```
> plot(f,0..5);
```



Hence, $f(x)$ has a local maximum 30 at $x = 1$ and has a local minimum 3 at $x = 4$.

Example 2. Find the critical points and local maximum and minimum values of the function $f(x) = 2 + 2x - 3x^{\frac{2}{3}}$.

Step1. Find the critical points.

```
> f:=x->2+2*x-3*(x^2)^(1/3);
```

$$f := x \rightarrow 2 + 2\,x - 3\,\left(x^2\right)^{(1/3)} \qquad (2.1.5)$$

Note that 0 is in the domain of the function, and at $x = 0$ the derivative of $f(x)$ does not exist. Hence, 0 is a critical point. To find other critical points, we solve the equation $(\mathrm{D}(f))(x) = 0$.

```
> solve(D(f)(x)=0, x);
```

$$4, 1 \qquad (2.1.6)$$

Thus, all critical points are 1 and 0.

```
> vecf:=(f(0),f(1));
```

$$vecf := 2, 1 \qquad (2.1.7)$$

Thus, the local minimum is 1 at $x = 1$ and the local maximum is 2 at $x = 0$, shown in the following graph.

```
> plot(f,-1..2);
```



4.2.2 Find the global extreme values of a function on a closed interval

Example 3. Find the global extreme values of $f(x) = x^3 - 3x^2 - 2x$ on interval $[-3, 3]$.

```
> f:=x->x^3-3*x^2-2*x;
```

$$f := x \to x^3 - 3\,x^2 - 2\,x \tag{2.2.1}$$

```
> solve(D(f)(x),x);
```

$$1 + \frac{1}{3}\sqrt{15},\ 1 - \frac{1}{3}\sqrt{15} \tag{2.2.2}$$

We now evaluate the function at all critical points and the endpoints of the given interval.

```
> map(f,[-3,1-1/3*sqrt(15),1+1/3*sqrt(15),3]);
```

$$\left[-48, \left(1 - \frac{1}{3}\sqrt{15}\right)^3 - 3\left(1 - \frac{1}{3}\sqrt{15}\right)^2 - 2 + \frac{2}{3}\sqrt{15}, \left(1 + \frac{1}{3}\sqrt{15}\right)^3 - 3\left(1 + \frac{1}{3}\sqrt{15}\right)^2 - 2 - \frac{2}{3}\sqrt{15}, -6\right] \tag{2.2.3}$$

```
> evalf(%);
```

$$[-48., 0.303314828, -8.303314837, -6.] \tag{2.2.4}$$

The maximum is 0.3033 at $x = 1 - \frac{\sqrt{15}}{3}$ and the minimum is -48 at $x = -3$.

Example 4. Find the maximum and minimum of $f(x) = \sqrt{1 + x^2} - 2x$ on the interval $[0, 2]$.

```
> f:=x->sqrt(1+x^2)-2*x;
```

$$f := x \to \sqrt{1 + x^2} - 2\,x \tag{2.2.5}$$

```
> df:= D(f);
```

$$df := x \to \frac{x}{\sqrt{1 + x^2}} - 2 \tag{2.2.6}$$

We now find the real zero of $f'(x)$.

```
> crt:= solve(df(x),x);
```

$$crt := \frac{2}{3}\,\mathrm{I}\sqrt{3} \qquad (2.2.7)$$

Remark. In MAPLE, the capital letter "I" denotes the imaginary unit $\sqrt{-1}$. When it occurs in a result, it indicates that the result is not a real number, i.e., the equation has no real solution. Here the result shows that the function has no critical points. This fact can also be shown by its graph:

```
> plot(f,0..2);
```



Thus, the maximum and the minimum of the function occur at the two endpoints. Evaluate the function at these two points:

```
> (f(0),f(2));
```

$$1, \sqrt{5}-4 \qquad (2.2.8)$$

```
> evalf(%);
```

$$1., -1.763932023 \qquad (2.2.9)$$

Hence, the function has the minimum $\sqrt{5}-4$ at $x=2$ and the maximum 1 at $x=0$.

4.2.3 Roll's Theorem

Roll's Theorem. Assume that $f(x)$ is continuous on $[a, b]$ and differentiable on (a, b). If $f(a)=f(b)$, then there is a number c between a and b such that $f'(c)=0$.

Example 5. Verify Roll's Theorem for $f(x)=x^4-x^2$.

```
> f:=x->x^4-x^2;
```

$$f := x \rightarrow x^4 - x^2 \qquad (2.3.1)$$

```
> (f(-2),f(2));
```

$$12, 12 \qquad (2.3.2)$$

The results show that $f(x)$ is continuous on $[-2, 2]$ and differentiable on $(-2, 2)$. Since $f(-2)=f(2)=12$, then there is a number c between a and b such that $f'(c)=0$.

Exercises

In Exercises 1−2, compute $\frac{\mathrm{d}f(x)}{\mathrm{d}x}$ using the limit definition.

1. $f(x)=\sqrt{x}$

2. $f(x) = \dfrac{1}{x}$

In Exercises3–4, compute the derivative using MAPLE syntax directly.

3. $\dfrac{dg(y)}{dy}$, $g(y) = (y+1)^{\frac{1}{3}}$

4. $\dfrac{du}{dx}$, $u(x) = x^{0.35}$

In Exercises 5−6, calculate the indicated derivative.

5. $f'(2)$, $f(x) = \dfrac{3}{x^4}$

6. $\dfrac{dC(100)}{dx}$, $C = 1500 + 120x - 0.01x^2$

In Exercises 7−8, sketch the graphs of $f(x)$ together with $f'(x)$.

7. $f(x) = x^2$, x in $[-3, 3]$.

8. $C(x) = 1500 + 120x - 0.01x^2$, x in $[1000, 10000]$.

9. Find the point on the curve $y = x^2 + 3x - 7$ at which the slope of the tangent line is equal to 4.

10. Find all values of x where the tangent lines to $y = x^3$ and $y = x^4$ are parallel.

Section 4.3 The Mean Value Theorem and Monotonicity

(1) **First derivative test for extrema:** Assume c is a critical point of $f(x)$. If $f'(x)$ changes the sign from + to − at c, then it has minimum at c, if it changes the sign from − to +, then it has maximum, if the sign holds, it is neither.
(2) $f'(x) > 0$ on (a, b) implies that it is increasing on (a, b); $f'(x) < 0$ on (a, b) implies that it is decreasing on (a, b).

4.3.1 Testing critical points

Example 1. Test the critical point of $h(x) = 3x^2(5x+1)$
Step 1. Find the critical points.

```
> h:=x->3*x^2*(5*x+1);
```

$$h := x \to 3\,x^2\,(5\,x+1) \qquad (3.1.1)$$

```
> dh:=D(h);
```

$$dh := x \to 6\,x\,(5\,x+1) + 15\,x^2 \qquad (3.1.2)$$

```
> solve(dh(x),x);
```

$$0, \frac{-2}{15} \qquad (3.1.3)$$

Step 2. Test the sign of $f'(x)$ on each interval by evaluating $f'(x)$ at the test points.

```
> TestP:=[-1,-1/15,1];
```

$$TestP := \left[-1, \frac{-1}{15}, 1\right] \qquad (3.1.4)$$

```
> dfsign:=map(D(h), TestP);
```

$$dfsign := \left[39, \frac{-1}{5}, 51\right] \quad (3.1.5)$$

Hence, $h(x)$ has a local maximum at $x = -\frac{2}{15}$ and has a local minimum at $x = 0$. This can be verified by its graph.

```
> plot(h,-3/15..3/15);
```



Example 2. Analyze the critical points of $f(x) = \frac{x^3}{3} - x^2 + x$

```
> f:=x->1/3*x^3-x^2+x;
```

$$f := x \rightarrow \frac{1}{3}x^3 - x^2 + x \quad (3.1.6)$$

```
> solve(D(f)(x),x);
```

$$1, 1 \quad (3.1.7)$$

```
> [D(f)(0),D(f)(2)];
```

$$[1, 1] \quad (3.1.8)$$

Since the sign of $f'(x)$ does not change, $x = 1$ is neither a local maximum nor a local minimum. This can be verified by its graph.

```
> plot(f,0..2);
```



4.3.2 Increasing and decreasing intervals

Example 3. Find intervals of increasing and decreasing for $f(x) = x^3 + 3x^2 - 2$

```
> f:=x->x^3+3*x^2-2;
```

$$f := x \rightarrow x^3 + 3\,x^2 - 2 \tag{3.2.1}$$

```
> solve(D(f)(x)>0);
```

$$RealRange(-\infty, Open(-2)), RealRange(Open(0), \infty) \tag{3.2.2}$$

```
> solve(D(f)(x)<0);
```

$$RealRange(Open(-2), Open(0)) \tag{3.2.3}$$

Hence, $f(x)$ is increasing on $(-\infty, -2)$ and $(0, \infty)$ while decreasing on $(-2, 0)$. This can be shown on its graph.

```
> plot(f,-3..2);
```



Example 4. Find intervals of increasing and decreasing for $f(x) = \dfrac{3x^2 - 2}{4x^2 + 1}$

```
> f:=x->(3*x^2-2)/(4*x^2+1);
```

$$f := x \rightarrow \frac{3\,x^2 - 2}{4\,x^2 + 1} \tag{3.2.4}$$

```
> solve(D(f)(x)>0);
```

$$RealRange(Open(0), \infty) \tag{3.2.5}$$

```
> solve(D(f)(x)<0);
```

$$RealRange(-\infty, Open(0)) \tag{3.2.6}$$

Hence, $f(x)$ is increasing on $(0, \infty)$ and decreasing on $(-\infty, 0)$. This can be shown on its graph.

```
> plot(f,-3..3);
```



Exercises

1. Test the critical point of $h(x) = x^3 - 6x^2$
2. Test the critical point of $f(x) = \dfrac{2x+1}{x^2+1}$
3. Use the first derivative test to determine whether the given critical point is a local maximum or a local minimum.

(1) $y = \sin(x)\cos(x)$, $c = \dfrac{\pi}{4}$

(2) $y = x^3 - 27x - 2$, $c = -3$

4. Find the intervals of increasing and decreasing of $f(x) = 3x^4 + 8x^3 - 6x^2 - 24x$. Then draw the graph to verify the answer.
5. Find the intervals of increasing and decreasing of $f(x) = x + \cos(x)$, on $[0, 2\pi]$. Then draw the graph to verify the answer.

Section 4.4 The Shape of a Graph

(1) Concavity: If $f'(x)$ is increasing on (a, b), then $f(x)$ is **concave up** on (a, b), and if $f'(x)$ is decreasing on (a, b), then $f(x)$ is **concave down** on (a, b). If $f(x)$ changes its concavity at $x = c$, then c is called a point of inflection of $f(x)$.
(2) Test of concavity: If $f''(x) > 0$ on (a, b), then $f(x)$ is **concave up** on (a, b), and if $f''(x) < 0$ on (a, b), then $f(x)$ is **concave down** on (a, b). If $f''(c)=0$ and $f''(x)$ changes its sign at c, then c is a point of inflection of $f(x)$.
(3) Second derivative test for extrema: If $f'(c)=0$ and $f''(c)$ exists, then $f(c)$ is a local maximum if $f''(c) < 0$, $f(c)$ is a local minimum if $f''(c) > 0$, and the test **fails** if $f''(c) = 0$.
Remark. If $f''(c) = 0$, then the second derivative test fails. It does not mean that $f(x)$ has no extremum at c. To determine whether $f(x)$ has a extremum at c, use the first derivative test.

4.4.1 Test of concavity

Example 1. Find the points of inflection of $f(x) = \cos(x)$ on $[0, 2\pi]$

```
> f:=x->cos(x);
```

$$f := x \rightarrow \cos(x) \quad (4.1.1)$$

```
> d2f:= D(D(f));
```

$$d2f := x \rightarrow -\cos(x) \quad (4.1.2)$$

```
> p1:=solve(d2f(x),x);
```

$$p1 := \frac{1}{2}\pi \quad (4.1.3)$$

Warning. *When you use the* "**solve**" *command to solve a trigonometric equation, MAPLE only lists the solution in the principal value because it applies the inverse trigonometric functions to solve the equation. Hence, you* ***must find the other solutions by yourself.***

Recall that $\frac{1}{2}\pi$ is a root of $f(x) = \cos(x)$on $[0, 2\pi]$. The another zero is

```
> p2:=p1+Pi;
```

$$p2 := \frac{3}{2}\pi \quad (4.1.4)$$

```
> evalf([p1,p2]);
```

$$[1.570796327, 4.712388981] \quad (4.1.5)$$

Draw the graph of the function.

```
> plot(f,0..2*Pi, -1..1);
```



The graph shows that both $\frac{\pi}{2}$ and $\frac{3\pi}{2}$ are points of inflection.

Example 2. Find the points of inflection of $f(x) = 3x^5 - 5x^4 + 1$ and determine the intervals where $f(x)$ is concave up and down.

```
> f:=x->3*x^5-5*x^4+1;
```

$$f := x \rightarrow 3x^5 - 5x^4 + 1 \quad (4.1.6)$$

```
> d2f:=D(D(f));
```

$$d2f := x \rightarrow 60x^3 - 60x^2 \quad (4.1.7)$$

```
> solve(d2f(x),x);
```

$$1, 0, 0 \quad (4.1.8)$$

```
> solve(d2f(x)>0);
```

$$RealRange(Open(1), \infty) \quad (4.1.9)$$

Hence, $x = 1$ is the only point of inflection and $f(x)$ is concave up on $(1, \infty)$ and concave down on $(-\infty, 1)$. See the following graph.

```
> plot(f, -1..2, -3..5);
```



4.4.2 Second derivative test of extrema

Example 3. Analyze the critical points of $f(x) = 2x^3 + 3x^2 - 12x$

```
> f:=x->2*x^3+3*x^2-12*x;
```

$$f := x \rightarrow 2\,x^3 + 3\,x^2 - 12\,x \quad (4.2.1)$$

```
> solve(D(f)(x),x);
```

$$1, -2 \quad (4.2.2)$$

```
> [D(D(f))(1),D(D(f))(-2)];
```

$$[18, -18] \quad (4.2.3)$$

Hence, $f(-2)$ is a local maximum and $f(1)$ is a local minimum.
Example 4. Analyze the critical points of $f(x) = x^5 - 5x^4$

```
> f:=x->x^5-5*x^4;
```

$$f := x \rightarrow x^5 - 5\,x^4 \quad (4.2.4)$$

```
> df:=D(f);
```

$$df := x \rightarrow 5\,x^4 - 20\,x^3 \quad (4.2.5)$$

```
> crtp:=solve(df(x),x);
```

$$crtp := 4, 0, 0, 0 \quad (4.2.6)$$

```
> map(D(D(f)),[crtp[1],crtp[2]]);
```

$$[320, 0] \quad (4.2.7)$$

Hence, $f(4)$ is a local minimum. However, the second derivative test fails for $x = 0$. We then use the first derivative test. Choose the test points in the intervals $(-\infty, 0)$ and $(0, 4)$ respectively, say

−1 and 1.

```
> tstp:=[-1,1];
```

$$tstp := [-1, 1] \tag{4.2.8}$$

```
> map(D(f),tstp);
```

$$[25, -15] \tag{4.2.9}$$

Since the first derivative changes the sign, $f(0)$ is a local maximum.

Exercises

1. Find the points of inflection of $f(x) = x^3 - 3t^2 + 1$
2. Determine the intervals on which $f(x) = 4x^5 - 5x^4$ is concave up or down and find the points of inflection.
3. Use the second derivative test to determine whether the critical points of the function are local minima or maxima.
(1) $f(x) = x^3 - 12x^2 + 45x$
(2) $f(x) = 3x^4 - 8x^3 + 6x^2$
(3) $f(x) = \sin(x)^2 + \cos(x)$, on $[0, \pi]$
(4) $f(x) = \dfrac{1}{x^2 - x + 2}$
4. Determine the intervals on which the function is concave up or down, find the points of inflection, and determine if the critical points are local maxima or minima.
(1) $f(x) = x^3 - 2x^2 + x$.
(2) $f(x) = 2x^4 - 3x^2 + 2$.
(3) $f(x) = x + \sin(x)$, on $[0, 2\pi]$

Section 4.5 Graph Sketching and Asymptotes

4.5.1 Graph sketching

Example 1. Sketch the curve $f(x) = \dfrac{x}{1 - x^2}$

```
> plot(x/(1-x^2),x=-3..3);
```



Some options can be added to "**plot**" for getting a better view, or for adding title, axis tickmarks, etc. The default resolution to plot a graph is 25 or 50 points. You can increase the number of points for drawing the graph and restrict the *y* region using the following way:

```
> plot(x/(1-x^2),x=-3..3,-30..30,numpoints=200);
```



where "**−30 . . 30**" is the *y*-region for the graph. You can also add a title, *x* tickmark, *y* tickmark etc., on the graph.

```
> str:="Graph of x/(1-x^2)":
> plot(x/(1-x^2),x=-3..3,-30..30,numpoints=200,title=str);
```




```
> plot(x/(1-x^2),x=-3..3,-30..30,numpoints=200,title=str, color=
  blue);
```



```
> plot(x/(1-x^2),x=-3..3,-30..30,numpoints=200,xtickmarks=7);
```



Example 2. Sketch the graph of $f(x) = 3x^4 - 8x^3 + 6x^2 + 1$
Step 1. Find all critical points and possible points of inflection.

```
> f:=x->3*x^4-8*x^3+6*x^2+1;
```

$$f := x \rightarrow 3\,x^4 - 8\,x^3 + 6\,x^2 + 1 \quad (5.1.1)$$

```
> solve(D(f)(x),x);
```

$$0,\ 1,\ 1 \quad (5.1.2)$$

```
> solve(D(D(f))(x),x);
```

$$1,\ \frac{1}{3} \quad (5.1.3)$$

Step 2. Determine the interval for displaying the graph according to the points found in **Step 1.** It is clear that the interval must contain the two points $1/3$ and 1. You can choose $[-1,2]$.

```
> plot(f,-1..2, title="graph of y=3*x^4-8*x^3+6*x^2+1",color=
  blue);
```



Example 3. Sketch the graph of $y = \sin(x) + \frac{x}{2}$ on $[0, 2\pi]$.

```
> f:=x->sin(x)+x/2;
```

$$f := x \rightarrow \sin(x) + \frac{1}{2}x \tag{5.1.4}$$

```
> plot(f,0..2*Pi);
```



4.5.2 Asymptotes

(1) If $\lim_{x\to\infty} f(x) = L$ or $\lim_{x\to-\infty} f(x) = L$, then $y = L$ is a horizontal asymptote of $f(x)$.

(2) If $\lim_{x\to c-} f(x) = \infty$ or $-\infty$, or if $\lim_{x\to c+} f(x) = \infty$ or $-\infty$, then $f(x)$ has the vertical asymptote $x = c$.

(3) If $\lim_{x\to\infty} \left(\frac{f(x)}{x}\right) = m$ and $\lim_{x\to\infty} (f(x) - mx) = b$, then $f(x)$ has a slant asymptote.

Example 4. Calculate $\lim_{x\to\infty} \left(\frac{2\,x^{5/3} + 7\,x^{-1/2}}{x - x^{1/2}}\right)$

```
> limit((2*x^(5/3)+7*x^(-1/2))/(x-x^(1/2)),x=infinity);
```

$$\infty \tag{5.2.1}$$

Example 5. Calculate $\lim_{x\to\infty} \left(\frac{2\,x^4 + 7x^2 + 1}{x^4 - x^2 + 2x + 1}\right)$

```
> limit((2*x^4+7*x^2+1)/(x^4-x^2+2*x+1),x=infinity);
```

$$2 \tag{5.2.2}$$

Example 6. Find the asymptotes of $f(x) = \frac{x}{x^2 - 1}$

```
> f:=x->x/(x^2-1);
```

$$f := x \rightarrow \frac{x}{x^2 - 1} \tag{5.2.3}$$

```
> m:=limit(f(x)/x,x=infinity);
```

$$m := 0 \quad (5.2.4)$$

```
> limit(f(x),x=infinity);
```

$$0 \quad (5.2.5)$$

```
> solve(x^2-1,x);
```

$$1, \ -1 \quad (5.2.6)$$

Hence, $f(x)$ has the horizontal asymptote $y = 0$, and the vertical asymptotes $x = 1$ and $x = -1$.

```
> plot(f,-5..5,-5..5);
```



Exercises

In Exercises 1−5, Sketch the graph on a suitable interval.

1. $f(x) = x^2 - 2x + 3$
2. $f(x) = x^3 + \frac{3x^2}{2}$
3. $y = x^2(x-4)^2$
4. $y = x^4 - 3x^3 + 4x$
5. $y = \sqrt{x} + \sqrt{9-x}$

In Exercises 6−8, find the limits.

6. $\lim_{x \to \infty} \left(\frac{3x^2 + 20x}{4x^2 + 7} \right)$
7. $\lim_{x \to \infty} \left(\frac{\sqrt{x^2+1}}{x+1} \right)$
8. $\lim_{x \to \infty} \left(\frac{x+1}{\left(x^2+1\right)^{\frac{1}{3}}} \right)$

9. Find all asymptotes of $f(x) = \frac{1}{x} + \frac{1}{x-2}$

10. Find the vertical asymptotes of $f(x) = \frac{4}{x^2 - 9}$

Section 4.6 Applied Optimization

4.6.1 Applied optimization

Example 1. A piece of wire of length L is bent into the shape of a rectangle. Which dimensions produce the rectangle of maximum area?
Step 1. Create the goal function.

```
> assum(L>0): f:=x->x*(L/2-x);
```

$$f := x \to x\left(\frac{1}{2}L - x\right) \tag{6.1.1}$$

Step 2. Maximize the area.

```
> solve(D(f)(x),x);
```

$$\frac{1}{4}L \tag{6.1.2}$$

```
> testp:=[0,L/4,L/2];
```

$$testp := \left[0, \frac{1}{4}L, \frac{1}{2}L\right] \tag{6.1.3}$$

```
> map(f,testp);
```

$$\left[0, \frac{1}{16}L^2, 0\right] \tag{6.1.4}$$

Hence, the maximum area is $\frac{L^2}{16}$ at $x = \frac{L}{4}$.
Example 2. Find the dimensions of a rectangle of maximum area that can be inscribed in a circle of radius r.

```
> assume(r>0): RArea:=x->x*sqrt((2*r)^2-x^2);
```

$$RArea := x \to x\sqrt{4r^2 - x^2} \tag{6.1.5}$$

```
> solve(D(RArea)(x),x);
```

$$\sqrt{2}\,r\sim, -\sqrt{2}\,r\sim \tag{6.1.6}$$

```
> RArea(sqrt(2)*r);
```

$$2\,r\sim^2 \tag{6.1.7}$$

Hence, the maximum area of the rectangle is $2r^2$, with two equal sides of length $\sqrt{2}\,r$.
Example 3. Given n numbers $x_1, x_2, \cdots, x_n$, find the value of t minimizing the sum of the squares: $f(t) = (t - x_1)^2 + (t - x_2)^2 + (t - x_3)^2$.

```
> f:=t->(t-x[1])^2+(t-x[2])^2+(t-x[3])^2;
```

$$f := t \to (t - x_1)^2 + (t - x_2)^2 + (t - x_3)^2 \tag{6.1.8}$$

```
> p:=solve(D(f)(t),t);
```

$$p := \frac{1}{3}x_1 + \frac{1}{3}x_2 + \frac{1}{3}x_3 \tag{6.1.9}$$

```
> f(p);
```

$$\left(-\frac{2}{3}x_1+\frac{1}{3}x_2+\frac{1}{3}x_3\right)^2+\left(\frac{1}{3}x_1-\frac{2}{3}x_2+\frac{1}{3}x_3\right)^2+\left(\frac{1}{3}x_1+\frac{1}{3}x_2-\frac{2}{3}x_3\right)^2 \quad (6.1.10)$$

Answer: Since $f(t)$ has the minimum and it has only one critical point $p=\frac{1}{3}x_1+\frac{1}{3}x_2+\frac{1}{3}x_3$, $f(t)$ has the minimal value at p and the minimal value is
$$\left(-\frac{2}{3}x_1+\frac{1}{3}x_2+\frac{1}{3}x_3\right)^2+\left(\frac{1}{3}x_1-\frac{2}{3}x_2+\frac{1}{3}x_3\right)^2+\left(\frac{1}{3}x_1+\frac{1}{3}x_2-\frac{2}{3}x_3\right)^2$$

Exercises

1. Find the dimensions of the rectangle of maximum area that can be formed from a 50-inch. piece of wire.
2. Find the positive number x such that $x+\frac{1}{x}$ is smallest.
3. The legs of a right triangle have lengths a and b satisfying $a+b=10$. Which values of a and b maximize the area of the triangle?
4. Find the radius and height of a cylindrical can of total surface area A whose volume is as large as possible.

Section 4.7 Newton's Method

Newton's method: It applies the recurrent formula
$$x_{n+1}=x_n-\frac{f(x_n)}{f'(x_n)}$$
to compute the numerical value of a root of $f(x)$, where the initial number x_0 is close to the zero, possibly referring to a graph.

3.7.1 Find roots by Newton's Method

Example 1. Calculate the first three approximation x_1, x_2, x_3 to a root of $f(x)=x^2-5$ using the initial guess $x_0=2$.

```
> f:=x->x^2-5: df:=D(f): Nf:=x->x-f(x)/df(x);
```

$$Nf:=x\rightarrow x-\frac{f(x)}{df(x)} \quad (7.1.1)$$

```
> x0:=2;
```

$$x0:=2 \quad (7.1.2)$$

```
> x1:=Nf(x0);
```

$$x1:=\frac{9}{4} \quad (7.1.3)$$

```
> x2:=Nf(x1);
```

$$x2:=\frac{161}{72} \quad (7.1.4)$$

```
> x3:=Nf(x2);
```

(7.1.5)

$$x3 := \frac{51841}{23184} \tag{7.1.5}$$

```
> evalf(x3);
```

$$2.236067978 \tag{7.1.6}$$

We compare the number to the numerical value of the exact root $\sqrt{5}$:

```
> evalf(sqrt(5));
```

$$2.236067977 \tag{7.1.7}$$

Example 2. Sketch the graph of $f(x) = x^3 - 4x + 1$ and use Newton's Method to approximate the largest positive root to within an error of at most 0.001.

```
> f:=x->x^3-4*x+1:
> plot(f,-3..3);
```




```
> df:=D(f): Nf:=x->x-f(x)/df(x);
```

$$Nf := x \to x - \frac{f(x)}{df(x)} \tag{7.1.8}$$

```
> x0:=2;
```

$$x0 := 2 \tag{7.1.9}$$

```
> x1:=evalf(Nf(x0));
```

$$x1 := 1.875000000 \tag{7.1.10}$$

```
> x2:=evalf(Nf(x1));
```

$$x2 := 1.860978520 \tag{7.1.11}$$

```
> x3:=evalf(Nf(x2));
```

$$x3 := 1.860805879 \tag{7.1.12}$$

```
> x4:=evalf(Nf(x3));
```

$$x4 := 1.860805853 \tag{7.1.13}$$

Hence, the approximation of the root is 1.861.

Exercises

In Exercises 1−3, calculate the first three approximation x_1, x_2, x_3 to a root of $f(x)$ using the given initial guess.

1. $f(x) = x^2 - 7$, $x_0 = 2.5$
2. $f(x) = \cos(x) - x$, $x_0 = 0.8$
3. $f(x) = x^4 - 8$, $x_0 = 1.6$

In Exercises 4–6, sketch the graph of $f(x)$ and use Newton's Method to approximate the root to within a given error .
4. $f(x) = x^4 + x^2 - 2x - 1$, find the unique positive root to within an error of 0.001.
5. $f(x) = \sin(x) - 0.9x$, find the smallest positive root to within an error of 0.0001.
6. $f(x) = x^4 - 6x^2 + x + 5$, find the largest positive root to within an error of 0.0001.

Section 4.8 Antiderivatives

A function $F(x)$ is an **antiderivative** of $f(x)$ on (a, b) if $F'(x) = f(x)$ on (a, b). The set of all antiderivatives of $f(x)$ is called the **indefinite integral** of $f(x)$ and denoted by $\int f(x)\mathrm{d}x$. It is know that

$$\int f(x)\mathrm{d}x = F(x) + C,$$

where C is an arbitrary constant. Since the indefinite integral of a function can be represented by one of its antiderivatives, MAPLE displays the indefinite integral of a function by one of its antiderivatives.

4.8.1 Find the antiderivatives (indefinite integrals) of functions

Syntax (for antiderivatives) : **int(expr,var).**

Example 1. Find an antiderivative of $f(x) = x^3 - 2x^2 + 4x - 2$

```
> f:=x^3-2*x^2+4*x-2;
```

$$f := x^3 - 2x^2 + 4x - 2 \tag{8.1.1}$$

```
> int(f,x);
```

$$\frac{1}{4}x^4 - \frac{2}{3}x^3 + 2x^2 - 2x \tag{8.1.2}$$

Example 2. Find the antiderivative of $f(x) = \dfrac{\cos(x)}{\sqrt{1 + \sin(x)}}$.

```
> f :=cos(x)/sqrt(1+sin(x));
```

$$f := \frac{\cos(x)}{\sqrt{1 + \sin(x)}} \tag{8.1.3}$$

```
> int(f,x);
```

$$2\sqrt{1 + \sin(x)} \tag{8.1.4}$$

Example 3. Find the indefinite integral of $\dfrac{x^2}{\sqrt{1 + x^3}}$

```
> int(x^2/sqrt(1+x^3), x);
```

$$\frac{2}{3}\sqrt{1 + x^3} \tag{8.1.5}$$

Example 4. Evaluate $\int \tan(x)\sec(x)^2\,dx$

```
> int(tan(x)*(sec(x))^2,x);
```

$$\frac{1}{2}\sec(x)^2 \tag{8.1.6}$$

4.8.2 Simple differential equations and initial problems

MAPLE has a command **"dsolve"** for solving ordinary differential equations. Its usage is the following.

Syntax: **dsolve(deqn, fnc(var))** is used for finding general solutions of an equation (i.e., no initial conditions).

Syntax: **dsolve({deqn, cond1, cond2, . . . ,condn}, fnc(var))** is used for solving initial problems, where "**cond1, cond2, . . . ,condn**" are the initial conditions for an equation.

Solve the General Differential Equations

Example 5. Find $y(x)$ from $\frac{dy}{dx} = x^3 - 2x + 2$

Method 1. By definition $y(x)$ is the indefinite integral of $x^3 - 2x + 2$. Hence,

```
> y:=int(x^3-2*x+2,x);
```

$$y := \frac{1}{4}x^4 - x^2 + 2x \tag{8.2.1}$$

```
> y:='y':
```

Method 2. Consider it as an equation.

Step 1. Define the equation.

```
>  de:= diff(y(x),x)=x^3-2*x-2;
```

$$de := \frac{d}{dx}y(x) = x^3 - 2x - 2 \tag{8.2.2}$$

Step 2. Solve the equation.

```
> dsolve(de,y(x));
```

$$y(x) = \frac{1}{4}x^4 - x^2 - 2x + \_C1 \tag{8.2.3}$$

where $\_C1$ stands for a general constant.

Example 6. Solve the equation $ydy = xdx$

Step 1. First, change it to the derivative form: $\frac{ydy}{dx} = x$, and then define the equation.

```
> de:=y(x)*diff(y(x),x)=x;
```

$$de := y(x)\left(\frac{d}{dx}y(x)\right) = x \tag{8.2.4}$$

Step 2. Solve the equation.

```
> dsolve(de,y(x));
```

$$y(x) = \sqrt{x^2 + \_C1},\ y(x) = -\sqrt{x^2 + \_C1} \tag{8.2.5}$$

Example 7. Solve $dy = 2xy^2\,dx$

Change it to $\frac{dy}{dx} = 2xy^2$

```
> de:=diff(y(x),x)=2*x*y(x)^2;
```

$$de := \frac{d}{dx}\,y(x) = 2\,x\,y(x)^2 \tag{8.2.6}$$

```
> dsolve(de,y(x));
```

$$y(x) = -\frac{1}{x^2 - \_C1} \tag{8.2.7}$$

Example 8. Solve the second order equation: $\frac{d^2y}{dx^2} = \sqrt{x}$

```
> de:= diff(y(x),x,x)=sqrt(x);
```

$$de := \frac{d^2}{dx^2}\,y(x) = \sqrt{x} \tag{8.2.8}$$

```
> dsolve(de,y(x));
```

$$y(x) = \frac{4}{15}\,x^{(5/2)} + \_C1\,x + \_C2 \tag{8.2.9}$$

where both $\_C1$ and $\_C2$ are arbitrary constants.

Solve the Initial Value Equations

Example 9. Solve $\frac{dy}{dx} = x^2 - 1,\, y(0) = 1$

```
> de:=diff(y(x),x)=x^2-1;
```

$$de := \frac{d}{dx}\,y(x) = x^2 - 1 \tag{8.2.10}$$

```
> ic:=y(0)=1;
```

$$ic := y(0) = 1 \tag{8.2.11}$$

```
> dsolve({de,ic},y(x));
```

$$y(x) = \frac{1}{3}\,x^3 - x + 1 \tag{8.2.12}$$

Example 10. Solve $2ydy = \sqrt{x}\,dx,\, y(0) = 2$

Step 1. Define the equation by changing the original form to $\frac{2ydy}{dx} = \sqrt{x},\, y(0) = 2$

```
> de:=2*y(x)*diff(y(x),x)=sqrt(x);
```

$$de := 2\,y(x)\left(\frac{d}{dx}\,y(x)\right) = \sqrt{x} \tag{8.2.13}$$

Step 2. Add the initial condition.

```
> ic:=y(0)=2;
```

$$ic := y(0) = 2 \tag{8.2.14}$$

Step 3. Solve the equation.

```
> dsolve({de,ic},y(x));
```

$$y(x) = \frac{1}{3}\sqrt{36 + 6\,x^{(3/2)}} \tag{8.2.15}$$

Example 11. Solve $\frac{d^2y}{dx^2} = -16\sin(4x), y(\pi) = \pi + 1\,,\ \ y\,'(\pi)=1$

Syntax 1: Use "D" form for differentiation.

```
> de:=D(D(y))(x)=-16*sin(4*x);
```

$$de := \left(\left(D^{(2)}\right)(y)\right)(x) = -16\sin(4\,x) \tag{8.2.16}$$

```
> ic:=(y(Pi)=Pi+1, D(y)(Pi)=1);
```

$$ic := y(\pi) = \pi + 1,\ (D(y))(\pi) = 1 \tag{8.2.17}$$

```
> dsolve({de,ic},y(x));
```

$$y(x) = \sin(4\,x) - 3\,x + 4\,\pi + 1 \tag{8.2.18}$$

Syntax 2. Use "diff" form for differentiation.

```
> de:=diff(y(x),x,x)=-16*sin(4*x);
```

$$de := \frac{d^2}{dx^2}\,y(x) = -16\sin(4\,x) \tag{8.2.19}$$

```
> ic:=(y(Pi)=Pi+1, D(y)(Pi)=1);
```

$$ic := y(\pi) = \pi + 1,\ (D(y))(\pi) = 1 \tag{8.2.20}$$

```
> dsolve({de,ic},y(x));
```

$$y(x) = \sin(4\,x) - 3\,x + 4\,\pi + 1 \tag{8.2.21}$$

Exercises

In Exercises 1−3, find an antiderivative of the function.

1. $f(x) = 2x^3 - 3x^2 + 3x + 2$
2. $f(x) = \sin(4x - 1)$
3. $g(z) = \frac{5z + z^2 - 3z^3}{z^2}$

In Exercises 4 − 7, calculate the indefinite integral.

4. $\int x + \frac{1}{x}\, dx$
5. $\int \tan(2x)\sec(2x)\, dx$

6. Solve the equation $\frac{dy}{dx} = \sqrt{x+1}$
7. Solve the equation $x^2 dy = 2\,y^3 dx$

In Exercises 8−10, solve the initial problem.

8. $\dfrac{dy}{dx} = x^2 + 2x, y(0) = 3$

9. $\dfrac{dy}{dt} = 3t^2 + \cos(t), y(0) = 12$

10. $f''(t) = t - \cos(t), f'(0) = 2,\ f(0) = -2$

Chapter 5 INTEGRAL

5.1 Approximating and Computing Area

(1) Right sum, Left sum, and Midpoint sum are used to approximate the area under the graph of $f(x)$ over $[a, b]$. Let $\Delta x = \frac{(b-a)}{N}$.

Right sum: $R_N = \Delta x \sum_{i=1}^{N} f(a + i\Delta x)$

Left sum: $L_N = \Delta x \sum_{i=0}^{N-1} f(a + i\Delta x)$

Midpoint sum: $M_N = \Delta x \sum_{i=1}^{N} f\left(a + \left(i - \frac{1}{2}\right)\Delta x\right)$

(2) If $f(x) \geq 0$ is continuous on $[a, b]$, then the area under the graph of $f(x)$ over $[a, b]$ is defined by

$$\lim_{N \to \infty} R_N = \lim_{N \to \infty} L_N = \lim_{N \to \infty} M_N.$$

5.1.1 Approximating area by Riemann sums

Example 1. Calculate Right-endpoint approximations of R_4 and R_6 for $f(x) = x^2$ on the interval $[1, 3]$.

```
> a:=1: b:=3: n:=4: f :=x->x^2: dx:=(b-a)/n:
> R4:=dx*sum(f(a+k*dx),k=1..n);
```

$$R4 := \frac{43}{4} \tag{1.1.1}$$

```
> n:=6: dx:=(b-a)/n:
> R6:=dx*sum(f(a+k*dx),k=1..n);
```

$$R6 := \frac{271}{27} \tag{1.1.2}$$

Example 2. Calculate Left-endpoint approximations of L_4 and L_6 for $f(x) = x^2$ on the interval $[1, 3]$.

```
> a:=1: b:=3: n:=4: f :=x->x^2: dx:=(b-a)/n:
> L4:=dx*sum(f(a+k*dx),k=0..n-1);
```

$$L4 := \frac{27}{4} \tag{1.1.3}$$

```
> n:=6: dx:=(b-a)/n:
> L6:=dx*sum(f(a+k*dx),k=0..n-1);
```

$$L6 := \frac{27}{4} \tag{1.1.4}$$

Example 3. Calculate midpoint approximations of $\boldsymbol{M_4}$ and $\boldsymbol{M_6}$ for $f(x) = x^2$ on the interval $[1, 3]$.

```
> a:=1: b:=3: n:=4: f :=x->x^2: dx:=(b-a)/n:
> M4:=dx*sum(f(a+(2*k-1)/2*dx),k=1..n);
```

$$M4 := \frac{69}{8} \tag{1.1.5}$$

```
> n:=6: dx:=(b-a)/n:
> M6:=dx*sum(f(a+(2*k-1)*dx),k=1..n);
```

$$M6 := \frac{556}{27} \tag{1.1.6}$$

Example 4. Calculate Right-endpoint approximations of $\boldsymbol{R_n}$ for $f(x) = x^2$ on the interval $[1, 3]$.

```
> a:=1: b:=3: n:='n': f :=x->x^2: dx:=(b-a)/n:
> Rn:=dx*sum(f(a+k*dx),k=1..n);
```

$$Rn := \frac{2\left(n + \frac{2(n+1)^2}{n} - \frac{2(n+1)}{n} + \frac{4}{3}\frac{(n+1)^3}{n^2} - \frac{2(n+1)^2}{n^2} + \frac{2}{3}\frac{n+1}{n^2}\right)}{n} \tag{1.1.7}$$

```
> Srn:=simplify(Rn);
```

$$Srn := \frac{2}{3}\frac{13n^2 + 12n + 2}{n^2} \tag{1.1.8}$$

5.1.2 Computing the area as the limit of approximation

Example 5. Find the area under the graph of $f(x) = 2x^2 - x + 3$ over [2, 4].
Step 1. Get the formula of the Right sum R_N .

```
> a:=2: b:=4: n:='n': f :=x->2*x^2-x+3: dx:=(b-a)/n:
> Rn:=dx*sum(f(a+k*dx),k=1..n);
```

$$Rn := \frac{2\left(9n + \frac{7(n+1)^2}{n} - \frac{7(n+1)}{n} + \frac{8}{3}\frac{(n+1)^3}{n^2} - \frac{4(n+1)^2}{n^2} + \frac{4}{3}\frac{n+1}{n^2}\right)}{n} \tag{1.2.1}$$

Step 2. Compute the limit $\lim_{N \to \infty} \boldsymbol{R_N}$.

```
> AreaF:=limit(Rn, n=infinity);
```

$$AreaF := \frac{112}{3} \tag{1.2.2}$$

Example 6. Find the area under the graph of $f(x) = \sin(x)$ over $[0, \pi]$.

```
> a:=0: b:=Pi: n:='n': f :=x->sin(x): dx:=(b-a)/n:
> Rn:=dx*sum(f(a+k*dx),k=1..n);
```

$$Rn := \frac{1}{n}\left(\pi\left(\frac{1}{2}\,\frac{\sin\left(\frac{\pi}{n}\right)\cos\left(\frac{(n+1)\pi}{n}\right)}{\cos\left(\frac{\pi}{n}\right)-1} - \frac{1}{2}\sin\left(\frac{(n+1)\pi}{n}\right) - \frac{1}{2}\,\frac{\sin\left(\frac{\pi}{n}\right)\cos\left(\frac{\pi}{n}\right)}{\cos\left(\frac{\pi}{n}\right)-1} + \frac{1}{2}\sin\left(\frac{\pi}{n}\right)\right)\right) \tag{1.2.3}$$

```
> AreaF:=limit(Rn, n=infinity);
```

$$AreaF := 2 \tag{1.2.4}$$

Exercises

1. Calculate Right-endpoint approximations of $\boldsymbol{R}_4$ and $\boldsymbol{L}_4$ for $f(x) = x^2 - x - 2$ on the interval $[1, 4]$.
2. Calculate Left-endpoint approximations of $\boldsymbol{R}_6$ and $\boldsymbol{L}_6$ for $f(x) = \frac{1}{x^2+1}$ on the interval $[0, 5]$.
3. Calculate Midpoint approximations of $\boldsymbol{M}_4$ and $\boldsymbol{M}_6$ for $f(x) = 2x^2 - x + 1$ on the interval $[0, 3]$.
4. Calculate Right-endpoint approximations of $\boldsymbol{R}_n$ for $f(x) = \frac{1}{x}$ on the interval $[1, 4]$.
5. Find the area under the graph of $f(x) = \tan(x)$ over $[0, \frac{\pi}{4}]$.
6. Find the limit: $\lim_{n \to \infty} \boldsymbol{R}_n$ where $\boldsymbol{R}_n = \frac{5}{N}\sum_{i=1}^{N}\left(-2 + \frac{5i}{N}\right)^4$.
7. Find the sum $\sum_{i=1}^{20}(2i+1)$.

5.2 The Definite Integral

(1) **Definite integral:** The definite integral of $f(x)$ over $[a, b]$ is the limit of Riemann sums of $f(x)$ over $[a, b]$. It is denoted by $\int_a^b f(x)\,\mathrm{d}x$.

(2) **Signed area of a region** = (area above x-axis - area below x-axis) = $\int_a^b f(x)\,\mathrm{d}x$.

Note: The unsigned area of a region enclosed by $f(x)$, $x = a$ and $x = b$, is the integral $\int_a^b |f(x)|\,\mathrm{d}x$.

In MAPLE, the definite integral can be calculated using the following syntax:
> **int(f(x), x = a..b)**, where the limit process of the Riemann sum is hidden.

5.2.1 Calculating the signed area

Example 1. Calculate the signed area of $f(x) = 2x - 5$ over the interval [0, 3].

```
> SignArea:=int(2*x-5,x=0..3);
```

$$SignArea := -6 \qquad (2.1.1)$$

Example 2. Calculate the unsigned area of $f(x) = 2x - 5$ over the interval [0, 3].

```
> UnsignArea:=int(abs(2*x-5),x=0..3);
```

$$UnsignArea := \frac{13}{2} \qquad (2.1.2)$$

Example 3. Calculate $\int_0^4 |3 - x| \mathrm{d}x$.

```
> int(abs(3-x),x=0..4);
```

$$5 \qquad (2.1.3)$$

Example 4. Calculate $\int_0^b x^2 \mathrm{d}x$.

```
> b:='b':
> int(x^2,x=0..b);
```

$$\frac{1}{3} b^3 \qquad (2.1.4)$$

Exercises

1. Calculate the signed area of $f(x) = x^2 - 3x$ over the interval [−2, 2].
2. Calculate the unsigned area of $f(x) = x^2 - 3x$ over the interval [−2, 2].
3. Calculate $\int_{-3}^{3} |6 - x^2| \mathrm{d}x$.
4. Calculate $\int_0^1 x^4 - 2x^2 \mathrm{d}x$.
5. Calculate $\int_0^\pi \sin\frac{x}{2} \mathrm{d}x$.

5.3 The Fundamental Theorem of Calculus, Part 1

The Fundamental Theorem of Calculus, Part 1:

Let $F(x)$ be an antiderivative of $f(x)$ on $[a, b]$. Then

$$\int_a^b f(x) \mathrm{d}x = F(b) - F(a)$$

MAPLE evaluates definite integrals of most functions based on the fundamental theorem, though the rule is hidden for display.

Example 1. Evaluate $\int_0^1 \left(x-x^2\right)\mathrm{d}x$.

```
> int(x-x^2,x=0..1);
```

$$\frac{1}{6} \tag{3.1}$$

Example 2. Evaluate $\int_{\frac{\pi}{4}}^{\frac{3\pi}{4}} \sin(x)\mathrm{d}x$.

```
> int(sin(x), x=Pi/4..3*Pi/4);
```

$$\sqrt{2} \tag{3.2}$$

Example 3. Evaluate $\int_{-x}^{x} \left(t-t^3\right)\mathrm{d}t$.

```
> int(t-t^3, t=-x..x);
```

$$0 \tag{3.3}$$

Exercises

1. Evaluate $\int_0^4 \sqrt{x}\,\mathrm{d}x$
2. Evaluate $\int_0^{\frac{\pi}{2}} \cos(y)\mathrm{d}y$
3. Evaluate $\int_0^1 \left(x-x^2\right)\mathrm{d}x$
4. Evaluate $\int_{-1}^{1} \left|x-x^2\right|\mathrm{d}x$

5.4 The Fundamental Theorem of Calculus, Part 2

(1) The Fundamental Theorem of Calculus, Part 2: Let $f(x)$ be a continuous function on $[a, b]$. Then $A(x)=\int_a^x f(t)\mathrm{d}t$ is an antiderivative of $f(x)$, that is

$$\frac{d}{dx}\int_a^x f(t)\mathrm{d}t = f(x)$$

(2) The Chain Rule for the integral $\int_{a(x)}^{b(x)} f(t)\mathrm{d}t$:

$$\frac{d}{dx}\int_{a(x)}^{b(x)} f(t)\mathrm{d}t = f(b(x))b'(x) - f(a(x))a'(x)$$

Example 1. Calculate the derivative $\frac{\mathrm{d}}{\mathrm{d}x}\int_0^x \left(t-t^2\right)\mathrm{d}t$.

Method 1. Apply the Fundamental Theorem of Calculus.

```
> IntGrnd:=t-t^2;
```

$$IntGrnd := t - t^2 \qquad (4.1)$$

```
> DrInt:= eval(IntGrnd, t=x);
```

$$DrInt := x - x^2 \qquad (4.2)$$

Method 2. Directly calculate it.

```
> Ax:=int(IntGrnd, t=0..x);
```

$$Ax := \frac{1}{2} x^2 - \frac{1}{3} x^3 \qquad (4.3)$$

```
> dA:= diff(Ax,x);
```

$$dA := x - x^2 \qquad (4.4)$$

Example 2. Calculate the derivative $\frac{d}{dx} \int_0^{x^2} \sin(2t) dt$.

Method 1. Apply the fundamental theorem of Calculus and the Chain Rule.

```
> IntGrnd:=sin(2*t); bx:=x^2:
```

$$IntGrnd := \sin(2\,t) \qquad (4.5)$$

```
> DrI:=eval(IntGrnd,t=bx)*diff(bx,x);
```

$$DrI := 2 \sin\left(2\, x^2\right) x \qquad (4.6)$$

Method 2. Directly calculate it.

```
> Ax:=int(IntGrnd, t=0..x^2);
```

$$Ax := \frac{1}{2} - \frac{1}{2} \cos\left(2\, x^2\right) \qquad (4.7)$$

```
> dA:=diff(Ax,x);
```

$$dA := 2 \sin\left(2\, x^2\right) x \qquad (4.8)$$

Example 3. Calculate the derivative $\frac{d}{dx} \int_{\sqrt{x}}^{x^2} \tan(t) dt$.

Method 1.

```
> IntGrnd:=tan(t): bx:=x^2: ax:=sqrt(x):
> DrI:=eval(IntGrnd,t=bx)*diff(bx,x)-eval(IntGrnd,t=ax)*diff(ax,x);
```

$$DrI := 2 \tan\left(x^2\right) x - \frac{1}{2} \frac{\tan\left(\sqrt{x}\right)}{\sqrt{x}} \qquad (4.9)$$

Method 2.

```
> Ax:=int(tan(t), t=sqrt(x)..x^2):
> diff(Ax,x);
```

$$0 \qquad (4.10)$$

Exercises

1. Calculate the derivative $\frac{d}{dx}\int_0^{x^2} \sin(2t)\,dt$.

2. Calculate the derivative $\frac{d}{dx}\int_{\sin(x)}^{\cos(x)} \arcsin(t)\,dt$.

3. Calculate the derivative $\frac{d}{dx}\int_x^{x^2} t^2\tan(t)\,dt$.

4. Calculate the derivative $\frac{d}{dx}\int_{-x}^{x} e^{t^2}\,dt$.

5.5 Substitution Method

Substitution method: If $F'(x) = f(x)$, then

$$\int f(u(x))\left(\frac{d}{dx}u(x)\right)dx = F(u(x)) + C$$

and

$$\int_a^b f(u(x))\left(\frac{d}{dx}u(x)\right)dx = \int_{u(a)}^{u(b)} f(u)\,du$$

Note: MAPLE performs substitution method automatically. Hence, when you apply MAPLE to calculate integrals, you need not make the substitution by yourself. In the examples of this section, we show the substitution method only for illustrating the concept.

5.5.1 Calculate integrals by substitution method

Example 1. Evaluate $\int 2x\,(x^2+9)^5\,dx$ by the substitution method.

Step 1. Choose u and compute its derivative.

```
> u:=x->x^2+9; du:=D(u); f:=u->u^5;
```

$$u := x \to x^2 + 9 \tag{5.1.1}$$

$$du := x \to 2\,x$$

$$f := u \to u^5$$

Step 2. Verify the integrand.

```
> Intg:=f(u(x))*du(x);
```

$$Intg := 2\,(x^2+9)^5\,x \tag{5.1.2}$$

Step 3. Apply the formula.

```
> Fx:=int(f(u),u);
```

$$Fx := \frac{1}{6}\,u^6 \tag{5.1.3}$$

```
> Asw:=eval(Fx, u=u(x));
```

$$Asw := \frac{1}{6}\,\left(x^2+9\right)^6 \tag{5.1.4}$$

Remark: In practice, we can evaluate the integral directly.

```
> Asw:=int(2*x*(x^2+9)^5,x);
```

$$Asw := \frac{1}{6}\,x^{12}+9\,x^{10}+\frac{405}{2}\,x^8+2430\,x^6+\frac{32805}{2}\,x^4+59049\,x^2 \tag{5.1.5}$$

The difference between these two answers is a constant, which is shown as follows:

```
> Asws:=expand(1/6*(x^2+9)^6);
```

$$Asws := \frac{1}{6}\,x^{12}+9\,x^{10}+\frac{405}{2}\,x^8+2430\,x^6+\frac{32805}{2}\,x^4+59049\,x^2+\frac{177147}{2} \tag{5.1.6}$$

Example 2. Evaluate $\displaystyle\int \frac{x^2+2x}{\left(x^3+3x^2+9\right)^4}\,\mathrm{d}x$.

```
> u:=x->x^3+3*x^2+9; du:=D(u); f:=u->1/u^4;
```

$$u := x \rightarrow x^3+3\,x^2+9 \tag{5.1.7}$$

$$du := x \rightarrow 3\,x^2+6\,x$$

$$f := u \rightarrow \frac{1}{u^4}$$

```
> fux:=f(u(x))*du(x);
```

$$fux := \frac{3\,x^2+6\,x}{\left(x^3+3\,x^2+9\right)^4} \tag{5.1.8}$$

```
> cnst:=simplify(((x^2+2*x)/(x^3+3*x^2+9)^4)/fux);
```

$$cnst := \frac{1}{3} \tag{5.1.9}$$

```
> Fx:=int(f(u),u);
```

$$Fx := -\frac{1}{3}\,\frac{1}{u^3} \tag{5.1.10}$$

```
> Asw:=cnst*eval(Fx, u=u(x));
```

$$Asw := -\frac{1}{9}\,\frac{1}{\left(x^3+3\,x^2+9\right)^3} \tag{5.1.11}$$

The result obtained by the direct evaluation is

```
> Asws:=int((x^2+2*x)/(x^3+3*x^2+9)^4,x);
```

$$Asws := -\frac{1}{9}\frac{1}{\left(x^3+3x^2+9\right)^3} \quad (5.1.12)$$

Example 3. Evaluate $\displaystyle\int_0^{\frac{\pi}{4}} \tan(x)^3 \sec(x)^2 \, dx$.

```
> Asw:=int(tan(x)^3*sec(x)^2,x=0..Pi/4);
```

$$Asw := \frac{1}{4} \quad (5.1.13)$$

If you apply the substitution method, the procedure is the following:

```
> u:=x->tan(x); du:=D(u); f:=u->u^3;
```

$$u := x \rightarrow \tan(x) \quad (5.1.14)$$

$$du := x \rightarrow 1 + \tan(x)^2$$

$$f := u \rightarrow u^3$$

```
> cnst:=simplify( (tan(x)^3*sec(x)^2)/(f(u(x))*du(x)) );
```

$$cnst := 1 \quad (5.1.15)$$

```
> ua:=u(0); ub:=u(Pi/4);
```

$$ua := 0 \quad (5.1.16)$$

$$ub := 1$$

```
> asw:=int(f(u),u=ua..ub);
```

$$asw := \frac{1}{4} \quad (5.1.17)$$

Example 4. Evaluate $\displaystyle\int_1^3 \frac{x}{\left(x^2+1\right)^3} \, dx$.

```
> f:=x->x/(x^2+1)^3;
```

$$f := x \rightarrow \frac{x}{\left(x^2+1\right)^3} \quad (5.1.18)$$

```
> Asw:=int(f(x),x=1..3);
```

$$Asw := \frac{3}{50} \quad (5.1.19)$$

Exercises

In Exercises 1−4 evaluate the integral (1) by the substitution method and (2) by direct evaluation.

1. $\int x^3 \cos(x^4)\, dx$

2. $\int \frac{x+1}{(x^2+2x)^3}\, dx$

3. $\int_0^1 \frac{x}{(x^2+1)^3}\, dx$

4. $\int_0^{\frac{\pi}{2}} \cos(x)^3 \sin(x)\, dx$

In Exercises 5−8 evaluate the integral directly.

5. $\int \frac{x^3}{(x^4+1)^4}\, dx$

6. $\int \sec(x)^2 \tan(x)\, dx$

7. $\int_0^4 x\sqrt{x^2+9}\, dx$

8. $\int_0^{\frac{\pi}{4}} \tan(x)^2 \sec(x)^2\, dx$

Chapter 6 APPLICATIONS OF INTEGRAL

6.1 Area between Two Curves

(1) The area between two curves $f(x)$ and $g(x)$ ($g(x) \le f(x)$) over $[a, b]$ is

$$A = \int_a^b f(x) - g(x)\,dx.$$

(2) Similarly, the area between two curves $h(x)$ and $l(y)$ ($l(y) \le h(y)$) over $[c, d]$ is

$$A = \int_c^d h(y) - l(y)\,dy\,.$$

6.1.1 Calculate the area between two curves $f(x)$ and $g(x)$

Example 1. Calculate the area of the region between the graphs of $f(x) = x^2 - 4x + 10$ and $g(x) = 4x - x^2$ over [1, 3].

```
> f:=x->x^2-4*x+10; g:=x->4*x-x^2;
```

$$f := x \to x^2 - 4\,x + 10$$

$$g := x \to 4\,x - x^2 \quad (1.1.1)$$

```
> plot([f,g],1..3, color=[blue,green]);
```



From the graphs, we can see that $g(x) < f(x)$.

```
> Areafg:=int(f(x)-g(x),x=1..3);
```

$$Areafg := \frac{16}{3} \quad (1.1.2)$$

Example 2. Calculate the area of the region between the graphs of $f(x) = x^2 - 5x - 7$ and $g(x) = x - 12$ over [−2, 5].

```
> f:=x->x^2-5*x-7; g:=x->x-12;
```

$$f := x \to x^2 - 5\,x - 7 \quad (1.1.3)$$

$$g := x \to x - 12$$

```
> plot([f,g],-2..5,color=[red,green]);
```




```
> solve(f(x)=g(x),x);
```

$$5, 1 \qquad (1.1.4)$$

From the graphs, we find that $g(x) < f(x)$ over $[-2, 1]$, and $f(x) < g(x)$ over $[1, 5]$. Hence the area is:

```
> Areafg:=int(f(x)-g(x), x=-2..1)+int(g(x)-f(x), x=1..5);
```

$$Areafg := \frac{113}{3} \qquad (1.1.5)$$

Remark: You can use the formula

$$\text{Area} = \int_a^b |f(x) - g(x)| \, dx.$$

Therefore, **Example 2** can also be solved by the following:

```
> Areafg:=int(abs(f(x)-g(x)),x=-2..5);
```

$$Areafg := \frac{113}{3} \qquad (1.1.6)$$

Example 3. Find the area of the region bounded by the graphs of $y = \frac{8}{x^2}$, $y = 8x$, and $y = x$.

Step 1. Draw the graph of the region.

```
> f:=x->8/x^2; g:=x->8*x; h:=x->x;
```

$$f := x \to \frac{8}{x^2} \qquad (1.1.7)$$

$$g := x \to 8\,x$$

$$h := x \to x$$

```
> plot([f(x),g(x),h(x)], x=0..3, 0..10, color=[red, green, blue]
  );
```



Step 2. Find all intersections. It is clear that $x = 0$ is an intersection.

```
> solve(f(x)-g(x),x);
```

$$1,\ -\frac{1}{2} + \frac{1}{2}\,\mathrm{I}\sqrt{3},\ -\frac{1}{2} - \frac{1}{2}\,\mathrm{I}\sqrt{3} \qquad (1.1.8)$$

```
> solve(f(x)=h(x),x);
```

$$2,\ -1 + \mathrm{I}\sqrt{3},\ -1 - \mathrm{I}\sqrt{3} \qquad (1.1.9)$$

The intersection of $f(x)$ and $g(x)$ is $x = 1$ and the intersection of $f(x)$ and $h(x)$ is $x = 2$.
Step 3. Set up the integral and evaluate.

```
> Ar:=int(g(x)-h(x),x=0..1)+int(f(x)-h(x),x=1..2);
```

$$Ar := 6 \qquad (1.1.10)$$

6.1.2 Calculate the area between two curves $h(y)$ and $l(y)$

Example 4. Calculate the area between the graphs of $h(y) = y^3 - 4y$ and $l(y) = y^3 + y^2 + 8$, for $-2 \le y \le 2$.

```
> h:=y->y^3-4*y; l:=y->y^3+y^2+8;
```

$$h := y \to y^3 - 4\,y \qquad (1.2.1)$$

$$l := y \to y^3 + y^2 + 8$$

```
> with(plots):
> implicitplot([x=h(y), x=l(y)], x=-10..20,y=-2..2, color=[red, blue]);
```



```
> Area:=int(l(y)-h(y),y=-2..2);
```

$$Area := \frac{112}{3} \tag{1.2.2}$$

You can also do the following without drawing the graph:

```
> Area:=int(abs(h(y)-l(y)),y=-2..2);
```

$$Area := \frac{112}{3} \tag{1.2.3}$$

Exercises

1. Calculate the area of the region between the graphs of $f(x) = 3x^2 + 12$ and $g(x) = 4x + 4$ over $[-3, 3]$.
2. Calculate the area of the region between the graphs of $f(x) = 2 - x^2$ and $g(x) = -2$ over $[-2, 2]$.
3. Calculate the area of the region between the graphs of $f(x) = x^3 - 2x^2 + 10$ and $g(x) = 3x^2 + 4x - 10$.
4. Find the area of the region enclosed by the curves of $y = x^3 - 6x$ and $y = 8 - 3x^2$.
5. Find the area of the region enclosed by the curves of $x = y^2 - 5$ and $x = 3 - y^2$.
6. Find the area of the region enclosed by the curves of $x = y^3 - 18y$ and $y + 28x = 0$.

6.2 Setting Up Integrals: Volume, Density, Average Value

6.2.1 Volume

Volume of a solid body. Assume that a solid body extends from height $y = a$ to $y = b$. Let $A(x)$ be the area of the horizontal cross section at height y. Then the volume of the solid body is

$$V = \int_a^b A(y)\mathrm{d}y$$

Example 1. The base of a solid is the region between the x-axis and the inverted parabola $y = 4 - x^2$. The vertical cross sections of the solid perpendicular to the y-axis are semicircles. Compute the volume of the solid. Its graph is shown in **FIGURE 1.**

Note: The following figure is used for illustrating the graph of the bogy in Example 1. Drawing a surface is not required in this section.

```
> plot3d(4-x^2-y, x=-2..2, y=0..4, view=0..4, axes=BOXED, title=
  "FIGURE 1");
```

FIGURE 1



For a fixed y, the area of the cross section is the semicircle with radius $r = \sqrt{4-y}$ so that its area is $\frac{\pi r^2}{2} = \frac{\pi(4-y)}{2}$. Therefore,

```
> Ay:=Pi*(4-y)/2;
```

$$Ay := \frac{1}{2}\pi(4-y) \tag{2.1.1}$$

```
> V:=int(Ay,y=0..4);
```

$$V := 4\pi \tag{2.1.2}$$

Example 2. Find the volume of a sphere with radius r.

```
> Ay:=Pi*(r^2-y^2);
```

$$Ay := \pi\left(r^2 - y^2\right) \tag{2.1.3}$$

```
> V:=int(Ay,y=-r..r);
```

$$V := \frac{4}{3}\pi r^3 \tag{2.1.4}$$

6.2.2 Density

(1) Let the linear mass density of a rod be $\rho(x)$ over an interval [a, b]. Then the total mass of the rod is

$$M = \int_a^b \rho(x)\mathrm{d}x$$

(2) Let the radius mass density of a disk with radius R be $\rho(x)$. Then the total mass of the disk is

$$M = 2\pi \int_0^R r\rho(r)\mathrm{d}r$$

Example 3. Find the total mass of a 2-m rod of linear density $\rho(x) = 1 + x(2-x)$ kg/m, where x is the distance from one end of the rod.

```
> Dnsty:=x->1+x*(2-x);
```

$$Dnsty := x \rightarrow 1 + x\,(2-x) \tag{2.2.1}$$

```
> M:=int(Dnsty(x),x=0..2);
```

$$M := \frac{10}{3} \qquad (2.2.2)$$

Hence, the total mass of the rod is 10/3 kg.

Example 4. Find the total mass of a circular plate of radius 20 cm whose mass density is the radial function $\rho(r) = 0.03 + 0.01\cos\left(\pi r^2\right)$ g/cm^2.

```
> Dnsty:=r->0.03+0.01*cos(Pi*r^2);
```

$$Dnsty := r \rightarrow 0.03 + 0.01\cos\left(\pi r^2\right) \qquad (2.2.3)$$

```
> M:=2*Pi*int(r*Dnsty(r),r=0..20);
```

$$M := 12.\ \pi \qquad (2.2.4)$$

6.2.3 Flow rate

Flow rate Q is the volume per unit time of fluid passing through the tube. Let the radius of the pipe be R, the velocity of the liquid travel be $v(r)$, where r is the distance of the particles of the liquid from the center of the tube. Then the flow rate has a formula similar to the total mass of a disk:

$$Q = 2\pi \int_0^R r v(r) \mathrm{d}r$$

Example 5. According to Poiseuille's Law, the velocity of blood flowing in a blood vessel of radius R cm is $v(r) = k\left(R^2 - r^2\right)$, where r is the distance from the center of the vessel and k is a constant. Calculate the flow rate Q as the function of R, assuming that k = 0.5 (cm-s)$^{-1}$.

```
> v:=r->(R^2-r^2)/2;
```

$$v := r \rightarrow \frac{1}{2} R^2 - \frac{1}{2} r^2 \qquad (2.3.1)$$

```
> Q:=2*Pi*int(r*v(r),r=0..R);
```

$$Q := \frac{1}{4} \pi R^4 \qquad (2.3.2)$$

Hence, $Q = \frac{\pi R^4}{4}$ cm^3/s.

6.2.4 Average value

The average value of an integrable function $f(x)$ on $[a, b]$ is

$$AVf = \frac{1}{b - a} \int_a^b f(x) \mathrm{d}x$$

Example 6. Find the average value of $f(x) = \sin(x)$ on $[0, \pi]$.

```
> Avf:=int(sin(x),x=0..Pi)/Pi;
```

(2.4.1)

$$Avf := \frac{2}{\pi} \qquad (2.4.1)$$

Exercises

1. The base of a pyramid is a square of side 8, and its height is 20. Compute the volume of the pyramid.
2. Develop the formula for the volume of a right circular cone of height H and with a radius of the base circle R.
3. Find the total mass of a 1-m rod of linear density $\rho(x) = 10(x+1)^2$ kg/m, where x is the distance from one end of the rod.
4. Find the total population within a 10-mile radius of the city center if the radial population density is $\rho(r) = 4\left(1+r^2\right)^{\frac{1}{3}}$.
5. Find the total mass of a circular plate of radius 2 cm with radial mass density $\rho(r) = \frac{4}{r}$ g/cm.
6. Find the flow rate through a tube of radius 4 cm, assuming that the velocity of fluid particles at a distance r cm from the center is $v(r) = 16 - r^2$ cm/s.
7. Which of $f(x) = x^2 \sin(x)$ and $g(x) = x\sin(x)$ has a large average value over [0, π]?
8. Find the average value of $f(x)$ over the given interval.
   (a) $f(x) = x^3$, $[0, 1]$.
   (b) $f(x) = \sec(x)^2$, $\left[0, \frac{\pi}{2}\right]$.
   (c) $f(x) = x^n$, $[0, 1]$.

6.3 Volumes of Revolution

Volume of a side of revolution: Disk method. If $f(x)$ is continuous and $f(x) \geq 0$ on $[a, b]$, then the volume obtained by rotating the region under the graph about the x-axis is

$$V = \pi \int_a^b f(x)^2 \, dx$$

If the rotated region is between two curves $f(x)$ and $g(x)$, then the volume of revolution is

$$V = \pi \int_a^b f(x)^2 - g(x)^2 dx$$

6.3.1 Compute the volumes of revolution

Example 1. Calculate the volume V of the solid obtained by rotating the region under $y = x^2$ about the x-axis over [0, 2].

```
> f:=x->x^2;
```

$$f := x \rightarrow x^2 \qquad (3.1.1)$$

```
> V:=Pi*int(f(x)^2,x=0..2);
```

$$V := \frac{32}{5}\pi \qquad (3.1.2)$$

Example 2. Find the volume V of the solid obtained by rotating the region between $y=x^2+4$ and $y=2$ about the x-axis over [1, 3].

```
> f:=x->x^2+4: g:=x->2:
> plot([f,g], 1..3, color=[red,green]);
```




```
> V:=Pi*int(f(x)^2-g(x)^2,x=1..3);
```

$$V := \frac{2126}{15}\pi \qquad (3.1.3)$$

Note: The volume of revolution about a horizontal line other than the* x*-axis, or about a vertical line, can be computed in a similar way.

Example 3. Find the volume V of the solid obtained by rotating the region between $y=x^2+2$ and $y=4-x^2$ about the horizontal line $y=-3$.

```
> f:=x->x^2+2: g:=x->4-x^2:
> plot([f(x),g(x),-3],x=-2..2,color=[red,green,blue]);
```



Find the intersections of $f(x)$ and $g(x)$.

```
> solve(f(x)-g(x)=0,x);
```

$$1, -1 \qquad (3.1.4)$$

Note that $f(x) < g(x)$ on $(-1, 1)$. Since the distance of $f(x)$ from $y=-3$ is $f(x)+3$ and the distance of $g(x)$ from $y=-3$ is $g(x)+3$, the formula of the volume of revolution here needs to be modified to the following:

```
> V:=Pi*int((g(x)+3)^2-(f(x)+3)^2, x=-1..1);
```

$$V := 32\pi \qquad (3.1.5)$$

Example 4. Find the volume V of the solid obtained by rotating the region between $y=x^3$ and $y=x^{\frac{1}{3}}$ about the y-axis.

```
> x:='x':
> f:=x->x^3: g:=x->x^(1/3):
> plot([f(x),g(x)], x=0..1.2, color=[red,green]);
```




```
> solve (f(x)-g(x)=0,x);
```

$$-\frac{1}{2}\sqrt{2}-\frac{1}{2}\,\mathrm{I}\sqrt{2},\,1,\,-\frac{1}{2}\sqrt{2}+\frac{1}{2}\,\mathrm{I}\sqrt{2},\,0 \qquad (3.1.6)$$

The real solutions are $x=0$, and $x=1$. Write these curves as a function of y:

```
> hx:=y^(1/3): ly:=y^3:
> V:=Pi*int(hx^2-ly^2,y=0..1);
```

$$V := \frac{16}{35}\,\pi \qquad (3.1.7)$$

Exercises

1. Calculate the volume V of the solid obtained by rotating the region under $y=\sqrt{x+1}$ about the x-axis over [0, 3].
2. Calculate the volume V of the solid obtained by rotating the region under $y=1+\cos(x)$ about the x-axis over $[0, \pi]$.
3. Find the volume V of the solid obtained by rotating the region between $y=x^2+2$ and $y=10-x^2$ about the x-axis.
4. Find the volume V of the solid obtained by rotating the region enclosed by $y=x^2$, $y=12-x$, and $x=0$ about the horizontal line $y=-2$.
5. Find the volume V of the solid obtained by rotating the region between $y=\frac{9}{x^2}$ and $y=10-x^2$ about the horizontal line $y=12$.
6. Find the volume V of the solid obtained by rotating the region between $y=\frac{1}{x}$ and $y=\frac{5}{2}-x$ about the y-axis.

6.4 The Method of Cylindrical Shells

Volume of a side of revolution: Shell method. If $f(x)$ is continuous and $f(x) \geq 0$ on $[a, b]$, then the volume obtained by rotating the region under the graph about the y-axis is

$$V = 2\pi \int_a^b x f(x) \mathrm{d}x$$

If the rotated region is between two curves $f(x)$ and $g(x)$, then the volume of revolution about the y-axis is

$$V = 2\pi \int_a^b x |f(x) - g(x)| \mathrm{d}x$$

You can easily develop the formula for $h(y)$ (or $h(y)$ and $l(y)$) rotating about x-axis.

6.4.1 Calculating volume of revolution by shell method

Example 1. Calculate the volume V of the solid obtained by rotating the region under $f(x) = 1 - 2x + 3x^2 - 2x^3$ over [0, 1] about the y-axis.

```
> f:=x->1-2*x+3*x^2-2*x^3;
```

$$f := x \rightarrow 1 - 2x + 3x^2 - 2x^3 \quad (4.1.1)$$

```
> V:=2*Pi*int(x*f(x),x=0..1);
```

$$V := \frac{11}{30}\pi \quad (4.1.2)$$

Example 2. Calculate the volume V of the solid obtained by rotating the region enclosed by the graphs of $f(x) = 9 - x^2$ and $g(x) = 9 - 3x$ about the y-axis.

```
> f:=x->9-x^2: g:=x->9-3*x:
> plot([f,g],-1..4,color=[red,green]);
```




```
> solve(f(x)-g(x),x);
```

$$0, 3 \quad (4.1.3)$$

```
> V:=2*Pi*int(x*(f(x)-g(x)),x=0..3);
```

(4.1.4)

$$V := \frac{27}{2}\pi \tag{4.1.4}$$

Example 3. Calculate the volume V of the solid obtained by rotating the region under $f(x) = 9 - x^2$ over [0, 3] about the x-axis.

```
> f:=x->9-x^2:
```

Method 1: Disk method.

```
> V:=Pi*int(f(x)^2,x=0..3);
```

$$V := \frac{648}{5}\pi \tag{4.1.5}$$

Method 2: Shell method.

```
> h:=y->sqrt(9-y);
```

$$h := y \rightarrow \sqrt{9-y} \tag{4.1.6}$$

```
> V:=2*Pi*int(y*h(y), y=0..9);
```

$$V := \frac{648}{5}\pi \tag{4.1.7}$$

Exercises

1. Calculate the volume V of the solid obtained by rotating the region under $f(x) = x^2$ over [0, 1] about the y-axis.
2. Calculate the volume V of the solid obtained by rotating the region under $f(x) = \frac{x}{\sqrt{1+x^3}}$ over [1, 4] about the y-axis.
3. Calculate the volume V of the solid obtained by rotating the region enclosed by the graphs of $f(x) = x^2$ and $g(x) = \sqrt{x}$ about the y-axis.
4. Calculate the volume V of the solid obtained by rotating the region enclosed by the graphs of $f(x) = 8 - x^2$, $y = x^2$ and $x = 0$ about the y-axis.
5. Use the shell method to calculate the volume V of the solid obtained by rotating the region under $f(x) = 9 - x^2$ over [0, 3] about the x-axis.
6. Use the shell method to calculate the volume V of the solid obtained by rotating the region under $f(x) = \frac{1}{x}$ over [1, 4] about the x-axis.

6.5 Work

Computing work:

(1) The work performed in moving an object along the x-axis from a to b applying a force of magnitude $F(x)$ is

$$W = \int_a^b F(x)\,dx$$

(2) The work against gravity: Assume that the work performed lifting the weight of the object at a layer y (with the unit height) is $W(y)$. Then the total work is

$$W = \int_c^d W(y)\,dy$$

where [*c, d*] is the range of the layers.

6.5.1 Computing work

Example 1. Assume that the spring constant is k = 150 kg/s^2. Compute the work (joules) required to stretch the spring from equilibrium to 12 cm past equilibrium.

```
> F:=x-> 150*x;
```

$$F := x \rightarrow 150\,x \quad (5.1.1)$$

Recall that 12 cm = 0.12 m.

```
> W:=int(F(x),x=0..0.12);
```

$$W := 1.080000000 \quad (5.1.2)$$

Hence, the work is 1.08 joules.

Example 2. A spherical tank of radius R meters with a small hole at the top is filled with water. How much work is done pumping the water out through the hole, assuming the density of water is 1,000 kg/m^3?
Hint: The key is finding the work performed lifting the weight of the object at a layer* y*. We set the range of the layers to* [−R, R]*, assuming the zero-layer of the water through the center of the sphere.

```
> F:=y-> 9.8*Pi*(R^2-y^2); d:=y->R-y;
```

$$F := y \rightarrow 9.8\,\pi\left(R^2 - y^2\right) \quad (5.1.3)$$

$$d := y \rightarrow R - y$$

```
> dW:=F*d;
```

$$dW := F\,d \quad (5.1.4)$$

```
> W:=int(dW(y), y=-R..R);
```

$$W := 41.05014401\,R^4 \quad (5.1.5)$$

Hence, the work is 41.05R^4 joules.

Exercises

1. Assume that the spring constant is k = 120 kg/s^2. Compute the work (joules) required to compress the spring from equilibrium to 4 cm past equilibrium.
2. Assume that the spring constant is k = 150 kg/s^2. Compute the work (joules) required to stretch the spring from the current position to 4 cm, assuming that it is already compressed 5 cm.
3. A semi-spherical tank of radius 10 meters with a small spout at the top is filled with water. How much work is done pumping the water out through the spout, assuming the density of water is 1,000 kg/m^3?
4. A conical tank 10 meters high with a radius of 5 meters at its top, and having a small hole at the top, is filled with water. How much work is done pumping the water out through the hole, assuming the density of water is 1,000 kg/m^3?

Chapter 7 THE EXPONENTIAL FUNCTION

7.1 Derivative of $f(x) = b^x$ and the Number *e*

(1) The function $f(x) = b^x$ is called an **exponential function,** where $b > 0$ and b is not equal to 1.
(2) If the base $b = e$, the function is called the natural exponential function, which in MAPLE is denoted by $f(x) = \mathrm{e}^x$, (not by e^x).

7.1.1 Exponential functions

Example 1. Solve for the unknowns: (a) $2^{3x+1} = 2^5$; (b) $b^3 = 5^6$; (c) $7^{t+1} = \left(\frac{1}{7}\right)^{2t}$.

```
> solve(2^(3*x+1)=2^5,x);
```

$$\frac{4}{3} \qquad (1.1.1)$$

```
> solve(b^3=5^6,b);
```

$$25, -\frac{25}{2} + \frac{25}{2} \mathrm{I}\sqrt{3}, -\frac{25}{2} - \frac{25}{2} \mathrm{I}\sqrt{3} \qquad (1.1.2)$$

The real solution of b is 25.

```
> solve(7^(t+1)=(1/7)^(2*t));
```

$$\frac{-1}{3} \qquad (1.1.3)$$

Example 2. Find $\lim_{h \to 0} \frac{\mathrm{e}^h - 1}{h}$.

```
> limit((exp(h)-1)/h,h=0);
```

$$1 \qquad (1.1.4)$$

Example 3. Evaluate numerically the number e.

```
> evalf(exp(1));
```

$$2.718281828 \qquad (1.1.5)$$

Example 4. Evaluate numerically the number e to 40 places.

```
> evalf[41](exp(1));
```

$$2.7182818284590452353602874713526624977572 \qquad (1.1.6)$$

7.1.2 Derivative of $f(x) = b^x$

Example 5. Find the derivative of $f(x) = b^x$.

```
> f:=x->b^x;
```

$$f := x \to b^x \qquad (1.2.1)$$

```
> df:=D(f);
```

$$df := x \to b^x \ln(b) \qquad (1.2.2)$$

Example 6. Find the derivative of $f(x) = e^{\tan(x)}$.

```
> f:=x->exp(tan(x));
```

$$f := x \to e^{\tan(x)} \tag{1.2.3}$$

```
> df:=D(f);
```

$$df := x \to \left(1 + \tan(x)^2\right) e^{\tan(x)} \tag{1.2.4}$$

Example 7. Find the critical points of $f(x) = x^2 e^x$ and determine whether they are local minima, maxima, or neither.

```
> f:=x->x^2*exp(x);
```

$$f := x \to x^2 e^x \tag{1.2.5}$$

```
> solve(D(f)(x),x);
```

$$0, -2 \tag{1.2.6}$$

```
> [D(D(f))(0), D(D(f))(-2)];
```

$$\left[2, -2 e^{(-2)}\right] \tag{1.2.7}$$

```
> [f(0),f(-2)];
```

$$\left[0, 4 e^{(-2)}\right] \tag{1.2.8}$$

Hence, $f(x) = x^2 e^x$ has a local minimum 0 at $x = 0$ and has a local maximum $4e^{-2}$ at $x = -2$. This can be verified by its graph.

```
> plot(f,-3..1);
```



7.1.3 Integrals involving e^x

Example 8. Evaluate the integrals (1) $\int e^{7x}\,dx$; (2) $\int e^{2x^2}\,dx$; (3) $\int \frac{e^t}{1 + 2e^t + e^{2t}}\,dt$.

```
> int(exp(7*x),x);
```

$$\frac{1}{7} e^{(7x)} \tag{1.3.1}$$

```
> int(x*exp(2*x^2),x);
```

$$\frac{1}{4} e^{(2x^2)} \tag{1.3.2}$$

```
> int(exp(t)/(1+2*exp(t)+exp(2*t)),t);
```

$$-\frac{1}{e^t+1} \qquad (1.3.3)$$

Example 9. Find the area between $y=e^x$ and $y=e^{2x}$ over [0, 1].

```
> plot([exp(x),exp(2*x)], x=0..1, color=[blue, green]);
```




```
> Ar:=int(exp(2*x)-exp(x), x=0..1);
```

$$Ar := \frac{1}{2}e^2 - e + \frac{1}{2} \qquad (1.3.4)$$

```
> evalf(Ar);
```

$$1.476246222 \qquad (1.3.5)$$

Exercises

1. Solve for the unknowns: (a) $9^{2x}=9^5$; (b) $3^x=\left(\frac{1}{3}\right)^{x+1}$; (c) $e^{4t-3}=e^{t^2}$.

2. Find $\lim_{h\to 0}\frac{e^h-1-\frac{h^2}{2}}{h^3}$.

3. Evaluate numerically the number e^2 to 30 places.

4. Find the derivative of $f(x)=e^{\pi x}$.

5. Find the derivative of $f(x)=e^{\sin(x)}$.

6. Find the critical points of $f(x)=\frac{e^x}{x^2+1}$ and determine whether they are local minima, maxima, or neither.

7. Evaluate the integrals (a) $\int e^x\sqrt{e^x+1}\,dx$; (b) $\int x e^{-4x^2}dx$; (c) $\int\frac{e^t}{\sqrt{1+e^t}}dt$.

8. Find the area between $y=e^x$ and $y=e^{-x}$ over [0, 2].

7.2 Inverse Functions

(1) Let $f(x)$ have the domain D and range R. The inverse function $f^{-1}(x)$ is the function with the domain R such that $f(f^{-1}(x))= x$ on D and $f^{-1}(f(x)) = x$ on R.
(2) Let g be the inverse of f. Then $g'(b) = 1/f'(g(b))$.

7.2.1 Find the inverse

Example 1. Find the inverse of $f(x) = 2x - 18$.

```
> invf:=solve(x=2*y-18,y);
```

$$invf := \frac{1}{2}\,x + 9 \qquad (2.1.1)$$

Example 2. Find the inverse of $f(x) = \dfrac{3x+2}{5x-1}$.

```
> invf:=solve(x=(3*y+2)/(5*y-1),y);
```

$$invf := \frac{2+x}{5\,x-3} \qquad (2.1.2)$$

Example 3. Sketch the graph of the function $f(x) = \sqrt{4-x}$ together with its inverse.

```
> f:=x->sqrt(4-x);
```

$$f := x \rightarrow \sqrt{4-x} \qquad (2.1.3)$$

```
> g:=solve(x=f(y),y);
```

$$g := 4 - x^2 \qquad (2.1.4)$$

```
> plot([f(x),g], x=0..4, 0..4, scaling=constrained);
```



7.2.2 Derivative of the inverse

Example 4. Calculate $g'(1)$, where $g(x)$ is the inverse of $f(x) = x + e^x$.
Step 1. Solve x for $f(x) = 1$.

```
> f:=x->x+exp(x);
```

$$f := x \to x + e^x \tag{2.2.1}$$

```
> g1:=solve(f(x)=1,x);
```

$$g1 := 0 \tag{2.2.2}$$

Step 2. Find the derivative of $f(x)$.

```
> df:=D(f);
```

$$df := x \to 1 + e^x \tag{2.2.3}$$

Step 3. Find $g'(1)$ by the formula $g'(1) = 1/f'(g(1))$.

```
> dg1:=1/df(g1);
```

$$dg1 := \frac{1}{2} \tag{2.2.4}$$

Example 5. Find $g'(x)$, where $g(x)$ is the inverse of $f(x) = \dfrac{x}{1+x}$.

Method 1.

```
> f:=x->x/(1+x);
```

$$f := x \to \frac{x}{x+1} \tag{2.2.5}$$

```
> g:=solve(x=f(y),y);
```

$$g := -\frac{x}{x-1} \tag{2.2.6}$$

```
> dg:=simplify(diff(g,x));
```

$$dg := \frac{1}{(x-1)^2} \tag{2.2.7}$$

Method 2.

```
> df:=D(f);
```

$$df := x \to \frac{1}{x+1} - \frac{x}{(x+1)^2} \tag{2.2.8}$$

```
> dg:=simplify(1/df(g));
```

$$dg := \frac{1}{(x-1)^2} \tag{2.2.9}$$

Exercises

1. Find the inverse of $f(x) = 7x - 4$.
2. Find the inverse of $f(x) = \frac{x-2}{x+3}$.
3. Sketch the graphs of the function $f(x) = \frac{1}{x+1}$ together with its inverse for x in [0, 4] and y in [0, 4].
4. Calculate $g'(3)$, where $g(x)$ is the inverse of $f(x) = x^7 + x + 1$.
5. Find $g'(x)$, where $g(x)$ is the inverse of $f(x) = \sqrt{3-x}$.

7.3 Logarithms and Their Derivatives

7.3.1 Logarithms

Example 1. Solve $1000e^{0.35t} = 5000$.

```
> fsolve(1000*exp(0.35*t)=5000,t);
```

$$4.598394036 \tag{3.1.1}$$

Note: "**fsolve**" is used for solving equations numerically.

Example 2. Solve $\ln(x^2+1) - 3\ln(x) = \ln(2)$.

```
> solve(ln(x^2+1)-3*ln(x)=ln(2),x);
```

$$1 \tag{3.1.2}$$

7.3.2 Calculus of logarithms

Example 3. Find the derivative of (a) $f(x) = 4^{3x}$; (b) $f(x) = 5^{x^2}$.

```
> aA:=diff(4^(3*x),x);
```

$$aA := 6\, 4^{(3x)} \ln(2) \tag{3.2.1}$$

```
> bA:=diff(5^(x^2),x);
```

$$bA := 2\, 5^{(x^2)}\, x \ln(5) \tag{3.2.2}$$

Example 4. Find the derivative of (a) $y = x\ln(x)$; (b) $y = \ln(x)^2$.

```
> aA:=diff(x*ln(x), x);
```

$$aA := \ln(x) + 1 \tag{3.2.3}$$

```
> bA:=diff(ln(x)^2,x);
```

$$bA := \frac{2\ln(x)}{x} \tag{3.2.4}$$

Logarithmic differentiation: If $f(x)$ is in a product form, or a quotient form, or a power form, then $f'(x) = f(x)\frac{d}{dx}\ln(f(x))$.

Example 5. Find the derivative of $f(x) = \frac{(x+1)^2(2x^2-3)}{\sqrt{x^2+1}}$.

```
> f:=(x+1)^2*(2*x^2-3)/sqrt(x^2+1);
```

$$f := \frac{(x+1)^2\left(2x^2-3\right)}{\sqrt{x^2+1}} \qquad (3.2.5)$$

Method 1: Logarithmic differentiation.

```
> lnf:= eval(ln(u),u=f);
```

$$lnf := \ln\left(\frac{(x+1)^2\left(2x^2-3\right)}{\sqrt{x^2+1}}\right) \qquad (3.2.6)$$

```
> df:=f*diff(lnf,x);
```

$$df := \frac{2\,(x+1)\left(2x^2-3\right)}{\sqrt{x^2+1}} + \frac{4\,(x+1)^2 x}{\sqrt{x^2+1}} - \frac{(x+1)^2\left(2x^2-3\right)x}{\left(x^2+1\right)^{(3/2)}} \qquad (3.2.7)$$

Method 2: Direct differentiation.

```
> df:=diff(f,x);
```

$$df := \frac{2\,(x+1)\left(2x^2-3\right)}{\sqrt{x^2+1}} + \frac{4\,(x+1)^2 x}{\sqrt{x^2+1}} - \frac{(x+1)^2\left(2x^2-3\right)x}{\left(x^2+1\right)^{(3/2)}} \qquad (3.2.8)$$

Example 6. Evaluate (a) $\int_1^3 \frac{x}{1+x}dx$; (b) $\int \tan(x)dx$.

```
> int(x/(1+x),x=1..3);
```

$$-\ln(2)+2 \qquad (3.2.9)$$

```
> int(tan(x),x);
```

$$-\ln(\cos(x)) \qquad (3.2.10)$$

Exercises

1. Solve $7e^{5t} = 100$.
2. Solve $\ln\left(x^4\right) - \ln\left(x^2\right) = 2$.
3. Find the derivative of (a) $f(x) = 4^x$ at $x = 2$; (b) $f(x) = 5^{x^2-2x+9}$ at $x = 1$.
4. Find the derivative of (a) $y = x\ln(2x)$; (b) $y = \ln\left(x^2\right) + \ln(x)^2$.
5. Find the derivative of $f(x) = \frac{(x+1)^3 x}{\sqrt{x+1}}$.
6. Evaluate (a) $\int_1^3 \frac{1}{2+x}dx$; (b) $\int \frac{\cos(x)}{2\sin(x)+3}dx$.

7.4 Exponential Growth and Decay

(1) **Exponential Model** is the following:

$$P(t) = P_0 e^{kt}$$

where P_0 is the initial quantity. When $k > 0$, it is an exponential growth model and k is called the growth constant (or the growth rate); when $k < 0$, it is an exponential decay model and $|k|$ is called the decay constant (or the decay rate).
(2) An exponential growth (or decay) model satisfies the following differential equation:

$$y' = ky.$$

(3) **Double time** for an exponential growth model:

$$\text{Double time} = \frac{\ln(2)}{k}.$$

(4) **Half-life** for an exponential decay model:

$$\text{Half-life} = \frac{\ln(2)}{|k|}.$$

7.4.1 Exponential growth and decay

Example 1. A certain bacteria population P obeys the exponential growth law $P(t) = 2000\mathrm{e}^{1.3t}$ (t in hours). (a) How many bacteria are present initially? (b) At what time will there be 10000 bacteria?

```
> p:=t->2000*exp(1.3*t);
```

$$p := t \rightarrow 2000\,\mathrm{e}^{(1.3\,t)} \tag{4.1.1}$$

```
> p0:=2000;
```

$$p0 := 2000 \tag{4.1.2}$$

```
> fsolve(p(t)=10000, t);
```

$$1.238029163 \tag{4.1.3}$$

Hence, (a) 2000 bacteria are present initially. (b) There will be 10000 bacteria after 1.238 hours.

Example 2. Find the decay constant of Radium-226, given that its half-life is 1622 years.

```
> k:=-fsolve(ln(2)/r=1622,r);
```

$$k := -0.0004273410484 \tag{4.1.4}$$

Example 3. The decay constant of Cobalt-60 is 0.13 years^{-1}. What is its half-life ?

```
> HT:=evalf( ln(2)/0.13);
```

$$HT := 5.331901389 \tag{4.1.5}$$

Hence, its half-life is 5.3319 years.

Example 4. Find the solution $y' = \sqrt{2}\, y$ satisfying $y(0) = 20$.

```
> deq:= diff(y(x),x)=sqrt(2)*y(x); inc:=y(0)=20;
```

$$deq := \frac{d}{dx}\, y(x) = \sqrt{2}\; y(x)$$

$$inc := y(0) = 20 \tag{4.1.6}$$

```
> dsolve({deq, inc}, y(x));
```

$$y(x) = 20\, e^{\left(\sqrt{2}\, x\right)} \tag{4.1.7}$$

Example 5. Find the solution $y' = 3y$ satisfying $y(2) = 4$.

```
> deq:= diff(y(x),x)=3*y(x); inc:=y(2)=4;
```

$$deq := \frac{d}{dx}\, y(x) = 3\, y(x) \tag{4.1.8}$$

$$inc := y(2) = 4$$

```
> dsolve({deq, inc}, y(x));
```

$$y(x) = \frac{4\, e^{(3\, x)}}{e^{6}} \tag{4.1.9}$$

7.4.2 Exercises

1. A quantity P obeys the exponential growth law $P(t) = e^{5t}$ (t in years). (a) At what time t is P = 10? (b) At what time t is $P = 20$? (c) What is the double time for P?
2. Find the decay constant of Radon-222, given that its half-life is 3.825 days.
3. The decay constant of Cobalt-60 is 0.13 years^{-1}. How long will it take for 75% of Cobalt-60 to decay?
4. Find the solution $y' = -2y$ satisfying $y(0) = 3$.
5. Find the solution $y' = 5y$ satisfying $y(1) = 5$.

7.5 Compound Interest and Present Value

(1) **Compound interest:** If P_0 is the initial deposit, r is the annual interest rate, and M is the number of times compounded per year, then the future value after t years is

$$P(t) = P_0\left(1 + \frac{r}{M}\right)^{Mt}$$

(2) **Continuous compounded interest:** If P_0 is the initial deposit and r is the annual interest rate, compounded continuously, then the future value after t years is

$$P(t) = P_0\, e^{rt}$$

(3) **Present value:** If the annual interest rate is r, and after t years P dollars will be received, then the present value is

$$PV = Pe^{-rt}$$

(4) **Present value of an income stream:** The present value at interest rate r of an income stream paying out $R(t)$ dollars/year continuously for T years is

$$PV = \int_0^T R(t)e^{-rt}\,dt$$

7.5.1 Compound interest

Example 1. $P_0 = \$10000$, $r = 6\%$. Find the balance after 3 years (a) if the interest is compounded quarterly; (b) if the interest is compounded continuously.

```
> p0:=10000; r:=0.06; M:=4; t:=3;
```

$$p0 := 10000$$
$$r := 0.06$$
$$M := 4$$
$$t := 3 \quad (5.1.1)$$

```
> Pa:=evalf(p0*(1+r/M)^(M*t));
```

$$Pa := 11956.18171 \quad (5.1.2)$$

```
> Pb:=evalf(p0*exp(r*t));
```

$$Pb := 11972.17363 \quad (5.1.3)$$

Example 2. A bank pays interest at a rate of 5%. What is the yearly multiplier if interest is compounded (a) yearly? (b) three times a year? (c) continuously?

```
> r:=0.05;
```

$$r := 0.05 \quad (5.1.4)$$

```
> Ma:=(1+r);
```

$$Ma := 1.05 \quad (5.1.5)$$

```
> Mb:=(1+r/3)^3;
```

$$Mb := 1.050837964 \quad (5.1.6)$$

```
> Mc:=exp(r);
```

$$Mc := 1.051271096 \quad (5.1.7)$$

7.5.2 Present value

Example 3. Is it better to receive \$2000 today or \$2200 in 2 years if (a) $r = 0.07$ and (b) $r = 0.03$?

Case (a):

```
> P:=2200; r:=0.07;
```

$$P := 2200 \quad (5.2.1)$$
$$r := 0.07$$

```
> PV:=P*exp(-r*2);
```

$$PV := 1912.588118 \quad (5.2.2)$$

Case (b):

```
> r:=0.03;
```

$$r := 0.03 \tag{5.2.3}$$

```
> PV:=P*exp(-r*2);
```

$$PV := 2071.881974 \tag{5.2.4}$$

Hence, in case (a) it is better to receive \$2000 today, but in case (b) it is not.

Example 4. An investment pays out \$800/year continuously for 5 years. Find the *PV* of the investment for (a) $r = 0.06$ and (b) $r = 0.04$.

Case (a):

```
> r:=0.06;t:='t':
```

$$r := 0.06 \tag{5.2.5}$$

```
> PVa:=int(800*exp(-r*t),t=0..5);
```

$$PVa := 3455.757058 \tag{5.2.6}$$

Case (b):

```
> r:=0.04;
```

$$r := 0.04 \tag{5.2.7}$$

```
> PVb:=int(800*exp(-r*t),t=0..5);
```

$$PVb := 3625.384938 \tag{5.2.8}$$

Exercises

1. $P_0 = \$2000$, $r = 9\%$. Find the balance after 20 years (a) if the interest is compounded quarterly; (b) if the interest is compounded monthly; (c) if the interest is compounded continuously.
2. How long will it take for \$4000 to double in value if it is deposited in an account bearing 7% interest, continuously compounded?
3. Compute the *PV* of \$5000 received in 3 years if (a) $r = 0.06$ and (b) $r = 0.11$. What is the *PV* in these two cases if the same sum is instead received in 5 years?
4. Is it better to receive \$1000 today or \$1300 in 4 years if (a) $r = 0.08$ and (b) $r = 0.03$?
5. An investment pays out \$2500/year continuously for 10 years. Find the PV of the investment for (a) $r = 0.08$ and (b) $r = 0.05$.

7.6 Models Involving $y' = k(y - b)$

Assume $k > 0$. Then the differential equation

$$\frac{d}{dt} y(t) = k(y(t) - b)$$

leads to the exponential growth model

$$y(t) = b + Ce^{kt}$$

and

$$\frac{d}{dt} y(t) = -k(y(t) - b)$$

leads to the exponential decay model

$$y(t) = b + Ce^{-kt}$$

7.6.1 Models of $y' = k(y - b)$ and $y' = -k(y - b)$

Example 1. At $t = 0$ we submerge a hot metal bar with cooling constant $k = 2.1\ \text{min}^{-1}$ in a large tank of water at temperature $T_0 = 55\ ^\circ\text{F}$. Let $y(t)$ be the bar's temperature at time *t*. (a) Find the differential equation for $y(t)$ and find its general solution. (b) What is the bar's temperature after 1 min if its initial temperature is $400\ ^\circ\text{F}$? (c) What was the bar's initial temperature if, after half a minute, it had cooled to $120\ ^\circ\text{F}$?

(a) The equation for $y(t)$ is

```
> Eq:= diff(y(t),t)=-2.1*(y(t)-55);
```

$$Eq := \frac{d}{dt}\, y(t) = -2.1\, y(t) + 115.5 \qquad (6.1.1)$$

Its general solution is

```
> dsolve(Eq, y(t));
```

$$y(t) = 55 + e^{\left(-\frac{21}{10}\, t\right)}\, \_C1 \qquad (6.1.2)$$

(b) Find the bar's temperature after 1 minute if its initial temperature is $400\ ^\circ\text{F}$.

```
> inc:=y(0)=400:
> dsolve({Eq,inc},y(t));
```

$$y(t) = 55 + 345\, e^{\left(-\frac{21}{10}\, t\right)} \qquad (6.1.3)$$

```
> yt:=t->55+345*exp(-21/10*t);
```

$$yt := t \rightarrow 55 + 345\, e^{\left(-\frac{21}{10}\, t\right)} \qquad (6.1.4)$$

```
> evalf(yt(1));
```

$$97.24746776 \qquad (6.1.5)$$

(c) Find the bar's initial temperature if, after half a minute, it had cooled to$120\ ^\circ\text{F}$.

```
> y:='y':
> inc:=y(1/2)=120;
```

$$inc := y\left(\frac{1}{2}\right) = 120 \qquad (6.1.6)$$

```
> dsolve({Eq,inc}, y(t));
```

$$y(t) = 55 + \frac{65\, e^{\left(-\frac{21}{10}\, t\right)}}{e^{\left(\frac{-21}{20}\right)}} \qquad (6.1.7)$$

```
> yt:=t->55+65*exp(-21/10*t)/exp((-21)/20);
```

$$yt := t \rightarrow 55 + \frac{65\, e^{\left(-\frac{21}{10}\, t\right)}}{e^{\left(\frac{-21}{20}\right)}} \qquad (6.1.8)$$

```
> evalf(yt(0));
```

$$240.7473227 \tag{6.1.9}$$

Example 2. Let $P(t)$ be the balance in an annuity earning interest at the rate $r = 0.07$, with withdrawals made continuously at a rate of $N = \$500$/year. (a) Assume that $P_0 = \$5000$. Find $P(t)$ and determine when the annuity runs out of money. (b) Assume that $P_0 = \$9000$. Find $P(t)$ and show that the balance increases indefinitely.
(a) Assume that $P_0 = \$5000$. Find $P(t)$ and determine when the annuity runs out of money.

```
> b:= 500/0.07;
```

$$b := 7142.857143 \tag{6.1.10}$$

```
> P:=t->b + C*exp(0.07*t);
```

$$P := t \rightarrow b + C\,e^{(0.07\,t)} \tag{6.1.11}$$

```
> C:= solve(P(0)=5000,C);
```

$$C := -2142.857143 \tag{6.1.12}$$

```
> t:=solve(P(t)=0,t);
```

$$t := 17.19961149 \tag{6.1.13}$$

The money runs out after 17.2 years.
(b) Assume that $P_0 = \$9000$. Find $P(t)$ and show that the balance increases indefinitely.

```
> t:='t': C:='C':
> C:= solve(P(0)=9000,C);
```

$$C := 1857.142857 \tag{6.1.14}$$

```
> limit(P(t), t=infinity);
```

$$Float(\infty) \tag{6.1.15}$$

Hence, $P(t)$ increases infinitely as t tends to infinity.

Exercises

1. Find the general solution of $y' = 2(y - 10)$. Then find the two solutions satisfying (a) $y(0) = 25$ and (b) $y(0) = 5$, and sketch their graphs.
2. Find the general solutions of $y' = -3(y - 12)$. Then find the two solutions satisfying $y(0) = 20$ and $y(0) = 0$, and sketch their graphs.
3. A hot metal bar is submerged in a large reservoir of water whose temperature is 60 F. The temperature of the bar 20 seconds after submersion is 100 $^\circ$ F. After 1 minute, the temperature has cooled to 80 $^\circ$ F. (a) Determine the cooling constant *k*. (b) What is the differential equation satisfied by the temperature $F(t)$ of the bar? (c) What is the formula for $F(t)$? (d) Determine the temperature of the bar at the moment it is submerged.
4. Let $P(t)$ be the balance in an annuity earning interest at the rate $r = 0.05$, with withdrawals made continuously at a rate of $N = \$1000$/year. (a) Assume that $P_0 = \$15000$. Find $P(t)$and determine when the annuity runs out of money. (b) Which initial deposit P_0 yields a constant balance?

7.7 L'Hopital's Rule

(1) **L'Hopital's Rule:** If $\lim_{x\to\infty} f(x)$ and $\lim_{x\to\infty} g(x)$ both are zero or infinite, then

$$\lim_{x\to\infty}\frac{f(x)}{g(x)} = \lim_{x\to\infty}\frac{\frac{d}{dx}f(x)}{\frac{d}{dx}g(x)}$$

provided that the limit on the right exists.
(2) In MAPLE, **L'Hopital's Rule** is hidden. You need not apply it explicitly.

7.7.1 Applying L'Hopital's Rule

Example 1. Use **L'Hopital's Rule** to evaluate $\lim_{x\to 1}\frac{x^3-1}{x-1}$.

```
> f:=x->x^3-1; g:=x->x-1;
```

$$f := x \to x^3 - 1$$
$$g := x \to x - 1 \quad (7.1.1)$$

```
> limit(f(x),x=1);
```

$$0 \quad (7.1.2)$$

```
> limit(g(x),x=1);
```

$$0 \quad (7.1.3)$$

```
> limit(D(f)(x)/D(g)(x), x=1);
```

$$3 \quad (7.1.4)$$

Hence, the limit is 3. *It can be verified directly:*

```
> limit(f(x)/g(x),x=1);
```

$$3 \quad (7.1.5)$$

In the following examples, we shall directly evaluate the limit using MAPLE.
Example 2. Find $\lim_{x\to 0}(x\ln(x))$.

```
> limit(x*ln(x),x=0);
```

$$0 \quad (7.1.6)$$

Example 3. Find $\lim_{x\to 0}\left(\frac{e^x-1-x}{\cos(x)-1}\right)$.

```
> limit((exp(x)-1-x)/(cos(x)-1), x=0);
```

$$-1 \quad (7.1.7)$$

Example 4. Find $\lim_{x\to 0+} x^x$.

```
> limit(x^x, x=0, right);
```

$$1 \quad (7.1.8)$$

Exercises

1. Find $\lim_{x \to 1} \frac{2x^2 + x - 3}{x - 1}$.

2. Find $\lim_{x \to 0} \frac{x^2}{1 - \cos(x)}$.

3. Find $\lim_{x \to 4} \left(\frac{1}{\sqrt{x} - 2} - \frac{4}{x - 4} \right)$.

4. Find $\lim_{x \to \infty} \left(\frac{1}{x^2} - \csc(x)^2 \right)$.

5. Find $\lim_{x \to 0} \sin(x)^x$.

7.8 Inverse Trigonometric Functions

(a) The derivative of the inverse trigonometric functions are listed as follows.

```
> diff(arcsin(x),x);
```

$$\frac{1}{\sqrt{1 - x^2}} \quad (8.1)$$

```
> diff(arccos(x),x);
```

$$-\frac{1}{\sqrt{1 - x^2}} \quad (8.2)$$

```
> diff(arctan(x),x);
```

$$\frac{1}{1 + x^2} \quad (8.3)$$

```
> diff(arccot(x),x);
```

$$-\frac{1}{1 + x^2} \quad (8.4)$$

```
> diff(arcsec(x),x);
```

$$\frac{1}{x^2 \sqrt{1 - \frac{1}{x^2}}} \quad (8.5)$$

```
> diff(arccsc(x),x);
```

$$-\frac{1}{x^2 \sqrt{1 - \frac{1}{x^2}}} \quad (8.6)$$

(b) Integration formulas.

```
> int(1/sqrt(1-x^2),x);
```

$$\arcsin(x) \tag{8.7}$$

```
> int(1/(1+x^2),x);
```

$$\arctan(x) \tag{8.8}$$

7.8.1 Calculus of inverse trigonometric functions

Example 1. Simplify $\cos(\arcsin(x))$.

```
> simplify(cos(arcsin(x)));
```

$$\sqrt{1-x^2} \tag{8.1.1}$$

Example 2. Find $f'(1/2)$ for $f(x) = \arcsin\left(x^2\right)$.

```
> f:=x->arcsin(x^2);
```

$$f := x \rightarrow \arcsin\left(x^2\right) \tag{8.1.2}$$

```
> simplify(D(f)(1/2));
```

$$\frac{4}{15}\sqrt{15} \tag{8.1.3}$$

Example 3. Find $\frac{\mathrm{d}}{\mathrm{d}x}\arctan(3x+1)$.

```
> diff(arctan(3*x+1),x);
```

$$\frac{3}{1+(3\,x+1)^2} \tag{8.1.4}$$

Example 4. Evaluate $\int \frac{e^x}{1+e^{2x}}\,\mathrm{d}x$.

```
> int(exp(x)/(1+exp(2*x)),x);
```

$$\arctan\left(e^x\right) \tag{8.1.5}$$

Example 5. Evaluate $\int_0^{\frac{1}{2}} \frac{1}{\sqrt{1-x^2}}\,\mathrm{d}x$.

```
> int(1/sqrt(1-x^2),x=0..1/2);
```

$$\frac{1}{6}\pi \tag{8.1.6}$$

Exercises

1. Find $\frac{d}{dx}(x\arctan(x))$.

2. Find $\frac{d}{dt}\arctan\left(\frac{t}{1-t^2}\right)$.

3. Find the critical points of $y=\arctan\left(x^2-x\right)$ and determine whether the function has local minima/maxima.

4. Evaluate $\int \frac{x+1}{\sqrt{1-x^2}}\,dx$.

5. Evaluate $\int_{\frac{2}{\sqrt{3}}}^{\sqrt{2}} \frac{1}{x\sqrt{x^2-1}}\,dx$.

7.9 Hyperbolic Functions

The definitions of hyperbolic functions:

$$\sinh(x)=\frac{e^x-e^{-x}}{2},\ \cosh(x)=\frac{e^x+e^{-x}}{2}$$
$$\tanh(x)=\frac{e^x-e^{-x}}{e^x+e^{-x}},\ \coth(x)=\frac{e^x-e^{-x}}{e^x+e^{-x}}$$

The basic differential formulas for hyperbolic functions are the following.

```
> diff(sinh(x),x);
```

$$\cosh(x) \tag{9.1}$$

```
> diff(cosh(x),x);
```

$$\sinh(x) \tag{9.2}$$

```
> diff(tanh(x),x);
```

$$1-\tanh(x)^2 \tag{9.3}$$

7.9.1 Calculus of hyperbolic functions

Example 1. Verify $\cosh(x)^2-\sinh(x)^2=1$

```
> simplify(cosh(x)^2-sinh(x)^2);
```

$$1 \tag{9.1.1}$$

Example 2. Verify the formula $\frac{d}{dx}\coth(x)=-\operatorname{csch}(x)^2$.

```
> simplify(diff(coth(x),x)+csch(x)^2);
```

$$0 \tag{9.1.2}$$

Example 3. Calculate (a) $\frac{d}{dx}\cosh\left(3x^2+1\right)$; (b) $\frac{d}{dx}\left(\sinh(x)\tanh(x)\right)$.

```
> diff(cosh(3*x^2+1),x);
```

$$6\sinh\left(3x^2+1\right)x \tag{9.1.3}$$

```
> diff(sinh(x)*tanh(x),x);
```

$$\cosh(x)\tanh(x)+\sinh(x)\left(1-\tanh(x)^2\right) \tag{9.1.4}$$

```
> simplify(%);
```

$$\frac{\left(\cosh(x)^2+1\right)\sinh(x)}{\cosh(x)^2} \tag{9.1.5}$$

Example 4. Calculate $\int x\cosh\left(x^2\right)\,dx$.

```
> int(x*cosh(x^2), x);
```

$$\frac{1}{2}\sinh\left(x^2\right) \tag{9.1.6}$$

Example 5. Calculate $\int \frac{1}{\sqrt{x^2-1}}\,dx$.

```
>  int(1/sqrt(x^2-1), x);
```

$$\ln\left(x+\sqrt{-1+x^2}\right) \tag{9.1.7}$$

Exercises

1. Find $\frac{d}{dx}\cosh(1-4x)$.
2. Find $\frac{d}{dx}\cosh(\sinh(x))$.
3. Calculate (a) $\frac{d}{dx}\operatorname{arccosh}\left(3x^2+1\right)$; (b) $\frac{d}{dx}\operatorname{arcsinh}\left(x^2+1\right)$.
4. Calculate $\int e^{-x}\cosh(x)\,dx$.
5. Calculate $\int_0^1 \frac{1}{\sqrt{x^2+1}}\,dx$.

Chapter 8 TECHNIQUES OF INTEGRATION

8.1 Numerical Integration

(1) Trapezoidal Rule $\boldsymbol{T_N}$:

$$\boldsymbol{T_N} = \frac{1}{2}\,\triangle x(y_0 + 2y_1 + \ldots + 2y_{N-1} + y_N),$$

which is the average of the left sum of the right sum.

(2) Midpoint Rule $\boldsymbol{M_N}$:

$$\boldsymbol{M_N} = \triangle x(f\left(c_1\right) + f\left(c_2\right) + \cdots + f\left(c_N\right)), \quad c_i = a + (i - \frac{1}{2})\triangle x.$$

(3) Simpson's Rule $\boldsymbol{S_N}$:

$$\boldsymbol{S_N} = \frac{1}{3}\,\triangle x(y_0 + 4y_1 + 2y_2 + \cdots + 4y_{N-3} + 2y_{N-2} + 4y_{N-1} + y_N).$$

Let $N = 2m$. Then

$$\boldsymbol{S_N} = \frac{1}{3}\boldsymbol{T}_m + \frac{2}{3}\boldsymbol{M_m}$$

(4) Error bound:

$$\text{Error}(\boldsymbol{T_N}) \le \max|f''(x)|\,\frac{(b-a)^3}{12N^2}$$

$$\text{Error}(\boldsymbol{M_N}) \le \max|f''(x)|\,\frac{(b-a)^3}{24N^2}$$

$$\text{Error}(\boldsymbol{S_N}) \le \max|f^{(4)}(x)|\,\frac{(b-a)^5}{180N^4}$$

8.1.1 Numerical integration

Example 1. Calculate $\boldsymbol{T_6}$ and $\boldsymbol{M_6}$ for $\displaystyle\int_1^4 \sqrt{x}\,\mathrm{d}x.$

```
> a:=1: b:=4: f:=x->sqrt(x): N:=6: dx:=(b-a)/N:
> Rsum :=dx*sum(f(a+j*dx),j=1..6): Lsum:= dx*sum(f(a+j*dx),j=0.
  .5):
> TN:=evalf(0.5*(Rsum+Lsum));
```

$$TN := 4.661488383 \qquad (1.1.1)$$

```
> MN:=evalf(dx*sum(f(a+(j-1/2)*dx),j=1..6));
```

$$MN := 4.669244676 \qquad (1.1.2)$$

Example 2. Use the error bound to estimate th error for T_6 and M_6 in **Example 1,** and compare them to the exact errors.

```
> d2f:=D(D(f))(x);
```

$$d2f := -\frac{1}{4}\frac{1}{x^{(3/2)}} \quad (1.1.3)$$

```
> K2:=maximize(abs(d2f),x=a..b);
```

$$K2 := \frac{1}{4} \quad (1.1.4)$$

```
> TNError:=evalf(K2*(b-a)^3/(12*N^2));
```

$$TNError := 0.01562500000 \quad (1.1.5)$$

```
> Exactv:=evalf(int(sqrt(x),x=1..4));
```

$$Exactv := 4.666666667 \quad (1.1.6)$$

```
> TNExactError:=evalf(abs(TN-Exactv));
```

$$TNExactError := 0.005178284 \quad (1.1.7)$$

```
> MNError:=evalf(K2*(b-a)^3/(24*N^2));
```

$$MNError := 0.007812500000 \quad (1.1.8)$$

```
> MNExactError:=evalf(abs(MN-Exactv));
```

$$MNExactError := 0.002578009 \quad (1.1.9)$$

Example 3. Find N such that M_N approximates $\int_0^3 e^{-x^2} dx$ with an error of at most 0.0001.

```
> a:=0: b:=3: f:=x->exp(-x^2): er:=0.0001: N:='N':
> K2:=maximize(abs( D(D(f))(x)),x=0..3);
```

$$K2 := 2 \quad (1.1.10)$$

```
> fsolve( K2*(b-a)^3/(24*N^2)=er, N=5);
```

$$150.0000000 \quad (1.1.11)$$

Hence, N must be ≥ 150.

Example 4. Use Simpson's Rule with $N = 8$ to approximate $\int_2^4 \sqrt{1+x^3}\, dx$.

```
> a:=2: b:=4: f:=x->sqrt(1+x^3): N:=8:
> m:=N/2: dx:=(b-a)/m:
> Rsum:=dx*sum(f(a+j*dx),j=1..m): Lsum:=dx*sum(f(a+j*dx),j=0..(m
  -1)):
> Tm:=1/2*(Rsum+Lsum): Mm:=dx*sum(f(a+(j-1/2)*dx),j=1..m):
> SN:=(1/3*Tm+2/3*Mm):
> SN:=evalf(SN);
```

$$SN := 10.74159295 \quad (1.1.12)$$

Example 5. Find $\boldsymbol{S_8}$ for $\displaystyle\int_1^3 \frac{1}{x}\,dx$. Then (a) find an error bound; (b) find N such that the error is at most 10^{-6}.

```
> a:=1: b:=3: f:=x->1/x: m:=4:  dx:=(b-a)/m:
> Rsum:=dx*sum(f(a+j*dx),j=1..m): Lsum:=dx*sum(f(a+j*dx),j=0..(m
  -1)):
> Tm:=1/2*(Rsum+Lsum): Mm:=dx*sum(f(a+(j-1/2)*dx),j=1..m):
> SN:=evalf((1/3*Tm+2/3*Mm));
```

$$SN := 1.098725349 \qquad (1.1.13)$$

(a) Find an error bound.

```
> K2:=maximize((D@@4)(f)(x),x=a..b);
```

$$K2 := 24 \qquad (1.1.14)$$

```
> S8er:=evalf(K2*(b-a)^5/(180*N^4));
```

$$S8er := 0.001041666667 \qquad (1.1.15)$$

(b) Find N such that the error is at most 10^{-6}.

```
> N:='N':
> fsolve(K2*(b-a)^5/(180*N^4)=10^(-6), N=5);
```

$$45.44877466 \qquad (1.1.16)$$

Hence, we need $N \geq 46$.

Exercises

1. Calculate $\boldsymbol{T_6}$ and $\boldsymbol{M_6}$ for $\int_0^2 \sqrt[3]{x}\;dx$.

2. Use the error bound to estimate the error for $\boldsymbol{T_6}$ and $\boldsymbol{M_6}$ in **Exercise 1,** and compare them to the exact errors.

3. Find N such that $\boldsymbol{M_N}$ approximates $\displaystyle\int_0^2 e^{-3x^2}\,dx$ with an error of at most 0.0001.

4. Use Simpson's Rule with $N = 8$ to approximate $\displaystyle\int_0^3 \frac{1}{\sqrt{1+x^3}}\,dx$.

5. Find $\boldsymbol{S_8}$ for $\displaystyle\int_0^1 \frac{1}{1+x^2}\,dx$. Then (a) find an error bound; (b) find N such that the error is at most 10^{-6}.

8.2 Integration by Part

Integration by parts formula:

$$\int f(x)g'(x)\mathrm{d}x \;= f(x)g(x) - \int f'(x)g(x)\mathrm{d}x\ .$$

When you use MAPLE for integration, you need not to use the rule for evaluating integrals.

8.2.1 Evaluate integrals

Example 1. Evaluate $\int x\cos(x)\mathrm{d}x$.

```
> int(x*cos(x),x);
```

$$\cos(x) + x\sin(x) \qquad (2.1.1)$$

Example 2. Evaluate $\int x e^{x} \mathrm{d}x$.

```
> int(x*exp(x),x);
```

$$(-1 + x)\,e^{x} \qquad (2.1.2)$$

Example 3. Calculate $\int e^{x}\sin(x)\mathrm{d}x$.

```
> int(exp(x)*sin(x),x);
```

$$-\frac{1}{2}\,e^{x}\cos(x) + \frac{1}{2}\,e^{x}\sin(x) \qquad (2.1.3)$$

Example 4. Calculate $\int \sqrt{x}\ln(x)\mathrm{d}x$.

```
> int(sqrt(x)*ln(x),x);
```

$$\frac{2}{3}\,x^{(3/2)}\ln(x) - \frac{4}{9}\,x^{(3/2)} \qquad (2.1.4)$$

Example 5. Find $\int_0^1 \arctan(x)\mathrm{d}x$.

```
> int(arctan(x),x=0..1);
```

$$\frac{1}{4}\,\pi - \frac{1}{2}\,\ln(2) \qquad (2.1.5)$$

Example 6. Find $\int_0^{\frac{\pi}{4}} x\sin(2x)\mathrm{d}x$.

```
> int(x*sin(2*x),x=0..Pi/4);
```

$$\frac{1}{4} \qquad (2.1.6)$$

Exercises

1. Find $\int x^2\, e^{-x} dx$.
2. Find $\int \ln(x) dx$.
3. Find $\int e^x \cos(2x) dx$.
4. Find $\int_0^1 \arcsin(x) dx$.
5. Find $\int_0^\pi \sin(x)\cos(3x) dx$.

8.3 Trigonometric Integrals

In this section we consider integrals such as $\int \sin(x)^m \cos(x)^n\, dx$.

8.3.1 Calculate trigonometric integrals

Example 1. Evaluate $\int \sin(x)^4 \cos(x)^5 dx$.

```
> int(sin(x)^4*cos(x)^5,x);
```

$$-\frac{1}{9}\sin(x)^3\cos(x)^6 - \frac{1}{21}\sin(x)\cos(x)^6 + \frac{1}{105}\cos(x)^4\sin(x) + \frac{4}{315}\cos(x)^2\sin(x) + \frac{8}{315}\sin(x) \quad (3.1.1)$$

Example 2. Evaluate $\int \sin(x)^4 dx$.

```
> int(sin(x)^4,x);
```

$$-\frac{1}{4}\sin(x)^3\cos(x) - \frac{3}{8}\cos(x)\sin(x) + \frac{3}{8}x \quad (3.1.2)$$

Example 3. Evaluate $\int \sin(x)^4 \cos(x)^2 dx$.

```
> int(sin(x)^4*cos(x)^2,x);
```

$$-\frac{1}{6}\sin(x)^3\cos(x)^3 - \frac{1}{8}\sin(x)\cos(x)^3 + \frac{1}{16}\cos(x)\sin(x) + \frac{1}{16}x \quad (3.1.3)$$

Example 4. Evaluate $\int \tan(x) dx$.

```
> int(tan(x),x);
```

$$-\ln(\cos(x)) \quad (3.1.4)$$

Example 5. Find $\int_0^{\frac{\pi}{4}} \tan(x)^3 dx$.

```
> int(tan(x)^3,x=0..Pi/4);
```

$$\frac{1}{2} - \frac{1}{2}\ln(2) \tag{3.1.5}$$

Example 6. Find $\int_0^{\pi} \sin(4x)\cos(3x) dx$.

```
> int(sin(4*x)*cos(3*x),x=0..Pi);
```

$$\frac{8}{7} \tag{3.1.6}$$

Exercises

1. Find $\int \sin(x)^2 \cos(x)^3 dx$.
2. Find $\int \sin(x)^4 \cos(x)^2 dx$.
3. Find $\int \tan(x)^2 dx$.
4. Find $\int_0^{\pi} \sin(x)^6 dx$.
5. Find $\int_{-\frac{\pi}{4}}^{\frac{\pi}{4}} \frac{1}{\cos(x)} dx$.

8.4 Trigonometric Substitution

To integrate functions involving square root expressions, a useful approach to the integration is trigonometric substitution. Once again, MAPLE hides all of these substitutions, and you can evaluate these integrals directly.

8.4.1 Evaluate integrals involving square root expressions

Example 1. Evaluate $\int \sqrt{1-x^2}\,dx$.

```
> int(sqrt(1-x^2),x);
```

$$\frac{1}{2}\,x\sqrt{1-x^2}+\frac{1}{2}\,\arcsin(x) \tag{4.1.1}$$

Example 2. Evaluate $\int \frac{x^2}{\left(4-x^2\right)^{\frac{3}{2}}}\,dx$.

```
> int( x^2/(4-x^2)^(3/2),x);
```

$$\frac{x}{\sqrt{4-x^2}}-\arcsin\left(\frac{1}{2}\,x\right) \tag{4.1.2}$$

Example 3. Evaluate $\int \sqrt{4x^2+20}\,dx$.

```
> int(sqrt(4*x^2+20),x);
```

$$x\sqrt{x^2+5}\;+5\,\operatorname{arcsinh}\left(\frac{1}{5}\,\sqrt{5}\,x\right) \tag{4.1.3}$$

Example 4. Evaluate $\int \frac{1}{\left(x^2-6x+1\right)^2}\,dx$.

```
> int(1/(x^2-6*x+1)^2, x);
```

$$-\frac{1}{32}\,\frac{2\,x-6}{x^2-6\,x+1}+\frac{1}{64}\,\sqrt{2}\,\operatorname{arctanh}\left(\frac{1}{8}\,(2\,x-6)\sqrt{2}\right) \tag{4.1.4}$$

Exercises

1. Evaluate $\int \sqrt{1-4x^2}\,dx$.
2. Evaluate $\int \sqrt{x^2-1}\,dx$.
3. Evaluate $\int \frac{1}{\sqrt{x^2+2x+2}}\,dx$.
4. Find $\int \frac{1}{\sqrt{x^2-x-1}}\,dx$.
5. Find $\int \sqrt{x^2+3x+1}\,dx$.

8.5 The Method of Partial Fractions

When the integrand is a rational function, then it can be represented as a partial fraction decomposition and be evaluated. Once again, MAPLE will help you do the partial fraction decomposition. You need not do it by yourself.

8.5.1 Evaluate the integrals of rational functions

Example 1. Evaluate $\int \frac{1}{x^2-7x+10}\,dx$.

```
> int(1/(x^2-7*x+10), x);
```

$$-\frac{1}{3}\ln(x-2)+\frac{1}{3}\ln(x-5) \qquad (6.1.1)$$

Example 2. Evaluate $\int \frac{(x^2+2)(2x-8)(x+2)}{x-1}\,dx$.

```
> int((x^2+2)/(x-1)*(2*x-8)*(x+2), x);
```

$$\frac{1}{2}x^4-\frac{2}{3}x^3-7x^2-22x-54\ln(x-1) \qquad (6.1.2)$$

Example 3. Evaluate $\int \frac{x^3+1}{x^4+1}\,dx$.

```
> int((x^3+1)/(x^4+1),x);
```

$$\frac{1}{8}\sqrt{2}\ln\left(\frac{x^2+x\sqrt{2}+1}{x^2-x\sqrt{2}+1}\right)+\frac{1}{4}\sqrt{2}\arctan\left(x\sqrt{2}+1\right)+\frac{1}{4}\sqrt{2}\arctan\left(x\sqrt{2}-1\right) \qquad (6.1.3)$$
$$+\frac{1}{4}\ln\left(x^4+1\right)$$

Example 4. Evaluate $\int \frac{3x-9}{(x-1)(x+1)^2}\,dx$.

```
> int((3*x-9)/((x-1)*(x+1)^2),x);
```

$$-\frac{6}{x+1}-\frac{3}{2}\ln(x-1)+\frac{3}{2}\ln(x+1) \qquad (6.1.4)$$

Example 5. Evaluate $\displaystyle\int \frac{4-x}{x\left(x^2+2\right)^2}\mathrm{d}x.$

```
> int((4-x)/(x*(x^2+2)^2),x);
```

$$\frac{1}{8}\,\frac{-2\,x+8}{x^2+2}-\frac{1}{8}\sqrt{2}\,\arctan\left(\frac{1}{2}\,x\sqrt{2}\right)+\ln(x)-\frac{1}{2}\,\ln\left(x^2+2\right) \qquad (6.1.5)$$

Exercises

1. $\displaystyle\int \frac{x^2}{x^2-2x-3}\mathrm{d}x.$

2. $\displaystyle\int \frac{2x-3}{(x+1)^2(x-2)}\,\mathrm{d}x.$

3. $\displaystyle\int \frac{x^2+1}{\left(x^2+2x+2\right)^2}\,\mathrm{d}x.$

4. $\displaystyle\int \frac{x+1}{x^2\left(x^2+4\right)^2}\,\mathrm{d}x.$

8.6 Improper Integrals

The improper integral of $f(x)$ over $[a, \infty)$ is defined as the limit

$$\int_a^{\infty} f(x)\mathrm{d}x = \lim_{R\to\infty}\int_a^{R} f(x)\mathrm{d}x$$

When the limit exists, we say that the improper integral is convergent. Otherwise, it is divergent.

8.6.1 Evaluate improper integrals

Example 1. Show that $\displaystyle\int_2^{\infty} \frac{1}{x^2}\mathrm{d}x$ converges and compute its value.

```
> InR:= limit(int(1/x^2,x=2..R),R=infinity);
```

$$InR := \frac{1}{2} \tag{8.1.1}$$

It can be evaluated as follows.

```
> InR:= int(1/x^2,x=2..infinity);
```

$$InR := \frac{1}{2} \tag{8.1.2}$$

Example 2. Determine if $\int_{-\infty}^{-1} \frac{1}{x}\, dx$ converges.

```
> InR:=int(1/x,x=-infinity..-1);
```

$$InR := -\infty \tag{8.1.3}$$

Hence, it diverges.

Example 3. Determine if $\int_{0}^{\infty} x e^{-x}\, dx$ converges. If so, find its value.

```
> int(x*exp(-x),x=0..infinity) ;
```

$$1 \tag{8.1.4}$$

It converges and the value is 1.

Example 4. Determine whether $\int_{1}^{\infty} \frac{1}{\sqrt{x} + e^{3x}}\, dx$ converges or diverges.

```
> int(1/(sqrt(x)+exp(3*x)), x=1..infinity);
```

$$\int_{1}^{\infty} \frac{1}{\sqrt{x} + e^{(3x)}}\, dx \tag{8.1.5}$$

We have $0 \le \frac{1}{\sqrt{x} + e^{3x}} \le \frac{1}{e^{3x}}$ and the following is converges.

```
> int(1/exp(3*x), x=1..infinity);
```

$$\frac{1}{3} e^{(-3)} \tag{8.1.6}$$

Hence, $\int_{1}^{\infty} \frac{1}{\sqrt{x} + e^{3x}}\, dx$ converges.

Example 5. Determine whether $\int_0^{\infty} \frac{1}{\sqrt{1+x^2}}\,dx$ converges or diverges.

```
> int(1/sqrt(1+x^2), x=0..infinity) ;
```

$$\infty \qquad (8.1.7)$$

Hence, $\int_0^{\infty} \frac{1}{\sqrt{1+x^2}}\,dx$ diverges.

Exercises

1. Show that $\int_2^{\infty} \frac{1}{x^3}\,dx$ converges and computer its value.

2. Determine if $\int_1^{\infty} \frac{1}{x+1}\,dx$ converges or diverges.

3. Evaluate $\int_0^{\infty} e^{-3x}\,dx$.

4. Determine whether $\int_1^{\infty} \frac{1}{\sqrt{x+x^2}}\,dx$ converges or diverges.

5. Determine whether $\int_3^{\infty} \frac{1}{\ln(x)}\,dx$ converges or diverges.

Chapter 9 FURTHER APPLICATIONS OF INTEGRALS AND TAYLOR POLYNOMIALS

9.1 Arc Length and Surface Area

(1) The arc length of the curve of $f(x)$ over $[a, b]$ is

$$L=\int_a^b \sqrt{1+\left(\frac{d}{dx} f(x)\right)^2}\, dx$$

(2) The area of the surface obtained by rotating the curve of $f(x)$ about the x-axes on $[a, b]$ is

$$Area = 2\pi \int_a^b f(x)\sqrt{1+\left(\frac{d}{dx} f(x)\right)^2}\, dx$$

9.1.1 Arc length

Example 1. Calculate the arc length of $f(x) = \frac{x^3}{12} + \frac{1}{x}$ over [1, 3].

```
> f:=x->1/12*x^3+1/x;
```

$$f := x \rightarrow \frac{1}{12}x^3 + \frac{1}{x} \quad (1.1.1)$$

```
> dL:=sqrt(1+D(f)(x)^2);
```

$$dL := \sqrt{1+\left(\frac{1}{4}x^2 - \frac{1}{x^2}\right)^2} \quad (1.1.2)$$

```
> L:=int(dL,x=1..3);
```

$$L := \frac{17}{6} \quad (1.1.3)$$

Example 2. Calculate the arc length of $f(x) = \cosh(x)$ over $[0, a]$.

```
> assume(a>0):
> dL:=simplify(sqrt(1+diff(cosh(x),x)^2));
```

$$dL := \operatorname{csgn}(\cosh(x))\cosh(x) \quad (1.1.4)$$

```
> L:=int(dL,x=0..a);
```

$$L := \sinh(a\sim) \quad (1.1.5)$$

Example 3. Find the arc length of $y = \sin(x)$ over $[0, \pi]$ numerically.

```
> dL:=simplify(sqrt(1+diff(sin(x),x)^2));
```

$$dL := \sqrt{1+\cos(x)^2} \quad (1.1.6)$$

```
> L=evalf(int(dL,x=0..Pi));
```

$$\sinh(a\sim) = 3.820197788 \quad (1.1.7)$$

9.1.2 Surface area

Example 4. Find the surface area of the cone obtained by rotating the line $y = 2x$ about the x-axis on [0, 4].

```
> f:=x->2*x;
```

$$f := x \rightarrow 2\,x \tag{1.2.1}$$

```
> dL:=sqrt(1+D(f)(x)^2);
```

$$dL := \sqrt{5} \tag{1.2.2}$$

```
> Ar:=2*pi*int(f(x)*dL,x=0..4);
```

$$Ar := 32\,\pi\,\sqrt{5} \tag{1.2.3}$$

Example 5. Find the surface area of the cone obtained by rotating $y = \sqrt{x} - \frac{\left(\sqrt{x}\right)^3}{3}$ about the x-axis on [1, 3].

```
> f:=x-> sqrt(x)-1/3*sqrt(x)^3;
```

$$f := x \rightarrow \sqrt{x} - \frac{1}{3}\left(\sqrt{x}\right)^3 \tag{1.2.4}$$

```
> dL:= sqrt(1+D(f)(x)^2);
```

$$dL := \sqrt{1 + \left(\frac{1}{2}\,\frac{1}{\sqrt{x}} - \frac{1}{2}\sqrt{x}\right)^2} \tag{1.2.5}$$

```
> Ar:=2*pi*int(f(x)*dL,x=1..3);
```

$$Ar := \frac{16}{9}\,\pi \tag{1.2.6}$$

Exercises

1. Calculate the arc length of $f(x) = \frac{x^{\frac{3}{2}}}{3} - x^{\frac{1}{2}}$ over [1, 9].
2. Calculate the arc length of $f(x) = \sqrt{1 - x^2}$ over $[-\frac{1}{2}, \frac{1}{2}]$.
3. Find the arc length of $y = x^2\cos(x)$ over $[0, \pi]$ numerically.
4. Find the surface area of the surface obtained by rotating the curve $y = x^3$ about the x-axis on [0, 2].
5. Find the surface area of the cone obtained by rotating the line $y = x + \sqrt{x}$ about the x-axis on [1, 2].

9.2 Fluid Pressure and Force

(1) Let $f(y)$ be the horizontal width of the side at depth y, w be the fluid density, and the object extend from depth $y=a$ to $y=b$. Then the force of a flat side of an object submerged vertically in a fluid is

$$F = w\int_a^b y\,f(y)\,\mathrm{d}y.$$

(2) The **density of water** is 62.5 lb/ft^3 in the normal condisions.

9.2.1 Compute fluid force

Example 1. Calculate the force F on the side of a box submerged with its top 3 ft below the surface of the water. The box has a height of 5 ft and a square base with a side of 2 ft.

```
> f:=y->2; w:=62.5;
```

$$f := y \to 2$$

$$w := 62.5 \qquad (2.1.1)$$

```
> F:=w*int(y*f(y),y=3..8);
```

$$F := 3437.5 \qquad (2.1.2)$$

The force is 3437.5 lb.

Example 2. A plate in the shape of an equilateral triangle of side 2 m is submerged vertically in a tank of oil of mass density 900 kg/m^3. Calculate the force F on one side of the plate.

```
> f:=y->2*y*cot(Pi/3); w:=900 *9.8; h:=2*sin(Pi/3);
```

$$f := y \to 2\,y\cot\left(\frac{1}{3}\pi\right)$$

$$w := 8820.0$$

$$h := \sqrt{3} \qquad (2.1.3)$$

```
> F:=w*int(y*f(y),y=0..h);
```

$$F := 17640.0 \qquad (2.1.4)$$

Hence, the force is 17640 newtons.

Exercises

1. Calculate the force F on the side of a box submerged with its top 2 ft below the surface of the water. The box has a height of 4 ft and a square base with a side of 1 ft.
2. A plate in the shape of a triangle, whose base is 6 m and whose height is 4 m, is submerged vertically in a tank of oil of mass density 870 kg/m^3. Calculate the force F on one side of the plate.
3. A tank truck hauls milk in a 6-ft-diameter horizontal right circular cylindrical tank. How much force does the milk exert on each end of the tank when the tank is half full?

9.3 Center of Mass

The center of a mass is

$$x_{CM} = \frac{M_y}{M}, \quad y_{CM} = \frac{M_x}{M},$$

where M is the total mass, and M_x and M_y are x- and y- moments of the mass.
If the mass with a constant density is symmetrical with respect to a vertical line $x = a$, then $x_{CM} = a$; and if it is symmetrical with respect to a horizontal line $y = b$, then $y_{CM} = b$.

9.3.1 Calculate center of mass

Example 1. Find the centroid of the region under the curve $y = e^x$ over [1, 3].

```
> M:=int(exp(x),x=1..3);
```

$$M := -e + e^3 \tag{3.1.1}$$

```
> Mx:=1/2*int(exp(x)^2,x=1..3);
```

$$Mx := -\frac{1}{4}\,e^2 + \frac{1}{4}\,e^6 \tag{3.1.2}$$

```
> My:=int(x*exp(x),x=1..3);
```

$$My := 2\,e^3 \tag{3.1.3}$$

```
> cnt:=[My/M, Mx/M];
```

$$cnt := \left[\frac{2\,e^3}{-e + e^3}, \frac{-\frac{1}{4}\,e^2 + \frac{1}{4}\,e^6}{-e + e^3}\right] \tag{3.1.4}$$

```
> cnt:=evalf(cnt);
```

$$cnt := [2.313035286, 5.700954691] \tag{3.1.5}$$

Example 2. Find the centroid of a semicircle with radius 3.

```
> f:=x->sqrt(9-x^2);
```

$$f := x \rightarrow \sqrt{9 - x^2} \tag{3.1.6}$$

```
> M:=int(f(x),x=-3..3);
```

$$M := \frac{9}{2}\,\pi \tag{3.1.7}$$

```
> My:=0;
```

$$My := 0 \tag{3.1.8}$$

```
> Mx:=1/2*int(f(x)^2, x=-3..3);
```

$$Mx := 18 \tag{3.1.9}$$

```
> cnr:=[My/M, Mx/M];
```

$$cnr := \left[0, \frac{4}{\pi}\right] \tag{3.1.10}$$

Example 3. Find the centroid of the region lying between $y = x^2$ and $y = \sqrt{x}$ over [0, 1].

```
> f:=x->x^2; g:=x->sqrt(x);
```

$$f := x \rightarrow x^2$$

$$g := x \rightarrow \sqrt{x} \qquad (3.1.11)$$

```
> plot([f,g],0..1,color=[red, blue]);
```




```
> M:=int(g(x)-f(x),x=0..1);
```

$$M := \frac{1}{3} \qquad (3.1.12)$$

```
> My:=int(x*(g(x)-f(x)),x=0..1);
```

$$My := \frac{3}{20} \qquad (3.1.13)$$

```
> Mx:=1/2*int(g(x)^2-f(x)^2,x=0..1);
```

$$Mx := \frac{3}{20} \qquad (3.1.14)$$

```
> cnr:=[My/M, Mx/M];
```

$$cnr := \left[\frac{9}{20}, \frac{9}{20}\right] \qquad (3.1.15)$$

Exercises

1. Find the centroid of the region under the curve $y = \sqrt{x}$ over [4, 9].
2. Find the centroid of the region under the curve $y = 9 - x^2$ over [0, 3].
3. Find the moments and center of mass of the lamina occupying the region between the curves $y = x$ and $y = x^2$ for $0 \le x \le 1$.
4. Find the centroid of the region lying between $y = x$ and $y = \sqrt{x}$ over [0, 1].
5. Find the centroid of the region lying between $y = e^x$ and $y = 1$ over [0, 1].
6. Find the centroid of the region lying between $y = \sin(x)$ and $y = \cos(x)$ over [0, π/4].

9.4 Taylor Polynomials

(1) The Taylor polynomial of $f(x)$ is the following:

$$T_n(x) = f(a) + f'(a)(x-a) + \frac{f''(a)(x-a)^2}{2!} + \cdots + \frac{f^{(n)}(a)(x-a)^n}{n!}$$

If $a = 0$, it is called a Maclaurin polynomial.

(2) Error bound : $\left|T_n(x) - f(x)\right| \le \max|f^{(n+1)}(x)| \frac{|x-a|^{n+1}}{(n+1)!}$

9.4.1 Calculate Taylor polynomials

Example 1. Find the Taylor polynomial of $f(x) = \sqrt{1+x}$ at $a = 3$ of degree 4.

```
> taylor(sqrt(1+x),x=3,5);
```

$$2 + \frac{1}{4}(x-3) - \frac{1}{64}(x-3)^2 + \frac{1}{512}(x-3)^3 - \frac{5}{16384}(x-3)^4 + O\left((x-3)^5\right) \qquad (4.1.1)$$

Example 2. Find the Maclaurin polynomial of $y = \cos(x)$ of degree 5.

```
> taylor(cos(x),x=0);
```

$$1 - \frac{1}{2}x^2 + \frac{1}{24}x^4 + O\left(x^6\right) \qquad (4.1.2)$$

Example 3. Find the Maclaurin polynomial of $y = \ln(1+x)$ of degree 8.

```
> taylor(ln(1+x), x=0, 9);
```

$$x - \frac{1}{2}x^2 + \frac{1}{3}x^3 - \frac{1}{4}x^4 + \frac{1}{5}x^5 - \frac{1}{6}x^6 + \frac{1}{7}x^7 - \frac{1}{8}x^8 + O\left(x^9\right) \qquad (4.1.3)$$

Example 4. Use the error bound formula to find a bound for the error $\left|T_3(1.2) - \ln(1.2)\right|$, where $T_3(1.2)$ is the third Taylor polynomial for $f(x) = \ln(x)$ at $a = 1$.

```
> f:=x->ln(x);
```

$$f := x \rightarrow \ln(x) \qquad (4.1.4)$$

```
> d4x:=(D@@4)(f)(x);
```

$$d4x := -\frac{6}{x^4} \qquad (4.1.5)$$

```
> K:=maximize(abs(d4x),x=1..(1.2));
```

$$K := 6 \qquad (4.1.6)$$

```
> erb:=K*0.2^4/4!;
```

$$erb := 0.0004000000000 \qquad (4.1.7)$$

We compute the exact error in the following:

```
> taylor(f(x),x=1,4);
```

$$x - 1 - \frac{1}{2}(x-1)^2 + \frac{1}{3}(x-1)^3 + O\left((x-1)^4\right) \qquad (4.1.8)$$

```
> T3 :=x->(x-1)-1/2*(x-1)^2+1/3*(x-1)^3;
```

$$T3 := x \to x - 1 - \frac{1}{2}\,(x-1)^2 + \frac{1}{3}\,(x-1)^3 \quad (4.1.9)$$

```
> erx:=abs(f(1.2)-T3(1.2));
```

$$erx := 0.0003451099 \quad (4.1.10)$$

Example 5. Use the error bound formula to find n such that the Maclaurin polynomial $T_n(x)$ of $f(x) = \cos(x)$ gives the error $|T_n(0.2) - \cos(0.2)|$ at most 10^{-5}.

```
> K:=1:
> fsolve(K*0.2^(n+1)/(n+1)!=10^(-5),n=1);
```

$$3.596789342 \quad (4.1.11)$$

Hence, we choose $n = 4$.

Exercises

1. Find the Taylor polynomial of $f(x) = \dfrac{1}{1+x^2}$ at $a = -1$ of degree 4.
2. Find the Maclaurin polynomial of $y = \cos(2x)$ of degree 8.
3. Find the Maclaurin polynomial of $y = e^{\sin(x)}$ of degree 6.
4. Find n such that $|T_n(1) - e| \le 10^{-6}$, where $T_n(x)$ is the Maclaurin polynomial for $f(x) = e^x$.
5. Use the error bound formula to find n such that the Maclaurin polynomial $T_n(x)$ of $f(x) = \sin(x)$ gives the error $|T_n(0.3) - \sin(0.3)|$ at most 10^{-5}.

Chapter 10 INTRODUCTION TO DIFFERENTIAL EQUATIONS

10.1 Solving Differential Equations

In MAPLE, we use the syntax "**dsolve(deqn, fnc(var))**" for finding general solutions of differential equations, and use the syntax "**dsolve({deqn, cond1, cond2,...,condn}, fnc(var))**" for solving initial-value problems, where "**deqn**" is the differential equation, "**fnc(var)**" is for the function to be solved such as $y(x)$, $z(t)$, etc., and "**cond1, ..., condn**" indicates the initial conditions.

10.1.1 Find the general solutions

Example 1. Find the general solution of $\frac{y\mathrm{d}y}{\mathrm{d}x} - x = 0$.

```
> deq:=y(x)*diff(y(x),x)-x=0;
```

$$deq := y(x)\left(\frac{\mathrm{d}}{\mathrm{d}x}y(x)\right) - x = 0 \qquad (1.1.1)$$

```
> dsolve(deq,y(x));
```

$$y(x) = \sqrt{x^2 + \_C1},\ y(x) = -\sqrt{x^2 + \_C1} \qquad (1.1.2)$$

Example 2. Find the general solutions of $(1 + x^2)y' = x^3 y$.

```
> deq:=(1+x^2)*diff(y(x),x)  = x^3*y(x);
```

$$deq := (1 + x^2)\left(\frac{\mathrm{d}}{\mathrm{d}x}y(x)\right) = x^3\,y(x) \qquad (1.1.3)$$

```
> dsolve(deq, y(x));
```

$$y(x) = \frac{\_C1\ \mathrm{e}^{\left(\frac{1}{2}x^2\right)}}{\sqrt{1 + x^2}} \qquad (1.1.4)$$

10.1.2 Solve initial-value problems

Example 3. Solve the initial-value problem $y' = -ty$ with $y(0) = 3$.

```
> deq:=diff(y(t),t)=-t*y(t); inc:=y(0)=3;
```

$$deq := \frac{\mathrm{d}}{\mathrm{d}t}y(t) = -t\,y(t) \qquad (1.2.1)$$

$$inc := y(0) = 3$$

```
> dsolve({deq,inc},y(t));
```

$$y(t) = 3\ \mathrm{e}^{\left(-\frac{1}{2}t^2\right)} \qquad (1.2.2)$$

Example 4. Solve $y' - 2y + 4 = 0$, $y(1) = 4$.

```
> deq:=diff(y(x),x)-2*y(x)+4=0; inc:=y(1)=4;
```

$$deq := \frac{\mathrm{d}}{\mathrm{d}x}y(x) - 2\,y(x) + 4 = 0 \qquad (1.2.3)$$

$$inc := y(1) = 4$$

```
> dsolve({deq,inc},y(x));
```

$$y(x) = 2 + \frac{2\,e^{(2x)}}{e^2} \qquad (1.2.4)$$

Example 5. Find the value of a such that $y = e^{ax}$ is a solution of $y'' + 2y' - 8y = 0$.

```
> dsolve(diff(y(x),x,x)+2*diff(y(x),x)-8*y(x),y(x));
```

$$y(x) = e^{(2x)}\,\_C1 + \_C2\,e^{(-4x)} \qquad (1.2.5)$$

Hence, $a = 2$ or -4.

Exercises

1. Find the general solution of $\frac{dy}{dx} = xy^2$.
2. Find the general solutions of $\frac{(1+t^2)dx}{dt} = x^2 + 1$.
3. Solve the initial-value problem $y' + 2y = 0$ with $y(\ln(2)) = 3$.

4. solve $yy' = xe^{-y^2}$, $y(0) = -1$.

5. Find the value of a such that $y = x^a$ is a solution of $y'' - \frac{6y}{x^2} = 0$.

10.2 The Logistic Equation

The logistic differential equation is

$$\frac{dy}{dt} = ky\left(1 - \frac{y}{A}\right).$$

10.2.1 Solve the logistic equation

Example 1. Solve $\frac{d}{dt}\,y(t) = 0.3y(4-y)$ with initial condition $y(0) = 1$.

```
> deq:=diff(y(t),t) = 0.3*y(t)*(4-y(t)); inc:=y(0)=1;
```

$$deq := \frac{d}{dt}\,y(t) = 0.3\,y(t)\,(4 - y(t)) \qquad (2.1.1)$$

$$inc := y(0) = 1$$

```
> dsolve({deq,inc},y(t));
```

$$y(t) = \frac{4}{1 + 3\,e^{\left(-\frac{6}{5}t\right)}} \qquad (2.1.2)$$

Example 2. A 10000-acre forest has a carrying capacity of 1000 deer. Assume that the deer population grows logistically with growth constant $k = 0.4\ \text{yr}^{-1}$. (a) Find the deer population $P(t)$ if the initial population $P_0 = 100$. (b) How long does it take for the deer population to reach 500?

```
> deq:=diff(P(t),t)=0.4*P(t)*(1-P(t)/1000); inc:=P(0)=100;
```

$$deq := \frac{d}{dt} P(t) = 0.4\, P(t) \left(1 - \frac{1}{1000} P(t)\right) \tag{2.1.3}$$

$$inc := P(0) = 100$$

```
> Pt:=dsolve({deq,inc},P(t));
```

$$Pt := P(t) = \frac{1000}{1 + 9\, e^{\left(-\frac{2}{5} t\right)}} \tag{2.1.4}$$

```
> P:=t->1000/(1+9*exp(-2/5*t));
```

$$P := t \to \frac{1000}{1 + 9\, e^{\left(-\frac{2}{5} t\right)}} \tag{2.1.5}$$

```
> fsolve(P(t)=500,t=1);
```

$$5.493061443 \tag{2.1.6}$$

Hence, it takes about 5.5 years for the deer population to reach 500.

Exercises

1. Solve $\frac{d}{dt} y(t) = 2y(3 - y)$ with initial condition $y(0) = 10$.
2. A population $P(t)$ of squirrels lives in a forest with a carrying capacity of 2000. Assume logistic growth with growth constant $k = 0.6\ \text{yr}^{-1}$. (a) Find a formula for the squirrel population $P(t)$ if the initial population $P_0 = 500$. (b) How long does it take for the squirrel population to double?
3. The population $P(t)$ of mosquito larvae growing in a tree hole increases according to the logistic equation with a growth constant $k = 0.3\ \text{day}^{-1}$ and carrying capacity of 500. (a) Find a formula for the larvae population $P(t)$ if the initial population $P_0 = 50$. (b) How long does it take for the larvae population to reach 200?

10.3 First-Order Linear Equations

(1) The first-order linear equation has the form

$$a(x)\left(\frac{d}{dx} y(x)\right) + b(x) y(x) = c(x).$$

(2) The general solution of

$$\frac{d}{dx} y(x) + A(x) y(x) = B(x)$$

is

$$y = a(x)^{-1}\left(\int a(x) B(x)\, dx + C\right), \text{ where } a(x) = e^{\int A(x)\, dx}.$$

10.3.1 Solve first-order linear equations

Example 1. Solve $x\left(\frac{d}{dx}y(x)\right) - 3y(x) = x^2, y(1) = 2.$

Method 1. Apply the formula.

```
> A(x):=-3/x; B(x):=x^2/x; a(x):=exp(int(A(x),x));
```

$$A(x) := -\frac{3}{x} \quad (3.1.1)$$

$$B(x) := x$$

$$a(x) := \frac{1}{x^3}$$

```
> Gy:=1/a(x)*(int(a(x)*B(x),x)+C);
```

$$Gy := x^3\left(-\frac{1}{x} + C\right) \quad (3.1.2)$$

```
> solve(eval(Gy,x=1)=2,C);
```

$$3 \quad (3.1.3)$$

Therefore, the solution is $y = x^3\left(3 - \frac{1}{x}\right)$.

Method 2. Directly use MAPLE dsolve.

```
> Deq:=x*diff(y(x),x)-3*y(x)=x^2; inc:=y(1)=2;
```

$$Deq := x\left(\frac{d}{dx}y(x)\right) - 3\,y(x) = x^2 \quad (3.1.4)$$

$$inc := y(1) = 2$$

```
> dsolve({Deq, inc},y(x));
```

$$y(x) = \left(-\frac{1}{x} + 3\right)x^3 \quad (3.1.5)$$

Example 2. Solve the initial-value problem $\frac{d}{dx}y(x) + \left(1 - \frac{1}{x}\right)y(x) = x^2, y(1) = 2.$

```
> Deq:=diff(y(x),x)+(1-1/x)*y(x)=x^2; y(1)=2;
```

$$Deq := \frac{d}{dx}y(x) + \left(1 - \frac{1}{x}\right)y(x) = x^2 \quad (3.1.6)$$

$$y(1) = 2$$

```
> dsolve({Deq,inc}, y(x));
```

$$y(x) = x^2 - x + \frac{2\,x\,e^{(-x)}}{e^{(-1)}} \quad (3.1.7)$$

Exercises

1. Solve the equation $x\left(\frac{d}{dx}y(x)\right)+y(x)=x$.
2. Solve the equation $e^{2x}\left(\frac{d}{dx}y(x)\right)=1-e^{x}y(x)$.
3. Solve the initial -value problem $\frac{d}{dx}y(x)+y(x)=\sin(x)$, $y(0)=1$.
4. Solve the initial-value problem $x\left(\frac{d}{dx}y(x)\right)+y(x)=e^{x}$, $y(1)=3$.
5. Solve the initial-value problem $\frac{d}{dx}y(x)+\frac{xy(x)}{1+x^2}=\frac{1}{\left(\sqrt{1+x^2}\right)^3}$, $y(1)=0$.

Chapter 11 INFINITE SERIES

11.1 Sequences

(1) If a sequence is defined recursively, in MAPLE we may use the **"for"** loop to compute its terms.
(2) If a sequence is defined by a function, you may use the limit of the function to determine the convergence of the sequence. (But the reverse may not be true.)
(3) Bounded monotonic sequences converge.

11.1.1 Sequences defined recursively

Example 1. Compute a_2, a_3, a_4 for the sequence defined recursively by

$$a_1 = 1,\ a_n = \frac{a_{n-1} + \frac{2}{a_{n-1}}}{2}.$$

```
> a[1]:=1: for n from 2 to 4 do a[n]:=1/2*(a[n-1]+2/a[n-1]) end do;
```

$$a_2 := \frac{3}{2} \tag{1.1.1}$$

$$a_3 := \frac{17}{12}$$

$$a_4 := \frac{577}{408}$$

Note: If a sequence is defined by a recursive formula, we may apply the **"for"** loop to evaluate its terms. The syntax of the **"for"** loop is the following:
> **for *n* from *starting index* by *step* to *ending number* do *recursive formula* end do.**
Here, if the ***step*** is 1, then **by *step*** can be omitted.

Example 2. Compute $a_2, \ldots, a_5$ for the sequence defined recursively by
$a_1 = \sqrt{2},\ a_n = \sqrt{2 + a_{n-1}}$.

```
> a[1]:=sqrt(2); for n from 2 to 5 do a[n]:=sqrt(2+a[n-1]) end do;
```

$$a_1 := \sqrt{2} \tag{1.1.2}$$

$$a_2 := \sqrt{2 + \sqrt{2}}$$

$$a_3 := \sqrt{2 + \sqrt{2 + \sqrt{2}}}$$

$$a_4 := \sqrt{2 + \sqrt{2 + \sqrt{2 + \sqrt{2}}}}$$

$$a_5 := \sqrt{2 + \sqrt{2 + \sqrt{2 + \sqrt{2 + \sqrt{2}}}}}$$

```
> evalf(seq(a[i],i=1..5));
```

$$1.414213562, 1.847759065, 1.961570561, 1.990369453, 1.997590912 \quad (1.1.3)$$

Observing the terms a_1 to a_5, you may guess that the defined sequence is an increasing bounded sequence with limit 2.

11.1.2 The convergence of sequences defined by functions

Example 3. Find the limit of the sequence $0, \frac{1}{2}, \frac{2}{3}, \frac{3}{4}, \frac{4}{5}, \cdots$.

Step 1. Find the general term of the sequence.

```
> f:=n->(n-1)/n;
```

$$f := n \rightarrow \frac{n-1}{n} \quad (1.2.1)$$

Step 2. Find its limit as the limit of the function $f(n)$.

```
> limit(f(x),x=infinity);
```

$$1 \quad (1.2.2)$$

You can merge them into a single step as follows:

Example 4. Find the limit of the sequence with $a_n = \frac{5n^2}{n^2 - 4}$, $n = 3, 4, \cdots$.

```
> n:='n': assume(n, integer):
> limit(5*n^2/(n^2-4), n=infinity);
```

$$5 \quad (1.2.3)$$

Example 5. Find the limit of the sequence with $a_n = \frac{R^n}{n!}$.

```
> limit(R^n/n!, n=infinity);
```

$$0 \quad (1.2.4)$$

Example 6. Calculate $\lim_{n \to \infty} e^{\frac{3n}{n+1}}$.

```
>  limit(exp(3*n/(n+1)), n=infinity);
```

$$e^3 \quad (1.2.5)$$

11.1.3 Monotonic sequences

Example 7. Verify that $a_n = \sqrt{n+1} - \sqrt{n}$ is decreasing and bounded below.

Method 1. Compute the derivative of the function defining the sequence.

```
> f:=x->sqrt(x+1)-sqrt(x);
```

$$f := x \rightarrow \sqrt{x+1} - \sqrt{x} \quad (1.3.1)$$

```
> solve(D(f)(x)<0, x);
```

$$RealRange(Open(0), \infty) \quad (1.3.2)$$

Since the derivative of $f(x)$ is < 0 on $(0, \infty)$, the sequence is decreasing. It is clear that $0 < a_n$ for all n. Hence it is decreasing and bounded below.

Method 2. Compute $a_{n+1} - a_n$.

```
> df:= f(n+1)-f(n);
```

$$df := \sqrt{n+2} - 2\sqrt{n+1} + \sqrt{n} \qquad (1.3.3)$$

```
> solve(df<0, n);
```

$$RealRange(0, \infty) \qquad (1.3.4)$$

Since the $a_{n+1} - a_n$ is < 0 on $(0, \infty)$, the sequence is decreasing. It is clear that $0 < a_n$ for all n. Hence it is decreasing and bounded below.

Example 8. The sequence is defined by $a_n = \dfrac{3n^2}{n^2+2}$. Show that it is increasing and bounded above.

```
> assume(x>0):
> f:=x->3*x^2/(x^2+1);
```

$$f := x \to \frac{3x^2}{x^2+1} \qquad (1.3.5)$$

```
> D(f);
```

$$x \to \frac{6x}{x^2+1} - \frac{6x^3}{(x^2+1)^2} \qquad (1.3.6)$$

```
> dfx:=simplify(D(f)(x));
```

$$dfx := \frac{6x\sim}{(x\sim^2+1)^2} \qquad (1.3.7)$$

It is clear that $f'(x) > 0$ for $x > 0$. Hence, it is an increasing sequence. It is clear that $\dfrac{3n^2}{n^2+2} < 3$. Hence, it is bounded above.

Exercises

1. Compute a_3, a_4, a_5, a_6 for the sequence defined recursively by $a_1 = 2, a_2 = 5, a_n = a_{n-1} + 2a_{n-2}, n = 3, 4, \cdots$.
2. Compute a_2, a_3, a_4 for the sequence defined recursively by $a_1 = 3, a_{n+1} = 1 + a_n^2, n = 1, 2, \cdots$.
3. Find the limit of the sequence with $a_n = \dfrac{3n+2}{n+4}, n = 1, 2, 3, 4, \cdots$.
4. Find the limit of the sequence with $a_n = \arctan\left(1 - \dfrac{2}{n}\right)$.
5. Calculate $\lim\limits_{n\to\infty} \dfrac{e^{3n}}{n^n}$.
6. Verify that $a_n = \dfrac{1}{n+1}$ is decreasing and bounded below.

11.2 Summing an Infinite Series

(1) An infinite series $\sum_{n=1}^{\infty} a_n$ converges to S if $\lim_{N \to \infty} \left(\sum_{n=1}^{N} a_n \right) = S$.

(2) **Sum of a Geometric Series:** A geometric series with common ratio r converges if $|r| < 1$ and diverges if $|r| \geq 1$. When $|r| < 1$, then

$$\sum_{n=1}^{\infty} cr^n = \frac{c}{1-r}.$$

$$\sum_{n=M}^{\infty} cr^n = \frac{cr^M}{1-r}.$$

11.2.1 Compute sums of series

Example 1. Find the sum of $\sum_{n=1}^{\infty} \frac{1}{n(n+1)} = \frac{1}{1 \cdot 2} + \frac{1}{2 \cdot 3} + \frac{1}{3 \cdot 4} + \cdots$.

```
> S:=sum(1/(n*(n+1)), n=1..infinity);
```

$$S := 1 \qquad (2.1.1)$$

Example 2. Find the sum of $\sum_{n=0}^{\infty} 3^{-n}$.

```
> sum(3^(-n), n=0..infinity);
```

$$\frac{3}{2} \qquad (2.1.2)$$

Example 3. Find the sum of $\sum_{n=3}^{\infty} 7\left(-\frac{3}{4}\right)^n$.

```
> sum(7*(-3/4)^(n), n=3..infinity);
```

$$\frac{-27}{16} \qquad (2.1.3)$$

Example 4. Show that $\sum_{n=1}^{\infty} \frac{1}{\sqrt{n}}$ is divergent.

```
> sum(1/sqrt(n), n=1..infinity);
```

$$\infty \qquad (2.1.4)$$

Example 5. Find the sum of $\sum_{n=0}^{\infty} \frac{2+3^n}{5^n}$.

```
> sum((2+3^n)/5^n, n=0..infinity);
```

$$5 \qquad (2.1.5)$$

Exercises

1. Find the sum of $\sum_{n=1}^{6} \frac{1}{n^2}$.

2. Find the sum of $\sum_{n=1}^{\infty} \frac{1}{4n^2 - 1}$.

3. Find the sum of $\sum_{n=0}^{\infty} \frac{8 + 2^n}{5^n}$.

4. Find the sum of $\sum_{n=0}^{\infty} e^{-n}$.

5. Show that $\sum_{n=2}^{\infty} \frac{1}{\ln(n)}$ is divergent.

11.3 The Ratio and Root Tests

(1) **Ratio Test:** Let $\rho = \lim_{n \to \infty} \left| \frac{a_{n+1}}{a_n} \right|$.

(a) If $\rho < 1$, then $\sum_{n=1}^{\infty} a_n$ converges absolutely.

(b) If $1 < \rho$, then $\sum_{n=1}^{\infty} a_n$ diverges.

(c) If $\rho = 1$, then the Ratio Test is inconclusive. (The series may converge or diverge. You have to use other methods to determine it.)

(2) **Root Test:** Let $L = \lim_{n \to \infty} \sqrt[n]{|a_n|}$.

(a) If $L < 1$, then $\sum_{n=1}^{\infty} a_n$ converges absolutely.

(b) If $1 < L$, then $\sum_{n=1}^{\infty} a_n$ diverges.

(c) If $L = 1$, then the Root Test is inconclusive. (The series may converge or diverge. You have to use other methods to determine it.)

11.3.1 Ratio test

Example 1. Prove that the series $S = \sum_{n=1}^{\infty} \frac{1}{n!}$ converges.

```
> rho:= limit( 1/(n+1)!/(1/n!), n=infinity );
```

$$\rho := 0 \qquad (3.1.1)$$

Hence, $\sum_{n=1}^{\infty} \frac{1}{n!}$ converges.

Example 2. Determine whether $\sum_{n=1}^{\infty} \frac{n^2}{2^n}$ converges.

```
> rho:=limit((n+1)^2/2^(n+1)/(n^2/2^n),n=infinity);
```

$$\rho := \frac{1}{2} \qquad (3.1.2)$$

Hence, $\sum_{n=1}^{\infty} \frac{n^2}{2^n}$ converges.

Example 3. Determine whether $\sum_{n=1}^{\infty} \frac{(-1)^n n!}{100^n}$ converges.

```
> limit((n+1)!/100^(n+1)/(n!/100^n), n=infinity);
```

$$\infty \qquad (3.1.3)$$

Hence, $\sum_{n=1}^{\infty} \frac{(-1)^n n!}{100^n}$ diverges.

11.3.2 Root test

Example 4. Determine whether $\sum_{n=1}^{\infty} \left(\frac{n}{2n+3}\right)^n$ converges.

```
> L:=limit(((n/(2*n+3))^n)^(1/n), n=infinity);
```

$$L := \frac{1}{2} \qquad (3.2.1)$$

Hence, $\sum_{n=1}^{\infty} \left(\frac{n}{2n+3}\right)^n$ converges.

Example 5. Determine whether $\sum_{n=1}^{\infty} \frac{n^3}{5^n}$ converges.

```
> L:=limit((n^3/5^n)^(1/n), n=infinity);
```

$$L := \frac{1}{5} \qquad (3.2.2)$$

Hence, $\sum_{n=1}^{\infty} \frac{n^3}{5^n}$ converges.

Exercises

1. Prove that the series $S = \sum_{n=1}^{\infty} \frac{e^n}{n!}$ converges.

2. Determine whether $\sum_{n=1}^{\infty} \frac{n!}{2^n}$ converges.

3. Determine whether $\sum_{n=1}^{\infty} \frac{n^3}{3^n}$ converges.

4. Determine whether $\sum_{n=1}^{\infty} \frac{1}{10^n}$ converges.

5. Determine whether $\sum_{n=1}^{\infty} \left(\frac{3}{2}\right)^n$ converges.

11.4 Power Series

(1) A power series with the center c is an infinite series of the form

$$F(x) = \sum_{n=0}^{\infty} a_n (x-c)^n F(x)$$

(2) If $F(x)$ converges in $(c-R, c+R)$ and diverges for $R < |x-c|$, then R is called the radius of convergence of $F(x)$.

(3) If $L = \lim_{n\to\infty} \left| \frac{a_{n+1}}{a_n} \right|$ exists, then $R = \frac{1}{L}$.

Note: We can directly compute the radius using

$$R = \lim_{n\to\infty} \left| \frac{a_n}{a_{n+1}} \right|.$$

(4) If $0 < R$, then $F(x)$ is differentiable in $(c-R, c+R)$ and may be differentiated and integrated in $(c-R, c+R)$ term by term:

$$\frac{d}{dx} F(x) = \sum_{n=1}^{\infty} n a_n (x-c)^{n-1}$$

$$\int_c^x F(x)dx = \sum_{n=0}^{\infty} \frac{a_n (x-c)^{n+1}}{n+1}$$

(5) To find the power series of a function $f(x)$ at $x=a$ up to degree n, use the following syntax

```
> series(f(x), x=a, n+1);
```

or

```
> taylor(f(x), x=a, n+1);
```

11.4.1 Find the radius of convergence of a power series

Note: When you work on the problems for series, you can assume that the index of a term is integer:

```
> assume(n, integer):
```

Example 1. Find the radius of convergence of the power series $\sum_{n=0}^{\infty} \frac{x^n}{2^n}$.

```
> f:=n->1/2^n;
```

$$f := n \to \frac{1}{2^n} \tag{4.1.1}$$

```
> R:=limit(f(n)/f(n+1),n=infinity);
```

$$R := 2 \tag{4.1.2}$$

Example 2. Find the radius of convergence of the power series $\sum_{n=0}^{\infty} \frac{x^n}{n!}$.

```
> f:=n->1/n!;
```

$$f := n \to \frac{1}{n!} \tag{4.1.3}$$

```
> R:=limit(f(n)/f(n+1),n=infinity);
```

$$R := \infty \tag{4.1.4}$$

Hence, the power series $\sum_{n=0}^{\infty} \frac{x^n}{n!}$ converges everywhere.

Example 3. Find the radius of convergence of the power series $\sum_{n=0}^{\infty} x^n n!$

```
> f:=n->n!;
```

$$f := n \to n! \tag{4.1.5}$$

```
> R:=limit(f(n)/f(n+1),n=infinity);
```

$$R := 0 \tag{4.1.6}$$

Hence, the power series $\sum_{n=0}^{\infty} x^n n!$ converges only at $x = 0$.

11.4.2 Calculus of power series

Example 4. Use the identity $\frac{1}{1-x} = \sum_{n=0}^{\infty} x^n$ to derive the power series of $\frac{1}{(1-x)^2}$ at $x = 0$.

```
> f:=x->1/(1-x);
```

$$f := x \to \frac{1}{1-x} \tag{4.2.1}$$

```
> Df:= D(f);
```

$$Df := x \to \frac{1}{(1-x)^2} \tag{4.2.2}$$

```
> assume(n>0):
> anx:=x^n;
```

$$anx := x^{n\sim} \tag{4.2.3}$$

```
> nanx:=simplify(diff(anx, x));
```

$$nanx := x^{(n\sim -1)}\, n\sim \tag{4.2.4}$$

Hence, $\frac{1}{(1-x)^2} = \sum_{n=1}^{\infty} nx^{n-1}$.

You can verify the result by the following:

```
> series(1/(1-x)^2, x=0, 7);
```

$$1 + 2x + 3x^2 + 4x^3 + 5x^4 + 6x^5 + 7x^6 + \mathrm{O}(x^7) \tag{4.2.5}$$

Or, you can try the following:

```
> sum(n*x^(n-1), n=1..infinity);
```

$$\frac{1}{(x-1)^2} \tag{4.2.6}$$

Note: In the display (4.2.5), $\mathrm{O}(x^7)$ stands for the term which is comparable to x^7 as x tends to 0.

Example 5. Using the identity $\frac{1}{1+x^2} = \sum_{n=0}^{\infty} (-1)^n x^{2n}$ to derive the power series of $\arctan(x)$ at $x = 0$.

```
> int(1/(1+x^2),x=0..x);
```

$$\arctan(x) \tag{4.2.7}$$

```
> nanx:=int(x^(2*n),x=0..x);
```

$$nanx := \frac{x^{(2\,n\sim+1)}}{2\,n\sim+1} \tag{4.2.8}$$

Hence, $\arctan(x) = \sum_{n=0}^{\infty} \frac{(-1)^n x^{2n+1}}{2n+1}$.

You can verify the result by the following:

```
> series(arctan(x), x=0, 11);
```

$$x - \frac{1}{3}x^3 + \frac{1}{5}x^5 - \frac{1}{7}x^7 + \frac{1}{9}x^9 + \mathrm{O}(x^{11}) \tag{4.2.9}$$

Warning: To find the sum of a series involves infinitely many additions, which cannot be performed by computers. Hence, MAPLE will search the basic formulas of the series representation of functions and then perform the function operations to determine whether there exists an explicit function whose series is the given one. It usually takes a long time to get the result, and the display of the result function is not unique, since a function may have infinitely many different representations.

For example, if you do the following:

```
> sum((-1)^n*x^(2*n+1)/(2*n+1), n=0.. infinity);
```

$$-\frac{1}{2}\,\mathrm{I}\ln\left(\frac{1+\mathrm{I}\,x}{1-\mathrm{I}\,x}\right) \tag{4.2.10}$$

Then the result is a complex function, which also represents arctan(x).

Example 6. Show that for $|x| < 1$,

$$\frac{1+2x}{1+x+x^2} = 1+x-2x^2+x^3+x^4-2x^5+x^6+x^7-2x^8+\cdots.$$

```
> series((1+2*x)/(1+x+x^2), x=0, 9);
```

$$1+x-2\,x^2+x^3+x^4-2\,x^5+x^6+x^7-2\,x^8+\mathrm{O}\left(x^9\right) \tag{4.2.11}$$

Exercises

1. Find the radius of convergence of the power series $\sum_{n=1}^{\infty}\frac{x^n}{3^n}$ and determine its convergence at the two endpoints of the convergent interval.

2. Find the radius of convergence of the power series $\sum_{n=0}^{\infty}(x-2)^n$ and determine its convergence at the two endpoints of the convergent interval.

3. Find the radius of convergence of the power series $\sum_{n=0}^{\infty}\frac{x^n}{n^2}$.

4. Use the identity $\frac{1}{1-x} = \sum_{n=0}^{\infty}x^n$ to expand the function $\frac{1}{1-3x}$ in a power series with center $c =$ 0 and determine the set of x for which the expansion is valid.

5. Evaluate $\sum_{n=1}^{\infty}\frac{n}{2^n}$ using (a) the formula $\frac{1}{(1-x)^2} = \sum_{n=1}^{\infty}nx^{n-1}$, and (b) the MAPLE syntax.

11.5 Taylor Series

(1) If $f(x)$ is represented by a power series $F(x)$ centered at c on an interval $(c-R, c+R)$ with $0 < R$, then $F(x)$ is the **Taylor series** of $f(x)$ centered at $x = c$.
(2) A Taylor series centered at $x = 0$ is called a **Maclaurin series**.
(3) If $f(x)$ is represented by a power series $F(x)$ centered at c on an interval $(c-R, c+R)$ and $T_k(x)$ is the Taylor polynomial of degree k, then for any x in $(c-R, c+R)$, the error bound for the Taylor polynomial $T_k(x)$ is

$$|R_k(x)| = \left|f(x)-T_k(x)\right| \le \frac{KR^{k+1}}{(k+1)!}.$$

where K is the maximum value of $\left|\frac{\mathrm{d}^k}{\mathrm{d}x^k}f(x)\right|$ on $[c, x]$, if $c < x$; or on $[x, c]$, if $x < c$.

▶ 11.5.1 Find Taylor polynomials

Example 1. Find the Maclaurin series of $f(x) = e^x$ and display it up to degree 6.

```
> T:=taylor(exp(x),x=0,7);
```

$$T := 1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3 + \frac{1}{24}x^4 + \frac{1}{120}x^5 + \frac{1}{720}x^6 + O\left(x^7\right) \tag{5.1}$$

Example 2. Find the Maclaurin series of $f(x) = x^2 e^x$ and display it up to degree 6.

```
> T:=taylor(x^2*exp(x), x=0, 7);
```

$$T := x^2 + x^3 + \frac{1}{2}x^4 + \frac{1}{6}x^5 + \frac{1}{24}x^6 + O\left(x^7\right) \tag{5.2}$$

Example 3. Find the Maclaurin series of $f(x) = e^x \cos(x)$ and display it up to degree 8.

```
> T8:=taylor(exp(x)*cos(x), x=0, 9);
```

$$T8 := 1 + x - \frac{1}{3}x^3 - \frac{1}{6}x^4 - \frac{1}{30}x^5 + \frac{1}{630}x^7 + \frac{1}{2520}x^8 + O\left(x^9\right) \tag{5.3}$$

Example 4. Find the Maclaurin series of $f(x) = (1+x)^{\frac{1}{3}}$ and display it up to degree 7.

```
> T8:=taylor((1+x)^(1/3), x=0,8);
```

$$T8 := 1 + \frac{1}{3}x - \frac{1}{9}x^2 + \frac{5}{81}x^3 - \frac{10}{243}x^4 + \frac{22}{729}x^5 - \frac{154}{6561}x^6 + \frac{374}{19683}x^7 + O\left(x^8\right) \tag{5.4}$$

Example 5. Let $J = \int_0^1 \sin\left(x^2\right) dx$. (a) Express J as an infinite series. (b) Determine J to within an error less than 10^{-4}.

(a) Express J as an infinite series.

```
> assume(n, integer); assume(n>0);
> ntermOfsinx:=x->(-1)^n*x^(2*n+1)/(2*n+1)!;
```

$$ntermOfsinx := x \rightarrow \frac{(-1)^n x^{(2n+1)}}{(2n+1)!} \tag{5.5}$$

```
> ntermOfsinxsqr:=ntermOfsinx(x^2);
```

$$ntermOfsinxsqr := \frac{(-1)^{n\sim} \left(x^2\right)^{(2n\sim+1)}}{(2n\sim+1)!} \tag{5.6}$$

```
> ntermOfJ:= int(ntermOfsinxsqr, x=0..1);
```

$$ntermOfJ := \frac{(-1)^{n\sim}}{(2n\sim)!\left(8n\sim^2 + 10n\sim + 3\right)} \tag{5.7}$$

```
> J:=Sum(ntermOfJ, n=0..infinity);
```

$$J := \sum_{n\sim=0}^{\infty} \frac{(-1)^{n\sim}}{(2n\sim)!\left(8n\sim^2 + 10n\sim + 3\right)} \tag{5.8}$$

Note: The syntax "**Sum**" introduces the display of a series without getting its sum. For example,

```
> sum(1/n!, n=0..infinity);
```

$$\mathrm{e} \tag{5.9}$$

```
> Sum(1/n!, n=0..infinity);
```

$$\sum_{n\sim=0}^{\infty} \frac{1}{n\sim!} \tag{5.10}$$

(b) Determine J to within an error less than 10^{-4}.

```
> map(n-> factorial(2*n)*(8*n^2+10*n+3), [1,2,3,4]);
```

$$[42, 1320, 75600, 6894720] \tag{5.11}$$

Since $10000 < 75600$, we have

```
> AppJ:=sum(ntermOfJ, n=0..2);
```

$$AppJ := \frac{2867}{9240} \tag{5.12}$$

```
> evalf(AppJ,5);
```

$$0.31028 \tag{5.13}$$

Example 6. Find the first 4 terms of the Taylor series of $f(x) = \dfrac{1}{1-x^2}$ at $x=3$.

```
> T4:= taylor(1/(1-x^2), x=3, 4);
```

$$T4 := -\frac{1}{8} + \frac{3}{32}(x-3) - \frac{7}{128}(x-3)^2 + \frac{15}{512}(x-3)^3 + \mathrm{O}\left((x-3)^4\right) \tag{5.14}$$

Exercises

1. Find the Maclaurin series of $f(x) = \dfrac{x}{1-x^4}$ and display it up to degree 6.
2. Find the Maclaurin series of $f(x) = x^2 \mathrm{e}^{x^2}$ and display it up to degree 10.
3. Find the Maclaurin series of $f(x) = \sqrt{1+x}$ and display it up to degree 6.
4. Find the first five terms of the Maclaurin series of $f(x) = \mathrm{e}^x \ln(1-x)$.
5. Find the first five terms of the Taylor series of $f(x) = \sqrt{x}$ at $x=4$.
6. Let $F(x) = \displaystyle\int_0^x \frac{\sin(t)}{t}\,\mathrm{d}t$. (a) Show that $F(x) = x - \dfrac{x^3}{3\cdot 3!} + \dfrac{x^5}{5\cdot 5!} - \dfrac{x^7}{7\cdot 7!} + \cdots$. (b) Evaluate $F(1)$ to three decimal places.

Chapter 12 PARAMETRIC EQUATIONS, POLAR COORDINATES, AND CONIC SECTIONS

12.1 Parametric Equations

(1) $x = x(t)$, $y = y(t)$, $a \le t \le b$, is a parametric equation of the plane path or curve.
(2) Each path may be parameterized in infinitely many ways. Eliminating the parameter t will lead to the curve equation of the form of $F(x, y) = 0$.

(3) The slope of the tangent line at $c(t) = (x(t), y(t))$ is $\frac{d}{dx}y(x) = \frac{\frac{d}{dt}y(t)}{\frac{d}{dt}x(t)}$ (valid if $\frac{d}{dt}x(t) \neq 0$).

12.1.1 Convert from/to parametric equations

Example 1. Eliminate the parameter t from $c(t) = (2t - 4, 3 + t^2)$.

```
> tx:=solve(x=2*t-4,t);
```

$$tx := \frac{1}{2}x + 2 \tag{1.1.1}$$

```
> eq:=subs(t=tx, y=3+t^2);
```

$$eq := y = 3 + \left(\frac{1}{2}x + 2\right)^2 \tag{1.1.2}$$

Note: If $x(t)$ and $y(t)$ are involved in trigonometric functions, we may apply the trigonometric identities to eliminate the parameter t.

Example 2. Eliminate the parameter t from $c(t) = (2\cos(t) - 4, 3\sin(t) + 2)$.

```
> cost:=solve(x=2*cos(t)-4,cos(t));
```

$$cost := \frac{1}{2}x + 2 \tag{1.1.3}$$

```
> sint:=solve(y=3*sin(t)+2, sin(t));
```

$$sint := \frac{1}{3}y - \frac{2}{3} \tag{1.1.4}$$

```
> eq:= cost^2+sint^2=1;
```

$$eq := \left(\frac{1}{2}x + 2\right)^2 + \left(\frac{1}{3}y - \frac{2}{3}\right)^2 = 1 \tag{1.1.5}$$

Note: Do not use cos(t) and sin(t) to replace the names of variables, **sint** and **cost**, above. Note that cos(t) and sin(t) are built-in functions of MAPLE; they cannot be reassigned for the names of variables. But **sint** and **cost** are not the built-in functions. Hence, they can be used as the names of variables.

Example 3. Show that the ellipse $\left(\frac{x}{a}\right)^2+\left(\frac{y}{b}\right)^2=1$ can be parameterized by $c(t)=(a\cos(t), b\sin(t)),\ 0\le t<2\pi$.

```
> xt:= a*cos(t); yt:=b*sin(t);
```

$$xt := a\cos(t) \quad (1.1.6)$$

$$yt := b\sin(t)$$

```
> simplify((xt/a)^2+(yt/b)^2, trig);
```

$$1 \quad (1.1.7)$$

12.1.2 Draw parametric curves

We may draw a plane curve in several ways depending on the form of the equation that describes the curve.

(a) If the curve is described by a function $y=f(x),\ x\in[a,b]$, then we may draw the curve using the following syntax:

> **plot(f(x), x=a..b);**

or

>**plot(f, a..b);**

(b) If the curve is described by an equation $F(x,y)=C$, then you have to use the **"implicitplot"** command in the package **"plots"**:

> **plots[implicitplot] (F(x,y)=C, x=a..b, y=c..d);**

or

> **with(plots):**

> **implicitplot(F(x,y)=C, x=a..b, y=c..d);**

(c) If the curve is given by a parametric equations $x=x(t),\ y=y(t),\ t\in[a,b]$, then the following syntax is applied.

> **plot([x, y, a..b])**, if functions are assigned by the input-output mode,

or

> **plot([x(t),y(t), t=a..b])**, if functions are assigned by the expression mode.

Example 4. Draw the graph of the ellipse $\left(\frac{x}{2}\right)^2+y^2=1$ using different methods. (a) Change it to an explicit function form. (b) Use implicit plot. (c) Use its parametric equation $x=2\cos(t),\ y=\sin(t),\ t\in[-\pi,\pi]$.

(a) Change $\left(\frac{x}{2}\right)^2+y^2=1$ to the function $y=f(x)$, then plot the graph.

```
> eq:=solve((x/2)^2+y^2=1,y);
```

$$eq := \frac{1}{2}\sqrt{-x^2+4},\ -\frac{1}{2}\sqrt{-x^2+4} \quad (1.2.1)$$

```
> plot({eq[1], eq[2]}, x=-2..2, scaling=constrained, color=[red,
  red]);
```



Or do the following:

```
> f:=x->1/2*(-x^2+4)^(1/2); g:=x->-1/2*(-x^2+4)^(1/2);
```

$$f := x \rightarrow \frac{1}{2}\sqrt{-x^2+4} \tag{1.2.2}$$

$$g := x \rightarrow -\frac{1}{2}\sqrt{-x^2+4}$$

```
> plot({f,g}, -2..2, scaling=constrained, color=[red,red]);
```



(b) Use implicit plot.

```
> plots[implicitplot]((x/2)^2+y^2=1, x=-2..2, y=-1..1, scaling=
  constrained);
```



(c) Use the parametric equation $x=2\cos(t)$, $y=\sin(t)$, $t\in[-\pi,\pi]$.

```
> plot([2*cos, sin,-Pi..Pi],scaling=constrained);
```



Or do the following.

```
> plot([2*cos(t), sin(t),t=-Pi..Pi],scaling=constrained);
```



Example 5. Sketch the graph of $c(t) = \left(t^2, \sin(t)\right)$ for $t \in [-2\pi, 2\pi]$.

```
> plot([t^2, sin(t), t=-2*Pi..2*Pi]);
```



12.1.3 Slope of parametric curves

Example 6. Let $c(t) = \left(t^2 + 1, t^3 - 4t\right)$. (a) Find the equations of the tangent line at $t = 3$ and draw the line and the curve in a figure. (b) Find the points where the tangent is horizontal.
(a) Find the equations of the tangent line at $t = 3$ and draw the line and the curve in a figure.

```
> xt:=t^2+1; yt:=t^3-4*t;
```

$$xt := t^2 + 1 \tag{1.3.1}$$

$$yt := t^3 - 4t$$

```
> dyx3:= eval(diff(yt,t)/diff(xt,t), t=3);
```

$$dyx3 := \frac{23}{6} \tag{1.3.2}$$

```
> x3:=eval(xt,t=3); y3:=eval(yt,t=3);
```

$$x3 := 10 \tag{1.3.3}$$

$$y3 := 15$$

```
> teq:=dyx3*(x-x3)+y3;
```

$$teq := \frac{23}{6}\,x - \frac{70}{3} \tag{1.3.4}$$

```
> plt1:=plot([xt,yt,t=1..4], color=red):
> plt2:=plot(teq, x=6..14, color=blue):
> plots[display]({plt1, plt2});
```



Note: When two objects need to be plotted in a screen, but they are defined in different ways. assign them to plotting objects as above, then use the **"display"** command in the package **"plots"** to display them.

(b) Find the points where the tangent is horizontal.

```
> ht:=solve(diff(yt,t)=0,t);
```

$$ht := \frac{2}{3}\sqrt{3}, -\frac{2}{3}\sqrt{3} \tag{1.3.5}$$

The two points are the following:

```
> p1:=eval([xt,yt], t=ht[1] );
```

$$p1 := \left[\frac{7}{3}, -\frac{16}{9}\sqrt{3}\right] \tag{1.3.6}$$

```
> p2:=eval([xt,yt], t=ht[2] );
```

$$p2 := \left[\frac{7}{3}, \frac{16}{9}\sqrt{3}\right] \tag{1.3.7}$$

Exercises

1. Eliminate the parameter t from $c(t) = \left(t^2, t^3 + 1\right)$.
2. Eliminate the parameter t from $c(t) = (2\sec(t), 3\tan(t) + 1)$. *Hint: Use the identity* $\sec(t)^2 - \tan(t)^2 = 1$.
3. Show that the ellipse $\left(\frac{x-c}{a}\right)^2 + \left(\frac{y-d}{b}\right)^2 = 1$ can be parameterized by $c(t) = (a\cos(t) + c, b\sin(t) + d), 0 \le t < 2\pi$.
4. Draw the graph of the cycloid $c(t) = (t - \sin(t), 1 - \cos(t))$ for $t \in [0, 2\pi]$.
5. Sketch the graph of $c(t) = \left(t^2 - 4t, 9 - t^2\right)$ for $t \in [-4, 10]$.
6. Let $c(t) = \left(\frac{t}{2}, \frac{t^2}{4} - t\right)$. (a) Find the equations of the tangent line at $t = 1$ and draw the line and the curve in a figure. (b) Find the points where the tangent is vertical.
7. Let $c(t) = \left(t^2 - 9, t^2 - 8t\right)$. (a) Find the equations of the tangent line at $t = 4$ and draw the line and the curve in a figure. (b) Find the points where the tangent has slope $\frac{1}{2}$.

12.2 Arc Length and Speed

(1) The arc length of the curve $y = f(x)$, $x \in [a, b]$, is

$$L = \int_a^b \sqrt{1 + \left(\frac{d}{dx} f(x)\right)^2}\, dx$$

and for the curve $x = x(t)$, $y = y(t)$, t in $[a, b]$ is

$$L = \int_a^b \sqrt{\left(\frac{d}{dt} x(t)\right)^2 + \left(\frac{d}{dt} y(t)\right)^2}\, dt$$

(2) The speed at time t of a particle with trajectory $c(t) = (x(t), y(t))$ is

$$sp := \sqrt{\left(\frac{d}{dt} x(t)\right)^2 + \left(\frac{d}{dt} y(t)\right)^2}$$

(3) The surface area S of the surface obtained by rotating the curve $c(t) = (x(t), y(t))$ about the x-axis for $t \in [a, b]$ is

$$S = 2\pi \int_a^b y(t) \sqrt{\left(\frac{d}{dt} x(t)\right)^2 + \left(\frac{d}{dt} y(t)\right)^2}\, dt$$

12.2.1 Arc length

Example 1. Calculate the length s of the arc $0 \le \theta \le \theta_0$ of a circle of radius R.

```
> xt:=R*cos(t): yt:=R*sin(t): assume(R>0):
> ds:=simplify(sqrt(diff(xt,t)^2+diff(yt,t)^2),trig);
```

$$ds := R\sim \tag{2.1.1}$$

```
> s:= int(ds, t=0..theta[0]);
```

$$s := R\sim\,\theta_0 \tag{2.1.2}$$

Hence, the length s is $R\theta_0$.

Example 2. Calculate the length of one arch of the cycloid generated by a circle of radius 2.

```
> xt:=2*(t-sin(t)): yt:=2*(1-cos(t)):
> ds:=simplify(sqrt(diff(xt,t)^2+diff(yt,t)^2),trig);
```

$$ds := 2\sqrt{2-2\cos(t)} \tag{2.1.3}$$

```
> s:= int(ds, t=0..2*Pi);
```

$$s := 16 \tag{2.1.4}$$

12.2.2 Speed

Example 3. A particle travels along the path $c(t) = \left(2t,\, 1+\sqrt{t^3}\right)$. (a) Find the speed at $t=1$. (b) Compute the distance traveled and the displacement during time from 0 to 4.

(a) Find the speed at $t=1$.

```
> xt:= t->2*t: yt:=t->1+t^(3/2):
> sp:= t->sqrt(D(xt)(t)^2 + D(yt)(t)^2);
```

$$sp := t \rightarrow \sqrt{(\mathrm{D}(xt))(t)^2 + (\mathrm{D}(yt))(t)^2} \tag{2.2.1}$$

```
> sp1:=simplify(sp(1));
```

$$sp1 := \frac{5}{2} \tag{2.2.2}$$

(b) Compute the distance traveled and the displacement during time from 0 to 4.

```
> s:=int(sp(t), t=0..4);
```

$$s := -\frac{16}{9}\,\frac{-\frac{13}{6}\sqrt{\pi}\sqrt{13}+\frac{4}{3}\sqrt{\pi}}{\sqrt{\pi}} \tag{2.2.3}$$

```
> nums:=evalf(s);
```

$$nums := 11.51767899 \tag{2.2.4}$$

```
> disp:=sqrt((xt(4)-xt(0))^2+(yt(4)-yt(0))^2);
```

$$disp := 8\sqrt{2} \tag{2.2.5}$$

Example 4. A particle travels along the path $c(t) = \left(t^3 - 4t, t^2 + 1\right)$ for $t \geq 0$. Find its minimal speed.

```
> xt:=t^3-4*t: yt:=t^2+1:
> sp:= sqrt(diff(xt,t)^2 + diff(yt,t)^2);
```

$$sp := \sqrt{9\,t^4 - 20\,t^2 + 16} \tag{2.2.6}$$

```
> minimize(sp, t, location);
```

$$\frac{1}{9}\sqrt{44}\sqrt{9}, \left\{\left[\left\{t = \frac{1}{3}\sqrt{10}\right\}, \frac{1}{9}\sqrt{44}\sqrt{9}\right], \left[\left\{t = -\frac{1}{3}\sqrt{10}\right\}, \frac{1}{9}\sqrt{44}\sqrt{9}\right]\right\} \tag{2.2.7}$$

The minimal speed is $\sqrt{\frac{44}{9}}$ at $x = \frac{\sqrt{10}}{3}$.

12.2.3 Surface area

Example 5. Calculate the surface area of the infinite surface generated by revolving the tractrix $c(t) = (t - \tanh(t), \operatorname{sech}(t))$ about the x-axis for $t \geq 0$.

```
> xt:= t-tanh(t): yt:=sech(t): assume(t>0);
> ds:= simplify( sqrt(diff(xt,t)^2 + diff(yt,t)^2));
```

$$ds := \frac{\sinh(t\sim)}{\cosh(t\sim)} \tag{2.3.1}$$

```
> S:=2*Pi*int(yt*ds, t=0..infinity);
```

$$S := 2\,\pi \tag{2.3.2}$$

Example 6. Calculate the surface area of the sphere of radius R.

```
> Assume(R>0): xt:=R*cos(t): yt:=R*sin(t):
> ds:= simplify( sqrt(diff(xt,t)^2 + diff(yt,t)^2), trig);
```

$$ds := R\sim \tag{2.3.3}$$

```
> S:=2*Pi*int(yt*ds, t=0..Pi);
```

$$S := 4\,\pi\,R\sim^2 \tag{2.3.4}$$

Exercises

1. Calculate the length s of the semicircle $c(t) = (3\cos(t), 3\sin(t))$.
2. Calculate the length of $c(t) = \left(2t^2, 3t^2 - t\right)$, for $t \in [1, 4]$.
3. A particle travels along the path $c(t) = (3\sin(5t), 8\cos(5t))$. (a) Find its speed at $t = \frac{\pi}{4}$. (b) Compute the distance traveled and the displacement during time from 0 to $\frac{\pi}{4}$ (round off to 3 decimal places).
4. A particle travels along the path $c(t) = \left(t^3, \frac{1}{t^2}\right)$ for $t \geq 1/2$. Find its minimal speed.
5. Calculate the surface area of the surface generated by revolving the astroid $c(t) = \left(\cos(t)^3, \sin(t)^3\right)$ about the x-axis for $t \in [0, \frac{\pi}{2}]$.
6. Calculate the surface area of the surface generated by revolving the cycloid $c(t) = (t - \sin(t), 1 - \cos(t))$ about the x-axis for $t \in [0, 2\pi]$.

12.3 Polar Coordinates

(1) Convert from Polar to Rectangular:

$$x = r\cos(\theta), y = r\sin(\theta).$$

(2) Convert from Rectangular to Polar:

$$r = \sqrt{x^2 + y^2}, \tan(\theta) = \frac{y}{x}.$$

and

$$\theta = \arctan\left(\frac{y}{x}\right), \quad \text{if } 0 < x$$
$$\theta = \arctan\left(\frac{y}{x}\right) + \pi, \quad \text{if } x < 0$$
$$\theta = \frac{\pi}{2} \text{ or } -\frac{\pi}{2}, \quad \text{if } x = 0$$

12.3.1 Equations in Polar

Example 1. Convert $(r, \theta) = \left(3, \frac{5\pi}{6}\right)$ to Rectangular.

```
> plar:=[3, 5*Pi/6];
```

$$plar := \left[3, \frac{5}{6}\pi\right] \quad (3.1.1)$$

```
> rct:=[plar[1]*cos(plar[2]), plar[1]*sin(plar[2])];
```

$$rct := \left[-\frac{3}{2}\sqrt{3}, \frac{3}{2}\right] \quad (3.1.2)$$

Example 2. Convert $(x, y) = (-1, 1)$ to Polar.
Recall that $x = -1 < 0$.

```
> rct:= [-1,1];
```

$$rct := [-1, 1] \quad (3.1.3)$$

```
> plar:=[sqrt(rct[1]^2+rct[2]^2), arctan(rct[2]/rct[1])+Pi];
```

$$plar := \left[\sqrt{2}, \frac{3}{4}\pi\right] \quad (3.1.4)$$

Example 3. Find the line $y = \frac{3x}{2}$ in Polar.

```
> tthe:=subs({y=r*sin(theta), x=r*cos(theta)}, y=3/2*x );
```

$$tthe := r\sin(\theta) = \frac{3}{2}r\cos(\theta) \quad (3.1.5)$$

```
> eq:= solve(tthe, theta);
```

$$eq := \arctan\left(\frac{3}{2}\right) \quad (3.1.6)$$

Hence, the equation is $\theta = \arctan\left(\frac{3}{2}\right)$.

To plot the graph of an equation in Polar system, add the option **"coords=polar"** in the plot.

Example 4. Draw the line of $\theta = \arctan\left(\frac{3}{2}\right)$.

```
> plot([r, arctan(3/2), r=-3..3], coords=polar);
```



If the graph is described by the polar function $r = r(\theta)$, then the graph can be plotted as a function.

Example 5. Draw the graph of the circle $r = 2$ in Polar.

```
> plot(2, theta=0..2*Pi, coords=polar, scaling =constrained);
```



Example 6. Draw the graph of the circle $r = 2\cos(\theta)$ in Polar.

```
> plot(2*cos(theta), theta=0..Pi, coords=polar, scaling =
  constrained);
```



Example 7. Draw the graph of $r = 2\sin(3\theta)$.

```
> plot(2*sin(3*theta), theta=0..Pi, coords=polar, scaling =
  constrained);
```



Exercises

1. Convert $(r, \theta) = \left(4, \frac{3\pi}{4}\right)$ to Rectangular.
2. Convert $(x, y) = \left(-1, \sqrt{3}\right)$ to Polar.
3. Convert $r = 2\sin(\theta)$ in Rectangular.
4. Convert $y = x^2$ to Polar equation $r = r(\theta)$.
5. Draw the graph of the circle $r = 2\sin(2\theta)$.
6. Draw the graph of $r = \cos(\theta) - 1$.

12.4 Area and Arc Length in Polar Coordinates

(1) **Area in Polar Coordinates:** If $r = f(\theta)$ is a continuous function, then the area bounded by a curve in polar form $r = f(\theta)$ and the rays $\theta = \alpha$, $\theta = \beta$ is

$$S = \int_{\alpha}^{\beta} \frac{f(\theta)^2}{2} \, d\theta .$$

and the area between two curves $r = f(\theta)$ and $r = g(\theta)$ within the section $\alpha \le \theta \le \beta$ is

$$S = \int_{\alpha}^{\beta} \frac{\left| f(\theta)^2 - g(\theta)^2 \right|}{2} \, d\theta .$$

(2) The arc length of the curve of $r = f(\theta)$ with $\alpha \le \theta \le \beta$ is

$$L = \int_{\alpha}^{\beta} \sqrt{f(\theta)^2 + \left(\frac{d}{d\theta} f(\theta) \right)^2} \, d\theta .$$

12.4.1 Area in polar coordinates

Example 1. Find the area of the right semicircle with equation $r = 4\sin(\theta)$.

```
> plot(4*sin(theta), theta=0..Pi/2, coords=polar, scaling=
  constrained);
```




```
> CArea:=1/2*int((4*sin(theta))^2, theta=0..Pi/2);
```

$$CArea := 2\,\pi \qquad (4.1.1)$$

Example 2. Sketch the graph of $r = \sin(3\theta)$ and find the area of one "petal".

```
> plot(sin(3*theta), theta=0..Pi, coords = polar, scaling=
  constrained);
```



One petal of the graph is from $t = 0$ to $t = \frac{2\pi}{3}$. Its graph is shown in the following:

```
> plot(sin(3*theta), theta=0..Pi/3, coords = polar, scaling=
  constrained);
```




```
> CArea:= 1/2*int(sin(3*theta)^2, theta=0..Pi/3);
```

$$CArea := \frac{1}{12}\pi \tag{4.1.2}$$

Example 3. Find the area of the region inside the circle $r=2\cos(\theta)$ but outside the circle $r=1$.

```
> plot([2*cos(theta), 1], theta=-Pi/2..Pi/2, color=[red, blue],
  coords=polar, scaling=constrained);
```



(a) Find the intersections of the two curves.

```
> solve(2*cos(theta)=1, theta);
```

$$\frac{1}{3}\pi \tag{4.1.3}$$

Since $\cos(\theta)$ is even, another point is for $t=-\frac{\pi}{3}$.

(b) Find the area between two curves for $t\in[-\frac{\pi}{3}, \frac{\pi}{3}]$.

```
> CArea:=1/2*int((2*cos(theta))^2-1, theta=-Pi/3..Pi/3);
```

$$CArea := \frac{1}{2}\sqrt{3}+\frac{1}{3}\pi \tag{4.1.4}$$

12.4.2 Curve length

Example 4. Find the length of the curve of $r=e^{\theta}$ for $\theta\in[0, \pi]$.

```
> assume(theta>0): r:= exp(theta); dr:=diff(r,theta);
```

$$r := e^{\theta\sim} \tag{4.2.1}$$

$$dr := e^{\theta\sim}$$

```
> ds :=simplify(sqrt(r^2+dr^2));
```

$$ds := \sqrt{2}\, e^{\theta\sim} \tag{4.2.2}$$

```
> L:=int(ds,theta=0..Pi);
```

$$L := -\sqrt{2}+\sqrt{2}\, e^{\pi} \tag{4.2.3}$$

Example 5. Find the length of one "petal" of the curve of $r = \sin(3\theta)$.

```
> r:=sin(3*theta); dr:=diff(r,theta);
```

$$r := \sin(3\,\theta\sim) \tag{4.2.4}$$

$$dr := 3\cos(3\,\theta\sim)$$

```
> ds:=simplify(sqrt(r^2 +dr^2),trig);
```

$$ds := \sqrt{8\cos(3\,\theta\sim)^2 + 1} \tag{4.2.5}$$

```
> L:=int(ds, theta=0..Pi/3);
```

$$L := 2\,\mathrm{EllipticE}\left(\frac{2}{3}\sqrt{2}\right) \tag{4.2.6}$$

```
> evalf(L);
```

$$2.227482204 \tag{4.2.7}$$

Exercises

1. Find the area of the one leaf of the "four-petaled rose" $r = \sin(2\theta)$.
2. Find the area enclosed by one loop of the lemniscate with equation $r^2 = \cos(2\theta)$.
3. Find the area enclosed by the cardioid $r = 2(1 + \cos(\theta))$.
4. Find the length of the curve of $r = \sin(2\,\theta)$ for $\theta \in [0, \pi]$.
5. Find the length of inner loop of the curve of $r = 2\cos(\theta) - 1$.

12.5 Conic Sections

(1) Equation of an **ellipse** in the standard position:

$$\left(\frac{x}{a}\right)^2 + \left(\frac{y}{b}\right)^2 = 1$$

eccentricity:

$$e = \frac{\sqrt{a^2 - b^2}}{a}$$

(2) Equation of a **hyperbola** in the standard position:

$$\left(\frac{x}{a}\right)^2 - \left(\frac{y}{b}\right)^2 = 1$$

or

$$-\left(\frac{x}{a}\right)^2 + \left(\frac{y}{b}\right)^2 = 1$$

eccentricity:

$$e = \frac{\sqrt{a^2 + b^2}}{a}$$

(3) Equation of a **parabola** in the standard position:

$$y = \frac{x^2}{4c}$$

or

$$x = \frac{y^2}{4c}$$

and the **eccentricity** of a parabola is 1.

12.5.1 Equations of conic sections

Example 1. Find the equation of the ellipse with foci $(\sqrt{11}, 0)$, $(-\sqrt{11}, 0)$ and $a = 6$. Then plot its graph.

```
> a:=6: c:=sqrt(11):
> b:=sqrt(a^2-c^2);
```

$$b := 5 \tag{5.1.1}$$

```
> Eq:= (x/a)^2+(y/b)^2=1;
```

$$Eq := \frac{1}{36}x^2 + \frac{1}{25}y^2 = 1 \tag{5.1.2}$$

```
> plots[implicitplot](Eq, x=-6..6, y=-5..5, scaling=constrained)
;
```



Example 2. Find the equation of the ellipse with vertical focal axis, center at $C = (6, 7)$, semimajor axis 5, and semiminor axis 3. Then sketch the graph.

```
> Eq:=((x-6)/3)^2+((y-7)/5)^2=1;
```

$$Eq := \left(\frac{1}{3}x - 2\right)^2 + \left(\frac{1}{5}y - \frac{7}{5}\right)^2 = 1 \tag{5.1.3}$$

```
> plots[implicitplot](Eq, x=0..12, y=0..15, scaling=constrained)
  ;
```



Example 3. Find the foci of the hyperbola $9x^2 - 4y^2 = 36$. Sketch its graph and asymptotes.

```
> Eq:=(9*x^2-4*y^2)/36=1;
```

$$Eq := \frac{1}{4}x^2 - \frac{1}{9}y^2 = 1 \qquad (5.1.4)$$

```
> a:=2: b:=3: c:=sqrt(a^2+b^2);
```

$$c := \sqrt{13} \qquad (5.1.5)$$

The foci are $(\sqrt{13}, 0)$ and $(-\sqrt{13}, 0)$.

```
> Asmp:=solve(9*x^2-4*y^2=0,y);
```

$$Asmp := -\frac{3}{2}x, \frac{3}{2}x \qquad (5.1.6)$$

```
> plt1:=plots[implicitplot]( Eq,  x=-5..5, y=-5..5, color=red):
> plt2:=plot([Asmp[1], Asmp[2]], x=-5..5, color=[green,green]):
> plots[display]({plt1,plt2});
```



Example 4. Find the equation of the standard parabola with directrix $y = -2$.

```
> c:=2:
> Eq:=y=x^2/(4*c);
```

$$Eq := y = \frac{1}{8}\, x^2 \tag{5.1.7}$$

Example 5. Find the equation of standard ellipse with eccentricity $e = 0.8$ and vertices (10, 0) and (−10, 0).

```
> a:=10: e:=0.8: b:='b':
> solve((a^2-b^2)/a^2=e^2, b);
```

$$-6., 6. \tag{5.1.8}$$

```
> Eq:=(x/a)^2+(y/6)^2=1;
```

$$Eq := \frac{1}{100}\, x^2 + \frac{1}{36}\, y^2 = 1 \tag{5.1.9}$$

Exercises

1. Find the vertices and foci of $\left(\frac{x}{9}\right)^2 + \left(\frac{y}{4}\right)^2 = 1$.
2. Find the equation of the ellipse with foci (6, 0) and (−6, 0) and vertices (10, 0) and (−10, 0).
3. Find the equation of the hyperbola with vertical foci (2, 2) and (2, −4), center at $C = (2, -1)$, and eccentricity 3/4. Then sketch the graph.
4. Find the equation of the standard parabola with focus (2, 0).
5. Find the vertices, foci, center, and asymptotes of the hyperbola $\left(\frac{x-3}{16}\right)^2 - \left(\frac{y+5}{49}\right)^2 = 1$.

Chapter 13 VECTOR GEOMETRY

13.1 Vectors in the plane

13.1.1 Vectors and their displays

In MAPLE, a **vector** in the plane can be denoted by **u** : = [u_1, u_2].

```
> u:=[u1,u2];
```

$$u := [u1, u2] \tag{1.1.1}$$

Its **length** (is also called **norm** or **magnitude**) is

```
> normu:= sqrt(u1^2 +u2^2);
```

$$normu := \sqrt{u1^2 + u2^2} \tag{1.1.2}$$

Remark 1. The vector norm (or length) introduced in Calculus is also called **2-norm** in general. Other vector norms will be introduced in advanced courses. For example, the 1-norm of the vector **u** = [u_1, u_2] is defined by $|u_1| + |u_2|$.

Remark 2. MAPLE has many **built-in packages**, which can be called when you use MAPLE. You can call a MAPLE package in two different ways. The first way is usually used for the case that you only use it for a few particular commands in a worksheet, but do not use the package very often. The second way is usually used for the case that a MAPLE package needs to be available for all the time you are working on a worksheet so that you can call it any time you want it.

For example, a package that is very often used in our course is "**linalg**", which stands for "linear algebra". It is used for vector and matrix operations. The first way to call "**linalg**" is using the following syntax:

> **linalg[command](variables,..);**

This syntax means that the **command** in the package "**linalg**" is called, but ONLY for this time. If you want to use the same command the next time, you have to re-call it. If you want any commends in the package "**linalg**" to be available when you are working on a MAPLE worksheet, you have to use the following syntax:

> **with(linalg):**

After that time, any commends in the package can be used directly in the following way:

> **command(variables,...);**

until you close the worksheet.

Example 1. Use the "**linalg**" package to get the norm of a vector **u** = [u1, u2].

```
> u:=[u1,u2]:
> linalg[norm](u,2);
```

$$\sqrt{|u1|^2 + |u2|^2} \tag{1.1.3}$$

If you now want to find the length of the vector **a** = [3, 4], you have to call the package again.

```
> a:=[3,4]:
> linalg[norm](a,2);
```

$$5 \tag{1.1.4}$$

Example 2. Add the linear algebra package to the worksheet and then compute the magnitudes of the vectors **u** and **a**.

```
> with(linalg):
> u:=[u1,u2]:
> norm(u,2);
```

$$\sqrt{|u1|^2+|u2|^2} \tag{1.1.5}$$

Find the length of the vector **a** = [3, 4].

```
> a:=[3,4]:
> norm(a,2);
```

$$5 \tag{1.1.6}$$

To display a vector, you have to set the vector to an arrow, then draw the arrow on the screen. Here another package called "**plots**" is needed.

Example 3. Draw the vector **a** = [1, 2] on the rectangular coordinate system.

```
> a:=[1,2]:
> aarw:=arrow(a, width=0.01, head_length=0.2, color=red):
> with(plots):
Warning, the name changecoords has been redefined
> display(aarw,scaling=CONSTRAINED, axes=FRAMED);
```



The command "**arrow**" has several options; the option "**width**" describes the width of the arrow, and the head_length describes the length of the arrow head. The command "**display**" also has several options. To learn its options, please use MAPLE HELP to find the details.

Example 4. Draw two vectors [1, 2] and [−1, −2] in rectangular coordinates.

```
> twarw:=arrow({[1,2],[-1,-2]},width=0.01,head_length=0.3,color=
  blue):
> display(twarw);
```



A vector is usually assumed to start from the origin. If you want to draw a vector not staring from the origin, but from another point, you have to add the start-point in the syntax. See the following example.

Example 5. Draw the vector [−1, −2] starting from [1, 2]. (Hence, it ends at the point [0, 0].)

```
> a:=arrow([1,2],[-1,-2],width=0.01,head_length=0.25,color=red):
> display(a);
```



Note that in the command "**arrow**" here, no { } is used for [1, 2], [−1, −2]. Then the vector [−1, −2] is drawn starting from the point $P(1, 2)$.

Example 6. Draw the vector [1, 2], and then the vector [1, −2] starting from the point $P(1, 2)$.

```
> a:=arrow([1,2],width=0.02,head_length=0.4,color=red):
> b:=arrow([1,2],[1,-2],width=0.02,head_length=0.4,color=green):
> display(a,b);
```



Remark. When you create a geometric object "ARROW", you can add some options for the arrow. The option "**width**" describes the width of the arrow, the option "**head_length**" describes the length of the head of the arrow, and "**color**" assigns a specific color to the arrow. Similarly, there are several options for "**display**". Use MAPLE HELP to find more options for "**display**".

13.1.2 Vector algebra

Let **v** and **u** be two vectors.

```
> v:=[v1,v2];
```

$$v := [v1, v2] \quad (1.2.1)$$

```
> u:=[u1,u2];
```

$$u := [u1, u2] \quad (1.2.2)$$

Then **v** + **u** and 3**v** are

```
> v+u;
```

$$[u1 + v1, u2 + v2] \quad (1.2.3)$$

and

```
> 3*v;
```

$$[3\, v1, 3\, v2] \quad (1.2.4)$$

Example 7. Find the sum $[1, 3] + [-2, 1]$.

```
> a:=[1,3]; b:=[-2,1];
```

$$a := [1, 3]$$
$$b := [-2, 1] \quad (1.2.5)$$

```
> a+b;
```

$$[-1, 4] \quad (1.2.6)$$

Or directly do the following.

```
> [1,3]+[-2,1];
```

$$[-1, 4] \quad (1.2.7)$$

Example 8. Assume $\mathbf{a} = 2\mathbf{i} + 3\mathbf{j}$ and $\mathbf{b} = -2\mathbf{i} + \mathbf{j}$. Find $3\mathbf{a} + 2\mathbf{b}$, where $\mathbf{i} = [1, 0]$, $\mathbf{j} = [0, 1]$.

$$i = [1, 0]$$
$$j = [0, 1] \quad (1.2.8)$$

```
> a:=[2,3]: b:=[-2,1]:
> 3*a+2*b;
```

$$[2, 11] \quad (1.2.9)$$

Example 9. Express $\mathbf{u} = [4, 4]$ as the linear combination of $\mathbf{v} = [6, 2]$ and $\mathbf{w} = [2, 4]$.

```
> v:=[6,2]: w:=[2,4]: u:=[4,4]:
> solve({x*6+y*2=4,x*v[2]+y*w[2]=u[2]},{x,y});
```

$$\left\{x = \frac{2}{5}, y = \frac{4}{5}\right\} \quad (1.2.10)$$

That is, $\mathbf{u} = \frac{2\mathbf{v}}{5} + \frac{4\mathbf{w}}{5}$.

Example 10. Find the unit vector in the direction of $\mathbf{v} = [3, 5]$.

```
> v:=[3,5];
```

$$v := [3, 5] \quad (1.2.11)$$

```
> untv:=v/norm(v,2);
```

$$untv := \frac{1}{34}\,[3, 5]\sqrt{34} \quad (1.2.12)$$

Exercises

The exercises 1 − 6 are based on vectors $\mathbf{a} = [1, -2]$, $\mathbf{b} = [3, 2]$, and $\mathbf{c} = [3, -5]$.

1. Find $2\mathbf{a} - 3\mathbf{b}$.
2. Find the lengths of **a**, **b**, and **c**, respectively.
3. Find the unit vector in the direction of **a**.
4. Draw the vectors **a**, **b**, and **c** on the same graph.
5. Draw the vectors **a**, **b**, and **a** + **b** on the same graph.
6. Express **c** as the linear combination of **a** and **b**.

13.2 Vector in three dimensions

13.2.1 Vector in space

A vector in three dimensions has three components. It is denoted by $\mathbf{v} = \left[v_1, v_2, v_3\right]$. The operations among the vectors and their displays are similar to the vectors in the plane. For example, let **v** and **u** be two vectors in three dimensions.

```
> v:=[v1,v2,v3];
```

$$v := [v1, v2, v3] \tag{2.1.1}$$

```
> u:=[u1,u2,u3];
```

$$u := [u1, u2, u3] \tag{2.1.2}$$

The sum **u** + **v** and 3**v** are

```
> v+u;
```

$$[u1 + v1, u2 + v2, u3 + v3] \tag{2.1.3}$$

and

```
> 3*v;
```

$$[3\ v1, 3\ v2, 3\ v3] \tag{2.1.4}$$

The length of **v** is

```
> vlength:=sqrt(v[1]^2+v[2]^2+v[3]^2);
```

$$vlength := \sqrt{v1^2 + v2^2 + v3^2} \tag{2.1.5}$$

```
> norm(u,2);
```

$$\sqrt{|u1|^2 + |u2|^2 + |u3|^2} \tag{2.1.6}$$

The unit vector in the **v** direction is

```
> vunit:=v/vlength;
>
```

$$vunit := \frac{[v1, v2, v3]}{\sqrt{v1^2 + v2^2 + v3^2}} \tag{2.1.7}$$

If the linear algebra package "**linalg**" is already available on the worksheet, then you can also compute the length of the vector **v** by

```
> vnrm:=norm(v,2);
```

$$vnrm := \sqrt{|v1|^2 + |v2|^2 + |v3|^2} \tag{2.1.8}$$

Otherwise, you have to call the package before using the command "norm".

```
> nv:=linalg[norm](v,2);
```

$$nv := \sqrt{|v1|^2 + |v2|^2 + |v3|^2} \tag{2.1.9}$$

The unit vector in **v** direction is

```
> vunit:=v/vlength;
```

$$vunit := \frac{[v1, v2, v3]}{\sqrt{v1^2 + v2^2 + v3^2}} \tag{2.1.10}$$

Example 1. Find the length of the vector **v** = [1, 2, 3].

```
> v:=[1,2,3];
```

$$v := [1, 2, 3] \tag{2.1.11}$$

```
> valength:=sqrt(v[1]^2+v[2]^2+v[3]^2);
```

$$valength := \sqrt{14} \tag{2.1.12}$$

Or use the following syntax:

```
> valength:=norm(v,2);
```

$$valength := \sqrt{14} \tag{2.1.13}$$

Example 2. Find the unit vector in the direction of **v** = [1, 2, 3].

```
> vunt:=v/valength;
```

$$vunt := \frac{1}{14}\,[1, 2, 3]\sqrt{14} \tag{2.1.14}$$

Example 3. Find the distance between $P(1, -2, 3)$ and $Q(2, 0, 3)$

```
> a:=[1,-2,3]; b:=[2,0,3];
```

$$a := [1, -2, 3]$$
$$b := [2, 0, 3] \tag{2.1.15}$$

```
> distab:=norm(a-b,2);
```

$$distab := \sqrt{5} \tag{2.1.16}$$

Example 4. Find the parametric equation of the line through $P_0(3, -1, 4)$ with the direction vector **v** = [2, 1, 7].

```
> p0:=[3,-1,4]: v:=[2,1,7]:
> LineEq:=p0+t*v;
```

$$LineEq := [3, -1, 4] + t\,[2, 1, 7] \tag{2.1.17}$$

Hence, its parametric equation is $x = 3 + 2t$, $y = -1 + t$, $z = 4 + 7t$. To draw the line on the three-dimensional coordinates, we do the following.

```
> spacecurve([3+2*t,-1+t,4+7*t],t=-2..2,axes=NORMAL,color=blue);
```



To draw a 3-D vector, use the similar syntax as for the vector in the plane.

Example 5. Draw the vector **a** = [2, 1, −1].

```
> aarw:=arrow([2,1,-1],width=0.01,head_length=0.3,color=red):
> display(aarw,scaling=CONSTRAINED,axes=FRAMED);
```



Exercises

The exercises 1−6 are based on vectors **a** = [1, −2, 1], **b** = [−1, 3, 2], **c** = [2, 3, −1].

1. Find **a** − 3**b** + 2**c**.
2. Find the lengths of **a**, **b**, and **c**, respectively.
3. Find the unit vector in the direction of **a**.
4. Draw the vectors **a**, **b**, and **c** on the same graph.
5. Draw the vectors **a**, **b**, and **a** + **b** on the same graph.
6. Express the vector [1, 1, 1] as the linear combination of **a**, **b**, and **c**.
7. Find the distance between $P(1, 2, 1)$ and $Q(-2, 1, 3)$.
8. Find the parametric equation of the line through P_0 (1, −1, 0) with the direction vector **v** = [1, 1, 2], and then draw the line on the rectangular coordinates.

13.3 Dot product and the angle between two vectors

13.3.1 Dot product

In literature, **dot product** sometimes is also called **inner product**. Note that in MAPLE the build-in procedure "**dotprod**" works on the COMPLEX mode, and "**innerprod**" works for the real mode. Hence, it is suggested that when you compute the dot product, you use the syntax "**innerprod**".

```
> v:=[v1,v2,v3]; u:=[u1,u2,u3];
```

$$v := [v1, v2, v3]$$
$$u := [u1, u2, u3] \quad (3.1.1)$$

To find their dot product, you can call "**innerprod**":

```
> DotPrdvu=innerprod(v,u);
```

$$DotPrdvu = v1\,u1 + v2\,u2 + v3\,u3$$

Example 1. Find the dot product of 2-D vectors **a** = [1, 5] and **b** = [2, 4].

```
> a:=[1,5]; b:=[2,4];
```

$$a := [1, 5]$$
$$b := [2, 4] \quad (3.1.2)$$

```
> innerprod(a,b);
```

$$22 \quad (3.1.3)$$

Example 2. Find the dot product of 3-D vectors **a** = [1, 2, 3] and **b** = [2, −4, 1].

```
> a:=[1,2,3]; b:=[2,-4,1];
```

$$a := [1, 2, 3]$$
$$b := [2, -4, 1] \quad (3.1.4)$$

```
> innerprod(a,b);
```

$$-3 \quad (3.1.5)$$

Example 3. Find x such that the vectors **u** = [x, 2, −1] and **v** = [2 x − 1, 1, 1] are perpendicular.

```
> u:=[x,-2,1]; v:=[2*x-1,1,1];
```

$$u := [x, -2, 1]$$
$$v := [2\,x - 1, 1, 1] \quad (3.1.6)$$

```
> Eq:=innerprod(u,v);
```

$$Eq := -1 + 2\,x^2 - x \quad (3.1.7)$$

```
> solve(Eq,x);
```

$$1, \frac{-1}{2} \quad (3.1.8)$$

Hence, when x = 1 or −1/2, **u** and **v** are perpendicular.

A **main application of dot product** is to determine whether two vectors are perpendicular or not. Two non-zero vectors **u** and **v** are perpendicular if and only if their dot product is zero.
Example 4. Determine whether two vectors **a** = [1, 2, 1] and **b** = [2, −1, 0] are perpendicular or not.

```
> a:=[1,2,1]: b:=[2,-1,0]:
> dtprd:=innerprod(a,b);
```

$$dtprd := 0 \quad (3.1.9)$$

Since their dot product is 0, they are perpendicular.

13.3.2 Angles and projections

(1) **Angle between two vectors.** The dot product of $\frac{\mathbf{a}}{\|\mathbf{a}\|}$ and $\frac{\mathbf{b}}{\|\mathbf{b}\|}$ is the cosine of the angle between **a** and **b**, assuming they are non-zero vectors. Therefore, if their dot product is zero, **a** and **b** are perpendicular.
(2) **Vector projection.** The projection of a vector **a** onto a non-zero vector **b** is given by

$$\text{Proj}_{\mathbf{b}}\mathbf{a} = \frac{\mathbf{a}\cdot\mathbf{b}}{\mathbf{b}\cdot\mathbf{b}}\mathbf{b}$$

Example 5. Find the angle between $\mathbf{a} = [1, 2, 3]$ and $\mathbf{b} = [2, -4, 1]$.

Method 1. By the formula Angle<**a**,**b**> = $\arccos\left(\frac{\mathbf{a}\cdot\mathbf{b}}{\|\mathbf{a}\|\cdot\|\mathbf{b}\|}\right)$

```
> a:=[1,2,3]: b:=[2,-4,1]:
```

```
> csa:=innerprod(a,b)/(norm(a,2)*norm(b,2));
```

$$csa := -\frac{1}{98}\sqrt{14}\sqrt{21} \quad (3.3.1)$$

```
> angab:=arccos(csa);
```

$$angab := \pi - \arccos\left(\frac{1}{98}\sqrt{14}\sqrt{21}\right) \quad (3.3.2)$$

Method 2. Directly apply the MAPLE command "**angle**".

```
> anglab:=angle(a,b);
```

$$anglab := \pi - \arccos\left(\frac{1}{98}\sqrt{14}\sqrt{21}\right) \quad (3.3.3)$$

To get the numerical value of the angle in RADIAN, use the following syntax:

```
> evalf(anglab);
```

$$1.746665077 \quad (3.3.4)$$

Example 5. Determine if $\mathbf{a} = [1, 2, 1]$ and $\mathbf{b} = [2, -1, 0]$ are perpendicular.

```
> a:=[1,2,1]: b:=[2,-1,0]:
> dtprd:=innerprod(a,b);
```

$$dtprd := 0 \quad (3.3.5)$$

Since their dot product is 0, they are perpendicular.

Example 6. Find the vector projection $\text{Proj}_{\mathbf{b}}\mathbf{a}$, where $\mathbf{a} = [1, 2, 3]$ and $\mathbf{b} = [2,-4,1]$.

```
> a:=[1,2,3]; b:=[2,-4,1];
```

$$a := [1, 2, 3]$$
$$b := [2, -4, 1] \quad (3.3.6)$$

By the formula, the vector projection $\text{Proj}_{\mathbf{b}}\mathbf{a}$ is the vector:

```
> innerprod(a,b)/innerprod(b,b)*b;
```

$$\left[\frac{-2}{7}, \frac{4}{7}, \frac{-1}{7}\right] \quad (3.3.7)$$

Exercises

Exercises 1 −6 are based on the vectors **a** = [1, 1, −2], **b** = [3, −2, 1], and **c** = [0, 1, −5].
1. Find the dot product of **a** and **b** and the dot product of **b** and **c**.
2. Find the angle between **a** and **b** and the angle between **a** and **c**.
3. Find the vector projection of **a** onto **c** and the vector projection of **c** onto **a**.
4. Find the vector projection of **a** onto **b**, where **a** = 5**i** + 7**j** − 4**k** and **v** = **k**.
5. Find the value of x such that the vectors **a** = [2, x, −4] and **b** = [3x, 1, 5] are perpendicular.

13.4 The Cross product

The cross product of two vectors **u** and **v** is a new vector **w** such that **w** is perpendicular to **u** and **v** in the way that **u**, **v**, and **w** make a right-hand system; and the length of **w** is the area of the parallelogram made by **u** and **v**. The cross product is denoted by **w = u** × **v**.

13.4.1 The cross product

```
> v:=[v1,v2,v3]; u:=[u1,u2,u3];
```

$$v := [v1, v2, v3]$$

$$u := [u1, u2, u3] \quad (4.1.1)$$

```
>  w:=crossprod(v,u);
```

$$w := [v2\,u3 - v3\,u2 \quad v3\,u1 - v1\,u3 \quad v1\,u2 - v2\,u1] \quad (4.1.2)$$

Example 1. Let **i** = [1, 0, 0], **j** = [0, 1, 0], and **k** = [0, 0, 1]. Find the cross product of **i** and **j**, **j** and **k**, **k** and **i**.

```
> i:=[1,0,0]: j:=[0,1,0]: k:=[0,0,1]:
> crossprod(i,j);
```

$$[0 \quad 0 \quad 1] \quad (4.1.3)$$

```
> crossprod(j,k);
```

$$[1 \quad 0 \quad 0] \quad (4.1.4)$$

```
> crossprod(k,i);
```

$$[0 \quad 1 \quad 0] \quad (4.1.5)$$

```
> i:'i': j:='j': k:='k':
```

Example 2. Find the cross product of 2**i** + **k** and 3**j** + 5**k**.

```
> u:=[2,1,0]: v:=[0,3,5]:
> crossprod(u,v);
```

$$[5 \quad -10 \quad 6] \quad (4.1.6)$$

13.4.2 Perpendicular vectors, areas, and volumes

(1) To find a vector that is perpendicular to two given vectors **u** and **v**, apply the cross product on **u** and **v**.
(2) The area of the parallelogram made by **u** and **v** is the length of the cross product of **u** and **v**.
(3) The volume of the parallelopiped made by three vectors **a**, **b**, and **c** is the triple product of **a**, **b**, and **c**.

Example 3. Find a vector perpendicular to **v** = [1, 2, −1], **u** = [3, −2, 1].

```
> v:=[1,2,-1]; u:=[3,-2,1];
```

$$v := [1, 2, -1]$$
$$u := [3, -2, 1] \quad (4.2.1)$$

```
> w:=crossprod(v,u);
```

$$w := [0 \quad -4 \quad -8] \quad (4.2.2)$$

The vector **w** now is perpendicular to both **u** and **v**. It can be verified by the dot products of **v** and **w**, and **u** and **w**.

```
> innerprod(w,v);
```

$$0 \quad (4.2.3)$$

```
> innerprod(w,u);
```

$$0 \quad (4.2.4)$$

Example 4. Find the area of the triangle with the vertices $A(2, 1, 3), B(0, 1, -1), C(1, 4, 0)$.

```
> v:=[0,1,-1]-[2,1,3];
```

$$v := [-2, 0, -4] \quad (4.2.5)$$

```
> u:=[1,4,0]-[2,1,3];
```

$$u := [-1, 3, -3] \quad (4.2.6)$$

```
> w:=crossprod(v,u);
```

$$w := [12 \quad -2 \quad -6] \quad (4.2.7)$$

```
> area:=1/2*norm(w,2);
```

$$area := \sqrt{46} \quad (4.2.8)$$

Example 5. Find the area of the triangle with the vertices $A(2, 1), B(-1, 1), C(4, 0)$.
We first extend the vectors to 3-D. Then apply the area formula.

```
> A:=[2,1,0]: B:=[-1,1,0]: C:=[4,0,0]:
> v:=B-C;
```

$$v := [-5, 1, 0] \quad (4.2.9)$$

```
> u:=A-C;
```

$$u := [-2, 1, 0] \quad (4.2.10)$$

```
> area:=1/2*(norm(crossprod(v,u),2));
```

$$area := \frac{3}{2} \quad (4.2.11)$$

Example 6. Find the volume of the parallelopiped made by the vectors **v** = [1, 2, 1], **u** = [2, −1, 3], and **w** = [2, 1, −1].

```
> v:=[1,2,1]; u:=[2,-1,3]; w:=[2,1,-1];
```

$$v := [1, 2, 1]$$
$$u := [2, -1, 3]$$
$$w := [2, 1, -1] \quad (4.2.12)$$

The following computation shows the property of vector triple product.

```
> innerprod(crossprod(v,u),w);
```

$$18 \quad (4.2.13)$$

```
> innerprod(crossprod(u,w),v);
```

$$18 \quad (4.2.14)$$

```
> innerprod(crossprod(w,v),u);
```

$$18 \quad (4.2.15)$$

```
> innerprod(crossprod(u,v),w);
```

$$-18 \quad (4.2.16)$$

Exercises

1. Find the cross product of the following vectors.
   (1) $\mathbf{v} = [1, 2, 1]$ and $\mathbf{u} = [3, 1, 1]$.
   (2) $\mathbf{v} = [2/3, 1, 1/2]$ and $\mathbf{u} = [-1, 1, 2]$.
2. Find a vector that is perpendicular to both $\mathbf{a} = [1, 2, -1]$ and $\mathbf{b} = [0, -1, 2]$.
3. Find a **unit** vector that is perpendicular to the plane through three points $P(1, -1, 0)$, $Q(2, 1, -1)$, and $R(-1, 1, 2)$.
4. Find the area of the triangles whose vertices are given.
   (1) $A(0, 0), B(-2, 3), C(3, 1)$.
   (2) $A(-5, 3), B(1, -2), C(6, -2)$.
5. Find the areas of the parallelograms whose vertices are given.
   (1) $A(1, 0), B(0, 1), C(-1, 0), D(0, -1)$.
   (2) $A(-1, 2), B(2, 0), C(7, 1), D(4, 3)$.
6. Find the area of the triangle with vertices $P(1, -1, 0)$, $Q(2, 1, -1)$, and $R(-1, 1, 2)$.

13.5 Planes in space

13.5.1 Lines and line segments in space

(1) A vector equation of a line L through a point $P_0(x_0, y_0, z_0)$ parallel to a vector $\mathbf{v}$ is the vector $\mathbf{u}(t)$:

$$\mathbf{u}(t) = P_0 + t\mathbf{v}.$$

(2) The standard parametrization of the line L is

$$x = x_0 + tv_1, \ y = y_0 + tv_2, \ z = z_0 + tv_3$$

(3) The distance from a point P to the line L is

$$d(P, L) = \frac{\left\|\left(P - P_0\right) \times \mathbf{v}\right\|}{\|\mathbf{v}\|}$$

Example 1. Find a line through the points $P_0(1, 2, -1)$ and in the direction of the vector $\mathbf{v} = [-1, 0, 1]$.

```
> P0:=[1,2,-1]; v:=[-1,0,1];
```

$$P0 := [1, 2, -1]$$
$$v := [-1, 0, 1] \quad (5.1.1)$$

The equation of the line is

```
> Leq=P0+t*v;
```

$$Leq = [1, 2, -1] + t\,[-1, 0, 1] \quad (5.1.2)$$

Example 2. Find a line through the points $P(1, 2, -1)$ and $Q(-1, 0, 1)$.

```
> P:=[1,2,-1]; Q:=[-1,0,1];
```

$$P := [1, 2, -1]$$
$$Q := [-1, 0, 1] \quad (5.1.3)$$

```
> v:=Q-P;
```

$$v := [-2, -2, 2] \quad (5.1.4)$$

The equation is

```
> Leq=P+t*v;
```

$$Leq = [1, 2, -1] + t\,[-2, -2, 2] \quad (5.1.5)$$

The parametric equation of the line is

```
> x=P[1]+t*v[1]; y=P[2]+t*v[2]; z=P[3]+t*v[3];
```

$$x = 1 - 2\,t$$
$$y = 2 - 2\,t$$
$$z = -1 + 2\,t \quad (5.1.6)$$

```
> x:='x': y:='y': z:='z':
```

Example 3. Find the distance from the point $R(3, 2, 4)$ to the line in **Example 2**.

```
> R:=[3,2,4]: u:=R-P:
> dist=norm(crossprod(v,u),2)/norm(v,2);
```

$$dist = \frac{1}{3}\sqrt{78}\,\sqrt{3} \quad (5.1.7)$$

13.5.2 A plane in space

(1) A **plane equation** of a plane S through a point P_0 normal to a vector **n** is the following:

$$\left(P - P_0\right)\cdot\mathbf{n} = 0,$$

where $P=[x, y, z]$ is a point on the plane.

Example 4. Find the plane through $P(1, 1, -1)$ normal to $\mathbf{v} = [2, 0, -2]$.

```
> P:=[1, 1, -1]: n:=[2, 0, -2]: R:=[x,y,z]:
> Peq=innerprod(R-P,n);
```

$$Peq = -4 + 2\,x - 2\,z \quad (5.2.1)$$

Example 5. Find the plane through $P(1, 1, -1)$, $Q(2, 0, 2)$, and $R(0, -2, 1)$.

```
> P:=[1,1,-1]; Q:=[2,0,2]; R:=[0,-2,1];
```

$$P := [1, 1, -1]$$
$$Q := [2, 0, 2]$$
$$R := [0, -2, 1] \quad (5.2.2)$$

```
> u:=Q-P; v:=R-P;
```

$$u := [1, -1, 3]$$
$$v := [-1, -3, 2] \quad (5.2.3)$$

```
> n:=crossprod(u,v);
```

$$n := [7 \quad -5 \quad -4] \quad (5.2.4)$$

The equation is

```
> Peq=innerprod([x,y,z]-P,n);
```

$$Peq = -6 + 7x - 5y - 4z \quad (5.2.5)$$

Example 6. Find the plane through $P(0, 2, -1)$ normal to $\mathbf{n} = 3\mathbf{i} - 2\mathbf{j} - \mathbf{k}$.

```
> P:=[0,2,-1]; n:=[3,-2,-1];
```

$$P := [0, 2, -1]$$
$$n := [3, -2, -1] \quad (5.2.6)$$

```
> Peq=innerprod([x,y,z]-P,n);
```

$$Peq = 3x + 3 - 2y - z \quad (5.2.7)$$

13.5.3 Intersection of plane and plane, or line and plane

(1) **The vector on the line of the intersection of two planes**: Let

$$A_1(x - x_1) + B_1(y - y_1) + C_1(z - z_1) = 0$$

and

$$A_2(x - x_2) + B_2(y - y_2) + C_2(z - z_2) = 0$$

be two planes. The vector on the intersection line of these two planes is the cross product of the vectors $\mathbf{u} = [A_1, B_1, C_1]$ and $\mathbf{v} = [A_2, B_2, C_2]$.

(2) **The intersection of a line and a plane**: Assume the line equation is

$$x = x_0 + tv_1,\ y = y_0 + tv_2,\ z = z_0 + tv_3,$$

and the plane equation is

$$A_1(x - x_1) + B_1(y - y_1) + C_1(z - z_1) = 0$$

Then the intersection of the line and the plane is the point $[x_0 + sv_1,\ y_0 + sv_2,\ z_0 + sv_3]$ with s being the solution of the equation

$$A_1\left(x_0 + sv_1 - x_1\right) + B_1\left(y_0 + sv_2 - y_1\right) + C_1\left(z_0 + sv_3 - z_1\right) = 0.$$

(3) **The distance from a point** $P(x_1, y_1, z_1)$ **to a plane** $Ax + Bx + Cz = D$: Let $P_0(x_0, y_0, z_0)$ be a point on the plane. Then the distance is the length of the projection of the vector $\mathbf{u} = [x_1 - x_0, y_1 - y_0, z_1 - z_0]$ onto the normal vector of the plane $\mathbf{n} = [A, B, C]$:

$$d = \text{Proj}_{\mathbf{n}}\mathbf{u} = \frac{\mathbf{u} \cdot \mathbf{n}}{\|\mathbf{n}\|}$$

Example 7. Find the equation of the intersection line of two planes $3x - 6y - 2z = 3$ and $2x + y - 2z = 2$.

Step 1. Find a point on the intersection.

```
> solve({3*x-6*y-2*z=3,2*x+y-2*z=2}, {x,y,z});
```

$$\left\{x = x,\ z = \frac{15}{14}x - \frac{15}{14},\ y = \frac{1}{7}x - \frac{1}{7}\right\} \quad (5.3.1)$$

We can set x to any real value, say $x = 0$, for getting a point on the line.

```
> P:=[0, -1/7, -15/14];
```

$$P := \left[0, \frac{-1}{7}, \frac{-15}{14}\right] \quad (5.3.2)$$

Step 2. Find the direction of the line.

```
> u:=[3,-6,-2]: v:=[2,1,-2]:
> w:=crossprod(u,v);
```

$$w := [14 \quad 2 \quad 15] \tag{5.3.3}$$

The line equation is

```
> x=14*t; y=-1/7+2*t; z=-15/14+15*t;
```

$$x = 14\,t$$
$$y = -\frac{1}{7} + 2\,t$$
$$z = -\frac{15}{14} + 15\,t \tag{5.3.4}$$

Example 8. Find the intersection of the line $x = 1 - t$, $y = 3t$, $z = 1 + t$ and the plane $2x - y + 3z = 6$.

```
> x:=1-t; y:=3*t; z:=1+t;
```

$$x := 1 - t$$
$$y := 3\,t$$
$$z := 1 + t \tag{5.3.5}$$

```
> eq:=2*x-y+3*z=6;
```

$$eq := 5 - 2\,t = 6 \tag{5.3.6}$$

```
> solve(eq, t);
```

$$\frac{-1}{2} \tag{5.3.7}$$

The t value of the point and the coordinates of the point are given in the following.

```
> t:=-1/2;
```

$$t := \frac{-1}{2} \tag{5.3.8}$$

```
> p:=[x,y,z];
```

$$p := \left[\frac{3}{2}, \frac{-3}{2}, \frac{1}{2}\right] \tag{5.3.9}$$

```
> t:='t': x:='x': y:='y': z:='z':
```

Example 9. Find the plane determined by the intersection lines of L_1: $x = -1 + t$, $y = 2 + t$, $z = 1 - t$, and L_2: $x = 1 - 4s$, $y = 1 + 2s$, $z = 2 - 2s$.

Step1. Find the intersection of two lines.

```
> solve({-1+t=1-4*s,2+t=1+2*s,1-t=2-2*s},{t,s});
```

$$\left\{t = 0, s = \frac{1}{2}\right\} \tag{5.3.10}$$

```
> P:=[-1,2,1]; u:=[1,1,-1];v:=[-4,2,-2];
```

$$P := [-1, 2, 1]$$
$$u := [1, 1, -1]$$
$$v := [-4, 2, -2] \tag{5.3.11}$$

```
> n:=crossprod(u,v);
```

$$n := [0 \quad 6 \quad 6] \tag{5.3.12}$$

```
> PlaneEq=innerprod([x,y,z]-P,n);
```

$$PlaneEq = -18 + 6\,y + 6\,z \tag{5.3.13}$$

Example 10. Find the distance from the point (2, −3, 4) to the plane $x + 2y + 2z = 13$.
Step 1. Find a point on the plane.

```
> solve(x+2*y+2*z=13,{x,y,z});
```

$$\left\{z = -\frac{1}{2}\,x - y + \frac{13}{2},\, y = y,\, x = x\right\} \tag{5.3.14}$$

Set $x = 1$ and $y = 2$. Then z is

```
> eval(-x/2-y+13/2,{x=1,y=2});
```

$$4 \tag{5.3.15}$$

Hence, the point (1, 2, 4) is on the plane.
Step 2. Find the distance as the norm of a projection or by the distance formula.
The distance is the norm of the projection of the vector, which starts from (1, 2, 4) and ends at (2, −3, 4), onto the vector [1, 2, 2].

```
> P:=[1,2,4]; v:=[1,2,2];
```

$$P := [1, 2, 4]$$
$$v := [1, 2, 2] \tag{5.3.16}$$

```
> u:=[2,-3,4]-P;
```

$$u := [1, -5, 0] \tag{5.3.17}$$

```
> Projuv:=innerprod(u,v)/innerprod(v,v)*v;
```

$$Projuv := [-1, -2, -2] \tag{5.3.18}$$

```
> Dist:=norm(Projuv,2);
```

$$Dist := 3 \tag{5.3.19}$$

You can also compute the distance by the distance formula:

```
> Dist:=abs(innerprod(u,v))/norm(v,2);
```

$$Dist := 3 \tag{5.3.20}$$

13.5.4 Exercises

1. Find a line through the points $P_0(1, 2, 2)$ and in the direction of the vector **v** = [−1, 2, −1].
2. Find a line through the points $P(-1, 1, -1)$ and $Q(1, 1, 1)$.
3. Find the distance from the point $R(2, 3, 1)$ to the line in **Exercise 2**.
4. Find the plane through $P(1, 0, -1)$ normal to **v** = [2, 1, −2].
5. Find the plane through $P(3, 1, 0)$, $Q(2, 0, 2)$, and $R(0, 2, -1)$.
6. Find the plane through $P(0, 2, -1)$ normal to the vector **n** = **i** − 2**k**.
7. Find the equation of the intersection line of two planes $3x - y - z = 1$ and $2x + y = 2$.
8. Find the intersection of the line $x = 1 - t,\ y = t,\ z = 2 + t$, and the plane $2x - y + z = 2$.
9. Find the plane determined by the intersection lines of L_1: $x = t,\ y = 2 + t,\ z = t$, and
   L_2: $x = 1 - 4s,\ y = 3 + 2s,\ z = 1 - s$.
10. Find the distance from the point (2, −1, 1) to the plane $3x + y + 2z = 5$.

13.6 Cylinders and Quadric Surfaces

An important way to understand the three-dimensional objectives defined by 3-D equations is by drawing their graphs. This section introduces the technique used in the plotting of 3-D graphs.
To plot the 3-D functions, we need the "**plots**" package. Before you draw 3-D graphs, first add the

package to your worksheet as the following:

> **with (plots):**

In the package, "**plot3d**" is the main command used for plotting the graph given by the explicit function $z = f(x, y)$ in space. Its syntax is
> **plot3d(f, x=a..b, y=c..d, *options*);**
where ***options*** mean some options you can add in the command to improve the display of the graph. The main options to plot3d are listed in the following.
title = plot title;
scaling = unconstrained / constrained;
style = point / hidden / patch / wireframe;
axes = boxed / normal / frame / none;
coords = cartesian / spherical / cylindrical;
view = zmin..zmax / [xmin..xmax,ymin..ymax,zmin..zmax];
numpoints = n.
The details of the usage of these options can be found in MAPLE HELP.

13.6.1 Plotting functions in rectangle coordinates

```
> with(plots):
```

Example 1. Plot the paraboloid $z = x^2 + y^2$ with view=0..9 (which means that the graph should be cut by z = 9 above and the *xy*-plane below).

```
> plot3d(x^2+y^2,x=-3..3,y=-3..3,view=0..9,axes=frame);
```



Remark. Adding the option "**view**" sometimes is quite critical. For example, if you plot the same function without "**view**" option, then the graph is the following.

Example 2. Plot $z = x^2 + y^2$ without "**view**".

```
> plot3d(x^2+y^2,x=-3..3,y=-3..3,title=`Graph of z=x^2+y^2`,axes=
  normal);
```



13.6.2 Plotting equations in rectangle coordinates

If the equation of a surface is a quadratic one, then the surface is called a quadratic surface. Most quadratic surfaces are formulated by an **equation**, **not a function**. The function defined by an equation is called an implicit function. To plot the surfaces described by equations, we have to use implicit plot, which is illustrated in the following examples.

Example 3. Draw the graph of the parabolic cylinder $y = x^2$.

```
> implicitplot3d(y=x^2,x=-2..2, y=-4..4,z=-3..3,axes=normal);
```



Example 4. Draw the graph of the hyperbolic paraboloid $y^2 - x^2 = z$.

```
> implicitplot3d(y^2-x^2=z,x=-3..3,y=-3..3,z=-5..5,axes=framed);
```



You can also use "**plot3d**" for the plot since z is an explicit function of x and y.

```
> plot3d(y^2-x^2,x=-3..3,y=-3..3,axes=framed);
```



Example 5. Plot $x^2 + 2y^2 + z^2 = 1$.

```
> implicitplot3d(x^2+2*y^2+z^2=1,x=-1..1,y=-1..1,z=-1..1,
  numpoints=400,scaling=constrained,axes=frame);
```



13.6.3 Exercises

Sketch the surface of the function.

1. $z = x^2 + 4y^2$
2. $y = 1 - x^2 - z^2$
3. $y^2 - x^2 = z$

Sketch the surface of the equation.

4. $z^2 - y^2 = 1$
5. $x^2 + y^2 = z^2$
6. $x^2 + y^2 - z^2 = 1$

13.7 Cylindrical and spherical coordinates

(1) The **cylindrical coordinates** of a point P in 3-D is (r, θ, z). The converting formula is the following.

Cylindrical to rectangular:

$$x = r\cos(\theta)$$
$$y = r\sin(\theta)$$
$$z = z$$

Rectangular to cylindrical:

$$r = \sqrt{x^2 + y^2}$$
$$\tan(\theta) = \frac{y}{x}$$
$$z = z$$

(2) The **spherical coordinates** of a point P in 3-D is (ρ, θ, φ). The converting formula is the following.

Spherical to rectangular:

$$x=\rho\cos(\theta)\sin(\varphi)$$
$$y=\rho\sin(\theta)\sin(\varphi)$$
$$z=\rho\cos(\varphi)$$

Rectangular to Spherical:

$$\rho=\sqrt{x^2+y^2+z^2}$$
$$\tan(\theta)=\frac{y}{x}$$
$$\cos(\varphi)=\frac{z}{\rho}$$

(3) In **cylindrical coordinates**, the function form is $r=f(z, \theta)$. In **spherical coordinates**, the function form is $\rho=f(\varphi,\theta)$. If you want to draw graphs of functions in these coordinates, you have to use the package "**plots**" on your worksheet, and then set the option "**coords**" of "**plot3d**" to the right coordinate system. If the graph is involving a sphere, a cube, etc., the option "**scaling=constrained**" should also be used.

Note. About coordinate choosing.

- If Cartesian coordinates are used and the graph is defined by a function $z=f(x, y)$, then it can be plotted by the following syntax:

```
> plot3d(z(x, y), x=a..b, y=c..d);
```

- For alternate coordinate systems this is interpreted differently. For example, when cylindrical coordinates are used, the syntax is the following:

```
> plot3d(r(theta, z), theta=a..b, z=c..d, coords=cylindrical);
```

and for spherical coordinates the syntax is:

```
> plot3d(r(theta,phi), theta=a..b, phi=c..d, coords=spherical);
```

13.7.1 Cylindrical coordinates

Example 1. Find the cylindrical coordinates for the point $(-3\sqrt{3},\ -3, 5)$ in the rectangular.

```
> eval([sqrt(x^2+y^2),arctan(y/x),z],[x=-3*sqrt(3),y=-3,z=5]);
```

$$\left[\sqrt{36}, \frac{1}{6}\pi, 5\right] \qquad (7.1.1)$$

Example 2. Convert cylindrical coordinates $(r, \theta, z) = (2, \frac{3\pi}{4}, 5)$ to rectangular.

```
> eval([r*cos(theta),r*sin(theta),z],[r=2,theta=3*Pi/4,z=5]);
```

$$\left[-\sqrt{2}, \sqrt{2}, 5\right] \qquad (7.1.2)$$

Example 3. Find the cylindrical equation $r=f(z, \theta)$ for the surface $x^2+y^2+z^2=9$. Then sketch its graph.

```
> Eq:=eval(x^2+y^2+z^2-9,[x=r*cos(theta),y=r*sin(theta),z=z]);
```

$$Eq := r^2\cos(\theta)^2+r^2\sin(\theta)^2+z^2-9 \qquad (7.1.3)$$

```
> Neq:=simplify (Eq);
```

$$Neq := z^2-9+r^2 \qquad (7.1.4)$$

```
> Newr:=solve(Neq,r);
```

$$Newr := \sqrt{-z^2+9}, -\sqrt{-z^2+9} \qquad (7.1.5)$$

We only need the equation that makes r nonnegative. Hence, the cylindrical equations are

```
> eq1:=Newr[1];
```

$$eq1 := \sqrt{-z^2+9} \tag{7.1.6}$$

The graph of these functions can be drawn in cylindrical coordinates.

```
> plot3d(eq1,theta=0..2*Pi,z=-3..3,axes=frame,coords=
  cylindrical,scaling=constrained);
```



Note. In "**plot3d**", the variable "**theta**" must be claimed before "**z**". Otherwise, the graph would not be plotted correctly. The following syntax, which interchanges the order of "**theta**" and "**z**", creates a totally different graph for the same function.

```
> plot3d(eq1,z=-3..3,theta=0..2*Pi,axes=frame,coords=
  cylindrical,scaling=constrained);
```



Example 4. Find the cylindrical equation $r = f(z, \theta)$ for the surface $x^2 + y^2 = 1$. Then sketch its graph.

```
> assume(theta,real);
> cylEq:=eval(x^2+y^2-1,[x=r*cos(theta),y=r*sin(theta),z=z]);
```

$$cylEq := r^2 \cos(\theta\sim)^2 + r^2 \sin(\theta\sim)^2 - 1 \tag{7.1.7}$$

```
> cylEq:=simplify(cylEq);
```

$$cylEq := -1 + r^2 \tag{7.1.8}$$

It yields the solution $r = 1$.

```
> plot3d(1,theta=0..2*Pi,z=-3..3,axes=frame,coords=cylindrical);
```



Remark. Some equations in cylindrical coordinates may not be easily represented in an explicit form $r = f(z, \theta)$. In the case, use "**implicitplot3d**" to plot them.
Example 5. Find the cylindrical equation for the surface $z = xy$. Then sketch its graph.

```
> cyleq:=eval(z=x*y,[x=r*cos(theta),y=r*sin(theta),z=z]);
```

$$cyleq := z = r^2 \cos(\theta\sim) \sin(\theta\sim) \quad (7.1.9)$$

```
> implicitplot3d(z=r^2*cos(theta)*sin(theta),r=0..2,theta=0..2*
  Pi,z=-1..1,axes=frame,coords=cylindrical);
```



13.7.2 Spherical coordinates

Example 6. Find the spherical coordinates for the point $(2, -2\sqrt{3},\ 3)$ in the rectangular.

```
> eval([sqrt(x^2+y^2+z^2),arctan(y/x),arccos(z/sqrt(x^2+y^2+z^2)
)],[x=2,y=-2*sqrt(3),z=3]);
```

$$\left[\sqrt{25},\ -\frac{1}{3}\ \pi,\ \arccos\left(\frac{3}{25}\ \sqrt{25}\right)\right] \quad (7.2.1)$$

Example 7. Convert the spherical coordinates $(\rho, \theta, \varphi) = (2, \frac{3\pi}{4}, \frac{\pi}{2})$ to the rectangular.

```
> eval([rho*cos(theta)*sin(phi),rho*sin(theta)*sin(phi),rho*cos
(phi)],[rho=2,theta=3*Pi/4,phi=Pi/2]);
```

$$\left[-\sqrt{2}, \sqrt{2}, 0\right] \quad (7.2.2)$$

Example 8. Find the spherical equation $\rho = f(\theta, \varphi)$ for the surface $x^2 + y^2 + z^2 = 9$. Then sketch its graph.

```
> assume(phi,real);
> sphEq:=eval((x^2+y^2+z^2)-9,[x=rho*cos(theta)*sin(phi),y=rho*
sin(theta)*sin(phi),z=rho*cos(phi)]);
```

$$sphEq := \rho^2 \cos(\theta\sim)^2 \sin(\phi\sim)^2 + \rho^2 \sin(\theta\sim)^2 \sin(\phi\sim)^2 + \rho^2 \cos(\phi\sim)^2 - 9 \quad (7.2.3)$$

```
> sphEq:=simplify(sphEq);
```

$$sphEq := \rho^2 - 9 \quad (7.2.4)$$

```
> solve(sphEq,rho);
```

$$3, -3 \quad (7.2.5)$$

```
> plot3d(3,theta=0..2*Pi,phi=0..Pi,axes=frame,coords=spherical,
scaling=constrained);
```



Example 9. Find the spherical equation $\rho = f(\theta, \varphi)$ for $z = x^2 + y^2$ and sketch its graph.

```
> sphEq:=eval(x^2+y^2,[x=rho*cos(theta)*sin(phi),y=rho*sin
  (theta)*sin(phi)]);
```

$$sphEq := \rho^2 \cos(\theta\sim)^2 \sin(\phi\sim)^2 + \rho^2 \sin(\theta\sim)^2 \sin(\phi\sim)^2 \quad (7.2.6)$$

```
> sphEq:=simplify(sphEq,trig);
```

$$sphEq := -\rho^2 \left(-1 + \cos(\phi\sim)^2\right) \quad (7.2.7)$$

```
> solve(rho*cos(phi)=sphEq,rho);
```

$$0, -\frac{\cos(\phi\sim)}{-1 + \cos(\phi\sim)^2} \quad (7.2.8)$$

```
> plot3d(cos(phi)/sin(phi)^2,theta=0..2*Pi,phi=0..Pi,view=[-3.
  .3,-3..3,0..4],axes=frame,coords=spherical,scaling=
  constrained);
```



Exercises

1. Find the cylindrical coordinates for the point $(-\sqrt{3}, 1, 2)$ in the rectangular.
2. Convert the cylindrical coordinates $(r, \theta, z) = (3, \frac{5\pi}{6}, -2)$ to the rectangular.
3. Find the cylindrical equation $r = f(z, \theta)$ for the surface $z = \sqrt{x^2 + y^2}$. Then sketch its graph.
4. Find the cylindrical equation $r = f(z, \theta)$ for the surface $z = x^2 + y^2$. Then sketch its graph.
5. Find the spherical coordinates for the point $(-2, 2\sqrt{3}, 3)$ in the rectangular.

6. Convert the spherical coordinates $(\rho, \theta, \varphi) = (3, \frac{5\pi}{4}, \frac{\pi}{3})$ to the rectangular.

7. Sketch the graph of $\varphi = \frac{\pi}{4}$ in the spherical coordinates. [Hint: Use "**plot3d**" for the vector [rho, theta, Pi/4], or use "**implicitplod3d**".]

8. Find the spherical equation $\rho = f(\theta, \varphi)$ for $z = \sqrt{x^2 + y^2}$ and sketch its graph.

Chapter 14 CALCULUS OF VECTOR-VALUED FUNCTIONS

14.1 Vector-Valued Functions

14.1.1 Vector-valued functions of single variable

A **vector-valued function** is a function whose output is a vector. For example, a vector-valued function from $I \subset R$ to R^3 has the form of

$$x = f(t),\ \ y = g(t),\ \ z = h(t),\ t \in I$$

You can also write it in the vector form:

$$\mathbf{r}(t) = f(t)\mathbf{i} + g(t)\mathbf{j} + \ h(t)\mathbf{k}$$

Example 1. Draw the curve of the vector-valued function $\mathbf{r}(t)$ and evaluate it at $t = \frac{\pi}{4}$, where $\mathbf{r}(t) = \sin(3t)\cos(t)\mathbf{i} + \sin(3t)\sin(t)\mathbf{j} + t\mathbf{k},\ 0 \le t \le 2\pi.$

```
> with(plots): with(linalg):
Warning, the name changecoords has been redefined
Warning, the protected names norm and trace have been redefined
and unprotected
> f:=t->sin(3*t)*cos(t); g:=t->sin(3*t)*sin(t); h:=t->t;
```

$$f := t \to \sin(3\,t)\cos(t)$$

$$g := t \to \sin(3\,t)\sin(t)$$

$$h := t \to t \qquad (1.1.1)$$

```
> spacecurve([f(t),g(t),h(t)],t=0..2*Pi, axes=normal, numpoints=
  400);
```



```
> rValue:=[f(Pi/4),g(Pi/4),h(Pi/4)];
```

$$rValue := \left[\frac{1}{2}, \frac{1}{2}, \frac{1}{4}\,\pi\right] \tag{1.1.2}$$

Example 2. Draw the curve of the vector-valued function $\mathbf{r}(t) = \cos(t)\mathbf{i} + \sin(t)\mathbf{j} + \sin(2t)\mathbf{k}$, $0 \le t \le 2\pi$.

```
> f:=t->cos(t); g:=t->sin(t); h:=t->sin(2*t);
```

$$f := t \rightarrow \cos(t)$$
$$g := t \rightarrow \sin(t)$$
$$h := t \rightarrow \sin(2\,t) \tag{1.1.3}$$

```
> spacecurve([f(t),g(t),h(t)], t=0..2*Pi, axes=normal,
  numpoints=400);
```



Example 3. Draw the curve of the vector-valued function
$\mathbf{r}(t) = (4 + \sin(20t))\cos(t)\mathbf{i} + (4 + \sin(20t))\sin(t)\mathbf{j} + \cos(20t)\mathbf{k}$, $0 \le t \le 2\pi$.

```
> spacecurve([(4+sin(20*t))*cos(t), (4+sin(20*t))*sin(t), cos
  (20*t)], t=0..2*Pi, axes=normal, numpoints=600);
```



Example 4. Represent the vector-valued function $\mathbf{v}(t) = t^2\mathbf{i} + \sqrt{t-1}\,\mathbf{j} + \sqrt{5-t}\,\mathbf{k}$ by its components and find the vector-valued function 2**v**(t), the function $v_2(t)$ (its second component) and evaluate **v** (t) at t = 3 and $v_2(t)$ at t = 3.

(1) Represent the vector-valued function **v**(t) by its components.

```
> v:=[t^2,sqrt(t-1),sqrt(5-t)];
```

$$v := \left[t^2, \sqrt{t-1}, \sqrt{5-t}\right]$$

(2) Find 2**v**(t).

```
> 2*v;
```

$$\left[2\,t^2, 2\sqrt{t-1}, 2\sqrt{5-t}\right]$$

(3) Extract the second component $v_2(t)$ from **v**(t).

```
> v[2];
```

$$\sqrt{t-1}$$

(4) Evaluate **v**(t) at t = 3.

```
> v2:=eval(v, t=3);
```

$$v2 := \left[9, \sqrt{2}, \sqrt{2}\right]$$

(5) Evaluate the component $v_2(2)$ at t = 3.

```
> eval(v[2], t=2);
```

$$1 \qquad (1.1.4)$$

14.1.2 Parametrizing the intersection of surfaces

Example 5. Parameterize the intersection of $z = \sqrt{x^2 + y^2}$ and $z = 1 + y$ and draw its graph.
Step 1. Let x be the parameter and then find y and z as its functions.

```
> solve({z^2=x^2+y^2,z=1+y},{y,z});
```

$$\left\{z = \frac{1}{2}x^2 + \frac{1}{2}, y = \frac{1}{2}x^2 - \frac{1}{2}\right\}$$

Step2. Write the curve as a parametric vector-valued function of x.

```
> r:=[x,x^2/2-1/2,x^2+1/2];
```

$$r := \left[x, \frac{1}{2}x^2 - \frac{1}{2}, x^2 + \frac{1}{2}\right] \qquad (1.2.1)$$

Step 3. Draw its graph.

```
> spacecurve(r,x=-3..3,axes=normal);
```



Example 6. Parameterize the intersection of $x^2 + y^2 = 4$ and $z = xy$ and draw its graph.
Step 1. Find its parametric equation. Since the first equation has no variable "z", we set $x = 2\cos(t)$, $y = 2\sin(t)$. Then the second equation becomes $z = 4\cos(t)\sin(t)$.

```
> r:=[2*cos(t), 2*sin(t), 4*cos(t)*sin(t)];
```

$$r := [2\cos(t), 2\sin(t), 4\cos(t)\sin(t)] \tag{1.2.2}$$

Step 2. Plot the graph.

```
> spacecurve(r,t=0..2*Pi, axes=normal);
```



Exercises

1. Draw the curve of the vector function $\mathbf{r}(t) = \sin(t)\mathbf{i} + \left(t^2 - \cos(t)\right)\mathbf{j} + e^t\mathbf{k}$, $0 \le t \le 2\pi$, and evaluate it at $t = \frac{\pi}{2}$.
2. Let $\mathbf{v}(t) = (t^2+1)\mathbf{i} + \sqrt{t-1}\,\mathbf{j} + (2-t)\mathbf{k}$. Find $2\mathbf{v}$, $\mathbf{v}(2)$, and $\mathbf{v}(t^2)$.
3. Find the points where the path $\mathbf{r}(t) = [\sin(t), \cos(t), \sin(t)\cos(2t)]$ intersects the xy-plane.
4. Parameterize the intersection of $y^2 - z^2 = x - 2$ and $y^2 + z^2 = 9$, and draw its graph.
5. Use sine and cosine to parameterize the intersection of $x^2 + y^2 = 1$ and $x^2 + z^2 = 1$ (use two vector-valued functions) and draw its graph.

14.2 Calculus of Vector-Valued Functions

14.2.1 Limits and continuity

(1) Let $\mathbf{r}(t) = f(t)\mathbf{i} + g(t)\mathbf{j} + h(t)\mathbf{k}$ and assume that $\mathbf{L} = [L_1, L_2, L_3]$ is a constant vector. We say that the vector-valued function $\mathbf{r}$ has **limit L** as t approaches t_0 and write

$$\lim_{t \to t_0} \mathbf{r}(t) = \mathbf{L}$$

if

$$\lim_{t \to t_0} f(t) = L_1,\quad \lim_{t \to t_0} g(t) = L_2,\quad \lim_{t \to t_0} h(t) = L_3$$

(2) A vector-valued function $\mathbf{r}(t)$ is **continuous at a point** $t = t_0$ if

$$\lim_{t \to t_0} \mathbf{r}(t) = \mathbf{r}(t_0)$$

Hence, if a vector-valued function $\mathbf{r}(t)$ is continuous at a point $t = t_0$ its limit $\lim_{t \to t_0} \mathbf{r}(t)$ can be obtained by evaluating $\mathbf{r}(t)$ at $t = t_0$.

Example 1. Find the limit of $\mathbf{v}(t) = \left[t^2, \sqrt{t-1}, \sqrt{5-t}\right]$ as t approaches 3.

Step 1. Define the vector-valued function $\mathbf{v}(t)$.

```
> v:=[t^2,sqrt(t-1),sqrt(5-t)];
```

$$v := \left[t^2, \sqrt{t-1}, \sqrt{5-t}\right]$$

Step 2. Find the limit.

```
>    map(limit,v,t=3);
```

$$\left[9, \sqrt{2}, \sqrt{2}\right]$$

It can also be obtained as the following:

```
> limv:=[limit(v[1],t=3),limit(v[2],t=3),limit(v[3],t=3)];
```

$$limv := \left[9, \sqrt{2}, \sqrt{2}\right]$$

or it can be obtained by directly evaluating the vector-valued function at $t = 3$, since it is continuous there:

```
> limv:=eval(v,t=3);
```

$$limv := \left[9, \sqrt{2}, \sqrt{2}\right] \quad (2.1.1)$$

Example 2. Let $\mathbf{r}(t) = \cos(t)\mathbf{i} + \sin(t)\mathbf{j} + \sin(2t)\mathbf{k}$. Find the limit $\lim_{t \to \frac{\pi}{6}} \frac{r(t) - r\left(\frac{\pi}{6}\right)}{t - \frac{\pi}{6}}$.

Note. The function $\frac{r(t) - r\left(\frac{\pi}{6}\right)}{t - \frac{\pi}{6}}$ has no definition at $t = \frac{\pi}{6}$. Hence, you cannot evaluate it at $t = \frac{\pi}{6}$.

```
> r:=[cos(t), sin(t), sin(2*t)];
```

$$r := [\cos(t), \sin(t), \sin(2\,t)] \tag{2.1.2}$$

```
> r0:=eval(r,t=Pi/6); dt:=t-Pi/6;
```

$$r0 := \left[\frac{1}{2}\sqrt{3}, \frac{1}{2}, \frac{1}{2}\sqrt{3}\right]$$

$$dt := t - \frac{1}{6}\pi \tag{2.1.3}$$

```
> nr:=[(r[1]-r0[1])/dt, (r[2]-r0[2])/dt, (r[3]-r0[3])/(dt)];
```

$$nr := \left[\frac{\cos(t) - \frac{1}{2}\sqrt{3}}{t - \frac{1}{6}\pi}, \frac{\sin(t) - \frac{1}{2}}{t - \frac{1}{6}\pi}, \frac{\sin(2\,t) - \frac{1}{2}\sqrt{3}}{t - \frac{1}{6}\pi}\right] \tag{2.1.4}$$

```
> lmtr:=map(limit,nr, t=Pi/6);
```

$$lmtr := \left[\frac{-1}{2}, \frac{1}{2}\sqrt{3}, 1\right] \tag{2.1.5}$$

14.2.2 Derivatives and motion

(1) The derivative of a vector-valued function $\mathbf{r}(t) = f(t)\mathbf{i} + g(t)\mathbf{j} + h(t)\mathbf{k}$ is

$$\mathbf{r}'(t) = f'(t)\mathbf{i} + g'(t)\mathbf{j} + h'(t)\mathbf{k}$$

(2) If the vector-valued function $\mathbf{r}(t)$ is the position function of a particle moving a curve in space, then its velocity is $\mathbf{r}'(t)$, which is usually denoted by $\mathbf{v}(t)$, its speed is $|\mathbf{v}(t)|$, and its acceleration is $\mathbf{a}(t) = \mathbf{v}'(t) = \mathbf{r}''(t)$.

Example 3. Find $\mathbf{v}'(t)$ of the vector-valued function $\mathbf{v}(t) = \left[t^2, \sqrt{t-1}, \sqrt{5-t}\right]$.

```
> v:=[t^2,sqrt(t-1),sqrt(5-t)];
```

$$v := \left[t^2, \sqrt{t-1}, \sqrt{5-t}\right] \tag{2.2.1}$$

```
> dv=[diff(v[1],t),diff(v[2],t),diff(v[3],t)];
```

$$dv = \left[2\,t, \frac{1}{2}\frac{1}{\sqrt{t-1}}, -\frac{1}{2}\frac{1}{\sqrt{5-t}}\right]$$

Or do the following:

```
> dv:=map(diff,v,t);
```

$$dv := \left[2\,t, \frac{1}{2}\frac{1}{\sqrt{t-1}}, -\frac{1}{2}\frac{1}{\sqrt{5-t}}\right] \tag{2.2.2}$$

The following way also works.

```
> dv:=diff(v,t);
```

$$dv := \left[2\,t,\ \frac{1}{2}\,\frac{1}{\sqrt{t-1}},\ -\frac{1}{2}\,\frac{1}{\sqrt{5-t}}\right]$$

You can also use the function-assignment notation to calculate the derivative of a vector-valued function.

```
> v:=[t->t^2,t->sqrt(t-1),t->sqrt(5-t)];
```

$$v := \left[t\to t^2,\ t\to\sqrt{t-1},\ t\to\sqrt{5-t}\right] \tag{2.2.3}$$

```
> D(v);
```

$$\left[t\to 2\,t,\ t\to\frac{1}{2}\,\frac{1}{\sqrt{t-1}},\ t\to-\frac{1}{2}\,\frac{1}{\sqrt{5-t}}\right] \tag{2.2.4}$$

```
> dv:=D(v)(t);
```

$$dv := \left[2\,t,\ \frac{1}{2}\,\frac{1}{\sqrt{t-1}},\ -\frac{1}{2}\,\frac{1}{\sqrt{5-t}}\right] \tag{2.2.5}$$

Example 4. Let $\mathbf{r}(t) = [t^2, 5t, 1]$ and $f(t) = e^t$. Calculate $\frac{d}{dt}f(t)\mathbf{r}(t)$ and $\frac{d}{dt}\mathbf{r}(f(t))$.

Note that the syntax of the function e^t **is** exp(t).

```
> r:=[t^2, 5*t,1]; f:=exp(t);
```

$$r := \left[t^2,\ 5\,t,\ 1\right]$$

$$f := e^t \tag{2.2.6}$$

```
> dfr:=simplify(diff(f*r,t));
```

$$dfr := e^t\left[2\,t+t^2,\ 5+5\,t,\ 1\right] \tag{2.2.7}$$

```
> rft:=eval(r,t=exp(t));
```

$$rft := \left[\left(e^t\right)^2,\ 5\,e^t,\ 1\right] \tag{2.2.8}$$

```
> drft:=diff(rft,t);
```

$$drft := \left[2\,\left(e^t\right)^2,\ 5\,e^t,\ 0\right] \tag{2.2.9}$$

Example 5. Prove the formula $\frac{d}{dt}(\mathbf{r}(t)\times\mathbf{r}'(t)) = \mathbf{r}(t)\times\mathbf{r}''(t)$.

```
> f:='f':g:='g': h:='h':
> r:=[f(t),g(t),h(t)];
```

$$r := [f(t),\ g(t),\ h(t)] \tag{2.2.10}$$

```
> dr:=diff(r,t);
```

$$dr := \left[\frac{d}{dt}f(t),\ \frac{d}{dt}g(t),\ \frac{d}{dt}h(t)\right] \tag{2.2.11}$$

```
> ddr:=diff(r,t,t);
```

$$ddr := \left[\frac{d^2}{dt^2}f(t),\ \frac{d^2}{dt^2}g(t),\ \frac{d^2}{dt^2}h(t)\right] \tag{2.2.12}$$

```
> crdr:= crossprod(r,dr);
```

$$crdr := \left[\left[g(t) \left(\frac{d}{dt} h(t) \right) - h(t) \left(\frac{d}{dt} g(t) \right), h(t) \left(\frac{d}{dt} f(t) \right) - f(t) \left(\frac{d}{dt} h(t) \right), f(t) \left(\frac{d}{dt} g(t) \right) - g(t) \left(\frac{d}{dt} f(t) \right) \right] \right] \quad (2.2.13)$$

```
> lft:=map(diff,crdr,t);
```

$$lft := \left[\left[g(t) \left(\frac{d^2}{dt^2} h(t) \right) - h(t) \left(\frac{d^2}{dt^2} g(t) \right), h(t) \left(\frac{d^2}{dt^2} f(t) \right) - f(t) \left(\frac{d^2}{dt^2} h(t) \right), f(t) \left(\frac{d^2}{dt^2} g(t) \right) - g(t) \left(\frac{d^2}{dt^2} f(t) \right) \right] \right] \quad (2.2.14)$$

```
> rght:=crossprod(r,ddr);
```

$$rght := \left[\left[g(t) \left(\frac{d^2}{dt^2} h(t) \right) - h(t) \left(\frac{d^2}{dt^2} g(t) \right), h(t) \left(\frac{d^2}{dt^2} f(t) \right) - f(t) \left(\frac{d^2}{dt^2} h(t) \right), f(t) \left(\frac{d^2}{dt^2} g(t) \right) - g(t) \left(\frac{d^2}{dt^2} f(t) \right) \right] \right] \quad (2.2.15)$$

```
> simplify(lft-rght);
```

$$0 \quad (2.2.16)$$

The formula is proved.

Example 6. Let the position function of a motion be given by $\mathbf{r}(t) = 3\cos(t)\mathbf{i} + 3\sin(t)\mathbf{j} + t^2\mathbf{k}$. Find (a) its velocity and acceleration; (b) the speed; (c) the times when the acceleration is orthogonal to its velocity.

```
> r:=[3*cos(t),3*sin(t),t^2];
```

$$r := \left[3\cos(t), 3\sin(t), t^2\right] \quad (2.2.17)$$

```
> v:=diff(r,t);
```

$$v := [-3\sin(t), 3\cos(t), 2t] \quad (2.2.18)$$

```
> a:=diff(v,t);
```

$$a := [-3\cos(t), -3\sin(t), 2] \quad (2.2.19)$$

```
> sp:=simplify(sqrt(v[1]^2+v[2]^2+v[3]^2));
```

$$sp := \sqrt{4t^2 + 9} \quad (2.2.20)$$

```
> tm:=solve(innerprod(v,a)=0,t);
```

$$tm := 0 \quad (2.2.21)$$

14.2.3 Tangent vector and tangent line of a curve

(1) The tangent vector of a vector-valued function $\mathbf{r}(t) = f(t)\mathbf{i} + g(t)\mathbf{j} + h(t)\mathbf{k}$ at $t = t_0$ is

$$\mathbf{r}'(t_0) = f'(t_0)\mathbf{i} + g'(t_0)\mathbf{j} + h'(t_0)\mathbf{k}$$

(2) The equation of the tangent line of the curve of $\mathbf{r}(t)$ at $t = t_0$ is

$$\mathbf{L}(t) = \mathbf{r}(t_0) + t\mathbf{r}'(t_0)$$

(3) If $\mathbf{r}$ is a differentiable vector-valued function of constant length $|\mathbf{r}(t)| = \text{const}$, then

$$\mathbf{r} \cdot \mathbf{r}' = 0$$

Example 7. Let a curve be defined by the vector-valued function $\mathbf{c}(t) = \left[\sin(t) - t\cos(t),\ \cos(t) + t\sin(t),\ t^2\right]$, $0 \le t \le 6\pi$. Find its tangent line at $t_0 = 3\pi$ and draw the curve and the tangent line on the same coordinates.

```
> c:=[sin(t)-t*cos(t), cos(t)+t*sin(t), t^2];
```

$$c := \left[\sin(t) - t\cos(t), \cos(t) + t\sin(t), t^2\right] \tag{2.3.1}$$

```
> r0:=eval(c,t=3*Pi/2);
```

$$r0 := \left[-1, -\frac{3}{2}\pi, \frac{9}{4}\pi^2\right] \tag{2.3.2}$$

```
> dc:=diff(c,t);
```

$$dc := [t\sin(t), t\cos(t), 2t] \tag{2.3.3}$$

```
> dcto:=eval(dc, t=3*Pi/2);
```

$$dcto := \left[-\frac{3}{2}\pi, 0, 3\pi\right] \tag{2.3.4}$$

```
> LnEQ:=r0+[t*dcto[1], t*dcto[2],t*dcto[3]];
```

$$LnEQ := \left[-\frac{3}{2}t\pi - 1, -\frac{3}{2}\pi, 3t\pi + \frac{9}{4}\pi^2\right] \tag{2.3.5}$$

```
> spacecurve({c, LnEQ}, t=0..6*Pi, axes=normal, numpoints=400);
```



Example 8. The vector-valued function $\mathbf{r}(t) = [t - \sin(t), 1 - \cos(t)]$ traces a cycloid on the plane. Find the points where (a) $\mathbf{r}'(t)$ is horizontal and nonzero; (b) $\mathbf{r}'(t)$ is the zero vector.

```
> r:=[t-sin(t), 1-cos(t)];
```

$$r := [t - \sin(t), 1 - \cos(t)] \tag{2.3.6}$$

```
> dr:=diff(r,t);
```

$$dr := [1 - \cos(t), \sin(t)] \tag{2.3.7}$$

Note that $\mathbf{r}'(t)$ is horizontal and nonzero, which means that $x'(t) \ne 0$ but $y'(t) = 0$, but $\mathbf{r}'(t) = \mathbf{0}$ if and only if $x'(t) = 0$ and $y'(t) = 0$.

```
> solve(dr[2]=0,t);
```

$$0 \tag{2.3.8}$$

As we mentioned before, MAPLE only gives the solution of trigonometric equations in the principal value interval, for sin(t), in the interval [−π/2, π/2]. Hence, we need to find other zeroes using the properties of trigonometric functions. It is clear that another zero of the equation in [0, 2π] is π.

```
> eval(dr[1], t=0);
```

$$0 \quad (2.3.9)$$

```
> eval(dr[1],t=Pi);
```

$$2 \quad (2.3.10)$$

Hence, (a) $\mathbf{r}'(t)$ is horizontal and nonzero at $t = \pi$. (b) $\mathbf{r}'(t)$ is the zero vector at $t = 0$.
Example 9. Show that $\mathbf{r}(t) = 3\cos(t)\mathbf{i} + 3\sin(t)\mathbf{j} + 4\mathbf{k}$ has constant length and is orthogonal to its derivative.

```
> r:=[3*cos(t), 3*sin(t), 4];
```

$$r := [3\cos(t), 3\sin(t), 4] \quad (2.3.11)$$

```
> lengthr:=simplify(sqrt(r[1]^2+r[2]^2+r[3]^2));
```

$$lengthr := 5 \quad (2.3.12)$$

```
> dr:=diff(r,t);
```

$$dr := [-3\sin(t), 3\cos(t), 0] \quad (2.3.13)$$

```
> innerprod(r,dr);
```

$$0 \quad (2.3.14)$$

14.2.4 Integrals of vector-valued functions

Let $\mathbf{r}(t) = f(t)\mathbf{i} + g(t)\mathbf{j} + h(t)\mathbf{k}$ be a vector-valued function, its **antiderivative** is the vector-valued function

$$\mathbf{R}(t) = F(t)\mathbf{i} + G(t)\mathbf{j} + H(t)\mathbf{k}$$

such that $\mathbf{R}'(t) = \mathbf{r}(t)$. The **definite integral** of $\mathbf{r}(t)$on the interval [a, b] is the vecter

$$\mathbf{I} = \int_a^b F(t)\,\mathrm{d}x\mathbf{i} + \int_a^b G(t)\,\mathrm{d}x\mathbf{j} + \int_a^b H(t)\,\mathrm{d}x\mathbf{k}$$

Example 10. Find the integral of $\mathbf{v}(t) = t^2\mathbf{i} + \sqrt{t-1}\,\mathbf{j} + \sqrt{5-t}\,\mathbf{k}$ for t from 1 to 4.

```
> v:=[t^2,sqrt(t-1),sqrt(5-t)];
```

$$v := \left[t^2, \sqrt{t-1}, \sqrt{5-t}\right] \quad (2.4.1)$$

```
> Iv=[int(v[1],t=1..4),int(v[2],t=1..4),int(v[3],t=1..4)];
```

$$Iv = \left[21, 2\sqrt{3}, \frac{14}{3}\right]$$

Or, you can apply the "**map**" command to evaluate the integral.

```
> map(int,v,t=1..4);
```

$$\left[21, 2\sqrt{3}, \frac{14}{3}\right]$$

Example 11. Find the integral of $\mathbf{r}(t) = \cos(t)\mathbf{i} + \sin(t)\mathbf{j} + \sin(2t)\mathbf{k}$ for t from 0 to π/2.

```
> r:=[cos(t), sin(t), sin(2*t)];
```

$$r := [\cos(t), \sin(t), \sin(2t)] \quad (2.4.2)$$

```
> map(int, r, t=0..Pi/2);
```

$$[1, 1, 1] \quad (2.4.3)$$

Exercises

1. Let $\mathbf{r}(t) = \frac{1}{t}\mathbf{i} + \sin(t)\mathbf{j} + 2t\mathbf{k}$. Find the limit: $\lim_{h \to 0} \frac{\mathbf{r}(t+h) - \mathbf{r}(t)}{h}$.
2. Let $\mathbf{v}(t) = (t^2+1)\mathbf{i} + \sqrt{t-1}\,\mathbf{j} + (2-t)\mathbf{k}$. Find $\mathbf{v}'(t)$ and $\mathbf{v}'(3)$.
3. Let $\mathbf{r}(t) = (t+1)\mathbf{i} + \left(t^2-1\right)\mathbf{j} + 2t\mathbf{k}$. Find $\frac{\mathrm{d}}{\mathrm{d}t}\mathbf{r}(t)$ and $\frac{\mathrm{d}^2}{\mathrm{d}t^2}\mathbf{r}(t)$.
4. Let $\mathbf{r}(t) = 2\cos(t)\mathbf{i} + 3\sin(t)\mathbf{j} + 4t\mathbf{k}$. Find $\frac{\mathrm{d}}{\mathrm{d}t}\mathbf{r}(t)$ and $\frac{\mathrm{d}^2}{\mathrm{d}t^2}\mathbf{r}(t)$ at $t = \pi/2$.
5. Let $\mathbf{r}(t) = [t^2,\ 2,\ 3t]$ and $f(t) = \sin(t)$. Calculate $\frac{\mathrm{d}}{\mathrm{d}t}(f(t)\mathbf{r}(t))$ and $\frac{\mathrm{d}}{\mathrm{d}t}\mathbf{r}(\mathrm{f}(t))$.
6. Find the integral of $\mathbf{r}(t) = \cos(2t)\mathbf{i} + \sin(2t)\mathbf{j} + t\mathbf{k}$ for t from 0 to $\pi/2$.

14.3 Arc Length and Speed

(1) The length of a smooth curve $\mathbf{r}(t) = x(t)\mathbf{i} + y(t)\mathbf{j} + z(t)\mathbf{k}$, $a \le t \le b$, is

$$L = \int_a^b \sqrt{\left(\frac{\mathrm{d}x}{\mathrm{d}t}\right)^2 + \left(\frac{\mathrm{d}x}{\mathrm{d}t}\right)^2 + \left(\frac{\mathrm{d}x}{\mathrm{d}t}\right)^2}\,\mathrm{d}t$$

$$= \int_a^b \|\boldsymbol{r}'(t)\|\mathrm{d}t$$

(2) The differential of arc is

$$ds = s'(t)\mathrm{d}t = \|\mathbf{r}'(t)\|\,\mathrm{d}t = \|\mathbf{v}(t)\|\,\mathrm{d}t$$

if $\mathbf{r}(t)$ describes a motion, then $\|\mathbf{v}(t)\|$ is the speed of the motion.

(3) The unit tangent vector $\mathbf{T}$ is

$$\mathbf{T} = \frac{\mathrm{d}\mathbf{r}}{\mathrm{d}s} = \frac{\mathbf{v}}{\|\mathbf{v}\|}$$

14.3.1 Arc length

Example 1. Find the length of arc with $\mathbf{r}(t) = \cos(t)\mathbf{i} + \sin(t)\mathbf{j} + t\mathbf{k}$ from point (1, 0, 0) to (1, 0, 2π).

```
> with(linalg):
> r:=[cos(t),sin(t),t];
```

$$r := [\cos(t), \sin(t), t]$$

```
> dirr:=diff(r,t);
```

$$dirr := [-\sin(t), \cos(t), 1]$$

```
> lengthr=int(norm(dirr,2),t=0..2*Pi);
```

$$5 = 2\pi\sqrt{2}$$

Example 2. Find the length of the curve described by $\boldsymbol{r}(t) = [\cos(t),\ \sin(t), \sin(2t)]$.

```
> r:=[sin(t), cos(t), sin(2*t)];
```

$$r := [\sin(t), \cos(t), \sin(2t)] \tag{3.1.1}$$

```
> dr:=diff(r,t);
```

$$dr := [\cos(t), -\sin(t), 2\cos(2t)] \tag{3.1.2}$$

```
> ds:=int(norm(dr,2),t=0..2*Pi);
```

$$ds := \int_0^{2\pi} \sqrt{|\cos(t)|^2 + |\sin(t)|^2 + 4|\cos(2t)|^2}\,\mathrm{d}t \tag{3.1.3}$$

```
> evalf(%);
```

$$10.54073433 \quad (3.1.4)$$

Example 3. Find the length of the curve $\boldsymbol{r}(t) = \left[-2\sin(t), 2\cos(t), \sqrt{5}\right]$ from $t = 0$ to π.

```
> r:=[-2*sin(t), 2*cos(t), sqrt(5)];
```

$$r := \left[-2\sin(t), 2\cos(t), \sqrt{5}\right] \quad (3.1.5)$$

```
> dr:=diff(r,t);
```

$$dr := [-2\cos(t), -2\sin(t), 0] \quad (3.1.6)$$

```
> L:=int(norm(dr,2), t=0..Pi);
```

$$L := 2\pi \quad (3.1.7)$$

14.3.2 Velocity and speed

Example 4. Assume the position vector-valued function of a motion is **r**(t) = [2cos(t), 3sin(t), 4t]. Find its velocity and speed at $t = \pi/2$.

```
> r:=[2*cos(t), 3*sin(t), 4*t];
```

$$r := [2\cos(t), 3\sin(t), 4t] \quad (3.2.1)$$

```
> v:=diff(r,t);
```

$$v := [-2\sin(t), 3\cos(t), 4] \quad (3.2.2)$$

```
> vt:= eval(v, t=Pi/2);
```

$$vt := [-2, 0, 4] \quad (3.2.3)$$

```
> sp:=norm(vt,2);
```

$$sp := 2\sqrt{5} \quad (3.2.4)$$

14.3.3 Unit tangent vector T

Example 5. Find the unit tangent vector of **r**(t) = $\left[t^2, \frac{2t^3}{3}, t\right]$ at (1, 2/3, 1).

```
> assume(t,real);
> r:=[t^2,2/3*t^3,t];
```

$$r := \left[t\sim^2, \frac{2}{3}t\sim^3, t\sim\right]$$

```
> dr:=diff(r,t);
```

$$dr := \left[2t\sim, 2t\sim^2, 1\right]$$

```
> ds:=sqrt(innerprod(dr,dr));
```

$$ds := 1 + 2t\sim^2 \quad (3.3.1)$$

```
> UnitTangent:=[dr[1]/ds,dr[2]/ds,dr[3]/ds];
```

$$UnitTangent := \left[\frac{2t\sim}{1+2t\sim^2}, \frac{2t\sim^2}{1+2t\sim^2}, \frac{1}{1+2t\sim^2}\right]$$

```
> Ut1:= simplify(eval(UnitTangent,t=1));
```

$$Ut1 := \left[\frac{2}{3}, \frac{2}{3}, \frac{1}{3}\right]$$

Example 6. Find the unit tangent vector of **r**(t) = [sin(4t), 3t, cos(4t)].

```
> r:=[sin(4*t),3*t,cos(4*t)];
```

$$r := [\sin(4\,t\sim),\ 3\,t\sim,\ \cos(4\,t\sim)]$$

```
> dr:=diff(r,t);
```

$$dr := [4\cos(4\,t\sim),\ 3,\ -4\sin(4\,t\sim)]$$

```
> ds:=sqrt(simplify(dr[1]^2+dr[2]^2+dr[3]^2));
```

$$ds := 5 \qquad (3.3.2)$$

```
> UnitTangent:=dr/ds;
```

$$UnitTangent := \left[\frac{4}{5}\cos(4\,t\sim),\ \frac{3}{5},\ -\frac{4}{5}\sin(4\,t\sim)\right]$$

Exercises

1. Let $\mathbf{r}(t) = (t+1)\mathbf{i} + (t^2-1)\mathbf{j} + 2t\mathbf{k}$ be the position function of a motion. Find its speed, the velocity, the direction of the motion (represented by a unit vector), and the acceleration, when $t = 1$.
2. Let $\mathbf{r}(t) = 2\cos(t)\mathbf{i} + 3\sin(t)\mathbf{j} + 4t\mathbf{k}$ be the position function of a motion. Find the speed, the velocity, the direction of the motion (represented by a unit vector), and the acceleration (1) at a general time t; (2) at the time $t = \pi/2$.
3. Let the position function of a motion be given by $\mathbf{r}(t) = (t^2+1)\mathbf{i} + \sqrt{3}\,t\mathbf{j} + (2-t)\mathbf{k}$. Find the angle between the velocity and acceleration vectors when time $t = 0$.
4. Let the position function of a motion be given by $\mathbf{r}(t) = (1 + \cos(t))\mathbf{i} + \sin(t)\mathbf{j} + t^2\mathbf{k}$.
   (a) Find its velocity, speed, and acceleration at a general time t.
   (b) Find the times when its acceleration is orthogonal to its velocity.
5. Let the curve be defined by the vector function $\boldsymbol{c}(t) = [6\cos(2\,t), 6\sin(2\,t), 5t]$, $0 \le t \le \pi$. Find its tangent line at $t_0 = \pi/4$ and draw the curve and the tangent line on the same coordinates.
6. The function $\mathbf{r}(t) = [t - \sin(t), 1 - \cos(t)]$ traces a cycloid on the plane. Find the points (a) where $\mathbf{r}'(t)$ is horizontal and nonzero; (b) where $\mathbf{r}'(t)$ is the zero vector.
7. Show that $\mathbf{r}(t) = 5\cos(2t)\mathbf{i} + 5\sin(2t)\mathbf{j} + 12\mathbf{k}$ has constant length and it is orthogonal to its derivative.
8. Find the length of arc with $\mathbf{r}(t) = 2\cos(t)\,\mathbf{i} + 2\sin(t)\,\mathbf{j} + \sqrt{5}\,t\mathbf{k}$ from point (2, 0, 0) to $(-2, 0, \sqrt{5}\,\pi)$.
9. Find the length of the curve described by $\mathbf{r}(t) = [\cos(t), \sin(t), \sin(2t)]$ from $t = 0$ to π.
10. Find the unit tangent vector **T** and the unit normal **N** of $\mathbf{r}(t) = [t^2, t^3, t]$ at (1, 1, 1).

14.4 Curvature

(1) If **T** is a unit vector of as smooth curve, the **curvature** function of the curve is

$$\kappa = \left\|\frac{d\mathbf{T}}{ds}\right\|$$

when the curve is defined by a vector-valued function **r**(t), then the curvature can be calculated by the formula

$$\kappa = \frac{1}{\|\mathbf{v}\|}\left\|\frac{d\mathbf{T}}{dt}\right\|, \text{ where } \mathbf{T} = \frac{\mathbf{v}}{\|\mathbf{v}\|}$$

(2) The **vector formula** for curvature:

$$\kappa = \frac{\|\mathbf{v}\times\mathbf{a}\|}{\|\mathbf{v}\|^3} = \frac{\|\mathbf{r}'\times\mathbf{r}''\|}{\|\mathbf{r}'\|^3}$$

(3) The curvature of a graph $y = f(x)$ in the xy-plane at the point $(x, f(x))$ is

$$\kappa(x) = \frac{|f''(x)|}{(1+f'(x)^2)^{\frac{3}{2}}}$$

(4) The **principal unit normal vector N** is defined by

$$\mathbf{N} = \frac{1}{\kappa}\frac{\mathrm{d}\mathbf{T}}{\mathrm{d}s}.$$

which can be calculated by the formula

$$\mathbf{N} = \frac{\frac{\mathrm{d}\mathbf{T}}{\mathrm{d}t}}{\left\|\frac{\mathrm{d}\mathbf{T}}{\mathrm{d}t}\right\|}$$

(5) The **circle of curvature** or **osculating circle** at a point of P on the curve is the circle with the radius $\rho = 1/\kappa$ and the center $C = P + \rho\mathbf{N}$.

14.4.1 Computing curvature and finding unit normal N

Example 1. Find **T** and **N** for $\mathbf{r}(t) = \cos(2t)\mathbf{i} + \sin(2t)\mathbf{j}$.

```
> assume(t, real);
> r:=[cos(2*t), sin(2*t)];
```

$$r := [\cos(2\,t\sim), \sin(2\,t\sim)] \qquad (4.1.1)$$

```
> dr:=diff(r,t);
```

$$dr := [-2\sin(2\,t\sim), 2\cos(2\,t\sim)] \qquad (4.1.2)$$

```
> ds:=sqrt(simplify(innerprod(dr,dr)));
```

$$ds := 2 \qquad (4.1.3)$$

```
> T:=dr/ds;
```

$$T := [-\sin(2\,t\sim), \cos(2\,t\sim)] \qquad (4.1.4)$$

```
> dT:=diff(T,t);
```

$$dT := [-2\cos(2\,t\sim), -2\sin(2\,t\sim)] \qquad (4.1.5)$$

```
> NormdT:=sqrt(simplify(innerprod(dT,dT)));
```

$$NormdT := 2 \qquad (4.1.6)$$

```
> N:=dT/NormdT;
```

$$N := [-\cos(2\,t\sim), -\sin(2\,t\sim)] \qquad (4.1.7)$$

Example 2. Find the curvature function of $\mathbf{r}(t) = t\mathbf{i} + t^2\mathbf{j} + t^3\mathbf{k}$ and its value at the point (0, 0, 0).

```
> assume(t, real);
> r:=[t,t^2,t^3];
```

$$r := \left[t\sim, t\sim^2, t\sim^3\right]$$

```
> dr:=diff(r,t);
```

$$dr := \left[1, 2\,t\sim, 3\,t\sim^2\right]$$

```
> ds:=sqrt(innerprod(dr,dr));
```

$$ds := \sqrt{1+4\,t\sim^2+9\,t\sim^4}$$

```
> T:= [dr[1]/ds,dr[2]/ds,dr[3]/ds];
```

$$T := \left[\frac{1}{\sqrt{1+4\,t\sim^2+9\,t\sim^4}}, \frac{2\,t\sim}{\sqrt{1+4\,t\sim^2+9\,t\sim^4}}, \frac{3\,t\sim^2}{\sqrt{1+4\,t\sim^2+9\,t\sim^4}}\right] \tag{4.1.8}$$

```
> dT:=diff(T,t);
```

$$dT := \left[-\frac{1}{2}\,\frac{8\,t\sim+36\,t\sim^3}{\left(1+4\,t\sim^2+9\,t\sim^4\right)^{(3/2)}}, \frac{2}{\sqrt{1+4\,t\sim^2+9\,t\sim^4}} - \frac{t\sim\left(8\,t\sim+36\,t\sim^3\right)}{\left(1+4\,t\sim^2+9\,t\sim^4\right)^{(3/2)}}, \frac{6\,t\sim}{\sqrt{1+4\,t\sim^2+9\,t\sim^4}} - \frac{3}{2}\,\frac{t\sim^2\left(8\,t\sim+36\,t\sim^3\right)}{\left(1+4\,t\sim^2+9\,t\sim^4\right)^{(3/2)}}\right] \tag{4.1.9}$$

```
> dTnorm:=simplify(sqrt(innerprod(dT,dT)));
```

$$dTnorm := \frac{2\sqrt{9\,t\sim^4+9\,t\sim^2+1}}{1+4\,t\sim^2+9\,t\sim^4} \tag{4.1.10}$$

The curvature is

```
> curvr:=dTnorm/ds;
```

$$curvr := \frac{2\sqrt{9\,t\sim^4+9\,t\sim^2+1}}{\left(1+4\,t\sim^2+9\,t\sim^4\right)^{(3/2)}}$$

The curvature at (0, 0, 0) is

```
> curvrat0:=eval(curvr,t=0);
```

$$curvrat0 := 2$$

Example 3. Find and graph the osculating circle of the parabola $y=x^2$ at the origin.

```
> assume(x,real):
> r:=[x,x^2];
```

$$r := \left[x\sim, x\sim^2\right] \tag{4.1.11}$$

```
> dr:=diff(r,x);
```

$$dr := [1, 2\,x\sim] \tag{4.1.12}$$

```
> ds:=sqrt(innerprod(dr,dr));
```

$$ds := \sqrt{1+4\,x\sim^2} \tag{4.1.13}$$

```
> T:=[dr[1]/ds,dr[2]/ds];
```

$$T := \left[\frac{1}{\sqrt{1+4\,x\sim^2}}, \frac{2\,x\sim}{\sqrt{1+4\,x\sim^2}}\right] \tag{4.1.14}$$

```
> dT:=diff(T,x);
```

$$dT := \left[-\frac{4\,x\sim}{\left(1+4\,x\sim^2\right)^{(3/2)}}, \frac{2}{\sqrt{1+4\,x\sim^2}} - \frac{8\,x\sim^2}{\left(1+4\,x\sim^2\right)^{(3/2)}}\right] \tag{4.1.15}$$

```
> NormdT:=simplify(sqrt(innerprod(dT,dT)));
```

$$NormdT := \frac{2}{1+4\,x\sim^2} \tag{4.1.16}$$

```
> Kap:=eval(NormdT/ds, x=0);
```

$$Kap := 2 \tag{4.1.17}$$

```
> N:=eval(dT/NormdT, x=0);
```

$$N := [0, 1] \tag{4.1.18}$$

```
> C:=N/Kap;
```

$$C := \left[0, \frac{1}{2}\right] \tag{4.1.19}$$

Hence, the osculating circle is $x^2 + \left(y - \frac{1}{2}\right)^2 = \left(\frac{1}{2}\right)^2$. To draw its graph, we do the following.

```
> implicitplot([x^2 + (y-1/2)^2 = (1/2)^2, y=x^2], x=-1..1, y=0.
  .2, color=[red,blue], scaling=constrained);
```



14.4.2 Calculate curvature using the vector formula

Example 4. Let the vector function be $\mathbf{r}(t) = t\mathbf{i} + 4\sin(t)\mathbf{j} + 4\cos(t)\mathbf{k}$. Find the curvature function of $\mathbf{r}(t)$ by (a) the formula $\kappa = \frac{1}{\|\mathbf{v}\|}\left\|\frac{d\mathbf{T}}{dt}\right\|$ and (b) the vector formula $\kappa = \frac{\|\mathbf{r}' \times \mathbf{r}''\|}{\|\mathbf{r}'\|^3}$.

(a) Use the formula $\kappa = \frac{1}{\|\mathbf{v}\|}\|\frac{d\mathbf{T}}{dt}\|$

```
> r:=[t, 4*sin(t), 4*cos(t)];
```
$$r := [t\sim, 4\sin(t\sim), 4\cos(t\sim)] \quad (4.2.1)$$

```
> v:=diff(r,t);
```
$$v := [1, 4\cos(t\sim), -4\sin(t\sim)] \quad (4.2.2)$$

```
> ds:= sqrt(simplify(innerprod(v,v)));
```
$$ds := \sqrt{17} \quad (4.2.3)$$

```
> T:=[v[1]/ds,v[2]/ds,v[3]/ds];
```
$$T := \left[\frac{1}{17}\sqrt{17}, \frac{4}{17}\cos(t\sim)\sqrt{17}, -\frac{4}{17}\sin(t\sim)\sqrt{17}\right] \quad (4.2.4)$$

```
> dT:=diff(T,t);
```
$$dT := \left[0, -\frac{4}{17}\sin(t\sim)\sqrt{17}, -\frac{4}{17}\cos(t\sim)\sqrt{17}\right] \quad (4.2.5)$$

```
> kap:=sqrt(simplify(innerprod(dT,dT)))/ds;
```
$$kap := \frac{4}{17} \quad (4.2.6)$$

(b) Use the vector formula $\kappa = \frac{\|\mathbf{r}' \times \mathbf{r}''\|}{\|\mathbf{r}'\|^3}$.

```
> r:=[t, 4*sin(t), 4*cos(t)];
```
$$r := [t\sim, 4\sin(t\sim), 4\cos(t\sim)] \quad (4.2.7)$$

```
> v:=diff(r,t);
```
$$v := [1, 4\cos(t\sim), -4\sin(t\sim)] \quad (4.2.8)$$

```
> ds:= sqrt(simplify(innerprod(v,v)));
```
$$ds := \sqrt{17} \quad (4.2.9)$$

```
> a:=diff(v,t);
```
$$a := [0, -4\sin(t\sim), -4\cos(t\sim)] \quad (4.2.10)$$

```
> vca:=simplify(crossprod(a,v));
```
$$vca := [16 \quad -4\cos(t\sim) \quad 4\sin(t\sim)] \quad (4.2.11)$$

```
> kap:=sqrt(simplify(innerprod(vca, vca)))/ds^3;
```
$$kap := \frac{4}{17} \quad (4.2.12)$$

Example 5. Let the vector function be $\mathbf{r}(t) = (t - \sin(t)\mathbf{i} + (1 - \cos(t))\mathbf{j} + \sqrt{-t}\,\mathbf{k}$. Find the curvature function of $\mathbf{r}(t)$ at $t = -3\pi$ by (a) the formula $\kappa = \frac{1}{\|\mathbf{v}\|}\|\frac{d\mathbf{T}}{dt}\|$, (b) the vector formula $\kappa = \frac{\|\mathbf{r}' \times \mathbf{r}''\|}{\|\mathbf{r}'\|3}$.

(a) Use the formula $\kappa = \frac{1}{\|\mathbf{v}\|}\|\frac{d\mathbf{T}}{dt}\|$.

```
> r:=[t-sin(t), 1-cos(t), sqrt(-t)];
```
$$r := \left[t\sim - \sin(t\sim), 1 - \cos(t\sim), \sqrt{-t\sim}\right] \quad (4.2.13)$$

```
> v:=diff(r,t);
```

$$v := \left[1 - \cos(t\sim), \sin(t\sim), -\frac{1}{2}\,\frac{1}{\sqrt{-t\sim}}\right] \tag{4.2.14}$$

```
> ds:= sqrt(simplify(innerprod(v,v)));
```

$$ds := \frac{1}{2}\sqrt{-\frac{-8\,t\sim + 8\,t\sim\cos(t\sim) + 1}{t\sim}} \tag{4.2.15}$$

```
> T:=[v[1]/ds,v[2]/ds,v[3]/ds];
```

$$T := \left[\frac{2\,(1 - \cos(t\sim))}{\sqrt{-\frac{-8\,t\sim + 8\,t\sim\cos(t\sim) + 1}{t\sim}}}, \frac{2\sin(t\sim)}{\sqrt{-\frac{-8\,t\sim + 8\,t\sim\cos(t\sim) + 1}{t\sim}}}, -\frac{1}{\sqrt{-t\sim}\,\sqrt{-\frac{-8\,t\sim + 8\,t\sim\cos(t\sim) + 1}{t\sim}}}\right] \tag{4.2.16}$$

```
> dT:=diff(T,t);
```

$$dT := \left[\frac{2\sin(t\sim)}{\sqrt{-\frac{-8\,t\sim + 8\,t\sim\cos(t\sim) + 1}{t\sim}}} - \frac{(1 - \cos(t\sim))\left(-\frac{-8 + 8\cos(t\sim) - 8\,t\sim\sin(t\sim)}{t\sim} + \frac{-8\,t\sim + 8\,t\sim\cos(t\sim) + 1}{t\sim^2}\right)}{\left(-\frac{-8\,t\sim + 8\,t\sim\cos(t\sim) + 1}{t\sim}\right)^{(3/2)}},\right.$$
$$\frac{2\cos(t\sim)}{\sqrt{-\frac{-8\,t\sim + 8\,t\sim\cos(t\sim) + 1}{t\sim}}} - \frac{\sin(t\sim)\left(-\frac{-8 + 8\cos(t\sim) - 8\,t\sim\sin(t\sim)}{t\sim} + \frac{-8\,t\sim + 8\,t\sim\cos(t\sim) + 1}{t\sim^2}\right)}{\left(-\frac{-8\,t\sim + 8\,t\sim\cos(t\sim) + 1}{t\sim}\right)^{(3/2)}}, -\frac{1}{2}$$
$$\left.\frac{1}{(-t\sim)^{(3/2)}\sqrt{-\frac{-8\,t\sim + 8\,t\sim\cos(t\sim) + 1}{t\sim}}} + \frac{1}{2}\,\frac{-\frac{-8 + 8\cos(t\sim) - 8\,t\sim\sin(t\sim)}{t\sim} + \frac{-8\,t\sim + 8\,t\sim\cos(t\sim) + 1}{t\sim^2}}{\sqrt{-t\sim}\left(-\frac{-8\,t\sim + 8\,t\sim\cos(t\sim) + 1}{t\sim}\right)^{(3/2)}}\right] \tag{4.2.17}$$

```
> kap:=eval(sqrt(simplify(innerprod(dT,dT)))/ds, t=-3*Pi);
```

$$kap := \frac{2\sqrt{2}\sqrt{-\frac{1}{3}\,\frac{-864\,\pi^3 - 18\,\pi^2 - 2}{\left(2304\,\pi^2 + 96\,\pi + 1\right)\pi}}\,\sqrt{3}}{\sqrt{\frac{48\,\pi + 1}{\pi}}} \tag{4.2.18}$$

```
> kap:=simplify(kap);
```

$$kap := \frac{4\sqrt{432\,\pi^3+9\,\pi^2+1}}{(48\,\pi+1)^{(3/2)}} \tag{4.2.19}$$

```
> evalf(kap);
```

$$0.2483622670 \tag{4.2.20}$$

(b) Use the vector formula $\kappa = \dfrac{\|\mathbf{r}'\times\mathbf{r}''\|}{\|\mathbf{r}'\|^3}$.

```
> r:=[t-sin(t), 1-cos(t), sqrt(-t)];
```

$$r := \left[t\sim-\sin(t\sim),\,1-\cos(t\sim),\,\sqrt{-t\sim}\right] \tag{4.2.21}$$

```
> v:=diff(r,t);
```

$$v := \left[1-\cos(t\sim),\,\sin(t\sim),\,-\frac{1}{2}\,\frac{1}{\sqrt{-t\sim}}\right] \tag{4.2.22}$$

```
> a:=diff(v,t);
```

$$a := \left[\sin(t\sim),\,\cos(t\sim),\,-\frac{1}{4}\,\frac{1}{(-t\sim)^{(3/2)}}\right] \tag{4.2.23}$$

```
> vt:=eval(v,t=-3*Pi);
```

$$vt := \left[2,\,0,\,-\frac{1}{6}\,\frac{\sqrt{3}}{\sqrt{\pi}}\right] \tag{4.2.24}$$

```
> at:=eval(a,t=-3*Pi);
```

$$at := \left[0,\,-1,\,-\frac{1}{36}\,\frac{\sqrt{3}}{\pi^{(3/2)}}\right] \tag{4.2.25}$$

```
> acvt:=crossprod(vt, at);
```

$$acvt := \left[-\frac{1}{6}\,\frac{\sqrt{3}}{\sqrt{\pi}}\quad \frac{1}{18}\,\frac{\sqrt{3}}{\pi^{(3/2)}}\quad -2\right] \tag{4.2.26}$$

```
> kap:=simplify(norm(acvt,2)/norm(vt,2)^3);
```

$$kap := \frac{4\sqrt{432\,\pi^3+9\,\pi^2+1}}{(48\,\pi+1)^{(3/2)}} \tag{4.2.27}$$

```
> kap:=evalf(kap);
```

$$kap := 0.2483622670 \tag{4.2.28}$$

Example 6. Find the curvature of $f(x) = x^3 - 3x^2 + 4$ at $x = 0, 1, 2, 3$.

```
> f:= x^3-3*x^2+4;
```

$$f := x^3 - 3\,x^2 + 4 \tag{4.2.29}$$

```
> kx:=abs(diff(f,x,x))/(1+diff(f,x)^2)^(3/2);
```

$$kx := \frac{|6\,x-6|}{\left(1+\left(3\,x^2-6\,x\right)^2\right)^{(3/2)}} \tag{4.2.30}$$

```
> for i from 1 to 4 do eval(kx, x=i-1) end do;
```

$$6$$
$$0$$
$$6$$
$$\frac{3}{1681}\sqrt{82} \quad (4.2.31)$$

Exercises

1. Find the **T** and **N** of **r**(t) = [sin(t), 3t, cos(t)].
2. Find the curvature function of $\mathbf{r}(t) = \cos(2t)\mathbf{i} + 2\sin(2t)\mathbf{j} + 2t\mathbf{k}$ and then evaluate it at the point (0, 0, 0).
3. Find the curvature function of $\mathbf{r}(t) = e^t\cos(t)\mathbf{i} + e^t\sin(t)\mathbf{j} + 2\mathbf{k}$.
4. Find the equation of the osculating circle of the curve $\mathbf{r}(t) = \sqrt{4-t^2}\,\mathbf{i} + t\mathbf{j}$, $-2 \le t \le 2$, at the origin and then plot its graph together with the curve.
5. Find a parametric equation of the osculating circle of the curve $\mathbf{r}(t) = (t - \sin(t))\mathbf{i} + (1 - \cos(t))\mathbf{j}$ at $t = \pi$.
6. Let the vector function be $\mathbf{r}(t) = 5t\mathbf{i} + t\sin(2t)\mathbf{j} + t\cos(2t)\mathbf{k}$. Find the curvature function of **r**(*t*) by
   (a) the formula $\kappa = \frac{1}{\|\mathbf{v}\|}\left\|\frac{d\mathbf{T}}{dt}\right\|$;
   (b) the vector formula $\frac{\|\mathbf{v}\times\mathbf{a}\|}{\|\mathbf{v}\|^3}$.
7. Let the vector function be $\mathbf{r}(t) = (3t - t^2)\mathbf{i} + (3t^2)\mathbf{j} + (3t + t^3)\mathbf{k}$. Find the curvature of **r**(*t*) at $t = 1$ by (a) the formula $\kappa = \frac{1}{\|\mathbf{v}\|}\left\|\frac{d\mathbf{T}}{dt}\right\|$; (b) the vector formula $\frac{\|\mathbf{v}\times\mathbf{a}\|}{\|\mathbf{v}\|^3}$.

14.5 Motion in Three-Space

(1) Let **r**(*t*) = [x(t), y(t), z(t)] be the path of the motion of a particle traveling along. Then the **velocity** vector is **v**(*t*) = **r**′(*t*), the speed is *v*(*t*) = ‖**v**(*t*)‖, and the **acceleration** is **a**(*t*) = **r**″(*t*).
(2) Newton's Second Law:

$$\mathbf{F}(\mathbf{r}(t)) = m\mathbf{a}(t) = m\mathbf{r}''(t)$$

(3) Tangential and normal components of **acceleration**:

$$\mathbf{a} = a_T\mathbf{T} + a_N\mathbf{N}$$

where

$$a_T = \frac{d}{dt}\|\mathbf{v}\| = \frac{\|\mathbf{a}\times\mathbf{v}\|}{\|\mathbf{v}\|},\quad a_N = \kappa\|\mathbf{v}\|^2 = \sqrt{\|\mathbf{a}\|^2 - a_T^2}$$

14.5.1 Velocity, speed, and acceleration

Example 1. Find the velocity, acceleration, and speed for $\mathbf{r}(t) = \sin(t)\mathbf{i} + t\mathbf{j} + \cos(t)\mathbf{k}$, and evaluate them at $t = 0$.

```
> assume(t, real);
> r:=[sin(t),t,cos(t)];
```

$$r := [\sin(t\sim), t\sim, \cos(t\sim)]$$

The velocity is

```
> v:=diff(r,t);
```

$$v := [\cos(t\sim), 1, -\sin(t\sim)]$$

The acceleration is

```
> a:=diff(v,t);
```

$$a := [-\sin(t\sim), 0, -\cos(t\sim)]$$

The speed is

```
> spd:= simplify(norm(v, 2));
```

$$spd := \sqrt{2} \quad (5.1.1)$$

The velocity at $t = 0$ is

```
> vat0:=eval(v,t=0);
```

$$vat0 := [1, 1, 0]$$

The acceleration at $t = 0$ is

```
> aat0:=eval(a,t=0);
```

$$aat0 = [0, 0, -1]$$

The speed at $t = 0$ is

```
> spdat0=norm(vat0,2);
```

$$spdat0 = \sqrt{2}$$

Example2. The acceleration $\mathbf{a}(t) = 10\mathbf{k}$, $\mathbf{v}(0) = \mathbf{i} + \mathbf{j} - \mathbf{k}$, and $\mathbf{r}(0) = 2\mathbf{i} + 3\mathbf{j}$. Find the velocity and position vectors.

```
> a:=[0,0,10];
```

$$a := [0, 0, 10]$$

```
> v:=map(int,a,s=0..t)+[1,1,-1];
```

$$v := [1, 1, -1 + 10\, t\sim]$$

```
> r:=map(int,v,t=0..t)+[2,3,0];
```

$$r := \left[2 + t\sim, 3 + t\sim, -t\sim + 5\, t\sim^2\right]$$

14.5.2 Newton's Second Law

Example 3. A billet is fired from the grand at an angle of 60 $^\circ$ above the horizontal. What initial speed v_0 must be bullet have in order to hit a point 500 ft high on a tower located 800 ft away?

Step 1. Use Newton's Second Law.

```
> g:=32: a:=[0, -g];
```

$$a := [0, -32] \quad (5.2.1)$$

```
> v:=map(int, a, t=0..t)+[v0*cos(60*Pi/180),v0*sin(60*Pi/180)];
```

$$v := \left[\frac{1}{2}\, v0, \frac{1}{2}\, v0\sqrt{3} - 32\, t\right] \quad (5.2.2)$$

```
> r:= map(int, v, t=0..t);
```

$$r := \left[\frac{1}{2}\, v0\, t, \frac{1}{2}\, v0\sqrt{3}\, t - 16\, t^2\right] \quad (5.2.3)$$

```
> solve({r[1]=800, r[2]=500}, {t, v0});
```

$$\left\{t=\frac{1}{2}\,\mathrm{RootOf}\left(-200\sqrt{3}+\_Z^2+125,\,label=\_L15\right),\ v0=\frac{128}{167}\,\mathrm{RootOf}\left(-200\sqrt{3}+\_Z^2+125,\,label=\_L15\right)\left(5+8\sqrt{3}\right)\right\} \tag{5.2.4}$$

```
> v0:=128/167*sqrt(200*sqrt(3)-125)*(5+8*sqrt(3)); t:=1/2*sqrt
  (200*sqrt(3)-125);
```

$$v0:=\frac{640}{167}\sqrt{-5+8\sqrt{3}}\left(5+8\sqrt{3}\right)$$

$$t:=\frac{5}{2}\sqrt{-5+8\sqrt{3}} \tag{5.2.5}$$

```
> v0:=evalf(v0); t:=evalf(t);
```

$$v0:=215.0558218$$

$$t:=7.439928788 \tag{5.2.6}$$

Hence, the initial speed v_0 must be 215 ft/s. After about 7.44 seconds the bullet hits a point 500 ft high on a tower located 800 ft away.

```
> g:='g': a:='a': v0:='v0': v:='v': t:='t':
```

14.5.3 Tangential and normal components of an acceleration vector

Example 4. Find the tangential and normal components of the acceleration vector of $\mathbf{r}(t) = t\mathbf{i} + 4\sin(t)\mathbf{j} + 4\cos(t)\mathbf{k}$.

Method 1. Use the formula $a_T = \frac{d}{dt}\|\mathbf{v}\|,\ a_N = \kappa\|\mathbf{v}\|^2$.

```
> r:=[t,4*sin(t),4*cos(t)];
```

$$r:=\left[t\sim,\ 4\sin(t\sim),\ 4\cos(t\sim)\right]$$

```
> v:=diff(r,t);
```

$$v:=\left[1,\ 4\cos(t\sim),\ -4\sin(t\sim)\right]$$

```
> sp:=sqrt(simplify(innerprod(v,v)));
```

$$sp:=\sqrt{17}$$

```
> aT:=diff(sp,t);
```

$$aT:=0$$

```
> T:=map(x->x/sp,v);
```

$$T:=\left[\frac{1}{17}\sqrt{17},\ \frac{4}{17}\cos(t\sim)\sqrt{17},\ -\frac{4}{17}\sin(t\sim)\sqrt{17}\right]$$

```
> dT:=diff(T,t);
```

$$dT:=\left[0,\ -\frac{4}{17}\sin(t\sim)\sqrt{17},\ -\frac{4}{17}\cos(t\sim)\sqrt{17}\right] \tag{5.3.1}$$

```
> NormdT:=simplify(norm(dT,2));
```

$$NormdT:=\frac{4}{17}\sqrt{17}$$

```
> kap:=NormdT/sp;
```

$$kap := \frac{4}{17}$$

```
> aN:=kap*sp^2;
```

$$aN := 4$$

Method 2. Use the formula $a_T = \frac{d}{dt}||\mathbf{v}||,\ a_N = \sqrt{|\boldsymbol{a}|^2 - a_T^2}$.
a_T can be computed in the same way as in **Method 1**. We now calculate a_N.

```
> a:=diff(v,t);
```

$$a := [0, -4\sin(t\sim), -4\cos(t\sim)] \tag{5.3.2}$$

```
> aN:=sqrt(simplify(innerprod(a,a))-aT^2);
```

$$aN := 4 \tag{5.3.3}$$

Example 5. Decompose the acceleration vector **a** of **r**(t) = $\left[t^2, 2t,\ \ln(t)\right]$ into tangential and normal components at t =1/2.

```
> r:=[t^2, 2*t, ln(t)];
```

$$r := \left[t^2, 2\,t, \ln(t)\right] \tag{5.3.4}$$

```
> v:=diff(r,t);
```

$$v := \left[2\,t, 2, \frac{1}{t}\right] \tag{5.3.5}$$

```
> a:=diff(v,t);
```

$$a := \left[2, 0, -\frac{1}{t^2}\right] \tag{5.3.6}$$

```
> vP:=eval(v, t=1/2); aP:=eval(a, t=1/2);
```

$$vP := [1, 2, 2]$$
$$aP := [2, 0, -4] \tag{5.3.7}$$

```
> spdP:=sqrt(innerprod(vP,vP));
```

$$spdP := 3 \tag{5.3.8}$$

```
> TP:=vP/spdP;
```

$$TP := \left[\frac{1}{3}, \frac{2}{3}, \frac{2}{3}\right] \tag{5.3.9}$$

```
> aTP:=innerprod(aP,TP);
```

$$aTP := -2 \tag{5.3.10}$$

```
> aNP:=sqrt((innerprod(aP,aP))-aTP^2);
```

$$aNP := 4 \tag{5.3.11}$$

```
> NP:=(aP-aTP*TP)/aNP;
```

$$NP := \left[\frac{2}{3}, \frac{1}{3}, \frac{-2}{3}\right] \tag{5.3.12}$$

Hence, $\mathbf{a}(1/2) = -2\cdot\left[\frac{1}{3}, \frac{2}{3}, \frac{2}{3}\right] + 4\cdot\left[\frac{2}{3}, \frac{1}{3}, \frac{-2}{3}\right]$.

Exercises

1. Find the unit tangential vector and unit normal vector of $\mathbf{r}(t) = [\cos(t) + t\sin(t), \sin(t) - t\cos(t), 3]$.
2. Find the unit tangential vector and unit normal vector of $\mathbf{r}(t) = \left[t^2, t + \frac{t^3}{3}, t - \frac{t^3}{3}\right]$ at $t = 0$.
3. Find equations of the normal plane, osculating plane and rectifying plane of $\mathbf{r}(t) = \cos(t)\mathbf{i} + \sin(t)\mathbf{j} + t\mathbf{k}$ at $t = 0$.
4. Find the velocity, acceleration, and speed for $\mathbf{r}(\mathrm{t}) = t\sin(t)\mathbf{i} + t^2\mathbf{j} + t\cos(t)\mathbf{k}$, and evaluate them at $t = 0$.
5. The acceleration vector is $\mathbf{a}(t) = 3\mathbf{j} - 2t\mathbf{k}$. Assume $\mathbf{v}(0) = 2\mathbf{i} + \mathbf{j} - 2\mathbf{k}$, $\mathbf{r}(0) = 4\mathbf{i} + 5\mathbf{j} - 3\mathbf{k}$. Find the velocity and position vectors.
6. Find the tangential and normal components of the acceleration vector of $\mathbf{r}(t) = t\mathbf{i} + t\sin(t)\mathbf{j} + t\cos(t)\mathbf{k}$.
7. Find the decomposition of $\mathbf{a}(t)$ into tangential and normal components at $t = 1$ of $\mathbf{r}(t) = (4 - t)\mathbf{i} + (t + 1)\mathbf{j} + t^2\mathbf{k}$.

Chapter 15 DIFFERENTIATION IN SEVERAL VARIABLES

15.1 Functions of Two or More Variables

15.1.1 Evaluate multivariate functions

Example 1. Find the value of $f(x, y) = x^2 y^3$ at $x = 3, y = 2$.

```
> f:=x^2*y^3;
```

$$f := x^2 y^3 \quad (1.1.1)$$

```
> eval(f,[x=3,y=2]);
```

$$72 \quad (1.1.2)$$

The following is another way to evaluate it.

```
> f:=(x,y)->x^2*y^3;
```

$$f := (x, y) \rightarrow x^2 y^3 \quad (1.1.3)$$

```
> f(3,2);
```

$$72 \quad (1.1.4)$$

Use "evalf" to get the decimal result.

```
> f(Pi,2*Pi);
```

$$8\,\pi^5 \quad (1.1.5)$$

```
> evalf(%);
```

$$2448.157480 \quad (1.1.6)$$

Example 2. Evaluate the value of $f(x, y, z) = \sqrt{x^2+y^2+z^2}$ at the point (3, 0, 4).

```
> f:=(x,y,z)->sqrt(x^2+y^2+z^2);
```

$$f := (x, y, z) \rightarrow \sqrt{x^2 + y^2 + z^2} \quad (1.1.7)$$

```
> f(3,0,4);
```

$$5 \quad (1.1.8)$$

15.1.2 Domains, level curves, and graphs of bivariate functions

Example 3. Find the domain of the function $f(x, y) = \sqrt{y - x}$, then draw its graph and its level curves, respectively.

```
> f:=(x,y)->sqrt(y-x);
```

$$f := (x, y) \rightarrow \sqrt{y - x} \quad (1.2.1)$$

The domain is $y - x \geq 0$.

```
> with(plots):
```

Warning, the name changecoords has been redefined

```
> plot3d(f(x,y), x=0..5, y=0..5, view=0..3, axes=normal);
```




```
> contourplot(f(x,y), x=-5..5,y=-5..5,contours=6,title=`level
  curve of f(x,y)=sqrt(y-x)`,axes=box);
```



Example 4. Plot the graph of $f(x, y) = 4x^2 + y^2$ and its level curves.

```
> f:=4*x^2+y^2;
```

$$f := 4x^2 + y^2 \quad (1.2.2)$$

```
> plot3d(f,x=-3..3,y=-3..3,view=0..9,title=`f(x,y)=4*x^2+y^2`,
  axes=normal);
```




```
> plots[contourplot](f,x=-3..3,y=-3..3,contours=15,title=`level
  curve of f(x,y)=4x^2+y^2`,axes=box);
```



Example 5. Sketch the level curves of $f(x, y) = \sqrt{9 - x^2 - y^2}$ for $k = 1, \sqrt{3}, \sqrt{5}, \sqrt{8}$.

```
> contourplot(sqrt(9-x^2-y^2),x=-3..3,y=-3..3,contours=[1,sqrt
  (3),sqrt(5),sqrt(8)], grid=[40,40],axes=normal, scaling=
  constrained);
```



15.1.3 Implicit surfaces and parameterized surfaces

Example 6. Plot the surface described by $4\ln\left(x^2+y^2+z^2\right)=1$.

```
> implicitplot3d(4*ln(x^2+y^2+z^2)=1, x=-2..2, y=-2..2, z=-2..2,
  axes=boxed, scaling=constrained);
```



Example 7. Plot the surface given by the parametric equations:
$x = u\cos(v),\ y = u\sin(v),\ z = u,\ 0 \le u \le 2,\ 0 \le v \le 2\pi.$

```
> plot3d([u*cos(v),u*sin(v),u],u=0..2,v=0..2*Pi,axes=frame);
```



Exercises

1. Find the value of $f(x, y) = xye^{\frac{y}{x}}$ at $x = 3, y = 2$.
2. (a) Draw the level curves of $f(x, y) = y^2 - 2x^2$ using the default number. (b) Draw the graph of the function.
3. Sketch the level curves of $f(x, y) = \sqrt{36 - 4x^2 - y^2}$ for $k = 1, 2, 3, 4, 5$.
4. Plot the level surface described by $\ln\left(x^2 + y^2 - z^2\right) = 1$.
5. Plot the surface given by the parametric equations:
   $x = u\cos(v),\ y = u\sin(v),\ z = u^2,\ 0 \le u \le 2, 0 \le v \le 2.$

15.2 Limits and Continuity in Several Variables

15.2.1 Limits

Find limits of a function at a point where it is continuous. MAPLE cannot find limits for multivariable functions. You have to determine if the limit exists and what is the value if it exists. When the function is continuous at a point, then the limit of the function at the point is the function value. Therefore, finding the limit at the point becomes evaluating the function at that point.

Example 1. Find the limit of $f(x, y) = e^{(x+2y)}$ at $(x, y) = (1, 4)$.
Since the function $f(x, y) = e^{(x+2y)}$ is continuous at the point, the limit is equal to the function value at the point.

```
> f:=exp(x+2*y);
```

$$f := e^{(x+2y)} \qquad (2.1.1)$$

```
> lmt:=eval(f,{x=1,y=4});
```

$$lmt := e^9 \tag{2.1.2}$$

It can be also obtained in the following way.

```
> lmt:=subs({x=1,y=4},exp(x+2*y));
```

$$lmt := e^9 \tag{2.1.3}$$

Sometimes you can apply the substitution method to reduce a limit of two variables to one variable. A useful substitution is $x = r\cos(t)$, $y = r\sin(t)$ for finding the limit as $(x, y) \to (0, 0)$. Then $(x, y) \to (0, 0)$ is equivalent to $r \to 0$.

Example 2. Find the limit of $\dfrac{(x^2+y^2)}{\sqrt{x^2+y^2+1}-1}$ at $(x, y) = (0, 0)$.

Hint: Use the substitution method: $x = r\cos(t)$, $y = r\sin(t)$.

```
> f:=(x^2+y^2)/(sqrt(x^2+y^2+1)-1);
```

$$f := \frac{x^2+y^2}{\sqrt{x^2+y^2+1}-1} \tag{2.1.4}$$

```
> assume(t>0):
> newf:=subs({x=r*cos(t), y=r*sin(t)},f);
```

$$newf := \frac{r^2\cos(t\sim)^2 + r^2\sin(t\sim)^2}{\sqrt{r^2\cos(t\sim)^2 + r^2\sin(t\sim)^2+1}-1} \tag{2.1.5}$$

```
> newf:=simplify(newf);
```

$$newf := \frac{r^2}{\sqrt{1+r^2}-1} \tag{2.1.6}$$

```
> limit(newf,r=0);
```

$$2 \tag{2.1.7}$$

Example 3. Find the limit or show that it does not exist: Limit of $\dfrac{(x+y)^2}{x^2+y^2}$ at $(x, y) = (0, 0)$.

```
> f:=(x+y)^2/(x^2+y^2);
```

$$f := \frac{(x+y)^2}{x^2+y^2} \tag{2.1.8}$$

```
> limit(subs(y=x,f),x=0);
```

$$2 \tag{2.1.9}$$

```
> limit(subs(y=2*x,f),x=0);
```

$$\frac{9}{5} \tag{2.1.10}$$

Hence, the limit does not exist.

Example 4. Discuss the limit of $f(x, y) = \dfrac{xy}{x^2+y^2}$ at $(x, y) = (0, 0)$.

We find the limits of $f(x, y)$ at $(x, y) = (0, 0)$ along the different directions:

```
> f:=(x,y)->x*y/(x^2+y^2);
```

$$f := (x, y) \to \frac{x\,y}{x^2 + y^2} \tag{2.1.11}$$

Its limit along the horizontal direction is

```
> limit(f(x,0), x=0);
```

$$0 \tag{2.1.12}$$

Its limit along the $y = x$ direction is

```
> limit(f(x,x), x=0);
```

$$\frac{1}{2} \tag{2.1.13}$$

In general, the limit along with the line $y = kx$ is

```
> assume(k>0):
> limit(f(x, k*x), x=0);
```

$$\frac{k\sim}{1 + k\sim^2} \tag{2.1.14}$$

Since its limits along the different directions have different values, its (general) limit does not exists.

Example 5. Discuss the limit of $f(x, y) = \dfrac{xy^2}{x^2+y^4}$ at $(x, y) = (0, 0)$. Find two different paths to show that the limits of the function along these two paths are different. Therefore its (general) limit does not exists.

```
> f:=(x,y)->x*y^2/(x^2+y^4);
```

$$f := (x, y) \to \frac{x\,y^2}{x^2 + y^4} \tag{2.1.15}$$

Its limit along with all directional lines are the same:

```
> limit(f(x, k*x), x=0);
```

$$0 \tag{2.1.16}$$

However, if we select the path $y = \sqrt{x}$, then the limit will be different from those along the directional lines:

```
> limit(f(x, sqrt(x)), x=0);
```

$$\frac{1}{2} \tag{2.1.17}$$

Therefore $\displaystyle\lim_{(x, y) \to (0, 0)} \frac{xy^2}{x^2+y^4}$ does not exist.

15.2.2 Continuity

Each elementary function is continuous on its domain. Hence, we only concentrate our discussion of the continuity at the points where a piecewise function joins.

Example 6. Show that the piecewise function $f(x, y)$ defined by

$f(x,y) = \dfrac{2xy}{x^2+y^2}$, $(x, y) \neq (0, 0)$, and $f(x, y) = 0$, $(x, y) = (0, 0)$,

is discontinuous at $(0, 0)$.

```
> f:=(x,y)->2*x*y/(x^2+y^2);
```

$$f := (x, y) \rightarrow \frac{2\,x\,y}{x^2+y^2} \qquad (2.2.1)$$

```
> limit(f(x, k*x), x=0);
```

$$\frac{2\,k\sim}{1+k\sim^2} \qquad (2.2.2)$$

Since its limits along the different directions have different values, its (general) limit does not exist, that is, the function is not continuous at $(x, y) = (0, 0)$.

Example 7. At what point (x, y, z) is the function $f(x, y, z) = \ln(xyz)$ continuous?
Since $f(x, y, z) = \ln(xyz)$ is an elementary function, it is continuous on its domain: $xyz > 0$.

Exercises

1. Find the limit of $\dfrac{\ln\left(1-x^2-y^2\right)}{\sqrt{x^2+y^2+4}-2}$ at $(x, y) = (0, 0)$.
2. Find the limit of $\dfrac{(x+y-2)}{x^2+y^2-1}$ at $(x, y) = (0, 0)$.
3. Show that the limit of $f(x, y) = \dfrac{x^4}{x^4+y^2}$ at $(x, y) = (0, 0)$ does not exists.
4. Show that $f(x, y) = \dfrac{y}{\sqrt{x^2+y^2}}$, $(x, y) \neq (0, 0)$, and $f(x, y) = 0$, $(x, y) = (0, 0)$, is discontinuous at $(0, 0)$.
5. Show that $f(x, y)$ is continuous at $(0, 0)$: $f(x, y) = \dfrac{xy^2}{x^2+y^2}$, $(x, y) \neq (0, 0)$ and $f(x, y) = 0$, $(x, y) = (0, 0)$.

15.3 Partial Derivatives

(1) The partial derivative of $f(x, y)$ with respect to x at the point (x_0, y_0) is

$$\left.\frac{\partial f}{\partial x}\right|_{(x_0,\ y_0)} = \lim_{h \to 0} \frac{1}{h}\left(f\left(x_0+h, y_0\right) - f(x_0, y_0)\right)$$

(2) The partial derivative of f (x, y) with respect to y at the point (x_0, y_0) is

$$\left.\frac{\partial f}{\partial y}\right|_{(x_0,\ y_0)} = \lim_{h \to 0} \frac{1}{h}\left(f\left(x_0, y_0+h\right) - \mathrm{f}(x_0, y_0)\right)$$

15.3.1 Find partial derivatives

Example 1. Find the partial derivative of $f(x, y) = \sin(xy)e^{x^2+y^2}$.

```
> f:=sin(x*y)*exp(x^2+y^2);
```

$$f := \sin(x\,y)\,e^{(x^2+y^2)} \qquad (3.1.1)$$

```
> dfx:=diff(f,x);
```

$$dfx := \cos(x\,y)\,y\,e^{(x^2+y^2)} + 2\sin(x\,y)\,x\,e^{(x^2+y^2)} \qquad (3.1.2)$$

```
> dfy:=diff(f,y);
```

$$dfy := \cos(x\,y)\,x\,e^{(x^2+y^2)} + 2\sin(x\,y)\,y\,e^{(x^2+y^2)} \qquad (3.1.3)$$

You can also do the following:

```
> f:=(x,y)->sin(x*y)*exp(x^2+y^2);
```

$$f := (x, y) \rightarrow \sin(y\,x)\,e^{(x^2+y^2)} \qquad (3.1.4)$$

```
> dfx:=D[1](f);
```

$$dfx := (x, y) \rightarrow \cos(y\,x)\,y\,e^{(x^2+y^2)} + 2\sin(y\,x)\,x\,e^{(x^2+y^2)} \qquad (3.1.5)$$

```
> dfy:=D[2](f);
```

$$dfy := (x, y) \rightarrow \cos(y\,x)\,x\,e^{(x^2+y^2)} + 2\sin(y\,x)\,y\,e^{(x^2+y^2)} \qquad (3.1.6)$$

You can use "**x$n**" to obtain the n-th partial derivative for x or y when $f(x, y)$ is represented as an expression. If $f(x, y)$ is represented in the form of a pair of input-output, then you can use "**D[1$n](f), D[2$n](f)**" and "**D[1$m,2$k](f)**" to calculate the high partial derivatives for $f(x, y)$.

Example 2. Find all second partial derivatives of the function $f(x, y) = xye^{xy}$.

```
> f:=x*y*exp(x*y);
```

$$f := x\,y\,e^{(x\,y)} \qquad (3.1.7)$$

```
> dfx2:=diff(f,x$2);
```

$$dfx2 := 2\,y^2\,e^{(x\,y)} + x\,y^3\,e^{(x\,y)} \qquad (3.1.8)$$

```
> dfy2:=diff(f,y$2);
```

$$dfy2 := 2\,x^2\,e^{(x\,y)} + x^3\,y\,e^{(x\,y)} \qquad (3.1.9)$$

```
> dfxy:=diff(f,y,x);
```

$$dfxy := e^{(x\,y)} + 3\,x\,y\,e^{(x\,y)} + x^2\,y^2\,e^{(x\,y)} \qquad (3.1.10)$$

You can also do the following:

```
> f:=(x,y)->x*y*exp(x*y);
```

$$f := (x, y) \rightarrow x\,y\,e^{(y\,x)} \qquad (3.1.11)$$

```
> dfx2:=D[1$2](f);
```

$$dfx2 := (x, y) \rightarrow 2\,y^2\,e^{(y\,x)} + x\,y^3\,e^{(y\,x)} \qquad (3.1.12)$$

```
> dy2:=D[2$2](f);
```

$$dy2 := (x, y) \rightarrow 2\,x^2\,e^{(y\,x)} + x^3\,y\,e^{(y\,x)} \qquad (3.1.13)$$

```
> dfxy:=D[1,2](f);
```

$$dfxy := (x, y) \to e^{(y x)} + 3\, x\, y\, e^{(y x)} + x^2\, y^2\, e^{(y x)} \quad (3.1.14)$$

```
>
```

Example 3. Find f_{yxyz} if $f(x, y, z) = 1 - 2xy^2z + x^2y$.

```
> f:=1-2*x*y^2*z+x^2*y;
```

$$f := 1 - 2\, x\, y^2\, z + x^2\, y \quad (3.1.15)$$

```
> d4fyxyz:=diff(f, x, y$2, z);
```

$$d4fyxyz := -4 \quad (3.1.16)$$

The following is another way:

```
> f:=(x,y,z)->1-2*x*y^2*z+x^2*y;
```

$$f := (x, y, z) \to 1 - 2\, x\, y^2\, z + x^2\, y \quad (3.1.17)$$

```
> d4fyxyz:=D[1,2$2,3](f);
```

$$d4fyxyz := -4 \quad (3.1.18)$$

When the partial derivative functions of $f(x, y)$, say $\frac{\mathrm{d}f(x, y)}{\mathrm{d}x}$ and $\frac{\mathrm{d}f(x, y)}{\mathrm{d}y}$, are continuous at a point (x_0, y_0), then you can calculate the partial derivatives at the point (x_0, y_0), say $\left.\frac{\mathrm{d}f}{\mathrm{d}x}\right|_{(x_0, y_0)}$ and $\left.\frac{\mathrm{d}f}{\mathrm{d}y}\right|_{(x_0, y_0)}$, by evaluating the partial derivative functions $\frac{\mathrm{d}f(x, y)}{\mathrm{d}x}$ and $\frac{\mathrm{d}f(x, y)}{\mathrm{d}y}$ at the given point (x_0, y_0).

Example 4. Find $\frac{\partial f}{\partial x}$ and $\frac{\partial f}{\partial y}$ of $f(x, y) = xye^{xy}$ at $(x, y) = (1, -1)$.

```
> f:=x*y*exp(x*y);
```

$$f := x\, y\, e^{(x y)} \quad (3.1.19)$$

```
> dfx:=diff(f,x): dfy:=diff(f,y):
> dfxp:=eval(dfx,{x=1,y=-1});
```

$$dfxp := 0 \quad (3.1.20)$$

```
> dfyp:=eval(dfy,{x=1,y=-1});
```

$$dfyp := 0 \quad (3.1.21)$$

You can also use the following way to evaluate the partial derivatives.

```
> f:=(x,y)->x*y*exp(x*y);
```

$$f := (x, y) \to x\, y\, e^{(y x)} \quad (3.1.22)$$

```
> dfxp:=D[1](f)(1,-1);
```

$$dfxp := 0 \quad (3.1.23)$$

```
> dfyp:=D[2](f)(1,-1);
```

$$dfyp := 0 \quad (3.1.24)$$

15.3.2 Find partial derivatives of implicit functions

If $z = z(x, y)$ is an implicit function defined by the equation $F(x, y, z) = c$, then write the equation as $F(x, y, z(x, y)) = c$ and differentiate the equation. The partial derivatives $\frac{\partial z}{\partial x}$ and $\frac{\partial z}{\partial y}$ can be obtained by solving the equation for them. It therefore yields the following formulas:

$$\frac{\partial z}{\partial x} = -\frac{D_x F}{D_z F}, \quad \frac{\partial z}{\partial y} = -\frac{D_y F}{D_z F}$$

Similarly, if $y = y(x)$ is defned by $F(x, y) = c$, then

$$\frac{dy}{dx} = -\frac{D_x F}{D_y F}$$

Example 5. If the equation $yz - \ln(z) = x + y$, define z as a function of x and y. Find $\frac{\partial z}{\partial x}$ and $\frac{\partial z}{\partial y}$.

Method 1. Solve the equation for the partial derivatives.

```
> Eq:=y*z(x,y)-ln(z(x,y))=x+y;
```

$$Eq := y\,z(x, y) - \ln(z(x, y)) = x + y \tag{3.2.1}$$

```
> deq:= diff(Eq, x);
```

$$deq := y\left(\frac{\partial}{\partial x} z(x, y)\right) - \frac{\frac{\partial}{\partial x} z(x, y)}{z(x, y)} = 1 \tag{3.2.2}$$

```
> dzx:=solve(deq, diff(z(x,y),x));
```

$$dzx := \frac{z(x, y)}{y\,z(x, y) - 1} \tag{3.2.3}$$

```
> deq:=diff(Eq,y);
```

$$deq := z(x, y) + y\left(\frac{\partial}{\partial y} z(x, y)\right) - \frac{\frac{\partial}{\partial y} z(x, y)}{z(x, y)} = 1 \tag{3.2.4}$$

```
> dzy:=solve(deq, diff(z(x,y),y));
```

$$dzy := -\frac{z(x, y)\,(z(x, y) - 1)}{y\,z(x, y) - 1} \tag{3.2.5}$$

Method 2. Apply the formula.

```
> F:= y*z-ln(z)-(x+y);
```

$$F := y\,z - \ln(z) - x - y \tag{3.2.6}$$

```
> dzx:=-diff(F,x)/diff(F,z);
```

$$dzx := \frac{1}{y - \frac{1}{z}} \tag{3.2.7}$$

```
> dzy:=-diff(F,y)/diff(F,z);
```

$$dzy := -\frac{z - 1}{y - \frac{1}{z}} \tag{3.2.8}$$

Example 6. Find the value of $\frac{\partial z}{\partial x}$ at $(1, 1, 1)$ if the equation $xy+z^3x - 2yz=0$ defines z as a function of x and y.

```
> F:=x*y+z^3*x-2*y*z;
```

$$F := x\,y + z^3\,x - 2\,y\,z \quad (3.2.9)$$

```
> dzx:=-diff(F,x)/diff(F,z);
```

$$dzx := -\frac{y+z^3}{3\,z^2\,x - 2\,y} \quad (3.2.10)$$

```
> dzxp:=eval(dzx, {x=1,y=1,z=1});
```

$$dzxp := -2 \quad (3.2.11)$$

Example 7. Find $\frac{dy}{dx}$ for $x^3+y^3=6xy$.

```
> F:=x^3+y^3-6*x*y;
```

$$F := x^3 + y^3 - 6\,x\,y \quad (3.2.12)$$

```
> dyx:=simplify(-diff(F,x)/diff(F,y));
```

$$dyx := \frac{x^2 - 2\,y}{-y^2 + 2\,x} \quad (3.2.13)$$

Example 8. Find $\frac{dz}{dx}$ and $\frac{dz}{dy}$ for $x^3+y^3+z^3+6xyz=1$.

```
> F:=x^3+y^3+z^3+6*x*y*z-1;
```

$$F := x^3 + y^3 + z^3 + 6\,x\,y\,z - 1 \quad (3.2.14)$$

```
> dzx:=-simplify(diff(F,x)/diff(F,z));
```

$$dzx := -\frac{x^2 + 2\,y\,z}{z^2 + 2\,x\,y} \quad (3.2.15)$$

```
> dzy:=-simplify(diff(F,y)/diff(F,z));
```

$$dzy := -\frac{y^2 + 2\,x\,z}{z^2 + 2\,x\,y} \quad (3.2.16)$$

Exercises

1. Find the partial derivative of $f(x, y) = \sin(x+y)e^{-x}$.
2. Find $\frac{\partial f}{\partial x}$ and $\frac{\partial f}{\partial y}$ of $f(x, y) = e^{xy}\ln(y)$ at $(x, y) = (-1, 1)$.
3. Find $\frac{\partial z}{\partial x}$ and $\frac{\partial z}{\partial x}$ for $xyz+zy\ln(x) - x^2y = 0$ at $(1, 1, 1)$.
4. Find $\frac{\partial z}{\partial x}$ if the equation $xyz+z^2x - y^2z = 0$ defines z as a function of x and y.
5. Find all second partial derivatives of the function $f(x, y) = \cos\left(x+y^2\right)$.
6. Find g_{xxyz} for $g(x, y, z) = x^4y^5z^3$.

15.4 Differentiability, Linear Approximation, and Tangent Planes

(1) For a given function $z = f(x, y)$, its tangent plane at (x_0, y_0) is

$$z - z_0 = \frac{\partial}{\partial x} f(x, y) \Bigg|_{(x_0, y_0)} (x - x_0) + \frac{\partial}{\partial y} f(x, y) \Bigg|_{(x_0, y_0)} (y - y_0)$$

(2) The surface is given by an equation $F(x, y, z) = 0$. Then the tangent plane at $P_0(x_0, y_0, z_0)$ is

$$F_x(x - x_0) + F_y(y - y_0) + F_z(z - z_0) = 0$$

The normal line equation in the symmetric form is

$$\frac{x - x_0}{F_x} = \frac{y - y_0}{F_y} = \frac{z - z_0}{F_z}$$

and in the parametric form is

$$x - x_0 = tF_x, \quad y - y_0 = tF_y, \quad z - z_0 = tF_z$$

(3) If the surface is given by a parametric function $x = x(u, v),\ y = y(u, v),\ z = z(u, v)$, its tangent plane at $P_0(x_0, y_0, z_0)$ is given by

$$n_x(x - x_0) + n_y(y - y_0) + n_z(z - z_0) = 0$$

where $n = [n_x, n_y, n_z]$ is the normal vector of the plane, which can be computed by

$$n = \mathrm{d}u \times \mathrm{d}v \text{ with } \mathrm{d}u = [x_u, y_u, z_u],\ \mathrm{d}v = [x_v, y_v, z_v]$$

15.4.1 Tangent planes and normal lines

Example 1. Find the tangent plane of $z = 2x^2 + y^2$ at (1, 1, 3).

Method 1. Apply the formula in (1) above.

```
> f:=(x,y)->2*x^2+y^2:
> d1f:=D[1](f)(1,1); d2f:=D[2](f)(1,1);
```

$$d1f := 4$$
$$d2f := 2 \qquad (4.1.1)$$

```
> PlnEq: z-3=d1f*(x-1)+d2f*(y-1);
```

$$z - 3 = 4x - 6 + 2y \qquad (4.1.2)$$

Method 2. Apply the formula in (2) above.

```
> F:=2*x^2+y^2-z;
```

$$F := 2x^2 + y^2 - z \qquad (4.1.3)$$

```
> with(linalg):
Warning, the protected names norm and trace have been redefined and unprotected
> grf:=grad(F, [x,y,z]);
```

$$grf := [4x \quad 2y \quad -1] \qquad (4.1.4)$$

```
> grfp:=eval(grf, [x=1,y=1]);
```

$$grfp := [4 \quad 2 \quad -1] \qquad (4.1.5)$$

```
> Eq:=innerprod(grfp,[x-1,y-1,z-3])=0;
```
$$Eq := 4\,x - 3 + 2\,y - z = 0 \quad (4.1.6)$$

Example 2. Find the tangent plane of the surface at (1, 4, 5), assuming the surface is given by the parametric equations $x = u^2, y = v^2, z = u + 2v$.

```
> r:=[u^2, v^2, u+2*v];
```
$$r := \left[u^2, v^2, u + 2\,v\right] \quad (4.1.7)$$

```
> solve({r[1]=1,r[2]=4,r[3]=5}, [u,v] );
```
$$[\,[u = 1, v = 2\,]\,] \quad (4.1.8)$$

```
> dru:=diff(r,u);
```
$$dru := [\,2\,u, 0, 1\,] \quad (4.1.9)$$

```
> drv:=diff(r,v);
```
$$drv := [\,0, 2\,v, 2\,] \quad (4.1.10)$$

```
> drup:=eval(dru,{u=1,v=2});
```
$$drup := [\,2, 0, 1\,] \quad (4.1.11)$$

```
> drvp:=eval(drv,{u=1,v=2});
```
$$drvp := [\,0, 4, 2\,] \quad (4.1.12)$$

```
> nvp:=crossprod(drup,drvp);
```
$$nvp := [\,-4 \quad -4 \quad 8\,] \quad (4.1.13)$$

```
> x:='x': y:='y': z:='z':r:='r':
> PlnEq:=-2*(x-1)-4*(y-4)+4*(z-5)=0;
```
$$PlnEq := -2\,x - 2 - 4\,y + 4\,z = 0 \quad (4.1.14)$$

Example 3. Find the tangent plane and normal line of $z+1 = xe^{y}\cos(z)$ at (1, 0, 0).

```
> F:=z+1-x*exp(y)*cos(z);
```
$$F := z + 1 - x\,e^{y}\cos(z) \quad (4.1.15)$$

```
> grd:=grad(F,[x,y,z]);
```
$$grd := \left[\,-e^{y}\cos(z) \quad -x\,e^{y}\cos(z) \quad 1 + x\,e^{y}\sin(z)\,\right] \quad (4.1.16)$$

```
> P:={x=1,y=0,z=0};
```
$$P := \{x = 1, y = 0, z = 0\} \quad (4.1.17)$$

```
> grdP:=map(eval,grd,P);
```
$$grdP := [\,-1 \quad -1 \quad 1\,] \quad (4.1.18)$$

The equation of the tangent plane is

```
> TgntPlh:=innerprod(grdP,[x-1,y,z])=0;
```
$$TgntPlh := -x + 1 - y + z = 0 \quad (4.1.19)$$

The equation of the normal line is $\frac{x-1}{-1} = \frac{y}{-1} = \frac{z}{1}$.

15.4.2 Change in a specific directions

Estimate the change of f in a direction u: To estimate the change in the value of a differentiable function f when we move a small distance ds from a point P_0 in a particular direction **u**, use the formula

$$\mathrm{d}f = \frac{\partial}{\partial \mathbf{u}} f(P_0)\mathrm{d}s = \left(\nabla f(P_0) \cdot \mathbf{u}\right)\mathrm{d}s$$

which is an estimate of the exact change $f(P_0 + \mathbf{u}\mathrm{d}s) - f(P_0)$.

Example 4. Estimate how much the value of $f(x, y) = y\sin(x) + 2yz$ will change if the point $P(x, y, z)$ moves 0.1 unit from (0, 1, 0) straight toward (2, 2, −2).

```
> f:=y*sin(x)+2*y*z;
```

$$f := y\sin(x) + 2\,y\,z \qquad (4.2.1)$$

```
> grf:=grad(f, [x,y,z]);
```

$$grf := [y\cos(x) \quad \sin(x) + 2\,z \quad 2\,y] \qquad (4.2.2)$$

```
> grfp:=map(eval,grf,[x=0,y=1,z=0]);
```

$$grfp := [1 \quad 0 \quad 2\,] \qquad (4.2.3)$$

```
> preu:=[2,2,-2]-[0,1,0];
```

$$preu := [2, 1, -2] \qquad (4.2.4)$$

```
> u:=preu/norm(preu,2); ds:=0.1;
```

$$u := \left[\frac{2}{3}, \frac{1}{3}, \frac{-2}{3}\right]$$

$$ds := 0.1 \qquad (4.2.5)$$

```
> df:=innerprod(grfp,u)*ds;
```

$$df := -0.06666666667 \qquad (4.2.6)$$

15.4.3 Linear approximation

Linear approximation formula:

$$L(x, y) = f(x, y) + \left.\frac{\partial}{\partial x} f(x, y)\right|_{(a,\,b)} (x - a) + \left.\frac{\partial}{\partial y} f(x, y)\right|_{(a,\,b)} (y - b)$$

Total differential: Let $z = f(x, y)$. Assume (x, y) changes from (x_0, y_0) to $(x_0 + \mathrm{d}x, y_0 + \mathrm{d}y)$. Then its change is

$$\Delta z = f(x_0 + \mathrm{d}x, y_0 + \mathrm{d}y) - f(x_0, y_0)$$

and its **total differential** is

$$\mathrm{d}z = \left.\frac{\partial}{\partial x} f(x, y)\right|_{(x_0,\, y_0)} \mathrm{d}x + \left.\frac{\partial}{\partial y} f(x, y)\right|_{(x_0,\, y_0)} \mathrm{d}y$$

When $\mathrm{d}x$ and $\mathrm{d}y$ are small quantities, we can use $\mathrm{d}z$ to estimate Δz.

Example 5. Find the linear approximation of $f(x, y) = x^2 + 3xy - y^2$ for $(x, y) = (2.05, 2.96)$ at $(2, 3)$ and give the error of the approximation.

```
> a:=2;b:=3;
```

$$a := 2$$

$$b := 3 \qquad (4.3.1)$$

```
> f:=(x,y)->x^2+3*x*y-y^2;
```

$$f := (x, y) \rightarrow x^2 + 3\,y\,x - y^2 \qquad (4.3.2)$$

```
> dfxp:=D[1](f)(a,b);
```

$$dfxp := 13 \qquad (4.3.3)$$

```
> dfyp:=D[2](f)(a,b);
```
$$dfyp := 0 \tag{4.3.4}$$

```
> LiAp:=f(a,b)+dfxp*(2.03-a)+dfyp*(2.96-b);
```
$$LiAp := 13.39 \tag{4.3.5}$$

```
> f(2.03,2.96);
```
$$13.3857 \tag{4.3.6}$$

```
> err:=abs(f(2.03,2.96)-LiAp);
```
$$err := 0.0043 \tag{4.3.7}$$

```
> a:='a': b:='b':
```

The following graph shows that the linearization is close to the function about the given point.

```
> LiEq:= f(2,3)+dfxp*(x-2)+dfyp*(y-3);
```
$$LiEq := -13 + 13\,x \tag{4.3.8}$$

```
> plot3d({f(x,y),LiEq},x=1.5..2.5,y=2.5..3.5, axes=frame);
```



Example 6. Let $z = x^2 - xy + 3y^2$. Assume (x, y) changes from $(3, -1)$ to $(2.96, -0.95)$. Compare the values dz and $\triangle z$.

```
> f:=(x,y)->x^2-x*y+3*y^2;
```
$$f := (x, y) \rightarrow x^2 - y\,x + 3\,y^2 \tag{4.3.9}$$

```
> x1:=2.96: y1:=-0.95: x0:=3: y0:=-1:
> dfxp:=D[1](f)(x0,y0);
```
$$dfxp := 7 \tag{4.3.10}$$

```
> dfyp:=D[2](f)(x0,y0);
```
$$dfyp := -9 \tag{4.3.11}$$

```
> dz:=dfxp*(x1-x0)+dfyp*(y1-y0);
```
$$dz := -.73 \tag{4.3.12}$$

```
> DZ:=f(x1,y1)-f(x0,y0);
```
$$DZ := -.7189 \tag{4.3.13}$$

```
> abserr:=abs(DZ-dz);
```
$$abserr := 0.0111 \tag{4.3.14}$$

The following graph shows that the linearization is close to the function about the given point.

```
> LiEq:=f(x0,y0)+dfxp*(x-x0)+dfyp*(y-y0);
```
$$LiEq := -15 + 7\,x - 9\,y \tag{4.3.15}$$

```
> plot3d({f(x,y),LiEq},x=2..4,y=-2..0, axes=frame);
```



Exercises

1. Find the tangent plane of $z=2x^2+3y^2$ at (1, 1, 5).
2. Find the tangent plane of the surface $x^2+z^2e^{y-x}$ at (2, 2, 1).
3. Find the tangent plane and normal line of the surface $xz+2x^2y+y^2z^3=11$ at (2, 1, 1).
4. Estimate how much the value of $f(x,y)=x+x\cos(y)-y\sin(z)+y$ will change if the point $P(x,y,z)$ moves 0.2 unit from (2, −1, 0) straight toward (0, 1, 2).
5. Find the linearization of $f(x,y)=(x+y+2)^2$ at (1, 2).

15.5 Gradient and Directional Derivatives

(1) The **Derivative of f at a point $P_0(x_0, y_0)$ in the direction of the unit vector $\mathbf{u}=u_1\mathbf{i}+u_2\mathbf{j}$** is the number

$$\left(\frac{df}{ds}\right)\Bigg|_{\mathbf{u},P_0}=\lim_{s\to 0}\frac{f(x_0+su_1, y_0+su_2)-f(x_0,y_0)}{s}$$

provided the limit exists.
(2) The **gradient vector** of $f(x,y)$ at a point $P_0(x_0,y_0)$ is the vector

$$\nabla f=\frac{\partial f}{\partial x}\mathbf{i}+\frac{\partial f}{\partial y}\mathbf{j}$$

obtained by evaluating $\frac{\partial f}{\partial x}$ and $\frac{\partial f}{\partial y}$ at (x_0,y_0).
(3) If $f(x,y)$ is differentiable in an open region containing $P_0(x_0,y_0)$, then

$$\frac{df}{ds}\Bigg|_{\mathbf{u},P_0}=\nabla f|_{P_0}\cdot\mathbf{u}$$

15.5.1 Compute gradients of multivariate functions

Example 1. (a) Find the gradient of $f(x,y,z)=xyz$. (b) Evaluate it at $P=(1,2,3)$. (c) Find the rate of change at P in the direction $\mathbf{u}=\left(\frac{1}{3},-\frac{2}{3},\frac{2}{3}\right)$.

(a) Gradient

```
> f:=x*y*z;
```

$$f := x\,y\,z \tag{5.1.1}$$

```
> grd:=grad(f,[x,y,z]);
```

$$grd := [y\,z \quad x\,z \quad x\,y] \tag{5.1.2}$$

(b) Gradient at (1, 2, 3).

```
> grdP:=map(eval, grd, [x=1,y=2,z=3]);
```

$$grdP := [6 \quad 3 \quad 2] \tag{5.1.3}$$

(c) The rate of change in **u** direction.

```
> u:=[1/2,-2/3,2/3];
```

$$u := \left[\frac{1}{2}, \frac{-2}{3}, \frac{2}{3}\right] \tag{5.1.4}$$

```
> ratechg:= innerprod(grdP,u);
```

$$ratechg := \frac{7}{3} \tag{5.1.5}$$

Example 2. Find the gradient of $f(x, y, z) = e^{x+y}\cos(z) + (y+1)\arcsin(x)$ at $\left(0, 0, \frac{\pi}{6}\right)$.

```
> f:=exp(x+y)*cos(z)+(y+1)*arcsin(x);
```

$$f := e^{(x+y)}\cos(z) + (y+1)\arcsin(x) \tag{5.1.6}$$

```
> grd:=grad(f,[x,y,z]);
```

$$grd := \left[e^{(x+y)}\cos(z) + \frac{y+1}{\sqrt{1-x^2}} \quad e^{(x+y)}\cos(z) + \arcsin(x) \quad -e^{(x+y)}\sin(z)\right] \tag{5.1.7}$$

```
> grdf:=map(eval,grd,[x=0,y=0,z=Pi/6]);
```

$$grdf := \left[\frac{1}{2}\sqrt{3}+1 \quad \frac{1}{2}\sqrt{3} \quad \frac{-1}{2}\right] \tag{5.1.8}$$

15.5.2 Directional derivative in a given direction

Example 3. Find the gradient vector of of $f(x, y, z) = x\sin(yz)$ and its directional derivative at (1, 3, 0) in the direction of $\mathbf{v} = \mathbf{i} + 2\mathbf{j} - \mathbf{k}$.

```
> f:=x*sin(y*z);
```

$$f := x\sin(y\,z) \tag{5.2.1}$$

(1) Its gradient at (1, 3, 0) is the following.

```
> grd:=grad(f,[x,y,z]);
```

$$grd := [\sin(y\,z) \quad x\cos(y\,z)\,z \quad x\cos(y\,z)\,y] \tag{5.2.2}$$

```
> grdatP:=map(eval,grd,{x=1,y=3,z=0});
```

$$grdatP := [0 \quad 0 \quad 3] \tag{5.2.3}$$

(2) Directional derivative at (1, 3, 0) in the direction of $\mathbf{v} = \mathbf{i} + 2\mathbf{j} - \mathbf{k}$.

```
> v:=[1,2,-1];
```

$$v := [1, 2, -1] \tag{5.2.4}$$

```
> unitv:=v/norm(v,2);
```

$$unitv := \frac{1}{6}\,[1, 2, -1]\,\sqrt{6} \qquad (5.2.5)$$

```
> dird:=innerprod(grdatP,unitv);
```

$$dird := -\frac{1}{2}\sqrt{6} \qquad (5.2.6)$$

```
> v:='v':
```

Example 4. Find the maximum rate of change of $f(x, y, z) = x+\frac{y}{z}$ at (4, 3, −1) and the direction in which it occurs.

```
> f:=x+y/z;
```

$$f := x + \frac{y}{z} \qquad (5.2.7)$$

```
> grd:=grad(f,[x,y,z]);
```

$$grd := \begin{bmatrix} 1 & \frac{1}{z} & -\frac{y}{z^2} \end{bmatrix} \qquad (5.2.8)$$

```
> grdp:=map(eval,grd,[x=4,y=3,z=-1]);
```

$$grdp := [1 \quad -1 \quad -3] \qquad (5.2.9)$$

```
> maxrate:=norm(grdp,2);
```

$$maxrate := \sqrt{11} \qquad (5.2.10)$$

The direction for the maximum rate of change is the gradient direction. The unit vector at this direction is the following.

```
> GrdDir:=[grdp[1]/norm(grdp,2),grdp[2]/norm(grdp,2),grdp[3]
   /norm(grdp,2)];
```

$$GrdDir := \left[\frac{1}{11}\sqrt{11}, -\frac{1}{11}\sqrt{11}, -\frac{3}{11}\sqrt{11}\right] \qquad (5.2.11)$$

The following is an alternate way to do it.

```
> GradDir:=map(x->x/norm(grdp,2),grdp);
```

$$GradDir := \begin{bmatrix} \frac{1}{11}\sqrt{11} & -\frac{1}{11}\sqrt{11} & -\frac{3}{11}\sqrt{11} \end{bmatrix} \qquad (5.2.12)$$

Example 5. Draw the graph of the level surface of $f(x, y, z) = x+\frac{y}{z}$ at (4, 3, −1) and the unit vector in its gradient directions.
Note that the unit vector is found in **Example 5** with the MAPLE NAME

```
> GrdDir;
```

$$\left[\frac{1}{11}\sqrt{11}, -\frac{1}{11}\sqrt{11}, -\frac{3}{11}\sqrt{11}\right] \qquad (5.2.13)$$

```
> fp:=eval(f,{x=4,y=3,z=-1});
```

$$fp := 1 \qquad (5.2.14)$$

```
> grph1:=implicitplot3d(x+y/z=1, x=2..5, y=2..5, z=-2..-0.5,
  axes=frame):
> grph2:=arrow(<4, 3, -1>, GrdDir, width=0.05, head_length=0.4,
  color=red):
> display({grph1,grph2}, scaling=constrained);
```



Example 6. Let $f(x, y) = \frac{x^2}{2} + \frac{y^2}{2}$. Find the directions of maximal, minimal, and zero change at the point $(1, 1)$ and draw the level curve at the point and all of the unit vectors in the directions.

```
> f:=x^2/2+y^2/2;
```

$$f := \frac{1}{2}x^2 + \frac{1}{2}y^2 \tag{5.2.15}$$

```
> grdf:=grad(f,[x,y]);
```

$$grdf := [x \quad y] \tag{5.2.16}$$

```
> grdfp:=map(eval, grdf, [x=1,y=1]);
```

$$grdfp := [1 \quad 1] \tag{5.2.17}$$

```
> mxdir:=map(x->x/norm(grdfp,2), grdfp);
```

$$mxdir := \left[\frac{1}{2}\sqrt{2} \quad \frac{1}{2}\sqrt{2}\right] \tag{5.2.18}$$

```
> mndir:=[-mxdir[1], -mxdir[2]];
```

$$mndir := \left[-\frac{1}{2}\sqrt{2}, -\frac{1}{2}\sqrt{2}\right] \tag{5.2.19}$$

```
> zrdir:=[-mxdir[2], mxdir[1]];
```

$$zrdir := \left[-\frac{1}{2}\sqrt{2}, \frac{1}{2}\sqrt{2}\right] \tag{5.2.20}$$

```
> g1:=implicitplot(x^2/2+y^2/2=1,x=-2..2,y=-2..2,color=black):
> g2:=arrow(<1,1>,mxdir,width=0.05,head_length=0.4,color=red):
> g3:=arrow(<1,1>,mndir,width=0.05,head_length=0.4,color=green):
> g4:=arrow(<1,1>,zrdir,width=0.05,head_length=0.4,color=blue):
> display({g1, g2, g3, g4}, scaling=constrained);
```



Example 7. Let $f(x, y, z) = x^3 - xy^2 - z$. Find the direction of the maximal change at the point $(1, 1, 0)$ and draw the level curve at the point and the unit vector in the direction of the maximal change.

```
> f:=x^2-x*y^2-z;
```

$$f := x^2 - x\,y^2 - z \tag{5.2.21}$$

```
> grdf:= grad(f,[x,y,z]);
```

$$grdf := \left[-y^2 + 2\,x \quad -2\,x\,y \quad -1 \right] \tag{5.2.22}$$

```
> grfp:= map(eval,grdf,{x=1,y=1,z=0});
```

$$grfp := [1 \quad -2 \quad -1] \tag{5.2.23}$$

```
> np:=map(x->x/norm(grfp,2), grfp);
```

$$np := \left[\frac{1}{6}\sqrt{6} \quad -\frac{1}{3}\sqrt{6} \quad -\frac{1}{6}\sqrt{6} \right] \tag{5.2.24}$$

```
> eval(f, [x=1,y=1,z=0]);
```

$$0 \tag{5.2.25}$$

```
> g1:=implicitplot3d(f=0, x=0..2, y=0..2, z=-1..1):
> g2:=arrow(<1,1,0>,np,width=0.05,head_length=0.4,color=red):
> display({g1,g2}, axes=frame, scaling=constrained);
```



15.5.3 Tangent lines to curves

Assume a plane curve is given by an equation $f(x, y) = c$. It can be considered as a level curve of the function $z = f(x, y)$. Since its tangent line is perpendicular to its gradient, the tangent line equation at a point $P_0\left(x_0, y_0\right)$ is

$$\nabla f|_{P_0} \cdot \left(P - P_0\right) = 0$$

Example 8. Find an equation for the tangent line to the ellipse $\frac{x^2}{4} + y^2 = 2$ at the point $(-2, 1)$.

```
> f:=x^2/4+y^2;
```

$$f := \frac{1}{4}x^2 + y^2 \qquad (5.3.1)$$

```
> grf:=grad(f, [x,y]);
```

$$grf := \left[\frac{1}{2}x \quad 2y\right] \qquad (5.3.2)$$

```
> grfp:=map(eval, grf, [x=-2,y=1]);
```

$$grfp := [-1 \quad 2] \qquad (5.3.3)$$

The equation of the tangent line is

```
> Eq:=innerprod(grfp, [x+2,y-1])=0;
```

$$Eq := -x - 4 + 2y = 0 \qquad (5.3.4)$$

Exercises

1. (a) Find gradient of $f(x, y, z) = \cos\left(x^2 + y\right)$.

   (b) Evaluate it at $P = \left(\sqrt{\frac{\pi}{3}}, \frac{\pi}{6}\right)$.

   (c) Find the rate of change at P in the direction $\mathbf{u} = \langle 1, \sqrt{3}\rangle$.
2. Find the gradient of $f(x, y, z) = x \ln(y + z)$ at $(2, \mathrm{e}, \mathrm{e})$.
3. Find the gradient vector of of $f(x, y, z) = x\sin(y - z)$ and its directional derivative at $(1, \frac{\pi}{2}, \frac{\pi}{6})$ in the direction of $\mathbf{v} = \boldsymbol{i} + 2\mathbf{j} + 2\mathbf{k}$.
4. Draw the graph of the surface of $x^2 + y^2 - z^2 = 6$ at (3, 1, 2) together with the unit vector in its normal directions.
5. Find the maximum rate of change of $f(x, y, z) = x - y^2 - xz$ at (4, 2, −1) and the direction in which it occurs.
6. Let $f(x, y) = \frac{x^2}{4} + \frac{y^2}{9}$. Find the directions of maximal, minimal, and zero change at the point $(2, 3)$ and draw the level curve at the point and all of the unit vectors in the directions.
7. Let $f(x, y, z) = xz + 2x^2y + y^2z^3$. Find the direction of the maximal change at the point $(2, 1, 1)$ and draw the level curve at the point and the unit vector in the direction of the maximal change.
8. Find an equation for the tangent line to the hyperbola $\frac{x^2}{4} - \frac{y^2}{2} = 1$ at the point $\left(3, \sqrt{\frac{5}{2}}\right)$.

15.6 The Chain Rule

If $w=f(x,y)$ has continuous partial derivatives f_x and f_y and if $x=x(t)$ and $y=y(t)$ are differentiable functions, then the composite function $w=f(x(t),y(t))$ is differentiable for t and

$$\frac{df}{dt}=f_x(x(t),y(t))x'(t)+f_y(x(t),y(t))y'(t)$$

Similarly, if $w=f(x,y,z)$ has all continuous partial derivatives, and if $x=g(r,s)$, $y=h(r,s)$ and $z=k(r,s)$ are differentiable functions, then

$$\frac{\partial w}{\partial r}=\frac{\partial w}{\partial x}\frac{\partial x}{\partial r}+\frac{\partial w}{\partial y}\frac{\partial y}{\partial r}+\frac{\partial w}{\partial z}\frac{\partial z}{\partial r}$$

$$\frac{\partial w}{\partial s}=\frac{\partial w}{\partial x}\frac{\partial x}{\partial s}+\frac{\partial w}{\partial y}\frac{\partial y}{\partial s}+\frac{\partial w}{\partial z}\frac{\partial z}{\partial s}$$

Other formulas can be derived in a similar way.

15.6.1 Calculate partial derivatives for composite functions

Example 1. (a) Use the chain rule to find the derivatives of $w=xy$ with respect to t along the path $x=\cos(t)$, $y=\sin(t)$. (b) Use the MAPLE syntax to find them again. (3) Find the derivative value at $t=\frac{\pi}{2}$.

(a) Use the chain rule.

```
> w:=x*y:  wx:=diff(w,x); wy:=diff(w,y);
```

$$wx := y$$

$$wy := x \qquad (6.1.1)$$

```
> dft:=subs({x=cos(t), y=sin(t)}, wx)*diff(cos(t),t)+subs({x=cos
  (t), y=sin(t)}, wy)*diff(sin(t),t);
```

$$dft := -\sin(t\sim)^2+\cos(t\sim)^2 \qquad (6.1.2)$$

(b) Use the MAPLE syntax.

```
> wt:=subs({x=cos(t), y=sin(t)}, x*y);
```

$$wt := \cos(t\sim)\sin(t\sim) \qquad (6.1.3)$$

```
> dft:=diff(wt,t);
```

$$dft := -\sin(t\sim)^2+\cos(t\sim)^2 \qquad (6.1.4)$$

(c) Evaluate the derivative at $t=\frac{\pi}{2}$.

```
> eval(dft, t=Pi/2);
```

$$-1 \qquad (6.1.5)$$

Example 2. Find $\frac{dw}{dt}$ if $w=xy+z$, $x=\cos(t)$, $y=\sin(t)$, $z=t$.

```
> Wxyz:=(x,y,z)->x*y+z;
```

$$Wxyz := (x,y,z)\rightarrow y\,x+z \qquad (6.1.6)$$

```
> Wt:=Wxyz(cos(t), sin(t), t);
```

$$Wt := \cos(t\sim)\sin(t\sim)+t\sim \qquad (6.1.7)$$

```
> dwt:=diff(Wt,t);
```

$$dwt := -\sin(t\sim)^2 + \cos(t\sim)^2 + 1 \tag{6.1.8}$$

Example 3. Find $\frac{\partial w}{\partial r}$ and $\frac{\partial w}{\partial s}$ if $w = x+2y+z^2,\ x = \frac{r}{s},\ y = r^2 + \ln(s),\ z = 2r.$

```
> Wxyz:=(x,y, z)->x+2*y+z^2;
```

$$Wxyz := (x, y, z) \rightarrow x + 2\,y + z^2 \tag{6.1.9}$$

```
> Wsr:=Wxyz(r/s, r^2+ln(s), 2*r);
```

$$Wsr := \frac{r}{s} + 6\,r^2 + 2\ln(s) \tag{6.1.10}$$

```
> dwr:=diff(Wsr,r);
```

$$dwr := \frac{1}{s} + 12\,r \tag{6.1.11}$$

```
> dws:=diff(Wsr, s);
```

$$dws := -\frac{r}{s^2} + \frac{2}{s} \tag{6.1.12}$$

Example 4. Find $\frac{\partial w}{\partial r}$ and $\frac{\partial w}{\partial s}$ if $w = x^2+y^2,\ x = r-s,\ y = r+s.$

```
> wxy:=(x,y)->x^2+y^2;
```

$$wxy := (x, y) \rightarrow x^2 + y^2 \tag{6.1.13}$$

```
> wrs:=wxy(r-s,r+s);
```

$$wrs := (r - s)^2 + (r + s)^2 \tag{6.1.14}$$

```
> dwr:=diff(wrs, r); dws:=diff(wrs,s);
```

$$dwr := 4\,r$$
$$dws := 4\,s \tag{6.1.15}$$

15.6.2 Partial derivatives of implicit functions

If $z = z(x, y)$ is an implicit function defined by the equation $F(x, y, z) = c$, then its partial derivatives $\frac{\partial z}{\partial x}$ and $\frac{\partial z}{\partial y}$ can obtained by the following formulas:

$$\frac{\partial z}{\partial x} = -\frac{D_x F}{D_z F},\quad \frac{\partial z}{\partial y} = -\frac{D_y F}{D_z F}$$

Similarly, if $z = z(x, y)$ is defined by $F(x, y) = c$, then

$$\frac{dy}{dx} = -\frac{D_x \mathrm{F}}{D_y \mathrm{F}}$$

Example 5. Find $\frac{dy}{dx}$ if $y^2 - x^2 - \sin(xy) = 0.$

```
> F:=y^2-x^2-sin(x*y);
```

$$F := y^2 - x^2 - \sin(x\,y) \tag{6.2.1}$$

```
> dyx:=-diff(F,x)/diff(F,y);
```

$$dyx := -\frac{-2\,x - \cos(x\,y)\,y}{2\,y - \cos(x\,y)\,x} \tag{6.2.2}$$

Example 6. Find $\frac{\partial z}{\partial x}$ and $\frac{\partial z}{\partial y}$ at $(1, \ln(2), \ln(3))$ if $xe^y + ye^z + 2\ln(x) - 2 - 3\ln(2) = 0$.

```
> F:=x*exp(y)+y*exp(z)+2*ln(x)-2-3*ln(2);
```

$$F := x\,e^{y} + y\,e^{z} + 2\ln(x) - 2 - 3\ln(2) \tag{6.2.3}$$

```
> dzx:=-diff(F,x)/diff(F,z); dzy:=-diff(F,y)/diff(F,z);
```

$$dzx := -\frac{e^{y} + \frac{2}{x}}{y\,e^{z}}$$

$$dzy := -\frac{x\,e^{y} + e^{z}}{y\,e^{z}} \tag{6.2.4}$$

Exercises

1. (a) Use the chain rule to find the derivatives of $w = x^2y + xy^2$ with respect to t along the path $x = \cos(t)$, $y = \sin(t)$.
   (b) Use the MAPLE syntax to find them again.
   (c) Find the derivative value a $t = \frac{\pi}{2}$.
2. Find $\frac{\partial w}{\partial r}$ and $\frac{\partial w}{\partial s}$ if $w = xy + xz + yz$, $x = \frac{r}{s}$, $y = r\ln(s+1)$, $z = sr$.
3. Find $\frac{dy}{dx}$ at $(0, \ln(2))$ if $xe^y + \sin(xy) + y - \ln(2) = 0$.
4. Let $z = x^2 + \frac{y}{x}$, $x = u - 2v + 1$, $y = 2u + v - 2$. Find $\frac{\partial z}{\partial u}$ and $\frac{\partial z}{\partial v}$ at $u = 0$, $v = 0$.
5. Let $z = xe^{\frac{x}{y}}$, $x = \cos(t)$, $y = e^{2t}$. Find $\frac{dz}{dt}$.
6. Let $z = x^2 - 3x^2y^3$, $x = se^t$, $y = se^{-t}$. Find $\frac{\partial z}{\partial s}$, $\frac{\partial z}{\partial t}$.
7. Let $z = y^2\tan(x)$, $x = t^2uv$, $y = u + tv^2$. Find $\frac{\partial z}{\partial t}$, $\frac{\partial z}{\partial u}$, $\frac{\partial z}{\partial v}$ at $t = 2$, $u = 1$, $v = 0$.
8. Find $\frac{dy}{dx}$ for $x^3 + y^3 = 6xy$.
9. Find $\frac{\partial z}{\partial x}$ and $\frac{\partial z}{\partial y}$ for $x^3 + y^3 + z^3 + 6xyz = 1$.
10. Find $\frac{\partial z}{\partial x}$ and $\frac{\partial z}{\partial y}$ for $x^2 + y^2 + z^2 = 1$ at $(x, y, z) = \left(\frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}\right)$.

15.7 Extreme Value and Saddle Points

(1) Let $f(x, y)$ be defined on a region R containing the point (a, b) and $f(a, b)$ be a **local maximum** value or a **local minimum** value of f. Assume that the first derivatives of f exist there. Then $f_x(a, b) = f_y(a, b) = 0$.
(2) If (a, b) is an interior point of the domain of f and $f_x(a, b) = f_y(a, b) = 0$, or where one or both of $f_x(a, b)$ and $f_y(a, b)$ do not exist, then (a, b) is called a **critical point** of f. A critical point is called a **saddle point** if in every neighborhood of (a, b) there are points (x, y) where $f(x, y) > f(a, b)$ and points (x, y) where $f(x, y) < f(a, b)$.
(3) The **Second Derivative Test**. Assume that (a, b) is an interior point of the domain of f and $f_x(a, b) = f_y(a, b) = 0$.

(a) f has a local maximum at (a, b) if $f_{xx} < 0$ and $f_{xx} f_{yy} - f_{xy}^2 > 0$ at (a, b).

(b) f has a local minimum at (a, b) if $f_{xx} > 0$ and $f_{xx} f_{yy} - f_{xy}^2 > 0$ at (a, b).

(c) f has a saddle point at (a, b) if $f_{xx} f_{yy} - f_{xy}^2 < 0$ at (a, b).

(d) **The test is inconclusive** at (a, b) if $f_{xx} f_{yy} - f_{xy} = 0$ at (a, b). In this case, you must find other ways to determine the behavior of f at (a, b).

15.7.1 Derivative tests for local extreme values

Find local maximum and minimum: To find local extrema, you first find all **critical points**, then use **second derivative** test to determine if it is an extremal point, if the test is conclusive.
Example 1. Find the local extreme values of $f(x, y) = x^2 + y^2$.
Step 1: Find all critical points.

```
> with(linalg): with(plots):
Warning, the protected names norm and trace have been redefined
and unprotected
Warning, the name changecoords has been redefined
> f:=x^2+y^2;
```

$$f := x^2 + y^2 \quad (7.1.1)$$

```
> dfx:=diff(f,x); dfy:=diff(f,y);
```

$$dfx := 2x$$
$$dfy := 2y \quad (7.1.2)$$

```
> solve({dfx=0, dfy=0}, {x,y});
```

$$\{x = 0, y = 0\} \quad (7.1.3)$$

```
> dfxx:=diff(dfx,x); dfyy:=diff(dfy,y): dfxy:=diff(dfx,y):
```

$$dfxx := 2 \quad (7.1.4)$$

```
> hnss:=eval(dfxx*dfyy-dfxy^2,[x=0,y=0]);
```

$$hnss := 4 \quad (7.1.5)$$

Step 2: Determine if the function has a local extremum at the point.
The only critical point is (0, 0), where $f(0, 0) = 0$. Since the function is nonnegative, f has its minimum at (0, 0). It can be verified by the graph of the function.

```
> plot3d(f, x=-2..2, y=-2..2, view=0..4, axes=frame, scaling=
  constrained);
```



Example 2. Classify the critical points of $f(x, y) = y^2 - x^2$.

```
> f:=y^2-x^2;
```

$$f := y^2 - x^2 \tag{7.1.6}$$

```
> dfx:=diff(f,x); dfy:=diff(f,y);
```

$$dfx := -2\,x$$
$$dfy := 2\,y \tag{7.1.7}$$

```
> solve({dfx,dfy},{x,y});
```

$$\{x = 0, y = 0\} \tag{7.1.8}$$

```
> dfxx:=diff(dfx,x); dfyy:=diff(dfy,y); dfxy:=diff(dfx,y);
```

$$dfxx := -2$$
$$dfyy := 2$$
$$dfxy := 0 \tag{7.1.9}$$

```
> hnss:=dfxx*dfyy-dfxy^2;
```

$$hnss := -4 \tag{7.1.10}$$

Hence, $(0, 0)$ is a saddle point of the function.

```
> plot3d(y^2-x^2,x=-2..2, y=-2..2, axes=frame);
```



Example 3. Find the local extreme values of the function $f(x, y) = xy - y^2 - x^2 - 2x - 2y + 4$.

```
> f:=x*y-y^2-x^2-2*x-2*y+4;
```

$$f := x\,y - y^2 - x^2 - 2\,x - 2\,y + 4 \qquad (7.1.11)$$

```
> fx:=diff(f,x); fy:=diff(f,y);
```

$$fx := y - 2\,x - 2$$
$$fy := x - 2\,y - 2 \qquad (7.1.12)$$

```
> solve({fx,fy},{x,y});
```

$$\{x = -2, y = -2\} \qquad (7.1.13)$$

```
> fxx:=diff(fx,x); fyy:=diff(fy,y); fxy:=diff(fx,y);
```

$$fxx := -2$$
$$fyy := -2$$
$$fxy := 1 \qquad (7.1.14)$$

```
> Hssn:=fxx*fyy-fxy^2;
```

$$Hssn := 3 \qquad (7.1.15)$$

```
> eval(f, [x=-2,y=-2]);
```

$$8 \qquad (7.1.16)$$

Hence, $f(x, y)$ has the local maximum 8 at $(-2, -2)$.

Example 4. Find the local extreme values (if any) of $f(x, y) = xy$.

```
> f:=x*y; dfx:=diff(f,x); dfy:=diff(f,y);
```

$$f := x\,y$$
$$dfx := y$$
$$dfy := x \qquad (7.1.17)$$

```
> dfxx:=diff(dfx,x); dfyy:=diff(dfy,y); dfxy:=diff(dfx,y);
```

$$dfxx := 0$$
$$dfyy := 0$$
$$dfxy := 1 \qquad (7.1.18)$$

```
> hnss:=dfxx*dfyy-dfxy^2;
```

$$hnss := -1 \tag{7.1.19}$$

```
> plot3d(x*y, x=-2..2, y=-2..2, axes=frame);
```



Example 5. Find and classify the critical points of $f(x, y) = 10x^2y - 5x^2 - 4y^2 - x^4 - 2y^4$.
Step 1: Find all critical points.

```
> x:='x': y:='y':
> f:=10*x^2*y-5*x^2-4*y^2-x^4-2*y^4;
```

$$f := 10\,x^2\,y - 5\,x^2 - 4\,y^2 - x^4 - 2\,y^4 \tag{7.1.20}$$

```
> dfx:=diff(f,x);
```

$$dfx := 20\,x\,y - 10\,x - 4\,x^3 \tag{7.1.21}$$

```
> dfy:=diff(f,y);
```

$$dfy := 10\,x^2 - 8\,y - 8\,y^3 \tag{7.1.22}$$

```
> solve({dfx,dfy},{x,y});
```

$$\{x=0, y=0\}, \left\{x=0, y=\mathrm{RootOf}\left(\_Z^2 + 1, label=\_L1\right)\right\}, \left\{x=\mathrm{RootOf}\left(-5\,\mathrm{RootOf}\left(\_Z^3 + 5 - 18\,\_Z + 3\,\_Z^2, label=\_L5\right) + 2\,\_Z^2, label=\_L3\right), y=\frac{1}{2} + \frac{1}{2}\,\mathrm{RootOf}\left(\_Z^3 + 5 - 18\,\_Z + 3\,\_Z^2, label=\_L5\right)\right\} \tag{7.1.23}$$

Remark. The output "$\left\{y=\mathrm{RootOf}\left(\_Z^2 + 1, label=\_L8\right), x=0\right\}$" means that y is the roots of the equation "$z^2 + 1 = 0$". The output

$$\left\{y=\frac{1}{2} + \frac{1}{2}\,\mathrm{RootOf}\left(\_Z^3 + 5 - 18\_Z + 3\,\_Z^2, label=\_L12\right), x=\mathrm{RootOf}\left(-5\mathrm{RootOf}\left(\_Z^3 + 5 - 18\_Z + 3\_Z^2, label=\_L12\right) + 2\_Z^2, label=\_L10\right)\right\}$$

means that $y = \frac{1}{2} + \frac{1}{2}z$, where z is the root of the equation $z^3 + 5 - 18z + 3z^2 = 0$, and then x is the root of the equation $-5y + 3z^2 = 0$.

```
> e1:=z^2+1;
```

$$e1 := z^2 + 1 \tag{7.1.24}$$

This equation has no real solutions.

```
> e2:=(z^3+5-18*z+3*z^2);
```

$$e2 := z^3 + 5 - 18\,z + 3\,z^2 \tag{7.1.25}$$

```
> zv:=fsolve(e2,z);
```

$$zv := -6.090313261,\ 0.2935443982,\ 2.796768863 \tag{7.1.26}$$

```
> yv1:=1/2+1/2*zv[1];
```

$$yv1 := -2.545156630 \tag{7.1.27}$$

```
> xv1:=fsolve(-5*yv1+2*z^2,z);
```

$$xv1 := \tag{7.1.28}$$

```
> yv2:=1/2+1/2*zv[2];
```

$$yv2 := 0.6467721991 \tag{7.1.29}$$

```
> xv2:=fsolve(-5*yv2+2*z^2);
```

$$xv2 := -1.271585820,\ 1.271585820 \tag{7.1.30}$$

```
> yv3:=1/2+1/2*zv[3];
```

$$yv3 := 1.898384432 \tag{7.1.31}$$

```
> xv3:=fsolve(-5*yv3+2*z^2,z);
```

$$xv3 := -2.178522683,\ 2.178522683 \tag{7.1.32}$$

All critical points are

```
> p1:=[x=0,y=0];p2:=[x=xv2[1],y=yv2];p3:=[x=xv2[2],y=yv2];p4:=
  [x=xv3[1],y=yv3];p5:=[x=xv3[2],y=yv3];
```

$$p1 := [x = 0, y = 0]$$
$$p2 := [x = -1.271585820, y = 0.6467721991]$$
$$p3 := [x = 1.271585820, y = 0.6467721991]$$
$$p4 := [x = -2.178522683, y = 1.898384432]$$
$$p5 := [x = 2.178522683, y = 1.898384432] \tag{7.1.33}$$

Step 2. Apply the Second Derivative Test.

```
> fxx:=diff(dfx,x);fyy:=diff(dfy,y);fxy:=diff(dfx,y);
```

$$fxx := 20\,y - 10 - 12\,x^2$$
$$fyy := -8 - 24\,y^2$$
$$fxy := 20\,x \tag{7.1.34}$$

```
> Hssn:=fxx*fyy-fxy^2;
```

$$Hssn := \left(20\,y - 10 - 12\,x^2\right)\left(-8 - 24\,y^2\right) - 400\,x^2 \tag{7.1.35}$$

Test all critical points.

```
> t1:=eval([fxx,Hssn],p1);
```

$$t1 := [-10, 80] \tag{7.1.36}$$

```
> t2:=eval([fxx,Hssn],p2);
```

$$t2 := [-16.46772200,\ -349.7020257] \tag{7.1.37}$$

```
> t3:=eval([fxx,Hssn],p3);
```
$$t3 := [-16.46772200, -349.7020257] \tag{7.1.38}$$

```
> t4:=eval([fxx,Hssn],p4);
```
$$t4 := [-28.98384432, 840.377936] \tag{7.1.39}$$

```
> t5:=eval([fxx,Hssn],p5);
```
$$t5 := [-28.98384432, 840.377936] \tag{7.1.40}$$

The local maxima (with the locations) are:

```
> {eval(f,p1), p1}; {eval(f,p4), p4}; {eval(f,p5),p5};
```
$$\{0, [x=0, y=0]\}$$
$$\{3.45151694, [x=-2.178522683, y=1.898384432]\}$$
$$\{3.45151694, [x=2.178522683, y=1.898384432]\} \tag{7.1.41}$$

The saddle points are:

```
> p2;p3;
```
$$[x=-1.271585820, y=0.6467721991]$$
$$[x=1.271585820, y=0.6467721991] \tag{7.1.42}$$

15.7.2 Absolute maxima and minima on closed bounded regions

If $f(x, y)$ is continuous on a closed and bounded region R, then $f(x, y)$ has the absolute maximum and minimum on R.

Example 6. Find the absolute maximum and minimum values of
$$f(x, y) = 2 + 2x + 2y - x^2 - y^2.$$
Method 1. Directly apply MAPLE syntax.

```
> f:=2+2*x+2*y-x^2-y^2;
```
$$f := 2 + 2x + 2y - x^2 - y^2 \tag{7.2.1}$$

```
> minimize(f, location);
```
$$-\infty, \{[\{y=\infty, x=\infty\}, -\infty], [\{y=\infty, x=-\infty\}, -\infty], [\{y=-\infty, x=-\infty\}, -\infty], [\{y=-\infty, x=\infty\}, -\infty]\} \tag{7.2.2}$$

Hence, $f(x, y)$ has no minimum on the plane.

```
> maximize(f, location);
```
$$4, \{[\{y=1, x=1\}, 4]\} \tag{7.2.3}$$

Method 2. Find all critical points in the region and then apply the **Second Derivative Test**.
Step 1. Find the critical point(s) in the region.

```
> dfx:=diff(f,x);
```
$$dfx := 2 - 2x \tag{7.2.4}$$

```
> dfy:=diff(f,y);
```
$$dfy := 2 - 2y \tag{7.2.5}$$

```
> solve({dfx,dfy},{x,y});
```
$$\{y=1, x=1\} \tag{7.2.6}$$

Step 2. Apply the **Second Derivative Test**.

```
> dfxx:=diff(dfx,x); dfyy:=diff(dfy, y); dfxy:=diff(dfx,y);
```

$$dfxx := -2$$
$$dfyy := -2$$
$$dfxy := 0 \qquad (7.2.7)$$

```
> hssn:=dfxx*dfyy-dfxy^2;
```

$$hssn := 4 \qquad (7.2.8)$$

Hence, the function has the maximum at $(1, 1)$.

Find the absolute maximum and minimum on rectangle. To find the absolute maximum and minimum on a rectangle, you can either use MAPLE syntaxes "**maximize**" and "**minimize**", or do the following: Find all critical points in the region, and then find the maximum and minimum on the boundary. Comparing all the function values at those points, the largest one is the maximum and the smallest one is the minimum.

Example 7. Find the absolute maximum and minimum of

$$f(x, y) = y\sqrt{x} - y^2 - x + 6y$$

on $D = \{(x, y) \mid 0 \le x \le 9, 0 \le y \le 5\}$.

Method 1: Directly apply the MAPLE syntaxes "**maximize**" and "**minimize**".

```
> f:=y*sqrt(x)-y^2-x+6*y;
```

$$f := y\sqrt{x} - y^2 - x + 6\,y \qquad (7.2.9)$$

```
> maximize(f,x=0..9,y=0..5, location);
```

$$4\sqrt{4} + 4, \left\{\left[\{x=4, y=4\}, 4\sqrt{4} + 4\right]\right\} \qquad (7.2.10)$$

```
> minimize(f,x=0..9,y=0..5, location);
```

$$-9, \{[\{y=0, x=9\}, -9]\} \qquad (7.2.11)$$

Method 2: Evaluate the function at all critical points and the extremal points on the boundary.

Step 1: Find the critical points of the function in the region.

```
> dfx:=diff(f,x);
```

$$dfx := \frac{1}{2}\frac{y}{\sqrt{x}} - 1 \qquad (7.2.12)$$

```
> dfy:=diff(f,y);
```

$$dfy := \sqrt{x} - 2\,y + 6 \qquad (7.2.13)$$

```
> solve({dfx,dfy},{x,y});
```

$$\{x=4, y=4\} \qquad (7.2.14)$$

Step 2: Find the critical points of $f(0, y)$ on the interval $0 < y < 5$, the critical points of $f(9, y)$ on the interval $0 < y < 5$, the critical points of $f(x, 0)$ on the interval $0 < x < 9$, and the critical points of $f(x, 5)$ on the interval $0 < x < 9$.

```
> fx0:=eval(f, y=0);
```

$$fx0 := -x \qquad (7.2.15)$$

No critical points for this function exist on the interval (0, 9).

```
> fx5:=eval(f, y=5);
```

$$fx5 := 5\sqrt{x} + 5 - x \qquad (7.2.16)$$

```
> solve(diff(fx5,x),x);
```

$$\frac{25}{4} \tag{7.2.17}$$

One critical point $x = \frac{25}{4}$ for the function $f(x, 5)$ is on the interval (0, 9).

```
> f0y:=eval(f,x=0);
```

$$f0y := -y^2 + 6\,y \tag{7.2.18}$$

```
> solve(diff(f0y,y)=0,y);
```

$$3 \tag{7.2.19}$$

One critical point $y = 3$ for the function $f(0, y)$ is on the interval (0, 5).

```
> f9y:=eval(f,x=9);
```

$$f9y := y\sqrt{9} - y^2 - 9 + 6\,y \tag{7.2.20}$$

```
> solve(diff(f9y,y)=0,y);
```

$$\frac{1}{2}\sqrt{9} + 3 \tag{7.2.21}$$

One critical point $y = \frac{1}{2}\sqrt{9} + 3$ for the function $f(9, y)$ is on the interval (0, 5).

```
> Plist:={[x=4,y=4],[x=25/4,y=5],[x=0,y=3],[x=9, y=sqrt(9)/2+3],
  [x=0,y=0],[x=0,y=5], [x=9,y=0], [x=9,y=5]};
```

$$Plist := \left\{ [x=0, y=0],\ [x=4, y=4],\ \left[x=\frac{25}{4}, y=5\right],\ [x=0, y=3],\ [x=0, y=5],\ [x=9, y=0],\ [x=9, y=5],\ \left[x=9, y=\frac{9}{2}\right] \right\} \tag{7.2.22}$$

```
> for i from 1 to 8 do simplify(eval(f, Plist[i])) end do;
```

$$0$$
$$12$$
$$\frac{45}{4}$$
$$9$$
$$5$$
$$-9$$
$$11$$
$$\frac{45}{4} \tag{7.2.23}$$

Hence, the absolute maximum is 12 at (4, 4), and the absolute minimum is −9 at (9, 0).

Find the absolute maximum and minimum on a general region. To find the absolute maximum and minimum on a general region, you have to find all critical points in the region, then find the maximum and minimum on the boundary of the region, comparing all the function values at those points.

Example 8. Find the absolute maximum and minimum values of

$$f(x, y) = 2 + 2x + 2y - x^2 - y^2$$

on the triangular region in the first quadrant bounded by the lines $x = 0,\ y = 0,\ y = 9 - x$.

```
> f:=2+2*x+2*y-x^2-y^2;
```

$$f := 2 + 2\,x + 2\,y - x^2 - y^2 \tag{7.2.24}$$

```
> dfx:=diff(f,x);
```

$$dfx := 2 - 2\,x \tag{7.2.25}$$

```
> dfy:=diff(f,y);
```

$$dfy := 2 - 2\,y \tag{7.2.26}$$

```
> solve({dfx,dfy},{x,y});
```

$$\{y = 1, x = 1\} \tag{7.2.27}$$

```
> fx0:=eval(f,y=0);
```

$$fx0 := 2 + 2\,x - x^2 \tag{7.2.28}$$

```
> solve(diff(fx0,x),x);
```

$$1 \tag{7.2.29}$$

```
> f0y:=eval(f,x=0);
```

$$f0y := 2 + 2\,y - y^2 \tag{7.2.30}$$

```
> solve(diff(f0y,y),y);
```

$$1 \tag{7.2.31}$$

```
> fxy:=eval(f, y=9-x);
```

$$fxy := 20 - x^2 - (9 - x)^2 \tag{7.2.32}$$

```
> solve(diff(fxy,x),x);
```

$$\frac{9}{2} \tag{7.2.33}$$

```
> Plist:={[x=1,y=1],[x=1,y=0],[x=0,y=1],[x=9/2,y=9/2],[x=0,y=0],
  [x=0,y=9],[x=9,y=0]};
```

$$Plist := \left\{ [x = 0, y = 0],\ [x = 1, y = 1],\ [x = 1, y = 0],\ [x = 0, y = 1],\ \left[x = \frac{9}{2}, y = \frac{9}{2}\right],\ [x = 0, y = 9],\ [x = 9, y = 0] \right\} \tag{7.2.34}$$

```
> for i from 1 to 7 do eval(f, Plist[i]) end do;
```

$$\begin{array}{c} 2 \\ 4 \\ 3 \\ 3 \\ \frac{-41}{2} \\ -61 \\ -61 \end{array} \tag{7.2.35}$$

Hence, the absolute maximum is 4 at the point (1, 1), and the absolute minimum is −61 at (9, 0) and (0, 9).

Exercises

1. Find the local extreme values of $f(x, y) = x^2 + y^2 + xy + 2x - 2y + 3$.
2. Classify the critical points of $f(x, y) = y^2 - x^2 - 2x + 6y - 2$.
3. Find the local extreme values of the function $f(x, y) = \dfrac{1}{x^2 + y^2 - 1}$.
4. Find the local extreme values (if any) of $f(x, y) = x\sin(y)$.
5. Find the absolute maximum and minimum values of $f(x, y) = x^2 - xy + y^2 + 1$ on $D = \{(x, y) \mid 0 \le x \le 4, 0 \le y \le 3\}$.
6. Find the absolute maximum and minimum of $f(x, y) = (4x - x^2)\cos(y)$, on $D = \{(x, y) \mid 1 \le x \le 3, -\pi/4 \le y \le \pi/4\}$.
7. Find the absolute maximum and minimum values of $f(x, y) = x^2 + xy + y^2 - 6x + 2$ on the triangular region in the first quadrant bounded by the lines $x = 0$, $y = 0$, $y = 9 - 3x$.
8. Find the absolute maximum and minimum values of $f(x, y) = 2x^2 - 4x + y^2 + 1$ on the triangular region in the first quadrant bounded by the lines $x = 0$, $y = 2$, $y = 2x$.

15.8 Lagrange Multipliers

15.8.1 Constrained maxima and minima

Find the absolute maximum and minimum with constraints. To find the extrema of a multivariate function with constraints. you can use the MAPLE syntax "**extrema**". The result only gives you the candidates of the extremum value. You should pick up the right values by yourself.

Example 1. Find the point $P(x, y, z)$ closest to the region on the plane $2x + y - z - 5 = 0$. Set the goal function to the **square of the distance** (that may simplify the computation).

```
> g:=x^2+y^2+z^2;
```

$$g := x^2 + y^2 + z^2 \qquad (8.1.1)$$

```
> cnstrn:=2*x+y-z-5=0;
```

$$cnstrn := 2x + y - z - 5 = 0 \qquad (8.1.2)$$

```
> extrema(g,cnstrn,{x,y,z},'s');
```

$$\left\{\frac{25}{6}\right\} \qquad (8.1.3)$$

```
> s;
```

$$\left\{\left\{y = \frac{5}{6}, x = \frac{5}{3}, z = \frac{-5}{6}\right\}\right\} \qquad (8.1.4)$$

Hence, the closest point P is $\left(\frac{5}{3}, \frac{5}{6}, \frac{-5}{6}\right)$ and the distance is $\sqrt{\frac{25}{6}} = \frac{5\sqrt{6}}{6}$.

Example 2. Find the point $P(x, y, z)$ closest to the region on the hyperbolic cylinder $x^2 - z^2 - 1 = 0$.

```
> cnstrn:=x^2-z^2-1=0;
```

$$cnstrn := x^2 - z^2 - 1 = 0 \tag{8.1.5}$$

```
> extrema(g,cnstrn,{x,y,z}, 's');
```

$$\{-1, 1\} \tag{8.1.6}$$

```
> s;
```

$$\{\{z=0, x=1, y=0\}, \{z=0, y=0, x=-1\}, \{x=0, y=0, z=\mathrm{RootOf}(\_Z^2 + 1, label=\_L2)\}\} \tag{8.1.7}$$

The z-coordinate of the first point is given by the root of $z^2 + 1 = 0$, which is not a real number. Hence, it is not counted.

```
> eval(g, [x=-1,y=0,z=0]); eval(g, [x=1,y=0,z=0]);
```

$$1$$

$$1 \tag{8.1.8}$$

Hence, the closest points are $(-1, 0, 0)$ and $(1, 0, 0)$.

Example 3. Find the extrema of the function xyz on the surface $x^2 + y^2 + z^2 = 1$.

```
> extrema(x*y*z,x^2+y^2+z^2=1,{x,y,z},'s');
```

$$\left\{\max\left(\frac{1}{3}\,\mathrm{RootOf}(3\,\_Z^2-1), -\frac{1}{3}\,\mathrm{RootOf}(3\,\_Z^2-1), 0\right), \min\left(\frac{1}{3}\,\mathrm{RootOf}(3\,\_Z^2-1), -\frac{1}{3}\,\mathrm{RootOf}(3\,\_Z^2-1), 0\right)\right\} \tag{8.1.9}$$

where RootOf(3_Z^2-1) indicates the roots of the equation $3z^2 - 1 = 0$. You may solve the equation to find the value(s).

```
> rf:=solve(3*z^2-1,z);
```

$$rf := \frac{1}{3}\sqrt{3}, -\frac{1}{3}\sqrt{3} \tag{8.1.10}$$

Therefore, the minimum value is

```
> mf:=min(1/3*rf[1],1/3*rf[2],0);
```

$$mf := -\frac{1}{9}\sqrt{3} \tag{8.1.11}$$

and the maximum value is

```
> Mf:=max(1/3*rf[1],1/3*rf[2],0);
```

$$Mf := \frac{1}{9}\sqrt{3} \tag{8.1.12}$$

If you also want to find the maximum and minimum points, you should add a string variable in the commend "extrema" as follows:

```
> extrema(x*y*z,x^2+y^2+z^2=1,{x,y,z},'s');
```

$$\left\{\max\left(\frac{1}{3}\,\mathrm{RootOf}(3\,\_Z^2-1), -\frac{1}{3}\,\mathrm{RootOf}(3\,\_Z^2-1), 0\right), \min\left(\frac{1}{3}\,\mathrm{RootOf}(3\,\_Z^2-1), -\frac{1}{3}\,\mathrm{RootOf}(3\,\_Z^2-1), 0\right)\right\} \tag{8.1.13}$$

The string variable "**s**" gives the candidates of the points, on which the extrema achieve.

```
> s;
```

$$\{\{y=0, x=1, z=0\}, \{y=0, z=0, x=-1\}, \{y=\mathrm{RootOf}(3\_Z^2-1), z=\mathrm{RootOf}(3\_Z^2-1), x=\mathrm{RootOf}(3\_Z^2-1)\}, \{z=\mathrm{RootOf}(3\_Z^2-1), x=-\mathrm{RootOf}(3\_Z^2-1), y=-\mathrm{RootOf}(3\_Z^2-1)\}, \{z=\mathrm{RootOf}(3\_Z^2-1), x=\mathrm{RootOf}(3\_Z^2-1), y=-\mathrm{RootOf}(3\_Z^2-1)\}, \{y=\mathrm{RootOf}(3\_Z^2-1), z=\mathrm{RootOf}(3\_Z^2-1), x=-\mathrm{RootOf}(3\_Z^2-1)\}, \{x=0, y=1, z=0\}, \{x=0, z=0, y=-1\}, \{x=0, y=0, z=1\}, \{x=0, y=0, z=-1\}\} \quad (8.1.14)$$

From the values of "**s**", you can find which point provides maximum or minimum. To find the maximum or minimum, evaluate the function as follows:

```
> for i from 1 to 10 do eval(x*y*z, s[i]) end do;
```

$$0$$
$$0$$
$$\mathrm{RootOf}(3\_Z^2-1)^3$$
$$\mathrm{RootOf}(3\_Z^2-1)^3$$
$$-\mathrm{RootOf}(3\_Z^2-1)^3$$
$$-\mathrm{RootOf}(3\_Z^2-1)^3$$
$$0$$
$$0$$
$$0$$
$$0 \quad (8.1.15)$$

The real solutions of $3z^2 - 1$ are $\frac{\pm\sqrt{3}}{3}$. Hence, the maximum is $\frac{1}{3}\frac{\sqrt{3}}{3} = \frac{\sqrt{3}}{9}$, which occurs at $\left(\frac{\pm\sqrt{3}}{3}, \frac{\pm\sqrt{3}}{3}, \frac{\pm\sqrt{3}}{3}\right)$ with even number of the negative sign, and the minimum is $-\frac{1}{3}\frac{\sqrt{3}}{9} = -\frac{\sqrt{3}}{9}$, which occurs at $\left(\frac{\pm\sqrt{3}}{3}, \frac{\pm\sqrt{3}}{3}, \frac{\pm\sqrt{3}}{3}\right)$ with odd number of the negative sign.

Example 4. Find the extrema of the function $f(x, y) = x^2 + y^2$ on the curve $x^4 + y^4 = 1$.

```
> f:=x^2+y^2; g:=x^4+y^4-1;
```

$$f := x^2 + y^2$$
$$g := x^4 + y^4 - 1 \quad (8.1.16)$$

```
> extrema(f,g,{x,y},'s');
```

$$\{\min(2\,\mathrm{RootOf}(2\_Z^4-1)^2, -1), \max(2\,\mathrm{RootOf}(2\_Z^4-1)^2, 1)\} \quad (8.1.17)$$

```
> rtf:=solve(2*z^4-1,z);
```

$$rtf := \frac{1}{2}\,2^{(3/4)}, \frac{1}{2}\,\mathrm{I}\,2^{(3/4)}, -\frac{1}{2}\,2^{(3/4)}, -\frac{1}{2}\,\mathrm{I}\,2^{(3/4)} \quad (8.1.18)$$

Recall that the function is non-negative. By the constraint $x^4 + y^4 = 1$, its minimum and maximum must be positive. Then the maximum and the minimum are the following:

```
> Mf:=max(2*(1/2*2^(3/4))^2, 1);
```

$$Mf := \sqrt{2} \tag{8.1.19}$$

```
> mf:=min(2*(1/2*2^(3/4))^2, 1);
```

$$mf := 1 \tag{8.1.20}$$

To find the points where the function has extrema, we do the following:

```
> s;
```

$$\{\{x=0, y=-1\}, \{x=0, y=\mathrm{RootOf}(\_Z^2+1)\}, \{y=0, x=1\}, \{y=0, x=-1\}, \{y=0, x=\mathrm{RootOf}(\_Z^2+1)\}, \{x=0, y=1\}, \{y=\mathrm{RootOf}(2\_Z^4-1), x=-\mathrm{RootOf}(2\_Z^4-1)\}, \{y=\mathrm{RootOf}(2\_Z^4-1), x=\mathrm{RootOf}(2\_Z^4-1)\}\} \tag{8.1.21}$$

Discarding all non-real solutions and evaluating the function on the remained points, the maximum $\sqrt{2}$ is obtained at $(\pm 2^{1/4}, \pm 2^{1/4})$, and the minimum 1 is obtained at (±1, 0) and $(0, \pm 1)$.

15.8.2 Change a constrained problem to an unconstrained one

If the constraint of a constraned problem can be easily re-written to an explicit function, then the problem can be converted to an unconstrained one by eleminating the constraint.

Example 5. Find the point $P(x, y, z)$ closest to the region on the plane $2x + y - z - 5 = 0$.

```
> G:=x^2+y^2+z^2; Cnstr:=2*x+y-z-5=0;
```

$$G := x^2 + y^2 + z^2$$

$$Cnstr := 2x + y - z - 5 = 0 \tag{8.2.1}$$

```
> zf:=solve(Cnstr, z);
```

$$zf := 2x + y - 5 \tag{8.2.2}$$

```
> nG:= subs(z=zf, G);
```

$$nG := x^2 + y^2 + (2x + y - 5)^2 \tag{8.2.3}$$

```
> minimize(nG, location);
```

$$\frac{25}{6}, \left\{\left[\left\{x = \frac{5}{3}, y = \frac{5}{6}\right\}, \frac{25}{6}\right]\right\} \tag{8.2.4}$$

```
> minD:=sqrt(25/6);
```

$$minD := \frac{5}{6}\sqrt{6} \tag{8.2.5}$$

Or, do the following:

```
> dnGx:=diff(nG,x); dnGy:=diff(nG,y);
```

$$dnGx := 10x + 4y - 20$$

$$dnGy := 4y + 4x - 10 \tag{8.2.6}$$

```
> solve({dnGx,dnGy}, {x,y});
```
$$\left\{x=\frac{5}{3}, y=\frac{5}{6}\right\} \tag{8.2.7}$$

```
> minD:=simplify(eval(sqrt(nG), [x=5/3,y=5/6]));
```
$$minD := \frac{5}{6}\sqrt{6} \tag{8.2.8}$$

Example 6. Find the maximum value of xy on the line $x+y=16$.

```
> G:=x*y; Cnstr:=x+y=16;
```
$$G := x\,y$$
$$Cnstr := x+y=16 \tag{8.2.9}$$

```
> extrema(G, Cnstr, {x,y},'s');
```
$$\{64\} \tag{8.2.10}$$

```
> s;
```
$$\{\{x=8, y=8\}\} \tag{8.2.11}$$

```
> yf:=solve(Cnstr, y);
```
$$yf := -x+16 \tag{8.2.12}$$

```
> nG:=subs(y=yf, G);
```
$$nG := x\,(-x+16) \tag{8.2.13}$$

```
> maximize(nG, location);
```
$$64,\ \{[\{x=8\}, 64]\} \tag{8.2.14}$$

```
> y:=16-8;
```
$$y := 8 \tag{8.2.15}$$

```
> y:='y': w:='w':
```

15.8.3 The method of Lagrange multipliers

Assume the goal function is $f(x, y, z)$ and the constraint is $g(x, y, z)=0$. The method of Lagrange multipliers changes the constrained extremal problem to the unconstrained problem:

$$\nabla f - \lambda \nabla g = 0 \ \text{ and } \ g(x, y, z)=0$$

Example 7. Find the maximum value and minimum value of the function $f(x, y)=xy$ with the constraint $\frac{x^2}{8}+\frac{y^2}{2}=1$.

```
> f:=x*y; g:=x^2/8+y^2/2-1;
```
$$f := x\,y$$
$$g := \frac{1}{8}x^2+\frac{1}{2}y^2-1 \tag{8.3.1}$$

```
> gdf:= grad(f,[x,y])-w*grad(g,[x,y]);
```
$$gdf := [y \quad x] - w\left[\frac{1}{4}x \quad y\right] \tag{8.3.2}$$

```
> solve({y-w/4*x=0,x-w*y=0,g=0}, {x,y,w});
```
$$\{y=1, x=2, w=2\},\ \{x=-2, y=-1, w=2\},\ \{y=1, w=-2, x=-2\},\ \{w=-2, y=-1, x=2\} \tag{8.3.3}$$

Hence, the maximum is 2 at (2, 1) and (−2, −1), and the minimum is −2 at (2, −1) and (1,−2).

Example 8. Find the maximum and minimum values of the function $f(x, y) = 3x + 4y$ on the circle $x^2 + y^2 = 1$.

```
> f:=3*x+4*y; g:=x^2+y^2-1;
```

$$f := 3x + 4y$$
$$g := x^2 + y^2 - 1 \qquad (8.3.4)$$

```
> gd:=grad(f,[x,y])-w*grad(g,[x,y]);
```

$$gd := [3 \quad 4] - w\,[2x \quad 2y] \qquad (8.3.5)$$

```
> solve({3-2*w*x, 4-2*w*y, g}, {x,y,w});
```

$$\left\{y = \frac{4}{5}, x = \frac{3}{5}, w = \frac{5}{2}\right\}, \left\{x = \frac{-3}{5}, w = \frac{-5}{2}, y = \frac{-4}{5}\right\} \qquad (8.3.6)$$

```
> eval(f, [x=3/5, y=4/5]); eval(f, [x=-3/5, y=-4/5]);
```

$$5$$
$$-5 \qquad (8.3.7)$$

Hence, the maximum is 5 at $\left(\frac{3}{5}, \frac{4}{5}\right)$, and the minimum is -5 at $(-\frac{3}{5}, -\frac{4}{5})$.

15.8.4 Lagrange multipliers with two constraints

Assume the goal function of an extremal problem is $f(x, y, z)$, which has two constraints $g(x, y, z) = 0$ and $h(x, y, z) = 0$. The method of Lagrange multipliers changes the constrained extremal problem to the following unconstrained problem:

$$\nabla f - \lambda \nabla g - \mu \nabla h = 0, \; g(x, y, z) = 0, \text{ and } \; h(x, y, z) = 0.$$

Example 9. The plane $x + y + z = 1$ cuts the cylinder $x^2 + y^2 = 1$ in an ellipse. Find the points on the ellipse that lie closest to and farthest from the origin.
Method 1: The method of Lagrange multiplier.

```
> f:=x^2+y^2+z^2; g:=x+y+z-1; h:=x^2+y^2-1;
```

$$f := x^2 + y^2 + z^2$$
$$g := x + y + z - 1$$
$$h := x^2 + y^2 - 1 \qquad (8.4.1)$$

```
> Gde:=grad(f, [x,y,z])-l*grad(g,[x,y,z])-m*grad(h,[x,y,z]);
```

$$Gde := [2x \quad 2y \quad 2z] - l\,[1 \quad 1 \quad 1] - m\,[2x \quad 2y \quad 0] \qquad (8.4.2)$$

```
> solve({2*x-l-2*x*m, 2*y-l-2*y*m, 2*z-l, g, h}, {x,y,z,l,m});
```

$$\{z = 0, x = 1, l = 0, y = 0, m = 1\}, \{z = 0, x = 0, l = 0, y = 1, m = 1\}, \{z = -2\,\mathrm{RootOf}(2\_Z^2 - 1) + 1, l = -4\,\mathrm{RootOf}(2\_Z^2 - 1) + 2, m = -2\,\mathrm{RootOf}(2\_Z^2 - 1) + 3, y = \mathrm{RootOf}(2\_Z^2 - 1), x = \mathrm{RootOf}(2\_Z^2 - 1)\} \qquad (8.4.3)$$

```
> solve(2*z^2-1,z);
```

$$\frac{1}{2}\sqrt{2}, -\frac{1}{2}\sqrt{2} \qquad (8.4.4)$$

```
> eval(f, [x=1/2*sqrt(2), y=1/2*sqrt(2), z=-2*1/2*sqrt(2)+1]);
```

$$1+\left(-\sqrt{2}+1\right)^2 \tag{8.4.5}$$

```
> simplify(%);
```

$$4-2\sqrt{2} \tag{8.4.6}$$

```
> evalf(%);
```

$$1.171572876 \tag{8.4.7}$$

```
> eval(f, [x=-sqrt(1/2), y=-sqrt(1/2), z=2*sqrt(1/2)+1]);
```

$$1+\left(\sqrt{2}+1\right)^2 \tag{8.4.8}$$

```
> simplify(%);
```

$$4+2\sqrt{2} \tag{8.4.9}$$

```
> eval(f, [x=1, y=0, z=0]);eval(f, [x=0, y=1, z=0]);
```

$$1$$
$$1 \tag{8.4.10}$$

Hence, the points on the ellipse closest to the origin are (1, 0, 0) and (0, 1, 0). The point on the ellipse farthest from the origin is $\left(\frac{-1}{\sqrt{2}}, \frac{-1}{\sqrt{2}}, 1+\sqrt{2}\right)$.

Method 2. Change the problem to a new one which has only one constraint.
Substitute $x^2+y^2=1$ into $f(x, y, z)$:

```
> nf:=z^2+1;
```

$$nf := z^2+1 \tag{8.4.11}$$

```
> zf:=solve(g, z);
```

$$zf := -x-y+1 \tag{8.4.12}$$

```
> nnf:=subs(z=zf, nf);
```

$$nnf := (-x-y+1)^2+1 \tag{8.4.13}$$

The problem then is to find the extrema of $(-x-y+1)^2+1$ with the constraint $x^2+y^2=1$. Applying the method of Lagrange multiplier, we can solve the problem as follows:

```
> nd:=grad(nnf,[x,y])-m*grad(x^2+y^2, [x,y]);
```

$$nd := [2x+2y-2 \quad 2x+2y-2] - m\,[2x \quad 2y] \tag{8.4.14}$$

```
> solve({2*x+2*y-2-2*x*m, 2*x+2*y-2-2*y*m, x^2+y^2-1},{x,y,m});
```

$$\{y=0, x=1, m=0\}, \{x=0, y=1, m=0\}, \tag{8.4.15}$$
$$\left\{x=\operatorname{RootOf}\left(2\_Z^2-1\right), y=\operatorname{RootOf}\left(2\_Z^2-1\right), m=-2\operatorname{RootOf}\left(2\_Z^2-1\right)+2\right\}$$

The same result is obtained.
Method 3. Directly use MAPLE syntax "**extrema**".

```
> f:=x^2+y^2+z^2; g:=x+y+z-1; h:=x^2+y^2-1;
```

$$f := x^2+y^2+z^2$$
$$g := x+y+z-1$$
$$h := x^2+y^2-1 \tag{8.4.16}$$

```
> extrema(f,{g=0,h=0}, {x,y,z},'s');
```

$$\left\{\min\left(4-4\,\mathrm{RootOf}\left(2\_Z^2-1\right),1\right),\max\left(4-4\,\mathrm{RootOf}\left(2\_Z^2-1\right),1\right)\right\} \tag{8.4.17}$$

```
> solve(2*z^2-1,z);
```

$$\frac{1}{2}\sqrt{2},\ -\frac{1}{2}\sqrt{2} \tag{8.4.18}$$

```
> maxv:=max(4-4*1/2*sqrt(2), 4-4*(-1/2*sqrt(2)),1);
```

$$maxv := 4+2\sqrt{2} \tag{8.4.19}$$

```
> minv:=min(4-4*1/2*sqrt(2), 4-4*(-1/2*sqrt(2)),1);
```

$$minv := 1 \tag{8.4.20}$$

```
> s;
```

$$\left\{\{z=0,x=0,y=1\},\ \left\{z=-2\,\mathrm{RootOf}\left(2\_Z^2-1\right)+1,y=\mathrm{RootOf}\left(2\_Z^2-1\right),x=\mathrm{RootOf}\left(2\_Z^2-1\right)\right\},\ \{z=0,x=1,y=0\}\right\} \tag{8.4.21}$$

```
> Plist:={[x=1,y=0,z=0],[x=0,y=1,z=0], [x=sqrt(1/2),y=sqrt(1/2),
  z=-2*sqrt(1/2)+1],[x=-sqrt(1/2),y=-sqrt(1/2),z=2*sqrt(1/2)+1]}
  ;
```

$$Plist := \left\{[x=1,y=0,z=0],\ [x=0,y=1,z=0],\ \left[x=\frac{1}{2}\sqrt{2},y=\frac{1}{2}\sqrt{2},z=-\sqrt{2}+1\right],\left[x=-\frac{1}{2}\sqrt{2},y=-\frac{1}{2}\sqrt{2},z=\sqrt{2}+1\right]\right\} \tag{8.4.22}$$

```
> for i from 1 to 4 do simplify(eval(f,Plist[i])) end do;
```

$$1$$
$$1$$
$$4-2\sqrt{2}$$
$$4+2\sqrt{2} \tag{8.4.23}$$

The same result is obtained.

Exercises

1. Find the extrema of the function $f(x, y) = 49 - x^2 - y^2$ on the line $x + 3y = 10$.
2. Find the point $P(x, y, z)$ closest to the region from the surface $x^2 + y^2 - z^2 - 1 = 0$.
3. Find the maximum value of $x^2 y$ on the line $x + y = 3$.
4. Find the maximum and minimum values of the function $f(x, y, z) = x - 2y + 5z$ with the constraint $x^2 + y^2 + z^2 = 30$.
5. Find the maximum and minimum values of the function $f(x, y, z) = x + 2\,y + 3\,z$ on the sphere $x^2 + y^2 + z^2 = 25$.
6. Find the macimum value of $f(x, y, z) = xyz$ on the line of the intersection of the two planes $x + y + z = 40$ and $x + y - z = 0$.
7. Find the extreme values of $f(x, y, z) = xy + z^2$ on the circle, in which the plane $y - x = 0$ intersects the sphere $x^2 + y^2 + z^2 = 4$.
8. Find the point closest to the origin on the curve of the intersection of the palne $2\,y + 4\,z = 5$ and the cone $z^2 = 4\,x^2 + 4\,y^2$.

Chapter 16 MULTIPLE INTEGRATION

16.1 Integration in Several Variables

Fubini's Theorem: If $f(x, y)$ is continuous throughout the rectangular region $D = [a, b] \times [c, d]$. Then

$$\iint_D f(x, y)\,\mathrm{d}x\mathrm{d}y = \int_a^b \int_c^d f(x, y)\,\mathrm{d}y\mathrm{d}x = \int_c^d \int_a^b f(x, y)\,\mathrm{d}x\mathrm{d}y$$

16.1.1 Use a double Riemann sum to estimate the double integral

The **double Riemann sum** is $\sum_{i=1}^{n} \sum_{j=1}^{m} f\left(c_i, d_j\right) \Delta x_i \Delta y_j$

Example 1. Assume $R = [-2, 2] \times [-1, 1]$. (a) Use the Riemann sum with $m = 4$ and $n = 4$ to estimate the value of the double integral of $2x^2 + x^2 y$, taking the sample points to be the lower left corners of the subrectangles. (b) Compare the sum to the exact integral value.
(a) Use the Riemann sum to estimate the integral.

```
> f:=(x,y)->2*x^2+x^2*y;x0:=-2;y0:=-1;
```

$$f := (x, y) \rightarrow 2\,x^2 + x^2\,y$$

$$x0 := -2$$

$$y0 := -1 \qquad (1.1.1)$$

```
> dx:=(2-(-2))/4; dy:=(1-(-1))/4;
```

$$dx := 1$$

$$dy := \frac{1}{2} \qquad (1.1.2)$$

```
> LLsum:=sum(sum(f(x0+(i-1)*dx,y0+(j-1)*dy), i =1..4), j=1..4)*
  dx*dy;
```

$$LLsum := 21 \qquad (1.1.3)$$

(b) Compare the estimate to the exact value.

```
> A:=int(int(2*x^2+x^2*y,x=-2..2),y=-1..1);
```

$$A := \frac{64}{3} \qquad (1.1.4)$$

```
> evalf(%);
```

$$21.33333333 \qquad (1.1.5)$$

The error is

```
> err:=abs(LLsum-A);
```

$$err := \frac{1}{3} \qquad (1.1.6)$$

Example 2. Use the Midpoint Rule to estimate the integral in **Example 1**.

```
> Midsum:=sum(sum(f(x0+(2*i-1)/2*dx,y0+(2*j-1)/2*dy) , i =1..4),
  j=1..4)*dx*dy;
```

$$Midsum := 20 \tag{1.1.7}$$

16.1.2 Use iterated integrals to evaluate double integrals over a rectangular region

Example 3. Calculate the double integral $\int_1^2 \int_0^3 (2y^2 - 3xy^3)\, dx dy$.

```
> IntV:=int(int(2*y^2-3*x*y^3, x=0..3),y=1..2);
```

$$IntV := \frac{-293}{8} \tag{1.2.1}$$

Example 4. Calculate the double integral $\int_0^2 \int_{-1}^1 \left(1 - 6x^2 y\right) dx dy$.

```
> IntV:=int(int(1-6*x^2*y, x=-1..1),y=0..2);
```

$$IntV := -4 \tag{1.2.2}$$

Example 5. Compute the integral of $f(x, y) = xy$ over the rectangular region $[0, 1] \times [3, 4]$.

```
> IntV:=int(int(x*y,x=0..1),y=3..4);
```

$$IntV := \frac{7}{4} \tag{1.2.3}$$

16.1.3 Reverse the order of integration

Example 6. Verify that $\int_0^1 \int_2^4 x^2 y^3 dy dx = \int_2^4 \int_0^1 x^2 y^3 dx dy$.

```
> I1:=int(int(x^2*y^3,y=2..4),x=0..1);
```

$$I1 := 20 \tag{1.3.1}$$

```
> I2:=int(int(x^2*y^3,x=0..1),y=2..4);
```

$$I2 := 20 \tag{1.3.2}$$

Exercises

1. Assume $R = [1, 2] \times [0, 1]$. (a) Use the Riemann sum with $m = 8$ and $n = 4$ to estimate the value of the double integral of $x^2 y$, taking the sample points to be the lower left corners of the subrectangles. (b) Compare the sum to the exact integral value.
2. Calculate the double integral $\iint_R 4 dA$, where $R = [2, 4] \times [1, 3]$.
3. Calculate the double integral $\iint_R \sin(x) dA$, where $R = [0, 2\pi] \times [0, 2\pi]$.
4. Calculate the double integral $\int_0^2 \int_0^1 (x + xy^3)\, dx dy$.
5. Calculate the double integral $\int_1^2 \int_0^4 \frac{1}{x + y} dx dy$.

6. Verify that $\int_0^1 \int_0^3 xy^2 \mathrm{d}y\mathrm{d}x = \int_0^3 \int_0^1 xy^2 \mathrm{d}x\mathrm{d}y$.

16.2 Double Integrals over More General Regions

Double integrals over general regions: If $f(x, y)$ is continuous on a region D defined by $a \le x \le b$, $c(x) \le y \le d(x)$, then

$$\iint_D f(x, y)\mathrm{d}x\mathrm{d}y = \int_a^b \int_{c(x)}^{d(x)} f(x, y)\mathrm{d}y\mathrm{d}x$$

and if $f(x, y)$ is continuous on a region D defined by $a(y) \le x \le b(y)$, $c \le y \le d$, then

$$\iint_D f(x, y)\mathrm{d}x\mathrm{d}y = \int_c^d \int_{a(y)}^{b(y)} f(x, y)\mathrm{d}x\mathrm{d}y$$

16.2.1 Use iterated integrals to evaluate double integrals over general regions

Example 1. Evaluate the double integral of $x^2 + y^2$ over the region bounded by $y = 2x$ and $y = x^2$.
Step1. Plot the region over which the integral needs to be found. If it is necessary, find the intersections.

```
> plot({2*x,x^2},x=-1..3);
```




```
> solve({y=2*x,y=x^2},{x,y});
```

$$\{x = 0, y = 0\}, \{x = 2, y = 4\} \quad (2.1.1)$$

Step 2. Set the iterated integral to evaluate it.

```
> Intv:=int(int(x^2+y^2,y=x^2..2*x),x=0..2);
```

$$Intv := \frac{216}{35} \quad (2.1.2)$$

Example 2. Evaluate the double integral of xy over the region bounded by $y = x - 1$ and $y^2 = 2x + 6$.

Step1. Graph the region and obtained the intersections.

```
> implicitplot({y=x-1,y^2=2*x+6},x=-5..5, y=-5..5);
```



Find the intersections.

```
> solve({y=x-1,y^2=2*x+6},{x,y});
```

$$\{y=-2, x=-1\}, \{y=4, x=5\} \tag{2.1.3}$$

We compute the iterated integral first for *x* then for *y*.
Step 2. Find the lower limit and upper limit for the inner integral.

```
> solve(y=x-1,x);solve(y^2=2*x+6,x);
```

$$y+1$$

$$\frac{1}{2}y^2-3 \tag{2.1.4}$$

Step 3. Use iterated integral to evaluate it.

```
> int(int(x*y,x=1/2*y^2-3..y+1),y=-2..4);
```

$$36 \tag{2.1.5}$$

Note: Selecting the right order of an iterated integral makes the evaluation of the integral easy. For example, if you use the revised order of the integral above, you have to do the following:

```
> Ione:=int(int(x*y,y=-sqrt(2*x+6)..sqrt(2*x+6)), x=-3..-1);
```

$$Ione := 0 \tag{2.1.6}$$

```
> Itwo:=int(int(x*y, y=x-1..sqrt(2*x+6)), x=-1..5);
```

$$Itwo := 36 \tag{2.1.7}$$

```
> Iwhole:=Ione+Itwo;
```

$$Iwhole := 36 \tag{2.1.8}$$

Example 3. Find the volume of the solid between the paraboloids $z=2x^2+y^2$ and $z=8-x^2-2y^2$ and inside the cylinder $x^2+y^2=1$.

```
> z1:=2*x^2-y^2:z2:=8-x^2-2*y^2:
```

The graphs of the two surfaces are shown in the following figure. (This step is OPTIONAL.)

```
> plot3d({z1,z2},x=-1..1,y=-1..1, axes=frame);
```



The integral is over the region $R = \{(x, y); x^2 + y^2 \leq 1\}$.

```
> implicitplot(x^2+y^2=1, x=-1..1, y=-1..1);
```



Assume that the iterated integral is taken for x first. Hence, we need to solve the equation for x.

```
> solve(x^2+y^2=1, x);
```

$$\sqrt{-y^2+1},\ -\sqrt{-y^2+1} \tag{2.1.9}$$

Finally, change the double integral to the following iterated integral:

```
> int(int(z2-z1,x=-sqrt(1-y^2)..sqrt(1-y^2)),y=-1..1);
```

$$7\,\pi \tag{2.1.10}$$

16.2.2 Reversing the order of integration

Example 4. Reverse the order of the integral $\int_0^1 \int_2^{4-2x} \mathrm{d}y\mathrm{d}x$.

Step 1. Plot the region for the integral.

```
> plot({2, 4-2*x}, x=0..1, y=0..4);
```



Step 2. Find the function $x=a(y), x=b(y)$ for the boundary curves.

```
> xb:= solve(y=4-2*x, x);
```

$$xb := -\frac{1}{2}\,y+2 \tag{2.2.1}$$

Step 3. Reverse the order of the integral.
Note: If you want to display the integral form, but not the answer of the integral, use "**Int**" in the syntax instead of "**int**".

```
> Integral:=Int(Int(1, x=0..xb), y=2..4);
```

$$Integral := \int_2^4 \int_0^{-\frac{1}{2}y+2} 1\,\mathrm{d}x\,\mathrm{d}y \tag{2.2.2}$$

Example 5. Reverse the order of the integral $\int_0^2 \int_{-\sqrt{4-y^2}}^{\sqrt{4-y^2}} 3y\mathrm{d}x\mathrm{d}y$.

```
> implicitplot({x=-sqrt(4-y^2),x=sqrt(4-y^2)},x=-2..2,y=0..2,
scaling=constrained);
```




```
> solve(x^2+y^2=4,y);
```

$$\sqrt{-x^2+4},\ -\sqrt{-x^2+4} \qquad (2.2.3)$$

```
> Intform:=Int(Int(3*y, y=0..sqrt(4-x^2)),x=-2..2);
```

$$Intform := \int_{-2}^{2}\int_{0}^{\sqrt{-x^2+4}} 3\,y\,\mathrm{d}y\,\mathrm{d}x \qquad (2.2.4)$$

Example 6. Reverse the order of the integral $\int_{0}^{\pi}\int_{x}^{\pi} \frac{\sin(x)}{x}\,\mathrm{d}y\mathrm{d}x$, and then evaluate the integral.

```
> plot({x,Pi}, x=0..Pi);
```




```
> Iform:=Int(Int(sin(x)/x, x=0..y), y=0..Pi);
```

$$Iform := \int_{0}^{\pi}\int_{0}^{y} \frac{\sin(x)}{x}\,\mathrm{d}x\,\mathrm{d}y \qquad (2.2.5)$$

```
> Ival:=int(int(sin(x)/x, x=0..y), y=0..Pi);
```

$$Ival := \frac{1}{2}\sqrt{\pi}\left(-\frac{4}{\sqrt{\pi}} + 2\sqrt{\pi}\ \mathrm{Si}(\pi)\right) \qquad (2.2.6)$$

```
> evalf(%);
```

$$3.818031839 \tag{2.2.7}$$

The same answer can be obtained by the original iterated integral.

```
> int(int(sin(x)/x, y=x..Pi),x=0..Pi);
```

$$-2 + \pi \operatorname{Si}(\pi) \tag{2.2.8}$$

```
> evalf(%);
```

$$3.818031838 \tag{2.2.9}$$

16.2.3 Area of bounded regions in the plane and average values

(1) The **area** of a closed, bounded place region R is $\iint_R \mathrm{d}x\mathrm{d}y$.

(2) The **average value** of f over the region R is

$$\frac{\iint_R f(x, y)\mathrm{d}x\mathrm{d}y}{\iint_R \mathrm{d}x\mathrm{d}y}$$

Example 7. Find the area of the region enclosed by the parabola $y = x^2$ and the line $y = x + 2$.

Step 1. Draw the graph of the two curves.

```
> plot({x^2, x+2}, x=-2..3);
```



Step 2. Find the intersections.

```
> solve(x^2=x+2, x);
```

$$2, -1 \tag{2.3.1}$$

Step 3, Evaluate the iterated integral.

```
> IntV:=int(int(1,y=x^2..x+2), x=-1..2);
```

$$IntV := \frac{9}{2} \tag{2.3.2}$$

Example 8. Find the area of the region enclosed by the parabolas $x = y^2 - 1$ and $x = 2y^2 - 2$.

```
> plot({y^2-1,2*y^2-2}, y=-2..2);
```




```
> solve(y^2-1=2*y^2-2, y);
```

$$-1, 1 \tag{2.3.3}$$

```
> IntV:=int(int(1,x=2*y^2-2..y^2-1), y=-1..1);
```

$$IntV := \frac{4}{3} \tag{2.3.4}$$

Example 9. Find the average value of $f(x, y) = x\cos(xy)$ over the region R: $0 \le x \le \pi$, $0 \le y \le 1$.

```
> NumI:=int(int(x*cos(x*y), x=0..Pi), y=0..1);
```

$$NumI := 2 \tag{2.3.5}$$

```
> Ar:=int(int(1, x=0..Pi), y=0..1);
```

$$Ar := \pi \tag{2.3.6}$$

```
> AvrV:=NumI/Ar;
```

$$AvrV := \frac{2}{\pi} \tag{2.3.7}$$

Exercises

1. Compute the integral of $f(x, y) = \left(x^2 y - 2xy\right)$ over the rectangular region $[0, 3] \times [-2, 0]$.
2. Evaluate the double integral of $x - \sqrt{y}$ over the triangular region cut from the first quadrant by the line $x + y = 1$.
3. Evaluate the double integral of $x + 4$ over the region bounded by $y = x$ and $y = 2 - x^2$.
4. Reverse the order of the integral $\int_0^1 \int_y^{\sqrt{y}} \mathrm{d}x\mathrm{d}y$.
5. Reverse the order of the integral $\int_0^1 \int_{-\sqrt{1-x^2}}^{\sqrt{1-x^2}} 3y\mathrm{d}y\mathrm{d}x$.
6. Find the average height of the cone $z = \sqrt{x^2 + y^2}$ above the disk $x^2 + y^2 \le a^2$ in the xy-plane.

16.3 Triple Integrals

16.3.1 Triple integrals over a rectangular solid

Example 1. Evaluate the triple integral of $\iiint_R x^2 + y\,z\,\mathrm{d}x\mathrm{d}y\mathrm{d}z$, where $R = [0, 2; -3, 0; -1, 1]$.

```
> Ival:=int(int(int(x^2+y*z,x=0..2),y=-3..0),z=-1..1);
```

$$Ival := 16 \quad (3.1.1)$$

Example 2. Evaluate the triple integral of $\iiint_R x^2 + y^2 + z^2\,\mathrm{d}x\mathrm{d}y\mathrm{d}z$, where $R = [0, 1; 0, 1; 0, 1]$.

```
> Ival:=int(int(int(x^2+y^2+z^2,x=0..1), y=0..1), z=0..1);
```

$$Ival := 1 \quad (3.1.2)$$

16.3.2 Evaluate triple iterated integrals

Example 3. Evaluate the iterated integral $\int_0^1 \int_x^1 \int_0^{1-y} xyz\,dz\,dy\,dx$.

```
> int(int(int(x*y*z, z=0..1-y),y=x..1),x=0..1);
```

$$\frac{1}{240} \tag{3.2.1}$$

Example 4. Evaluate the iterated integral $\int_0^1 \int_0^{2-x} \int_0^{2-x-y} dz\,dy\,dx$.

```
> int(int(int(1, z=0..2-x-y),y=0..2-x),x=0..1);
```

$$\frac{7}{6} \tag{3.2.2}$$

16.3.3 Triple integrals over a general solid

Example 5. Find the integral $\iiint_E (x+2y)\,dx\,dy\,dz$, where E lies under the plane $z = x + 2y$ and above the region in the xy-plane bounded by the curves $y = x^2$, $y = 0$, and $x = 1$.

Draw the graph of the solid E.

```
> with (plots):
Warning, the name changecoords has been redefined
> implicitplot3d({z=x+2*y, y=x^2,y=0,x=1},x=0..1,y=0..1,z=0..3,
  axes=normal);
```



You can draw the solid graph in an alternate way.

```
> psf1:=piecewise(y>0 and y<x^2,x+2*y);
```

$$psf1 := \begin{cases} x+2\,y & 0 < y \text{ and } y < x^2 \\ 0 & otherwise \end{cases} \quad (3.3.1)$$

```
> plot3d({0,psf1},x=0..1,y=0..x^2, axes=normal);
```



Change it to iterated integral.

```
> plot(x^2, x=0..1);
```




```
> Iva:=int(int(int(x+2*y,z=0..x+2*y),y=0..x^2),x=0..1);
```

$$Iva := \frac{76}{105} \quad (3.3.2)$$

Example 6. Find the volume of the solid bounded by the paraboloid $x = 4y^2 + 4z^2$ and the plane $x = 4$.

Step 1. Plot the solid.

```
> implicitplot3d({x=4*y^2+4*z^4,x=4},x=0..4,y=-1..1,z=-1..1,
  axes=frame, scaling=constrained);
```



You can consider x as a function of y and z, so plot it by applying the (explicit) **"plot3d"**.

```
> plot3d({4*y^2+4*z^2,4},y=-1..1, z=-1..1,view=0..4, axes=frame,
  scaling=constrained);
```



```
> Ceq:=4*y^2+4*z^2=4;
```

$$Ceq := 4\,y^2 + 4\,z^2 = 4 \tag{3.3.3}$$

```
> implicitplot(Ceq, z=-2..2, y=-2..2, scaling=constrained);
```




```
> solve(Ceq, z);
```

$$\sqrt{1-y^2},\ -\sqrt{1-y^2} \tag{3.3.4}$$

Step 2. Compute the iterated integral.

```
> V:=int(int(int(1,x=4*y^2+4*z^2..4),z=-sqrt(1-y^2)..sqrt(1-y^2)
  ),y=-1..1);
```

$$V := 2\,\pi \tag{3.3.5}$$

Example 7. Rewrite the iterated integral $\int_0^1 \int_{\sqrt{x}}^1 \int_0^{1-y} f(x, y, z)\mathrm{d}z\mathrm{d}y\mathrm{d}x$ in other orders.

Plot the solid which the integral takes over.

```
> implicitplot3d({z=1-y,y=sqrt(x),x=1},x=0..1,y=0..1,z=0..1,
  axes=frame);
```



```
> plot(sqrt(x), x=0..1);
```



(a) Change the integral to the iterated integral with the order of $\{z, x, y\}$.

```
> Int(Int(Int(f(x,y,z),z=0..1-y),y=sqrt(x)..1),x=0..1)=int(int
  (int(f(x,y,z),z=0..1-y),x=0..y^2),y=0..1);
```

$$\int_0^1 \int_{\sqrt{x}}^1 \int_0^{1-y} f(x, y, z)\, \mathrm{d}z\, \mathrm{d}y\, \mathrm{d}x = \int_0^1 \int_0^{y^2} \int_0^{1-y} f(x, y, z)\, \mathrm{d}z\, \mathrm{d}x\, \mathrm{d}y \tag{3.3.6}$$

To verify the correction of the change, we take $f = xy^2z^3$ as a test function.

```
> func:=x*y^2*z^3;
```

$$func := x\, y^2\, z^3 \tag{3.3.7}$$

The original integral with the order $\{z, y, x\}$ is evaluated by the following:

```
> Int(Int(Int(f(x,y,z),z=0..1-y),y=sqrt(x)..1),x=0..1)=int(int
  (int(func,z=0..1-y),y=sqrt(x)..1),x=0..1);
```

$$\int_0^1 \int_{\sqrt{x}}^1 \int_0^{1-y} f(x, y, z)\, \mathrm{d}z\, \mathrm{d}y\, \mathrm{d}x = \frac{1}{18480} \tag{3.3.8}$$

After it is changed to the order {z, y, x}, the integral is evaluated by the following:

```
> Int(Int(Int(func,z=0..1-y),x=0..y^2),y=0..1)=int(int(int(func,
  z=0..1-y),x=0..y^2),y=0..1);
```

$$\int_0^1 \int_0^{y^2} \int_0^{1-y} x\, y^2\, z^3\, \mathrm{d}z\, \mathrm{d}x\, \mathrm{d}y = \frac{1}{18480} \tag{3.3.9}$$

They have the same value.

(b) Change to the order $\{x, y, z\}$.

First, consider the x-direction, The x-coordinate changes from 0 to y^2. Then consider the changes of the y-coordinate provided that x is fixed. It shows that the change for y is from 0 to $1 - z$. Hence, we have the following:

```
> Int(Int(Int(f(x,y,z),z = 0..1-y),y=sqrt(x)..1),x =0 ..1)=int
  (int(int(f(x,y,z),x=0..y^2),y=0..1-z),z=0..1);
```

$$\int_0^1 \int_{\sqrt{x}}^1 \int_0^{1-y} f(x, y, z)\, \mathrm{d}z\, \mathrm{d}y\, \mathrm{d}x = \int_0^1 \int_0^{1-z} \int_0^{y^2} f(x, y, z)\, \mathrm{d}x\, \mathrm{d}y\, \mathrm{d}z \tag{3.3.10}$$

Use the test function $f(x, y, z) = xy^2z^3$ to test the correction of the reversing.

```
> Int(Int(Int(func,x=0..y^2),y=0..1-z),z=0..1)=int(int(int(func,
  x=0..y^2),y=0..1-z),z=0..1);
```

$$\int_0^1 \int_0^{1-z} \int_0^{y^2} x\, y^2 z^3 \,\mathrm{d}x\,\mathrm{d}y\,\mathrm{d}z = \frac{1}{18480} \tag{3.3.11}$$

The other orders are {*x, z, y*}, {*y, x, z*}, {*y, z, x*} respectively.

(c) Change to the order {*x, z, y*}:

```
> Int(Int(Int(f(x,y,z),z=0..1-y),y=sqrt(x)..1),x =0..1)=int(int
  (int(f(x,y,z),x=0..y^2),z=0..1-y),y=0..1);
```

$$\int_0^1 \int_{\sqrt{x}}^1 \int_0^{1-y} f(x, y, z)\,\mathrm{d}z\,\mathrm{d}y\,\mathrm{d}x = \int_0^1 \int_0^{1-y} \int_0^{y^2} f(x, y, z)\,\mathrm{d}x\,\mathrm{d}z\,\mathrm{d}y \tag{3.3.12}$$

Check the correction of the reversing:

```
> Int(Int(Int(func, x=0..y^2),z=0..1-y),y=0..1)=int(int(int
  (func, x=0..y^2),z=0..1-y),y=0..1);
```

$$\int_0^1 \int_0^{1-y} \int_0^{y^2} x\, y^2 z^3 \,\mathrm{d}x\,\mathrm{d}z\,\mathrm{d}y = \frac{1}{18480} \tag{3.3.13}$$

(d) Change to the order {*y, x, z*}:

In the y-direction, we first meet $y = \sqrt{x}$ and then the plane $y = 1 - z$. To find the change of x, for a randomly fixed y, combining $1 - z = y$ and $y = \sqrt{x}$, we have $1 - z = \sqrt{x}$. Solving it for z, we have $x = (1 - z)^2$. Hence, we have

```
> Int(Int(Int(f(x,y,z),z=0..1-y),y= sqrt(x)..1),x=0..1)=int(int
  (int(f(x,y,z),y=sqrt(x)..1-z),x=0..1),z=0..1);
```

$$\int_0^1 \int_{\sqrt{x}}^1 \int_0^{1-y} f(x, y, z)\,\mathrm{d}z\,\mathrm{d}y\,\mathrm{d}x = \int_0^1 \int_0^{(1-z)^2} \int_{\sqrt{x}}^{1-z} f(x, y, z)\,\mathrm{d}y\,\mathrm{d}x\,\mathrm{d}z \tag{3.3.14}$$

Check the correction of the reversing:

```
> Int(Int(Int(func,y=sqrt(x)..1-z),x=0..(1-z)^2),z=0..1)=int(int
  (int(func,y=sqrt(x)..1-z),x=0..(1-z)^2),z=0..1);
```

$$\int_0^1 \int_0^{(1-z)^2} \int_{\sqrt{x}}^{1-z} x\, y^2 z^3 \,\mathrm{d}y\,\mathrm{d}x\,\mathrm{d}z = \frac{1}{18480} \tag{3.3.15}$$

(e) Change to the order {*y, z, x*}:

In the y-direction, we first meet $y = \sqrt{x}$ and then the plane $y = 1 - z$. For a randomly fixed y, combining $1 - z = y$ and $y = \sqrt{x}$, we have $1 - z = \sqrt{x}$. Solve it for x, we have $z = 1 - \sqrt{x}$.

```
> Int(Int(Int(f(x,y,z),z=0..1-y),y=sqrt(x)..1),x = 0..1)=int(int
  (int(f(x,y,z),y=sqrt(x)..1-z),z=0..1-sqrt(x)),x=0..1);
```

$$\int_0^1 \int_{\sqrt{x}}^1 \int_0^{1-y} f(x, y, z)\,\mathrm{d}z\,\mathrm{d}y\,\mathrm{d}x = \int_0^1 \int_0^{1-\sqrt{x}} \int_{\sqrt{x}}^{1-z} f(x, y, z)\,\mathrm{d}y\,\mathrm{d}z\,\mathrm{d}x \tag{3.3.16}$$

Check the correction of the reversing:

```
> Int((Int(Int(func,y=sqrt(x)..1-z),z=0..1-sqrt(x))),x=0..1)=int
  ((int(int(func,y=sqrt(x)..1-z),z=0..1-sqrt(x))),x=0..1);
```

$$\int_0^1 \int_0^{1-\sqrt{x}} \int_{\sqrt{x}}^{1-z} x\, y^2 z^3 \,\mathrm{d}y\, \mathrm{d}z\, \mathrm{d}x = \frac{1}{18480} \tag{3.3.17}$$

16.3.4 Masses and moments in three dimensions

(1) **Mass:**

$$M = \iiint_R \delta(x, y, z)\mathrm{d}x\mathrm{d}y\mathrm{d}z$$

where $\delta(x, y, z)$ is the material's density function over the solid R.

(2) **The first moments:**

$$M_{xy} = \iiint_R z\delta(x, y, z)\mathrm{d}x\mathrm{d}y\mathrm{d}z,\ M_{yz} = \iiint_R x\delta(x, y, z)\mathrm{d}x\mathrm{d}y\mathrm{d}z,\ M_{xz} = \iiint_R y\delta(x, y, z)\mathrm{d}x\mathrm{d}y\mathrm{d}z$$

(3) **Center of mass:**

$$x = \frac{M_{yz}}{M},\ y = \frac{M_{xz}}{M},\ z = \frac{M_{xy}}{M}$$

(4) **The second moment (moment of inertia).**

about the x-axis:

$$I_x = \iiint_R \left(y^2 + z^2\right) \delta(x, y, z)\mathrm{d}x\mathrm{d}y\mathrm{d}z$$

about the y-axis:

$$I_y = \iiint_R \left(x^2 + z^2\right) \delta(x, y, z)\mathrm{d}x\mathrm{d}y\mathrm{d}z$$

about the z-axis:

$$I_z = \iiint_R \left(x^2 + y^2\right) \delta(x, y, z)\mathrm{d}x\mathrm{d}y\mathrm{d}z$$

(5) **The second moment about a line L:**

$$I_L = \iiint_R r^2(x, y, z)\delta(x, y, z)\mathrm{d}x\mathrm{d}y$$

where $r(x, y, z)$ is the distance from (x, y, z) to L.

(6) **Radius of gyration about a line L:**

$$R_L = \sqrt{\frac{I_L}{M}}$$

Example 8. Find the center of mass of a solid of constant density δ bounded below by the disk $x^2 + y^2 \le 4$ in the plane $z = 0$ and above by the paraboloid $z = 4 - x^2 - y^2$.

```
> plot3d(4-x^2-y^2, x=-2..2, y=-2..2,view=0..4, axes=frame,
  scaling = constrained);
```




```
> solve(x^2+y^2=4, y);
```

$$\sqrt{-x^2+4}, -\sqrt{-x^2+4} \quad (3.4.1)$$

```
> M:=int(int(int(1, z=0..4-x^2-y^2), y=-sqrt(4-x^2)..sqrt(4-x^2)
  ),x=-2..2);
```

$$M := 8\,\pi \quad (3.4.2)$$

```
> Myz:=int(int(int(x, z=0..4-x^2-y^2), y=-sqrt(4-x^2)..sqrt(4-
  x^2)),x=-2..2);
```

$$Myz := 0 \quad (3.4.3)$$

```
> Mxz:=int(int(int(y, z=0..4-x^2-y^2), y=-sqrt(4-x^2)..sqrt(4-
  x^2)),x=-2..2);
```

$$Mxz := 0 \quad (3.4.4)$$

```
> Mxy:=int(int(int(z, z=0..4-x^2-y^2), y=-sqrt(4-x^2)..sqrt(4-
  x^2)),x=-2..2);
```

$$Mxy := \frac{32}{3}\,\pi \quad (3.4.5)$$

```
> cntr:=[Myz/M, Mxz/M, Mxy/M];
```

$$cntr := \left[0, 0, \frac{4}{3}\right] \quad (3.4.6)$$

Example 9. Find the center of mass of the first octant of the unit sphere, assuming the density $\delta(x, y, z) = y$.

```
> plot3d(sqrt(1-x^2-y^2), x=0..1, y=0..1, axes=normal, scaling =
  constrained);
```




```
> M:=int(int(int(y, z=0..sqrt(1-x^2-y^2)), y=0..sqrt(1-x^2)), x=
  0..1);
```

$$M := \frac{1}{16}\pi \quad (3.4.7)$$

```
> Mxy:=int(int(int(z*y, z=0..sqrt(1-x^2-y^2)), y=0..sqrt(1-x^2))
  , x=0..1);
```

$$Mxy := \frac{1}{15} \quad (3.4.8)$$

```
> Mxz:=int(int(int(y*y, z=0..sqrt(1-x^2-y^2)), y=0..sqrt(1-x^2))
  , x=0..1);
```

$$Mxz := \frac{1}{30}\pi \quad (3.4.9)$$

```
> Myz:=int(int(int(x*y, z=0..sqrt(1-x^2-y^2)), y=0..sqrt(1-x^2))
  , x=0..1);
```

$$Myz := \frac{1}{15} \quad (3.4.10)$$

```
> cntr:=[Myz/M,Mxz/M, Mxy/M];
```

$$cntr := \left[\frac{16}{15}\,\frac{1}{\pi}, \frac{8}{15}, \frac{16}{15}\,\frac{1}{\pi}\right] \quad (3.4.11)$$

Example 10. Find the center of mass of the cylinder $x^2 + y^2 = 1$ for $0 \le z \le 1$, assuming a mass density $\delta(x, y, z) = x^2 + y^2$.

```
> dens:=x^2+y^2;
```

$$dens := x^2 + y^2 \quad (3.4.12)$$

```
> M:=int(int(int(dens, z=0..1), y=-sqrt(1-x^2)..sqrt(1-x^2)), x=
-1..1);
```

$$M := \frac{1}{2}\pi \tag{3.4.13}$$

```
> Mxy:=int(int(int(z*dens, z=0..1), y=-sqrt(1-x^2)..sqrt(1-x^2))
, x=-1..1);
```

$$Mxy := \frac{1}{4}\pi \tag{3.4.14}$$

```
> Mxz:=int(int(int(y*dens, z=0..1), y=-sqrt(1-x^2)..sqrt(1-x^2))
, x=-1..1);
```

$$Mxz := 0 \tag{3.4.15}$$

```
> Myz:=int(int(int(y*dens, z=0..1), y=-sqrt(1-x^2)..sqrt(1-x^2))
, x=-1..1);
```

$$Myz := 0 \tag{3.4.16}$$

```
> cntr:=[Myz/M,Mxz/M,Mxy/M];
```

$$cntr := \left[0, 0, \frac{1}{2}\right] \tag{3.4.17}$$

Exercises

1. Compute the integral of $f(x, y, z) = x^2 + y^2 + z^2$ over the solid $0 \le x \le 2, 0 \le y \le 1, 1 \le z \le 2$.
2. Evaluate the iterated integral $\int_0^{\sqrt{2}} \int_0^{3y} \int_{x^2+3y^2}^{8-x^2-y^2} \mathrm{d}z\mathrm{d}x\mathrm{d}y$.
3. Find the volume of the region in the first octant bounded by the coordinate plane, the plane $y + z = 2$, and the cylinder $x = 4 - y^2$.
4. Find the volume of the wedge cut from the cylinder $x^2 + y^2 = 1$ by the planes $z = -y$ and $z = 0$.
5. Find the volume of the solid bounded by $y = x^2$, $x = y^2$, $z = x + y + 5$, and $z = 0$.
6. Find the integral $\iiint_E y\,\mathrm{d}V$, where V is the solid above $z = x^2 + y^2$, below $z = 5$, and bounded by $y = 0$, $y = 1$.
7. Find the mass and the mass center of the solid bounded by the planes $x + y + z = 1$, $x = 0$, $y = 0$, and $z = 0$, assuming a mass density of $\delta(x, y, z) = \sqrt{z}$.
8. Find the mass and the mass center of the cylinder $x^2 + y^2 = 1$, for $0 \le z \le 1$, assuming a mass density of $\delta(x, y, z) = x^2 + y^2$.

16.4 Integrals in Polar, Cylindrical, and Spherical Coordinates

16.4.1 Double Integrals in Polar Coordinates

(1) **Change Cartesian integrals into polar integrals:**

$$\iint_R f(x, y)\,\mathrm{d}x\mathrm{d}y = \iint_G f(r\cos(\theta), r\sin(\theta))\,r\mathrm{d}r\mathrm{d}\theta$$

(2) **Area in Polar Coordinates:**

$$A = \iint_G r\mathrm{d}r\mathrm{d}\theta$$

Example 1. Find the volume of the solid bounded by the plane $z = 0$ and the paraboloid $z = 1 - x^2 - y^2$.

The graph of the solid is the following:

```
> plot3d(1-x^2-y^2, x=-1..1, y=-1..1, view=0..1, axes=frame);
```



The integral is taken over the region $x^2 + y^2 \le 1$. We compute it using the polar coordinates.

```
> r:='r': theta:='theta':
> polarf:=subs({x=r*cos(theta), y=r*sin(theta)}, 1-x^2-y^2);
```

$$polarf := 1 - r^2\cos(\theta)^2 - r^2\sin(\theta)^2 \qquad (4.1.1)$$

```
> sf:=simplify(polarf);
```

$$sf := 1 - r^2 \qquad (4.1.2)$$

```
> volume:=int(int(r*sf,r=0..1),theta=0..2*Pi);
```

$$volume := \frac{1}{2}\pi \qquad (4.1.3)$$

Example 2. Find the volume above the cone $z = \sqrt{x^2 + y^2}$ and below the sphere $x^2 + y^2 + z^2 = 1$.

The graph of the solid is the following:

```
> plot3d({sqrt(x^2+y^2),sqrt(1-(x^2+y^2))},x=-1..1,y=-1..1,view=
  -1..1,axes=frame,scaling=constrained);
```



The region of the integral is taken over the region bounded by the *xy*-projection of the intersection of two surfaces $z=\sqrt{x^2+y^2}$ and $x^2+y^2+z^2=1$. Then it has the equation $x^2+y^2=1-x^2-y^2$, which yields $x^2+y^2=\frac{1}{2}$. Hence, the graph of the region is

```
> implicitplot(x^2+y^2=1/2,x=-1..1,y=-1..1,scaling=constrained);
```



The volume can be computed in the polar coordinates, where the functions of the surfaces are the following:

```
> assume(r>0): assume(theta, real):
> z1:=simplify(subs({x=r*cos(theta), y=r*sin(theta)}, sqrt(x^2+
  y^2)));
```

$$z1 := r\sim \tag{4.1.4}$$

```
> z2:=simplify(subs({x=r*cos(theta), y=r*sin(theta)},sqrt(1-x^2-
  y^2)));
```

$$z2 := \sqrt{1-r\sim^2} \tag{4.1.5}$$

```
> volume:=int(int((z2-z1)*r,r=0..sqrt(1/2)),theta=0..2*Pi);
```

$$volume := \frac{2}{3}\pi - \frac{1}{3}\sqrt{2}\,\pi \tag{4.1.6}$$

Example 3. Find the area that lies inside the cardioid $r = 1 + \cos(\theta)$ and outside the circle $r = 1$.

```
> polarplot({1+cos(theta),1}, theta=-Pi..Pi);
```




```
> Ar:=2*int(int(r, r=1..1+cos(theta)), theta=0..Pi/2);
```

$$Ar := 2 + \frac{1}{4}\pi \tag{4.1.7}$$

Example 4. Find the area enclosed by the lemniscate $r^2 = 4\cos(2\theta)$.

```
> polarplot({-2*sqrt(cos(2*theta)), 2*sqrt(cos(2*theta))},
  theta=-Pi/2..Pi/2);
```




```
> solve(cos(2*theta),theta);
```

$$\frac{1}{4}\pi \tag{4.1.8}$$

```
> Ar:=4*int(int(r,r=0..2*sqrt(cos(2*theta))), theta=0..Pi/4);
```

$$Ar := 4 \tag{4.1.9}$$

16.4.2 Triple integrals in cylindrical coordinates

(1) The **cylindrical coordinates** of a point P in 3-D are (r, θ, z). The converting formulas are the followings:

Cylindrical to rectangular:

$$x = r\cos(\theta)$$
$$y = r\sin(\theta)$$

$$z = z$$

Rectangular to cylindrical:

$$r = \sqrt{x^2 + y^2}$$

$$\tan(\theta) = \frac{y}{x}$$

$$z = z$$

(2) **Triple integrals in cylindrical coordinates:** For a region R of the form

$$\theta_1 \le \theta \le \theta_2,\ a(\theta) \le r \le b(\theta),\ z_1(r, \theta) \le z \le z_2(r, \theta)$$

the triple integral is

$$\iiint_R f(x, y, z)\mathrm{dV} = \int_{\theta_1}^{\theta_2}\int_{a(\theta)}^{b(\theta)}\int_{z_1(r,\theta)}^{z_2(r,\theta)} f(r\cos(\theta), r\sin(\theta), z) r\mathrm{d}z\mathrm{d}r\mathrm{d}\theta$$

Example 5. Evaluate $\displaystyle\int_0^{2\pi}\int_0^3\int_0^{\frac{r}{3}} r^3 \mathrm{d}z\mathrm{d}r\mathrm{d}\theta.$

```
> IntV:=int(int(int(r^3, z=0..r/3),r=0..3),theta=0..2*Pi);
```

$$IntV := \frac{162}{5}\pi \qquad (4.2.1)$$

Example 6. Evaluate $\displaystyle\int_0^{2\pi}\int_0^1\int_0^{\frac{1}{\sqrt{2-r^2}}} 3r\mathrm{d}z\mathrm{d}r\mathrm{d}\theta.$

```
> IntV:=int(int(int(3*r, z=0..1/sqrt(1-r^2)),r=0..1),theta=0..2*
  Pi);
```

$$IntV := 6\pi \qquad (4.2.2)$$

Example 7. Find the integral of $\displaystyle\iiint_E K\sqrt{x^2 + y^2}\,\mathrm{d}x\mathrm{d}y\mathrm{d}z$ over the solid E, which lies within the cylinder $x^2 + y^2 = 1$, below the plane $z = 4$ and above the paraboloid $z = 1 - x^2 - y^2$.

Step 1. Plot the graph of the region.

```
> implicitplot3d({z=4, z=1-x^2-y^2, x^2+y^2=1}, x=-1..1, y=-1.
  .1, z=0..4, axes=frame, scaling=constrained);
```



Step 2. Change to cylindrical coordinates and take the integral.

```
> assume(r>0):assume(theta,real):
> intg:=simplify(K*simplify(subs({x=r*cos(theta),y=r*sin(theta)}
  , sqrt(x^2+y^2))));
```

$$intg := K\, r\sim \tag{4.2.3}$$

```
> Int(Int(Int(K*sqrt(x^2+y^2),x = E .. ` `),y),z)=int(int(int
  (intg*r, z=1-r^2..4),r=0..1),theta=0..2*Pi);
```

$$\iiint_E K\sqrt{x^2+y^2}\,\mathrm{d}x\,\mathrm{d}y\,\mathrm{d}z = \frac{12}{5}K\pi \tag{4.2.4}$$

Example 8. Integrate $f(x, y, z) = z\sqrt{x^2+y^2}$ over the cylinder $x^2+y^2 \leq 4$ for $1 \leq z \leq 5$.

```
> intg:=z*r;
```

$$intg := z\, r \tag{4.2.5}$$

```
> IntV:=int(int(int(intg*r, z=1..5), r=0..2), theta=0..2*Pi);
```

$$IntV := 64\,\pi \tag{4.2.6}$$

16.4.3 Triple integrals in spherical coordinates

(1) The **spherical coordinates** of a point P in 3-D are (ρ, θ, φ). The converting formulas are the following:

Spherical to rectangular:

$$x = \rho\cos(\theta)\sin(\varphi)$$
$$y = \rho\sin(\theta)\sin(\varphi)$$
$$z = \rho\cos(\varphi)$$

Rectangular to spherical:

$$\rho = \sqrt{x^2+y^2+z^2}$$
$$\tan(\theta) = \frac{y}{x}$$
$$\cos(\varphi) = \frac{z}{\rho}$$

(2) **Triple integrals in spherical coordinates:** For a region R of the form

$$\theta_1 \le \theta \le \theta_2,\ \phi_1 \le \phi \le \phi_2,\ \rho_1(\theta,\phi) \le \rho \le \rho_2(\theta,\phi)$$

the triple integral is

$$\iiint_R f(x,y,z)\mathrm{dV} =$$

$$\int_{\theta_1}^{\theta_2}\int_{\phi_1}^{\phi_2}\int_{\rho_1(\theta,\phi)}^{\rho_2(\theta,\phi)} f(\rho\cos(\theta)\sin(\phi), \rho\sin(\theta)\sin(\phi), \rho\cos(\phi))\rho^2\sin(\phi)\mathrm{d}\rho\mathrm{d}\phi\mathrm{d}\theta$$

Example 9. Evaluate the triple integral $\int_0^{\pi}\int_0^{\pi}\int_0^{\frac{1-\cos(\phi)}{2}} \rho^2\sin(\phi)\mathrm{d}\rho\mathrm{d}\phi\mathrm{d}\theta.$

```
> int(int(int(rho^2*sin(phi),rho=0..(1-cos(phi))/2),phi=0..Pi),
  theta=0..Pi);
```

$$\frac{1}{6}\pi \qquad (4.3.1)$$

Example 10. Evaluate the triple integral $\int_0^{2\pi}\int_0^{\frac{\pi}{4}}\int_0^{\sec(\phi)} (\rho\cos(\phi))\rho^2\sin(\phi)\mathrm{d}\rho\mathrm{d}\phi\mathrm{d}\theta.$

```
> int(int(int(rho^3*cos(phi)*sin(phi),rho=0..sec(phi)),phi=0..
  Pi/4),theta=0..2*Pi);
```

$$\frac{1}{4}\pi \qquad (4.3.2)$$

Example 11. Evaluate the triple integral $\iiint_E \mathrm{e}^{\left(\sqrt{x^2+y^2+z^2}\right)^3}\mathrm{d}x\mathrm{d}y\mathrm{d}z$, where E is the unit ball $E = \left\{ [x,y,z], x^2+y^2+z^2 \le 1 \right\}$.

Step 1. Change the integrand to spherical coordinates.

```
> assume(rho>0): assume(phi,real):x:='x': y:='y': z:='z':
> x:=rho*cos(theta)*sin(phi);y:=rho*sin(theta)*sin(phi);z:=rho*
  cos(phi);
```

$$x := \rho\sim\cos(\theta\sim)\sin(\phi\sim)$$

$$y := \rho\sim\sin(\theta\sim)\sin(\phi\sim)$$

(4.3.3)

$$z := \rho\sim \cos(\phi\sim) \quad (4.3.3)$$

```
> fun:=subs({x=rho*cos(theta)*sin(phi),y=rho*sin(theta)*sin(phi)
,z=rho*cos(phi)},exp((x^2+y^2+z^2)^(3/2) ));
```

$$fun := e^{\left(\left(\rho\sim^2 \cos(\theta\sim)^2 \sin(\phi\sim)^2 + \rho\sim^2 \sin(\theta\sim)^2 \sin(\phi\sim)^2 + \rho\sim^2 \cos(\phi\sim)^2\right)^{(3/2)}\right)} \quad (4.3.4)$$

```
> sfun:=simplify(fun);
```

$$sfun := e^{\left(\rho\sim^3\right)} \quad (4.3.5)$$

Step 2. Take the integral.

```
> spJ:=rho^2*sin(phi);
```

$$spJ := \rho\sim^2 \sin(\phi\sim) \quad (4.3.6)$$

```
> Int(Int(Int(exp(sqrt('x'^2+'y'^2+'z'^2)^3),'x' = E .. ` `),
'y'),'z')=int(int(int(sfun*spJ,rho=0..1),theta=0..2*Pi),phi=0.
.Pi);
```

$$\iiint_E e^{\left(\left(x^2+y^2+z^2\right)^{(3/2)}\right)} \, dx \, dy \, dz = -\frac{4}{3}\pi + \frac{4}{3} e \pi \quad (4.3.7)$$

Example 12. Find the volume of the solid that lies above the cone $z = \sqrt{x^2 + y^2}$ and below the sphere $x^2 + y^2 + z^2 = z$.
Step 1. Get the equations of the surfaces in spherical coordinates: The cone function $z = \sqrt{x^2 + y^2}$ is changed to a spherical equation $\tan(\phi) = 1$ and the sphere $x^2 + y^2 + z^2 = z$ is changed to $\rho = \cos(\theta)$.

Step 2. Draw the graph of the solid. Note that the intersection of the two surfaces is $r = \frac{\sqrt{2}}{2}$.

```
> sph:=rho=cos(phi); conz:=phi=Pi/4;
```

$$sph := \rho\sim = \cos(\phi\sim)$$

$$conz := \phi\sim = \frac{1}{4}\pi \quad (4.3.8)$$

```
> grp1:=implicitplot3d(sph,rho=0..1,theta=0..2*Pi,phi=0..Pi/4,
coords=spherical,axes=frame):
> grp2:=implicitplot3d(conz,rho=0..sqrt(2)/2,theta=0..2*Pi, phi=
Pi/4..Pi/2,coords=spherical,axes=frame):
> display([grp1,grp2]);
```



Remark. Although you can use the following way to plot the solid, the graph is not as good as the previous one.

```
> implicitplot3d({rho=cos(phi),phi=Pi/4},rho=0..1,theta=0..2*Pi,
  phi=0..Pi/4+0.01,coords=spherical, scaling=constrained, axes=
  frame);
```



Step 3. Compute the integral in spherical coordinates.

```
> V=int(int(int(rho^2*sin(phi),rho=0..cos(phi)),theta=0..2*pi),
  phi=0..Pi/4);
```

$$V = \frac{1}{8}\pi \qquad (4.3.9)$$

Example 13. Find the mass of a sphere S of radius 4 and center at the origin with mass density $f(x, y, z) = x^2 + y^2$.

```
> x:=rho*cos(theta)*sin(phi): y:=rho*sin(theta)*sin(phi): z:=
  rho*cos(phi):
```

```
> spf:=x^2+y^2;
```

$$spf := \rho\sim^2 \cos(\theta\sim)^2 \sin(\phi\sim)^2 + \rho\sim^2 \sin(\theta\sim)^2 \sin(\phi\sim)^2 \quad (4.3.10)$$

```
> spf:=simplify(spf);
```

$$spf := -\rho\sim^2 \left(-1 + \cos(\phi\sim)^2\right) \quad (4.3.11)$$

```
> spJ:=rho^2*sin(phi);
```

$$spJ := \rho\sim^2 \sin(\phi\sim) \quad (4.3.12)$$

```
> M:=int(int(int(spf*spJ, rho=0..4), phi=0..Pi), theta=0..2*Pi);
```

$$M := \frac{8192}{15}\pi \quad (4.3.13)$$

Exercises

1. Find the area enclosed by one leaf of the rose $r = 12\cos(3\theta)$.
2. Use integral in polar coordinates to find the volume of the cone $z = \sqrt{x^2 + y^2}$ above the disk $x^2 + y^2 \le a^2$ in the xy-plane.
3. Find the area of the region cut from the first quadrant by the cardioid $r = 1 + \sin(\theta)$.
4. Evaluate the iterated integral $\int_0^{2\pi}\int_0^1\int_r^{\sqrt{2-r^2}} r\,dz\,dr\,d\theta$.
5. Use an integral in cylindrical coordinates to find the volume of the wedge cut from the cylinder $x^2 + y^2 = 1$ by the planes $z = -y$ and $z = 0$.
6. Use an integral in cylindrical coordinates to find the volume of the region common to the interiors of the cylinders $x^2 + y^2 = 1$ and $x^2 + z^2 = 1$. (*Hint: Find its volume in the first octant, then multiply it by* 8. *The graph of the solid can be plotted by* "**implisitplot3d**").
7. Use an integral in cylindrical coordinates to find the volume of the region enclosed by the cylinder $x^2 + y^2 = 4$ and the plane $z = 0$ and $x + y + z = 4$.
8. Use an integral in spherical coordinates to find the volume of the portion of the solid sphere $\rho \le 1$ that lies between the cones $\phi = \frac{\pi}{3}$ and $\phi = \frac{2\pi}{3}$.
9. Use the spherical coordinates to calculate the triple integral of $f(x, y, z) = y$ over the region $x^2 + y^2 + z^2 \le 1$, $x, y, z \le 0$.

16.5 Change of Variables

16.5.1 Change of variables in double integrals

(1) The **Jacobian determinant** or **Jacobian** of the coordinate transformation $x = g(u, v)$, $y = h(u, v)$ is

$$J(u,v) = \left| \frac{\partial(x,y)}{\partial(u,v)} \right| = \begin{vmatrix} \frac{\partial}{\partial u} g(u,v) & \frac{\partial}{\partial v} g(u,v) \\ \frac{\partial}{\partial u} h(u,v) & \frac{\partial}{\partial v} h(u,v) \end{vmatrix}$$

$$= \frac{\partial}{\partial u} g(u,v) \frac{\partial}{\partial v} h(u,v) - \frac{\partial}{\partial v} g(u,v) \frac{\partial}{\partial u} h(u,v)$$

If $\frac{\partial(x,y)}{\partial(u,v)}$ is invertible, then

$$\frac{\partial(u,v)}{\partial(x,y)} = \left(\frac{\partial(x,y)}{\partial(u,v)} \right)^{-1}$$

(2) The integral for the variable changes is

$$\iint_R f(x,y)\,\mathrm{d}x\mathrm{d}y = \iint_D \mathrm{abs}\left(\left| \frac{\partial(x,y)}{\partial(u,v)} \right| \right) \mathrm{d}u\mathrm{d}v$$

where abs(x) denotes the absolute value of x.

Example 1. Find the Jacobian of the transformation T given by $x = g(u, v)$ and $y = h(u, v)$.

```
> with(linalg):
```

Warning, the protected names norm and trace have been redefined and unprotected

```
> x:=g(u,v): y:=h(u,v): A:=vector([x,y]):
> Juv:=jacobian(A,[u,v]);
```

$$Juv := \begin{bmatrix} \frac{\partial}{\partial u} g(u,v) & \frac{\partial}{\partial v} g(u,v) \\ \frac{\partial}{\partial u} h(u,v) & \frac{\partial}{\partial v} h(u,v) \end{bmatrix} \quad (5.1.1)$$

```
> Jac:=det(Juv);
```

$$Jac := \left(\frac{\partial}{\partial u} g(u,v)\right)\left(\frac{\partial}{\partial v} h(u,v)\right) - \left(\frac{\partial}{\partial v} g(u,v)\right)\left(\frac{\partial}{\partial u} h(u,v)\right) \quad (5.1.2)$$

```
> x:='x': y:='y':
```

Example 2. Find the Jacobian of the variable change from rectangular coordinates to the polar coordinates $x = r\cos(\theta)$, $y = r\sin(\theta)$.

```
> x:=r*cos(theta); y:=r*sin(theta);
```

$$x := r\cos(\theta)$$
$$y := r\sin(\theta) \quad (5.1.3)$$

```
> Jac:=det(jacobian(vector([x,y]),[r,theta]));
```

$$Jac := \cos(\theta)^2 r + r\sin(\theta)^2 \quad (5.1.4)$$

```
> Jac:=simplify(Jac);
```

$$Jac := r \quad (5.1.5)$$

```
> x:='x': y:='y':
```

Example 3. Find the Jacobian of the transformation $x = u^2 - v^2$, $y = u^2 + v^2$.

```
> x:=u^2-v^2: y:=u^2+v^2:
> Jac:=det(jacobian(vector([x,y]),[u,v]));
```

$$Jac := 8\,u\,v \tag{5.1.6}$$

```
> x:='x': y:='y':
```

Example 4. Apply the transformation $u = \frac{2x - y}{2}$, $v = \frac{y}{2}$ to the integral $\int_0^4 \int_{\frac{y}{2}}^{\frac{y}{2}+1} \frac{2x - y}{2}\, dxdy$.

```
> solve({u=(2*x-y)/2, v=y/2}, {x,y});
```

$$\{x = u + v, y = 2\,v\} \tag{5.1.7}$$

```
> x:=u+v; y:=2*v;
```

$$x := u + v$$
$$y := 2\,v \tag{5.1.8}$$

```
> Jac:= det(jacobian(vector([x,y]), [u,v]));
```

$$Jac := 2 \tag{5.1.9}$$

```
> x:='x': y:='y':
```

Since the region in the uv-plane is bounded by $v = 0$, $v = 2$, $u = 0$, and $u = 1$, the integral is changed as following:

```
> Int(Int((2*x-y)/2,x=y/2..y/2+1),'y'=0..4)=Int(Int((2*(u+v)-2*
  v)/2*Jac, u=0..1), v=0..2);
```

$$\int_0^4 \int_{\frac{1}{2}y}^{\frac{1}{2}y+1} x - \frac{1}{2}\,y\, dx\, dy = \int_0^2 \int_0^1 2\,u\, du\, dv \tag{5.1.10}$$

It can be verified by the following:

```
> int(int((2*x-y)/2,x=y/2..y/2+1),y=0..4);
```

$$2 \tag{5.1.11}$$

```
> int(int((2*(u+v)-2*v)/2*Jac, u=0..1), v=0..2);
```

$$2 \tag{5.1.12}$$

Example 5. Apply the transformation $u = x - y$, $v = 2x + y$ to evaluate the integral $\iint_R \left(2x^2 - xy - y^2\right) dxdy$ for the region R bounded by the lines $y = -2x + 4$, $y = -2x + 7$, $y = x - 2$, and $y = x + 1$.

```
> x:='x': y:='y':
> plot({-2*x+4,-2*x+7, x-2, x+1}, x=0..4);
```




```
> solve({u=x-y, v=2*x+y}, {x,y});
```

$$\left\{y=\frac{1}{3}\,v-\frac{2}{3}\,u,\,x=\frac{1}{3}\,u+\frac{1}{3}\,v\right\} \tag{5.1.13}$$

```
> x:=(v-2*u)/3: y:=(u+v)/3:
> Jac:= abs(det(jacobian(vector([x,y]), [u,v])));
```

$$Jac := \frac{1}{3} \tag{5.1.14}$$

Since the four lines have the equation in the *uv*-plane $u = -1,\ u = 2,\ v = 4,\ v = 7$, the integral is changed to the following:

```
> Ivar := int(int((2*x^2-x*y-y^2)*Jac, u=-1..2),v=4..7);
```

$$Ivar := \frac{-21}{4} \tag{5.1.15}$$

```
> x:='x': y:='y':
```

Example 6. Use change of variables to integrate the function $f(x, y) = x^2 + y^2$ over the region D: $-3 \le x^2 - y^2 \le 3,\ 1 \le xy \le 4$.

```
> u:=x^2-y^2; v:=x*y;
```

$$u := x^2 - y^2$$
$$v := x\,y$$
$$x^2 - y^2 \tag{5.1.16}$$

```
> Jac:= det(jacobian(vector([u,v]), [x,y]));
```

$$Jac := 2\,x^2 + 2\,y^2 \tag{5.1.17}$$

```
> ifunc:=simplify((x^2+y^2)/Jac);
```

$$ifunc := \frac{1}{2} \tag{5.1.18}$$

Release u and v to independent variables for the integral.

```
> u:='u': v:='v':
> Ival:=int(int(ifunc,u=-3..3),v=1..4);
```

$$Ival := 9 \tag{5.1.19}$$

Example 7. Find the volume of the solid bounded by the intersection of the paraboloids $z = x^2 + y$ and $z = 1 - x^2 - y^2$.

Step 1: Draw the graph of the solid.

```
> plot3d({x^2+y,1-x^2-y^2},x=-1..1,y=-2..1, view=-2..1,scaling=
  constrained,axes=frame);
```



The integral region is the project on xy-plane of the intersection of the two surfaces: $1 - x^2 - y^2 = x^2 + y$. The following is its graph.

```
> implicitplot(1-x^2-y^2=x^2+y, x=-1..1,y=-2..1,axes=normal);
```



Step 2: Evaluate the integral.

Method A. Use rectangle coordinates to solve the problem.

(1) To get the integral limits for the iterated integral with the order of $\{y, x\}$, do the following:

```
> solve(1-2*x^2-y^2-y,y);
```

$$-\frac{1}{2} + \frac{1}{2}\sqrt{5 - 8x^2}, -\frac{1}{2} - \frac{1}{2}\sqrt{5 - 8x^2} \tag{5.1.20}$$

```
> solve(1-2*x^2+1/4,x);
```

$$\frac{1}{4}\sqrt{10}, -\frac{1}{4}\sqrt{10} \tag{5.1.21}$$

The volume is obtained by computing the following integral:

```
> int(int(1-2*x^2-y^2-y,y=-1/2*sqrt(5-8*x^2)-1/2..1/2*sqrt(5-8*
  x^2)-1/2),x=-1/4*sqrt(10)..1/4*sqrt(10));
```

$$\frac{25}{64}\sqrt{2}\,\pi \tag{5.1.22}$$

(2) To get the integral limits of the iterated integral with the order of {*x, y*}, do the following:

```
> solve(1-2*x^2-y^2-y,x);
```

$$\frac{1}{2}\sqrt{2-2y^2-2y},\ -\frac{1}{2}\sqrt{2-2y^2-2y} \tag{5.1.23}$$

```
> solve(1-y^2-y,y);
```

$$\frac{1}{2}\sqrt{5}-\frac{1}{2},\ -\frac{1}{2}-\frac{1}{2}\sqrt{5} \tag{5.1.24}$$

The volume is obtained by computing the following integral:

```
> int(int(1-2*x^2-y^2-y,x=-1/2*sqrt(2-2*y^2-2*y)..1/2*sqrt(2-2*
  y^2-2*y)),y=-1/2*sqrt(5)-1/2..1/2*sqrt(5)-1/2);
```

$$\frac{25}{64}\sqrt{2}\,\pi \tag{5.1.25}$$

Method B. Change the variables in the integral in the following way:

```
> x:=r*sqrt(5/8)*cos(theta);y:=r*sqrt(5/4)*sin(theta)-1/2;
```

$$x := \frac{1}{4}\,r\sim\sqrt{10}\,\cos(\theta\sim)$$

$$y := \frac{1}{2}\,r\sim\sqrt{5}\,\sin(\theta\sim)-\frac{1}{2} \tag{5.1.26}$$

```
> Jac:=det(jacobian(vector([x,y]),[r,theta]));
```

$$Jac := \frac{1}{8}\sqrt{10}\,\cos(\theta\sim)^2\,r\sim\sqrt{5}+\frac{1}{8}\,r\sim\sqrt{10}\,\sin(\theta\sim)^2\sqrt{5} \tag{5.1.27}$$

```
> jc:=simplify(Jac);
```

$$jc := \frac{5}{8}\sqrt{2}\,r\sim \tag{5.1.28}$$

```
> grt:=simplify(1-2*x^2-y^2-y);
```

$$grt := \frac{5}{4}-\frac{5}{4}\,r\sim^2 \tag{5.1.29}$$

```
> int(int(grt*jc,r=0..1),theta=0..2*Pi);
```

$$\frac{25}{64}\sqrt{2}\,\pi \tag{5.1.30}$$

```
> x:='x': y:='y':
```

16.5.2 Change of variables in triple integrals

(1) The **Jacobian determinant** or **Jacobian** of the coordinate transformation
$x = x(u, v, w),\ y = y(u, v, w),\ z = z(u, v, w)$ is

$$J(u, v, v) = \begin{vmatrix} \frac{\partial x}{\partial u} & \frac{\partial x}{\partial v} & \frac{\partial x}{\partial w} \\ \frac{\partial y}{\partial u} & \frac{\partial y}{\partial v} & \frac{\partial y}{\partial w} \\ \frac{\partial z}{\partial u} & \frac{\partial z}{\partial v} & \frac{\partial z}{\partial w} \end{vmatrix} = \frac{\partial(x, y, z)}{\partial(u, v, w)}$$

If $\frac{\partial(x, y, z)}{\partial(u, v, w)}$ is invertible, then

$$\frac{\partial(u, v, w)}{\partial(x, y, z)} = \left(\frac{\partial(x, y, z)}{\partial(u, v, w)}\right)^{-1}$$

(2) The formula of integral for the variable changes is

$$\iiint_R f(x, y, z) \mathrm{d}x\mathrm{d}y\mathrm{d}z = \iiint_D g(u, v, w) \mathrm{abs}\left(\left|\frac{\partial(x, y, z)}{\partial(u, v, w)}\right|\right) \mathrm{d}u\mathrm{d}v\mathrm{d}w$$

where

$$g(u, v, w) = f(x(u, v, w), y(u, v, w), z(u, v, w))$$

and abs(x) denotes the absolute value of x.

Example 8. Find the Jacobian for the transformation from rectangular coordinates to the cylindrical coordinates.

```
> x:=r*cos(theta); y:=r*sin(theta);
```

$$x := r\cos(\theta)$$
$$y := r\sin(\theta) \tag{5.2.1}$$

```
> A:=vector([x,y,z]):
> Jmtr:=jacobian(A, [r, theta, z]);
```

$$Jmtr := \begin{bmatrix} \cos(\theta) & -r\sin(\theta) & 0 \\ \sin(\theta) & r\cos(\theta) & 0 \\ 0 & 0 & 1 \end{bmatrix} \tag{5.2.2}$$

```
> Jdet:=simplify(det(Jmtr));
```

$$Jdet := r \tag{5.2.3}$$

```
> x:='x': y:='y':
```

Example 9. Find the Jacobian for the transformation from rectangular coordinates to spherical coordinates.

```
> x:=rho*cos(theta)*sin(phi); y:=rho*sin(theta)*sin(phi); z:=
  rho*cos(phi);
```

$$x := \rho\cos(\theta)\sin(\phi)$$
$$y := \rho\sin(\theta)\sin(\phi)$$
$$z := \rho\cos(\phi) \tag{5.2.4}$$

```
> A:=vector([x,y,z]):
> Jmtr:=jacobian(A, [rho, phi, theta]);
```

$$Jmtr := \begin{bmatrix} \cos(\theta)\sin(\phi) & \rho\cos(\theta)\cos(\phi) & -\rho\sin(\theta)\sin(\phi) \\ \sin(\theta)\sin(\phi) & \rho\sin(\theta)\cos(\phi) & \rho\cos(\theta)\sin(\phi) \\ \cos(\phi) & -\rho\sin(\phi) & 0 \end{bmatrix} \quad (5.2.5)$$

```
> Jdet:= simplify(det(Jmtr));
```

$$Jdet := \sin(\phi)\,\rho^2 \quad (5.2.6)$$

```
> x:='x': y:='y': z:='z':
```

Example 10. Find the Jacobian of the transformation $x = 2u$, $y = 3v^2$, $z = 4w^3$.

```
> A:=vector([2*u,3*v^2,4*w^3]);
```

$$A := \begin{bmatrix} 2\,u & 3\,v^2 & 4\,w^3 \end{bmatrix} \quad (5.2.7)$$

```
> Jac:=det(jacobian(A,[u,v,w]));
```

$$Jac := 144\,v\,w^2 \quad (5.2.8)$$

Example 11. Apply the transformation $u = \frac{2x-y}{2}$, $v = \frac{y}{2}$, $w = \frac{z}{3}$ to evaluate the integral

$$\int_0^3\int_0^4\int_{x=\frac{y}{2}}^{x=\frac{y}{2}+1}\left(\frac{2x-y}{2}+\frac{z}{3}\right)\mathrm{d}x\mathrm{d}y\mathrm{d}z.$$

```
> solve({u=(2*x-y)/2, v=y/2, w=z/2}, {x,y,z});
```

$$\{z = 2\,w, y = 2\,v, x = u + v\} \quad (5.2.9)$$

```
> x:=u+v: y:=2*v: z:=2*w:
> Jac:=det(jacobian(vector([x,y,z]),[u,v,w]));
```

$$Jac := 4 \quad (5.2.10)$$

```
> Ival:=int(int(int(((2*x-y)/2+z/3)*Jac,u=0..1),v=0..2),w=0..1);
```

$$Ival := \frac{20}{3} \quad (5.2.11)$$

Exercises

1. Find the Jacobian of the transformation given by $x = 3u + 4v$ and $y = 2u - v$.
2. Find the Jacobian of the variable change $x = 3u\cos(2v)$, $y = u - \sin(2v)$.
3. Find the Jacobian of the variable change $x = rs$, $y = r + s$.
4. Apply the transformation $u = x - 2y$, $v = 2x + y$ to evaluate the integral $\int\int_R \left(x^2 + xy + y^2\right)\mathrm{d}x\mathrm{d}y$ over the region R, which is bounded by the lines $y = -2x + 4$, $y = -2x + 7$, $y = \frac{x}{2} - 1$, and $y = \frac{x}{2} + 1$.
5. Apply the transformation $u = x + 2y$, $v = y$ to the integral $\int_1^3\int_{6-2y}^{10-2y}(x + 3y)\mathrm{d}x\mathrm{d}y$.
6. Use change of variables to integrate the function $f(x, y) = x + y$ over the region D: $0 \le x \le y \le 2x$, $1 \le xy \le 4$.

7. Find the Jacobian of the transformation $x = 2u + v - w$, $y = 3v - 2w$, $z = 4u + w$.
8. Find the Jacobian of the transformation $x = (u + v)\cos(\theta)$, $y = (u + v)\sin(\theta)$, $z = \theta$.
9. Find the Jacobian of the transformation $x = uv$, $y = u^2v^2$, $z = w$.
10. Apply the transformation $u = x - y$, $v = y$, $w = z$ to evaluate the integral
$\int_0^3 \int_0^4 \int_{x-y-1}^{x-y+1} (x - y + z)\,dx\,dy\,dz$.

Chapter 17 INTEGRATION IN VECTOR FIELDS

17.1 Vector Fields

(1) A two-dimensional **vector field** is defined by

$$\mathbf{F}(x, y) = M(x, y)\mathbf{i} + N(x, y)$$

and a three dimensional **vector field** is defined by

$$\mathbf{F}(x, y, z) = M(x, y, z)\mathbf{i} + N(x, y, z)\mathbf{j} + P(x, y, z)\mathbf{k}$$

(2) The **gradient field** of a differentiable function $f(x, y)$ defined on a two-dimensional region R is the field of gradient vectors

$$\nabla f = \frac{\partial f}{\partial x}\mathbf{i} + \frac{\partial f}{\partial y}\mathbf{j}$$

The **gradient field** of a differentiable function $f(x, y, z)$ defined on a three-dimensional region R is the field of gradient vectors

$$\nabla f = \frac{\partial f}{\partial x}\mathbf{i} + \frac{\partial f}{\partial y}\mathbf{j} + \frac{\partial f}{\partial z}\mathbf{k}$$

(3) If $\mathbf{F} = \nabla f$ for some scalar function f on D, then f is called a **potential function for F** on D.
(4) Let $\mathbf{F}(x, y, z) = M(x, y, z)\mathbf{i} + N(x, y, z)\mathbf{j} + P(x, y, z)\mathbf{k}$ be a vector field whose components have continuous first partial derivatives. Then **F** is a gradient field of a potential function if and only if

$$\frac{\partial P}{\partial y} = \frac{\partial N}{\partial z},\ \frac{\partial M}{\partial z} = \frac{\partial P}{\partial x},\ \frac{\partial N}{\partial x} = \frac{\partial M}{\partial y}.$$

17.1.1 Vector fields

Example 1. Plot the vector field $\mathbf{F}(x,y) = [\sin(x), \cos(y)]$.

```
> with(plots): with(linalg):
Warning, the name changecoords has been redefined
Warning, the protected names norm and trace have been redefined
and unprotected
> fieldplot([sin(x),cos(y)],x=-5..5,y=-5..5,arrows=SLIM, color=
  x);
```



You can also plot it in the following way.

```
> f:=(x,y)->sin(x); g:=(x,y)->cos(y);
```

$$f := (x, y) \to \sin(x)$$
$$g := (x, y) \to \cos(y)$$

```
> fieldplot([f,g],-5..5,-5..5,arrows=SLIM, color=g);
```



Example 2. Plot field $\mathbf{F}(x, y, z) = x\mathbf{i} + y\mathbf{j} + z\mathbf{k}$.

```
> fieldplot3d([x,y,z],x=-3..3,y=-3..3,z=-3..3,arrows=SLIM, axes=
  boxed, color=red);
```



17.1.2 Gradient fields

Example 3. Create a gradient field for $f(x, y) = \sin(x + y)$ and then plot it.

```
> f:=sin(x+y);
```

$$f := \sin(x + y) \qquad (1.2.1)$$

```
> grdf:=[diff(f,x), diff(f,y)];
```

$$grdf := [\cos(x + y), \cos(x + y)] \qquad (1.2.2)$$

```
> fieldplot(grdf, x=0..2*Pi, y=0..2*Pi, arrows=SLIM, color=x);
```



Example 4. Create a gradient field for $f(x, y, z) = xy + yz^2$ and then plot it.

```
> f:=x*y+y*z^2;
```

$$f := x\,y + y\,z^2 \quad (1.2.3)$$

```
> grdf:=grad(f, [x,y,z]);
```

$$grdf := \left[y \quad x + z^2 \quad 2\,y\,z\right] \quad (1.2.4)$$

```
> fieldplot3d(grdf,x=-3..3,y=-3..3,z=-3..3,arrows=SLIM,axes=
  boxed);
```



17.1.3 Determine whether or not a field F is a gradient field

Example 5. Determine if $\mathbf{F}(x, y) = (2x - 3y)\mathbf{i} + (2y - 3x)\mathbf{j}$ is a gradient field.

```
> M:=2*x-3*y; N:=2*y-3*x;
```

$$M := 2x - 3y$$
$$N := 2y - 3x$$

```
> testf:= diff(N,x)-diff(M,y);
```

$$testf := 0$$

Hence **F** is a gradient field.

Example 6. Determine if $\mathbf{F}(x, y) = (ye^x + \sin(y))\mathbf{i} + (xe^y + x\cos(y))\mathbf{j}$ is a gradient field.

```
> M:=y*exp(x)+sin(y); N:=x*exp(y)+x*cos(y);
```

$$M := y\,e^x + \sin(y)$$
$$N := x\,e^y + x\cos(y)$$

```
> testf:=diff(N,x)-diff(M,y);
```

$$testf := e^y - e^x$$

Hence **F** is not a gradient field.

Example 7. Determine if $\mathbf{F}(x, y, z) = 4xe^z\mathbf{i} + \cos(y)\mathbf{j} + 2x^2e^z\mathbf{k}$ is a gradient field.

```
> M:=4*x*exp(z); N:=cos(y); P:=2*x^2*exp(z);
```

$$M := 4x\,e^z$$
$$N := \cos(y)$$
$$P := 2x^2\,e^z$$

```
> testf1:=diff(N,x)-diff(M,y);
```

$$testf1 := 0$$

```
> testf2:=diff(P,x)-diff(M,z);
```

$$testf2 := 0$$

```
> testf3:=diff(P,y)-diff(N,z);
```

$$testf3 := 0$$

Hence **F** is a gradient field.

Exercises

1. Plot the vector field $\mathbf{F}(x, y) = [y,\ x]$.
2. Plot field $\mathbf{F}(x, y, z) = yz^2\mathbf{i} + xz^2\mathbf{j} + 2xyz\mathbf{k}$.
3. Create a gradient field for $f(x, y) = \sin(xy)$ and then plot it.
4. Create a gradient field for $f(x, y) = x^2 + y^2$ and then plot it.
5. Determine if $\mathbf{F}(x, y, z) = ye^{xy}\mathbf{F}(x, y, z)\mathbf{i} + xe^{xy}\mathbf{j}$ is a gradient field.
6. Determine if $\mathbf{F}(x, y, z) = yz^2\mathbf{i} + xz^2\mathbf{j} + 2xyz\mathbf{k}$ is a gradient field.

17.2 Line Integrals

17.2.1 Evaluate scalar line integrals

Evaluate a line integral for a function $f(x, y, z)$ over a curve C:
(1) Find a smooth parametrization of C,

$$\mathbf{r}(t) = g(t)\mathbf{i} + h(t)\mathbf{j} + k(t)\mathbf{k},\ \ a \le t \le b.$$

(2) Evaluate the integral as

$$\int_C f(x, y, z)\,\mathrm{d}s = \int_a^b f(x(t), y(t), z(t))\|\mathbf{v}(t)\|\mathrm{d}t,$$

where $\mathbf{v}(t) = \frac{\mathrm{d}}{\mathrm{d}t}\mathbf{r}(t)$.

Example 1. Evaluate the line integral of the function $\int_C \left(2+x^2 y\right)\mathrm{d}s$, where C is the upper half of the unit circle $x^2 + y^2 = 1$.

Step 1. Find the parametric equation of the curve C.

```
> assume(t,real): r:=[cos(t),sin(t)];
```

$$r := [\cos(t\sim), \sin(t\sim)]$$

Step 2. Determine the low limit and the up limit for the parameter.

```
> lowlmt:=0: uplmt:=Pi:
```

Step 3. Use the formula to get the arc differential $\mathrm{d}s = \|\mathbf{v}(t)\|\mathrm{d}t$.

```
> dr:= diff(r,t);
```

$$dr := [-\sin(t\sim), \cos(t\sim)] \quad (2.1.1)$$

```
> arcd:=sqrt(simplify(dr[1]^2+dr[2]^2, trig));
```

$$arcd := 1$$

```
> ds:=sqrt(simplify(innerprod(dr,dr)));
```

$$ds := 1 \quad (2.1.2)$$

```
> ds:=simplify(norm(dr,2));
```

$$ds := 1 \quad (2.1.3)$$

Step 4. Set the integrand and evaluate the line integral.

```
> func:=subs({x=cos(t), y=sin(t)},2+x^2*y);
```

$$func := 2 + \cos(t\sim)^2 \sin(t\sim)$$

```
> x:=cos(t): y:=sin(t):
> f:=2+x^2*y;
```

$$f := 2 + \cos(t\sim)^2 \sin(t\sim) \quad (2.1.4)$$

```
> LI:=int(f*ds,t=lowlmt..uplmt);
```

$$LI := \frac{2}{3} + 2\pi$$

```
> x:='x': y:='y':r:='r':
```

Example 2. Integrate $f(x, y, z) = x - 3y^2 + z$ over the line segment C joining the origin to the point (1, 1, 1).

```
> r:=[t,t,t]; lowlmt:=0; uplmt:=1;
```

$$r := [t\sim, t\sim, t\sim]$$
$$lowlmt := 0$$
$$uplmt := 1 \quad (2.1.5)$$

```
> dr:=diff(r,t);
```

$$dr := [1, 1, 1] \quad (2.1.6)$$

```
> ds:=norm(dr,2);
```

$$ds := \sqrt{3} \quad (2.1.7)$$

```
> f:=subs({x=t,y=t,z=t},x-3*y^2+z);
```

$$f := 2\,t\sim - 3\,t\sim^2 \quad (2.1.8)$$

```
> LI:=int(f*ds,t=lowlmt..uplmt);
```

$$LI := 0 \quad (2.1.9)$$

Example 3. Calculate $\int_C (x + y + z)\mathrm{d}s$, where C is the helix $\mathbf{c}(t) = (\cos(t), \sin(t), t)$, for $0 \le t \le \pi$.

```
> r:=[cos(t), sin(t), t];
```

$$r := [\cos(t\sim), \sin(t\sim), t\sim] \quad (2.1.10)$$

```
> dr:=diff(r,t);
```

$$dr := [-\sin(t\sim), \cos(t\sim), 1] \quad (2.1.11)$$

```
> ds:=simplify(norm(dr, 2));
```

$$ds := \sqrt{2} \quad (2.1.12)$$

```
> f:=subs({x=cos(t),y=sin(t),z=t}, x+y+z);
```

$$f := \cos(t\sim) + \sin(t\sim) + t\sim \quad (2.1.13)$$

```
> LI:=int(f*ds, t=0..Pi);
```

$$LI := 2\sqrt{2} + \frac{1}{2}\sqrt{2}\,\pi^2 \quad (2.1.14)$$

Example 4. Evaluate the line integral of the function $\int_C x^2 z\mathrm{d}s$, where the curve C:

$x = \sin(2t), y = 3t, z = \cos(2t), 0 \le t \le \dfrac{\pi}{4}$.

```
> x:=sin(2*t): y:=3*t: z:=cos(2*t): r:=[x,y,z]:
> lowlmt:=0: uplmt:=Pi/4:
> func:=x^2*z;
```

$$func := x^2 z$$

```
> arcd:=simplify(norm(diff(r,t),2));
```

$$arcd := \sqrt{13}$$

```
> LI:=int(func*arcd,t=lowlmt..uplmt);
```

$$LI := \frac{1}{6}\sqrt{13}$$

```
> x:='x': y:='y': z:='z': r:='r':
```

17.2.2 The vector line integral

(1) The linear integral of a vector field

$$\mathbf{F}(x, y, z) = M(x, y, z)\mathbf{i} + N(x, y, z)\mathbf{j} + P(x, y, z)\mathbf{k}$$

along the curve C:

$$\mathbf{r}(t) = x(t)\mathbf{i} + y(t)\mathbf{j} + z(t)\mathbf{k},\ \ a \le t \le b,$$

is calculated by

$$\int_C \mathbf{F} \cdot \mathrm{d}\mathbf{r} = \int_a^b (M(x(t), y(t), z(t))x'(t) + N(x(t), y(t), z(t))y'(t)$$

$$+ M(x(t), y(t), z(t)z'(t))\mathrm{d}t$$

(2) The **work** done by a force $\mathbf{F} = M\mathbf{i} + N\mathbf{j} + P\mathbf{k}$ over a smooth curve $\mathbf{r}(t)$ from $t = a$ to $t = b$ is

$$W = \int_C \mathbf{F}\cdot\mathrm{d}\mathbf{r} = \int_C \mathbf{F}\cdot\mathbf{T}\mathrm{d}s,$$

where $\mathbf{T} = \dfrac{\mathrm{d}\mathbf{r}}{\|\mathrm{d}\mathbf{r}\|}$ is the unit vector in the tangent direction.

Example 5. Evaluate the line integral of vector field $\mathbf{F}(x, y) = x\sqrt{y}\,\mathbf{i} + 2y\sqrt{x}\,\mathbf{j}$ over C, where C consists of the shortest arc of the circle $x^2 + y^2 = 1$ from (1, 0) to (0, 1) and the line segment from (1, 0) to (4, 3).

The curve is a piecewise-smooth one. Its graph is the following.

```
> P1:=plot(sqrt(1-x^2),x=0..1): P2:=plot(x-1,x=1..4):
> display(P1,P2,axes=normal);
```



We have to split the integral into two parts according to the curve equations.

The first part of the integral. The integral along the shortest arc of the circle $x^2 + y^2 = 1$ from (1, 0) to (0, 1).

```
> x:=cos(t):y:=sin(t):r:=[x,y]:
> lowlmt:=0: uplmt:=Pi/2:
> fieldf:=[x*sqrt(y),2*y*sqrt(x)];
```

$$fieldf := \left[\cos(t)\sqrt{\sin(t)},\ 2\sin(t)\sqrt{\cos(t)}\right]$$

```
> dr:=diff(r,t);
```

$$dr := [-\sin(t), \cos(t)]$$

```
> ifunc:=innerprod(fieldf,dr);
```

$$ifunc := -\cos(t)\sin(t)^{(3/2)} + 2\sin(t)\cos(t)^{(3/2)}$$

```
> LI1:=int(ifunc,t=lowlmt..uplmt);
```

$$LI1 := \frac{2}{5}$$

```
> x:='x': y:='y': r:='r':
```

The second part. Evaluate the integral along the line segment from (1, 0) to (4, 3). The line equation is $y = x - 1$.

```
> y:=x-1: r:=[x,y]:
> lowlmt:=1: uplmt:=4:
> fieldf:=[x*sqrt(y),2*y*sqrt(x)];
```

$$fieldf := \left[x\sqrt{x-1}, 2\,(x-1)\sqrt{x}\right]$$

```
> dr:=diff(r,x);
```

$$dr := [1, 1]$$

```
> ifunc:=innerprod(fieldf,dr);
```

$$ifunc := x\sqrt{x-1} + 2\,x^{(3/2)} - 2\sqrt{x}$$

```
> LI2:=int(ifunc,x=lowlmt..uplmt);
```

$$LI2 := \frac{28}{5}\sqrt{3} + \frac{232}{15}$$

```
> LI:=LI1+LI2;
```

$$LI := \frac{238}{15} + \frac{28}{5}\sqrt{3}$$

```
> x:='x': y:='y': r:='r':
```

Example 6. Evaluate the line integral of the vector field **F** over the curve C, where $\mathbf{F}(x, y, z) = (y + z)\mathbf{i} - x^2\mathbf{j} - 4y^2\mathbf{k}$, and C is determined by $\mathbf{r}(t) = t\mathbf{i} + t^2\mathbf{j} + t^4\mathbf{k}$, $0 \le t \le 1$.

```
> x:=t: y:=t^2: z:=t^4: r:=[x,y,z]:
> lowlmt:=0: uplmt:=1:
> fieldf:=[y+z,-x^2,-4*y^2];
```

$$fieldf := \left[t\sim^2 + t\sim^4, -t\sim^2, -4\,t\sim^4\right]$$

```
> dr:=diff(r,t);
```

$$dr := \left[1, 2\,t\sim, 4\,t\sim^3\right]$$

```
> ifunc:=innerprod(fieldf,dr);
```

$$ifunc := t\sim^2 + t\sim^4 - 2\,t\sim^3 - 16\,t\sim^7$$

```
> LI:=int(ifunc,t=lowlmt..uplmt);
```

$$LI := \frac{-59}{30}$$

```
> x:='x': y:='y': r:='r':
```

Example 7. Find the work done by $\mathbf{F} = (y - x^2)\mathbf{i} + (z - y^2)\mathbf{j} + (x - z^2)\mathbf{k}$ over the curve $\mathbf{r}(t) = t\mathbf{i} + t^2\mathbf{j} + t^3\mathbf{k}$, $0 \le t \le 1$, from (0, 0, 0) to (1, 1, 1).

```
> x:=t: y:=t^2: z:=t^3: r:=[x,y,z]:
> fieldf:=[y-x^2, z-y^2, x-z^2];
```

$$fieldf := \left[0, t\sim^3 - t\sim^4, t\sim - t\sim^6\right] \quad (2.2.1)$$

```
> dr:=diff(r,t);
```

$$dr := \left[1, 2\,t\sim, 3\,t\sim^2\right] \quad (2.2.2)$$

```
> ifunc:=innerprod(fieldf, dr);
```

$$ifunc := 2\,t\sim^4 - 2\,t\sim^5 + 3\,t\sim^3 - 3\,t\sim^8 \quad (2.2.3)$$

```
> Wrk:=int(ifunc, t=0..1);
```

$$Wrk := \frac{29}{60} \quad (2.2.4)$$

```
> x:='x': y:='y': z:='z': r:='r':
```

Example 8. Evaluate $\int_C xy\mathrm{d}x + (x+y)\mathrm{d}y$ along the curve $y = x^2$ from (−1, 1) to (2, 4).

```
> y:=x^2; r:=[x, y];
```

$$y := x^2$$

$$r := \left[x, x^2\right]$$

```
> fieldf:=[x*y, x+y];
```

$$fieldf := \left[x^3, x + x^2\right]$$

```
> ifunc:=innerprod(fieldf,diff(r,x));
```

$$ifunc := 3\,x^3 + 2\,x^2 \quad (2.2.5)$$

```
> LI:=int(ifunc, x=-1..2);
```

$$LI := \frac{69}{4} \quad (2.2.6)$$

Exercises

1. Evaluate the line integral of the function $\int_C (x+y)\mathrm{d}s$, where C is the straight line segment $x = t,\ y = 1 - t,\ z = 0$, from (0, 1, 0) to (1, 0, 0).

2. Evaluate $\int_C (xy + y + z)\mathrm{d}s$, where C is the curve $\mathbf{r}(t) = [2t, t, 2 - 2t],\ 0 \le t \le 1$.

3. Evaluate the line integral of the vector field $\mathbf{F}$ over the curve C, where $\mathbf{F}(x, y, z) = z^2\mathbf{i} + x\mathbf{j} + y\mathbf{k}$, and C is determined by $\mathbf{r}(t) = (3 + 5t^2)\mathbf{i} - t^2\mathbf{j} + t\mathbf{k}$, $0 \le t \le 1$.

4. Find the work done by the force $\mathbf{F} = (x - y)\mathbf{i} + (y - z)\mathbf{j} + z\mathbf{k}$ over the curve $\mathbf{r}(t) = (t + t^2)\mathbf{i} + t^3\mathbf{j} + t\mathbf{k}$, $0 \le t \le 1$.

5. Find the work done by the force $\mathbf{F} = xy\mathbf{i} + (x + y)\mathbf{j}$ around the circle $\mathbf{r}(t) = \cos(t)\mathbf{i} + \sin(t)\mathbf{j}$, $0 \le t \le 2\pi$.

17.3 Conservative Vector Fields

(1) Let **F** be a vector field defined on an open domain D in space, and suppose that for any two points A and B in D the work $\int_A^B \mathbf{F}\cdot d\mathbf{r}$ done in moving from A to B is the same over all paths. Then the integral $\int_A^B \mathbf{F}\cdot d\mathbf{r}$ is said to be **path independent in** D and the field **F** is said to be **conservative on** D.

(2) If $\mathbf{F} = \nabla f$ for some scalar function f on D, then f is called a **potential function for F**.

(3) **The fundamental theorem of line integrals:** Let $\mathbf{F}(x, y, z) = M(x, y, z)\mathbf{i} + N(x, y, z)\mathbf{j} + P(x, y, z)\mathbf{k}$ be a vector field whose components are continuous throughout an open connected region D in space. Then there exists a differentiable function f such that

$$\mathbf{F} = \nabla f = \frac{\partial f}{\partial x}\mathbf{i} + \frac{\partial f}{\partial y}\mathbf{j} + \frac{\partial f}{\partial z}\mathbf{k}$$

if and only if for all points A and B in D the value of $\int_A^B \mathbf{F}\cdot d\mathbf{r}$ is independent of the path joining A to B in D.

(4) If the integral $\int_A^B \mathbf{F}\cdot d\mathbf{r}$ is independent of the path from A to B, its value is

$$\int_A^B \mathbf{F}\cdot d\mathbf{r} = f(B) - f(A), \text{ where } \nabla f = \mathbf{F}$$

(5) The following statements are equivalent:

1. $\int \mathbf{F}\cdot d\mathbf{r} = 0$ around every closed loop in D.
2. The field **F** is conservative on D.

(6) Let $\mathbf{F}(x, y, z) = M(x, y, z)\mathbf{i} + N(x, y, z)\mathbf{j} + P(x, y, z)\mathbf{k}$ be a vector field whose components have continuous first partial derivatives. Then **F** is conservative if and only if

$$\frac{\partial P}{\partial y} = \frac{\partial N}{\partial z},\ \frac{\partial M}{\partial z} = \frac{\partial P}{\partial x},\ \frac{\partial N}{\partial x} = \frac{\partial M}{\partial y}$$

(7) An expression $M(x, y, z)dx + N(x, y, z)dy + P(x, y, z)dz$ is called a **differential form**. A differential form is **exact** on a domain D in space if

$$M dx + N dy + P dz = \frac{\partial f}{\partial x}dx + \frac{\partial f}{\partial y}dy + \frac{\partial f}{\partial z}dz$$

for some scalar function f throughout D.

17.3.1 Path independence and conservative field

Example 1. Find the work done by the conservative field $\mathbf{F} = yz\mathbf{i} + xz\mathbf{j} + xy\mathbf{k} = \nabla xyz$ along any smooth curve C joining the point $(-1, 3, 9)$ to the point $(1, 6, -4)$.

```
> f:=(x,y,z)->x*y*z;
```

$$f := (x, y, z) \rightarrow x\,y\,z \qquad (3.1.1)$$

```
> Wk:=f(1,6,-4)-f(-1,3,9);
```

$$Wk := 3 \qquad (3.1.2)$$

17.3.2 Determine whether or not a field F is a conservative vector field

Example 2. $\mathbf{F}(x, y) = (2x - 3y)\mathbf{i} + (2y - 3x)\mathbf{j}$.

```
> M:=2*x-3*y; N:=2*y-3*x;
```

$$M := 2\,x - 3\,y$$
$$N := 2\,y - 3\,x$$

```
> testf:= diff(N,x)-diff(M,y);
```

$$testf := 0$$

Hence **F** is a conservative field.

Example 3. $\mathbf{F}(x, y) = (ye^x + \sin(y))\mathbf{i} + (xe^y + x\cos(y))\mathbf{j}$.

```
> M:=y*exp(x)+sin(y); N:=x*exp(y)+x*cos(y);
```

$$M := y\,e^x + \sin(y)$$
$$N := x\,e^y + x\cos(y)$$

```
> testf:=diff(N,x)-diff(M,y);
```

$$testf := e^y - e^x$$

Hence **F** is not a conservative field.

Example 4. $\mathbf{F}(x, y, z) = 4xe^z\mathbf{i} + \cos(y)\mathbf{j} + 2x^2e^z\mathbf{k}$.

```
> M:=4*x*exp(z); N:=cos(y); P:=2*x^2*exp(z);
```

$$M := 4\,x\,e^z$$
$$N := \cos(y)$$
$$P := 2\,x^2\,e^z$$

```
> testf1:=diff(N,x)-diff(M,y);
```

$$testf1 := 0$$

```
> testf2:=diff(P,x)-diff(M,z);
```

$$testf2 := 0$$

```
> testf3:=diff(P,y)-diff(N,z);
```

$$testf3 := 0$$

Hence **F** is a conservative field.

17.3.3 Find the potential function f for a conservative vector field F

Example 5. Find the potential function f for $\mathbf{F}(x, y) = ye^x + \sin(y)\mathbf{i} + (e^x + x\cos(y) + \sin(y))\mathbf{j}$.

Step 1. Determine whether or not **F** is a conservative vector field.

```
> M:=y*exp(x)+sin(y): N:=exp(x)+x*cos(y)+sin(y):
> testf:=diff(N,x)-diff(M,y);
```

$$testf := 0$$

Step 2. Get an antiderivative of M with respect to x.

```
> pref:=int(M,x);
```

$$pref := y\,e^x + \sin(y)\,x$$

Step 3. Let g be the function such that $g = f -$pref. The derivative of g with respective to y is

```
> dg:=N-diff(pref,y);
```

$$dg := \sin(y)$$

Step 4. Obtain $f(x, y)$ using the formula $f =$ pref $+ g$.

```
> f:=pref+int(dg,y);
```

$$f := y\,e^x + \sin(y)\,x - \cos(y)$$

Example 6. Find the potential function f for $\mathbf{F}(x, y, z) = (4xe^z + x^2)\mathbf{i} + \cos(y)\mathbf{j} + (2x^2e^z + \cos(z))\mathbf{k}$.

Step 1. Determine whether or not **F** is a conservative vector field.

```
> M:=4*x*exp(z)+x^2; N:=cos(y); P:=2*x^2*exp(z)+cos(z);
```

$$M := 4\,x\,e^z + x^2$$

$$N := \cos(y)$$

$$P := 2\,x^2\,e^z + \cos(z)$$

```
> testf1:=diff(N,x)-diff(M,y);
```

$$testf1 := 0$$

```
> testf2:=diff(P,x)-diff(M,z);
```

$$testf2 := 0$$

```
> testf3:=diff(P,y)-diff(N,z);
```

$$testf3 := 0$$

Step 2. Get an antiderivative of P with respect to z.

```
> pref:=int(P,z);
```

$$pref := 2\,x^2\,e^z + \sin(z)$$

Step3. Obtain ($\frac{\partial c}{\partial x}$, $\frac{\partial c}{\partial y}$), the partial derivatives of function $c(x, y)$ with respect to x and y, where $c = f -$ pref.

```
> diffcx:=M-diff(pref, x);
```

$$diffcx := x^2$$

```
> diffcy:=N-diff(pref,y);
```

$$diffcy := \cos(y)$$

Step 4. Use the way in **Example 4** to find the function $c(x, y)$.

```
> prec:=int(diffcx,x);
```

$$prec := \frac{1}{3}\,x^3$$

```
> dg:=diffcy-diff(prec,y);
```

$$dg := \cos(y)$$

```
> c:=prec+int(dg,y);
```

$$c := \frac{1}{3}x^3 + \sin(y)$$

Step 5. Find f using the formula $f = \text{pref} + c$.

```
> f:=pref+c;
```

$$f := 2x^2 e^z + \sin(z) + \frac{1}{3}x^3 + \sin(y)$$

Step 6 (optional). Verify the correction of the answer.

```
> diff(f,x); diff(f,y); diff(f,z);
```

$$4xe^z + x^2$$
$$\cos(y)$$
$$2x^2 e^z + \cos(z) \quad (3.3.1)$$

Example 7. Find the potential function f for

$\mathbf{F}(x, y, z) = (e^x\cos(y) + yz)\mathbf{i} + (xz - e^x\sin(y))\mathbf{j} + (xy + z)\mathbf{k}$.

```
> M:=exp(x)*cos(y)+y*z; N:=x*z-exp(x)*sin(y); P:=x*y+z;
```

$$M := e^x\cos(y) + yz$$
$$N := xz - e^x\sin(y)$$
$$P := xy + z \quad (3.3.2)$$

```
> diff(M,y)-diff(N,x);
```

$$0 \quad (3.3.3)$$

```
> diff(M,z)-diff(P,x);
```

$$0 \quad (3.3.4)$$

```
> diff(N,z)-diff(P,y);
```

$$0 \quad (3.3.5)$$

```
> pref:=int(P,z);
```

$$pref := xyz + \frac{1}{2}z^2 \quad (3.3.6)$$

```
> diffcx:=M-diff(pref,x);
```

$$diffcx := e^x\cos(y) \quad (3.3.7)$$

```
> diffcy:=N-diff(pref,y);
```

$$diffcy := -e^x\sin(y) \quad (3.3.8)$$

```
> prec:=int(diffcx,x);
```

$$prec := e^x\cos(y) \quad (3.3.9)$$

```
> dg:=diffcy-diff(prec,y);
```

$$dg := 0 \quad (3.3.10)$$

```
> f:=pref+prec;
```

$$f := xyz + \frac{1}{2}z^2 + e^x\cos(y) \quad (3.3.11)$$

Verify the correction of the answer.

```
> diff(f,x); diff(f,y); diff(f,z);
```

$$e^x \cos(y) + y z$$
$$x z - e^x \sin(y)$$
$$x y + z \qquad (3.3.12)$$

17.3.4 Line integrals in conservative fields

Example 8. Evaluate the line integral of the vector field **F** over the curve *C*, from (0, 1) to (e, 2), where $\mathbf{F}(x, y) = e^{2y}\mathbf{i} + (1 + 2xe^{2y})\mathbf{j}$ and *C* is defined by $\mathbf{r}(t) = te^t\mathbf{i} + (1+t)\mathbf{j}$, $0 \le t \le 1$.

Step1. Verify that **F** is conservative.

```
> M:=exp(2*y);N:=1+2*x*exp(2*y);
```

$$M := e^{(2y)}$$
$$N := 1 + 2\,x\,e^{(2y)}$$

```
> testf:=diff(N,x)-diff(M,y);
```

$$testf := 0$$

Step 2. Find the potential function for **F**.

```
> pref:=int(M,x);
```

$$pref := x\,e^{(2y)}$$

```
> dg:=N-diff(pref,y);
```

$$dg := 1$$

```
> f:=pref+int(dg,y);
```

$$f := x\,e^{(2y)} + y$$

Step 3. Use the Fundamental Theorem to evaluate the integral.

```
> LI:=eval(f,[x=exp(1),y=2])-eval(f,[x=0,y=1]);
```

$$LI := e\,e^4 + 1$$

```
> LI:=simplify(LI);
```

$$LI := e^5 + 1 \qquad (3.3.1.1)$$

Exercises

1. Let $f(x, y, z) = xy \sin(yz)$. Evaluate $\int_C \nabla f \cdot d\mathbf{r}$, where *C* is any smooth curve joining the point (0, 0, 0) to the point (1, 1, π).
2. Find the work done by the gradient field $\mathbf{F} = 2xyz\mathbf{i} + x^2z\mathbf{j} + x^2y\mathbf{k} = \nabla x^2yz$ along any smooth curve *C* joining the point (1, 1, 2) to the point (2, 1, 4).
3. Determine whether or not the field $\mathbf{F}(x, y, z) = (z + y)\mathbf{i} + z\,\mathbf{j} + (y + x)\mathbf{k}$ is a conservative vector field.
4. Find the potential function f for $\mathbf{F}(x, y) = \sin(y^2)\mathbf{i} + 2xy\cos(y^2)\mathbf{j}$.

5. Find the potential function f for a conservative vector field $\mathbf{F}(x, y, z) = (y\sin(z))\mathbf{i} + (x\sin(z))\mathbf{j} + (xy\cos(z))\mathbf{k}$.

6. Show that $\mathbf{F}(x, y, z) = 2x\mathbf{i} - y^2\mathbf{j} - \frac{4}{1+z^2}\mathbf{k}$ is a conservative field and evaluate the integral $\int_C \mathbf{F}\cdot d\mathbf{r}$, where C is any smooth curve joining the point (0, 0, 0) to the point (3, 3, 1).

7. Evaluate the line integral of the vector field $\mathbf{F}$ over the curve C from (0, 1) to (e, 2), where $\mathbf{F}(x, y) = ye^{xy}\mathbf{i} + xe^{xy}\mathbf{j}$, and C is defined by $\mathbf{r}(t) = e^t\mathbf{i} + (1-t)\mathbf{j}$, $0 \le t \le 1$.

17.4 Parameterized Surfaces and Surface Integrals

(1) The area of the surface $f(x, y, z) = c$ over a closed and bounded plane region R is

$$\text{Area} = \iint_S dS = \iint_R \frac{\|\nabla f\|}{\|\nabla f\cdot\mathbf{p}\|}dA$$

where $\mathbf{p}$ is a unit vector normal to R and $\nabla f\cdot\mathbf{p} \ne 0$. If the surface is given by the function $z = f(x, y)$, then the surface area differential is

$$dS = \frac{\|\nabla f\|}{\|\nabla f\cdot\mathbf{p}\|}dA = \sqrt{1 + \left(\frac{\partial f}{\partial x}\right)^2 + \left(\frac{\partial f}{\partial y}\right)^2}$$

(2) The surface integral of g over S is

$$\iint_S g\,dS = \iint_R \frac{g(x, y, z)\|\nabla f\|}{\|\nabla f\cdot\mathbf{p}\|}dA$$

If the surface is given by the function $z = f(x, y)$, then

$$\iint_S g\,dS = \iint_R g(x, y, z)\sqrt{1 + \left(\frac{\partial f}{\partial x}\right)^2 + \left(\frac{\partial f}{\partial y}\right)^2}\,dA$$

(3) If the **parametric vector form of a surface** is

$$\mathbf{r}(u, v) = f(u, v)\mathbf{i} + g(u, v)\mathbf{j} + h(u, v)\mathbf{k},\ a \le u \le b,\ c \le v \le d,$$

then the **differential of a parametric surface area** is

$$dS = \|\mathbf{r}_u \times \mathbf{r}_v\|du\,dv$$

the **area of a parametric surface** is

$$A = \int_c^d\int_a^b \|\mathbf{r}_u \times \mathbf{r}_v\|du\,dv$$

and the parametric **surface integral** of a function $G(x, y, z)$ over S is

$$\text{Int} = \int_c^d\int_a^b G(x(u, v), y(u, v), z(u, v))\|\mathbf{r}_u \times \mathbf{r}_v\|du\,dv.$$

17.4.1 Parameterization of surfaces

Example 1. Find a parametrization of the cone $z = \sqrt{x^2 + y^2}$, $0 \le z \le 1$.

```
> assume(rr>0):assume(theta, real):
```

The parametric equation of the surface is

```
> x:=rr*cos(theta);y:=rr*sin(theta);z:=simplify(sqrt(x^2+y^2));
```

$$x := rr\sim \cos(\theta\sim)$$
$$y := rr\sim \sin(\theta\sim)$$
$$z := rr\sim \qquad (4.1.1)$$

with $0 \le \theta \le 2\pi$, $0 \le rr \le 1$.

```
> x:='x': y:='y': z:='z':
```

Example 2. Find a parametrization of the sphere $x^2 + y^2 + z^2 = a^2$.

```
> assume(a>0): assume(phi, real):
> x:=a*cos(theta)*sin(phi);y:=a*sin(theta)*sin(phi);z:=a*cos
  (phi);
```

$$x := a\sim \cos(\theta\sim) \sin(\phi\sim)$$
$$y := a\sim \sin(\theta\sim) \sin(\phi\sim)$$
$$z := a\sim \cos(\phi\sim) \qquad (4.1.2)$$

with $0 \le \theta \le 2\pi$, $0 \le \varphi \le \pi$.

```
> x:='x': y:='y': z:='z':
```

Example 3. Find a parametrization of the cylinder $x^2 + (y-3)^2 = 9$, $0 \le z \le 5$.

```
> x:=3*cos(theta); y:=3*sin(theta)+3; z:=z;
```

$$x := 3\cos(\theta\sim)$$
$$y := 3\sin(\theta\sim) + 3$$
$$z := z \qquad (4.1.3)$$

with $0 \le \theta \le 2\pi$, $0 \le z \le 5$.

```
> x:='x': y:='y':
```

17.4.2 Surface areas

Example 4. Find the area of the paraboloid $x = y^2 + z^2$ that lies inside the cylinder $y^2 + z^2 = 9$. The graph of the surface is shown as follows:

```
> plot3d(y^2+z^2,y=-3..3,z=-3..3,view=0..9,axes=normal,scaling=
  constrained);
```



Step 1. Get the integrand.

```
> s:=x^2+y^2;
```

$$s := x^2 + y^2 \tag{4.2.1}$$

```
> ds:=sqrt(1+diff(s,y)^2+diff(s,z)^2);
```

$$ds := \sqrt{1 + 4\,y^2} \tag{4.2.2}$$

Step 2. Get the equation(s) of the boundary of the region. It is better to use the polar coordinates.

```
> y:=r*cos(theta); z:=r*sin(theta);
```

$$y := r\cos(\theta\sim)$$
$$z := r\sin(\theta\sim) \tag{4.2.3}$$

```
> func:=r*ds;
```

$$func := r\sqrt{1 + 4\,r^2\cos(\theta\sim)^2} \tag{4.2.4}$$

Step 3. Evaluate the integral.

```
> Ar:=int(int(func,r=0..3),theta=0..2*Pi);
```

$$Ar := \frac{37}{3}\,\mathrm{EllipticK}(6\,\mathrm{I}) + \frac{35}{3}\,\mathrm{EllipticE}(6\,\mathrm{I}) \tag{4.2.5}$$

```
> Ar:=evalf(Ar);
```

$$Ar := 80.06978028 - 0.\,\mathrm{I} \tag{4.2.6}$$

Hence, the area is 80.07.

```
> y:='y': z:='z':
```

Example 5. Find the area of the cap cut from the hemisphere $x^2 + y^2 + z^2 = 2$, $z \geq 0$, by the cylinder $x^2 + y^2 = 1$.

```
> s:=sqrt(2-x^2-y^2);
```

$$s := \sqrt{2 - x^2 - y^2} \tag{4.2.7}$$

```
> ds:=sqrt(1+diff(s,x)^2+diff(s,y)^2);
```

$$ds := \sqrt{1 + \frac{x^2}{2 - x^2 - y^2} + \frac{y^2}{2 - x^2 - y^2}} \tag{4.2.8}$$

Evaluate the integral in polar coordinates.

```
> x:=r*cos(theta); y:=r*sin(theta); func:=simplify(ds)*r;
```

$$x := r\cos(\theta\sim)$$
$$y := r\sin(\theta\sim)$$
$$func := \sqrt{2}\,\sqrt{-\frac{1}{-2 + r^2}}\;r \tag{4.2.9}$$

```
> Ar:=int(int(func, r=0..1),theta=0..2*Pi);
```

$$Ar := 4\,\pi - 2\sqrt{2}\;\pi \tag{4.2.10}$$

```
> x:='x': y:='y': r:='r':
```

17.4.3 Parametric surface areas

Example 6. Find the surface area of a sphere of radius a.

```
> assume(a>0): assume(theta, real): assume(phi,real):
```

The parametric equation for the sphere is

```
> x:=a*cos(theta)*sin(phi); y:=a*sin(theta)*sin(phi); z:=a*cos
  (phi);
```

$$x := a\sim \cos(\theta\sim)\sin(\phi\sim)$$
$$y := a\sim \sin(\theta\sim)\sin(\phi\sim)$$
$$z := a\sim \cos(\phi\sim) \quad (4.3.1)$$

```
> rvec:=[x,y,z];
```

$$rvec := [a\sim \cos(\theta\sim)\sin(\phi\sim), a\sim \sin(\theta\sim)\sin(\phi\sim), a\sim \cos(\phi\sim)] \quad (4.3.2)$$

```
> drt:=diff(rvec, theta);
```

$$drt := [-a\sim \sin(\theta\sim)\sin(\phi\sim), a\sim \cos(\theta\sim)\sin(\phi\sim), 0] \quad (4.3.3)$$

```
> drp:=diff(rvec, phi);
```

$$drp := [a\sim \cos(\theta\sim)\cos(\phi\sim), a\sim \sin(\theta\sim)\cos(\phi\sim), -a\sim \sin(\phi\sim)] \quad (4.3.4)$$

```
> dA:=simplify(norm(crossprod(drt,drp),2));
```

$$dA := a\sim^2 |\sin(\phi\sim)| \quad (4.3.5)$$

```
> Ar:=int(int(dA, theta=0..2*Pi),phi=0..Pi);
```

$$Ar := 4\, a\sim^2 \pi \quad (4.3.6)$$

Example 7. Find the surface area of the portion of the cone $z = 2\sqrt{x^2 + y^2}$ between the planes $z = 2$ and $z = 6$.

The parametric equation for the cone is

```
> x:=r*cos(theta); y:=r*sin(theta); z:=2*r;
```

$$x := r\cos(\theta\sim)$$
$$y := r\sin(\theta\sim)$$
$$z := 2\, r \quad (4.3.7)$$

```
> rvec:=[x,y,z];
```

$$rvec := [r\cos(\theta\sim), r\sin(\theta\sim), 2\, r] \quad (4.3.8)$$

```
> drr:=diff(rvec, r);
```

$$drr := [\cos(\theta\sim), \sin(\theta\sim), 2] \quad (4.3.9)$$

```
> drt:=diff(rvec, theta);
```

$$drt := [-r\sin(\theta\sim), r\cos(\theta\sim), 0] \quad (4.3.10)$$

```
> dA:=simplify(norm(crossprod(drr,drt),2));
```

$$dA := \sqrt{5}\, |r| \quad (4.3.11)$$

```
> Ar:=int(int(dA, r=1..3),theta=0..2*Pi);
```

$$Ar := 8\sqrt{5}\, \pi \quad (4.3.12)$$

Example 8. Find the area of the part of helicoid with vector equation $\mathbf{r}(u, v) = u\cos(v)\mathbf{i} + u\sin(v)\mathbf{j} + v\mathbf{k}$, $0 \le u \le 1$, $0 \le v \le \pi$.

```
> assume(v,real):
> x:=u*cos(v); y:=u*sin(v); z:=v;
```

$$x := u\cos(v\sim)$$
$$y := u\sin(v\sim)$$
$$z := v\sim \qquad (4.3.13)$$

```
> rvec:=[x, y, z];
```

$$rvec := [u\cos(v\sim), u\sin(v\sim), v\sim] \qquad (4.3.14)$$

```
> dru:=map(diff, rvec,u);
```

$$dru := [\cos(v\sim), \sin(v\sim), 0] \qquad (4.3.15)$$

```
> drv:=map(diff, rvec,v);
```

$$drv := [-u\sin(v\sim), u\cos(v\sim), 1] \qquad (4.3.16)$$

```
> dA:=simplify(norm(crossprod(dru,drv),2));
```

$$dA := \sqrt{1 + |u|^2} \qquad (4.3.17)$$

```
> Ar:= int(int(dA,u=0..1), v=0..Pi);
```

$$Ar := \frac{1}{2}\sqrt{2}\,\pi - \frac{1}{2}\ln\left(\sqrt{2} - 1\right)\pi \qquad (4.3.18)$$

```
> x:='x': y:='y': z:='z': r:='r':
```

17.4.4 Surface integrals

Example 9. Evaluate the surface integral of $g(x, y, z) = y$ over the surface $z = x+y^2$, $0 \le x \le 1$, $0 \le y \le 2$.

```
> f:=x+y^2:
> plot3d(f,x=0..1,y=0..2,axes=normal);
```



Step 1. Calculate the surface area differential.

```
> ds:=sqrt(1+diff(f,x)^2+diff(f,y)^2);
```

$$ds := \sqrt{2+4y^2}$$

Step 2. Evaluate the surface integral.

```
> int(int(y*ds,x=0..1),y=0..2);
```

$$\frac{13}{3}\sqrt{2}$$

Example 10. Evaluate the surface integral of $g(x, y, z) = xyz$ over the surface of the cube, $-1 \le x \le 1, -1 \le y \le 1, -1 \le z \le 1$, cut from the first octant (the coordinate planes are not included).

The surface contains three squares: $S_1 = \{x = 1, 0 \le y \le 1, 0 \le z \le 1\}$, $S_2 = \{y = 1, 0 \le x \le 1, 0 \le z \le 1\}$,
$S_3 = \{z = 1, 0 \le x \le 1, 0 \le z \le 1\}$. Since $g(x, y, z) = xyz$ is symmetric with respect to x, y, and z, we have

```
> IntV:=3*int(int(y*z, y=0..1),z=0..1);
```

$$IntV := \frac{3}{4} \qquad (4.4.1)$$

Example 11. Evaluate the surface integral of $g(x, y, z) = z - x$ over the surface of $z = x + y^2$, where $0 \le x \le y \le 1$.

```
> z:=x+y^2;
```

$$z := x + y^2 \qquad (4.4.2)$$

```
> ds:=sqrt(1+diff(z,x)^2+diff(z,y)^2);
```

$$ds := \sqrt{2+4y^2} \qquad (4.4.3)$$

```
> g:=z-x;
```

$$g := y^2 \qquad (4.4.4)$$

```
> IntV:=int(int(g*ds, y=x..1), x=0..1);
```

$$IntV := \frac{1}{30}\sqrt{2} + \frac{1}{5}\sqrt{2}\sqrt{3} \qquad (4.4.5)$$

17.4.5 Parametric surface integrals

Example 12. Evaluate the surface integral of x^2 over the unit sphere.
Step 1. Get the parametric equation of the unit sphere.

```
> assume(phi,real): assume(theta,real):
> x:=sin(phi)*cos(theta); y:=sin(phi)*sin(theta); z:=cos(phi);
  rvec:=[x,y,z];
```

$$x := \sin(\phi\sim)\cos(\theta\sim)$$
$$y := \sin(\phi\sim)\sin(\theta\sim)$$
$$z := \cos(\phi\sim)$$

$$rvec := [\sin(\phi\sim)\cos(\theta\sim), \sin(\phi\sim)\sin(\theta\sim), \cos(\phi\sim)]$$

```
> thetallmt:=0: thetaulmt:=2*Pi:
> phillmt:=0: phiulmt:=Pi:
```

Step 2. Obtain the surface differential and change the surface integral to the double integral.

```
> drt:=diff(rvec,theta);
```

$$drt := [-\sin(\phi\sim)\sin(\theta\sim), \sin(\phi\sim)\cos(\theta\sim), 0]$$

```
> drp:=diff(rvec,phi);
```

$$drp := [\cos(\phi\sim)\cos(\theta\sim), \cos(\phi\sim)\sin(\theta\sim), -\sin(\phi\sim)]$$

```
> vds:=simplify(crossprod(drt,drp));
```

$$vds := \left[-\sin(\phi\sim)^2\cos(\theta\sim) \quad -\sin(\phi\sim)^2\sin(\theta\sim) \quad -\sin(\phi\sim)\cos(\phi\sim) \right]$$

```
> ds:=(norm(vds,2));
```

$$ds := \sqrt{\sin(\phi\sim)^4\,|\cos(\theta\sim)|^2 + \sin(\phi\sim)^4\,|\sin(\theta\sim)|^2 + |\sin(\phi\sim)\cos(\phi\sim)|^2}$$

```
> ds:=simplify(ds);
```

$$ds := |\sin(\phi\sim)|$$

```
> func:=x^2;
```

$$func := \sin(\phi\sim)^2\cos(\theta\sim)^2$$

```
> int(int(func*ds,theta=thetallmt..thetaulmt),phi=phillmt..
  phiulmt);
```

$$\frac{4}{3}\pi$$

```
> x:='x': y:='y': z:='z': r:='r':
```

Example 13. Evaluate the surface integral of x^2 over the cone $z = \sqrt{x^2 + y^2}$, $0 \le z \le 1$.

```
> x:=r*cos(theta); y:=r*sin(theta); z:= r;
```

$$x := r\cos(\theta\sim)$$
$$y := r\sin(\theta\sim)$$
$$z := r \qquad (4.5.1)$$

```
> rvec:=[x, y, z];
```

$$rvec := [r\cos(\theta\sim), r\sin(\theta\sim), r] \qquad (4.5.2)$$

```
> drr:= diff(rvec,r);
```

$$drr := [\cos(\theta\sim), \sin(\theta\sim), 1] \qquad (4.5.3)$$

```
> drt:=diff(rvec,theta);
```

$$drt := [-r\sin(\theta\sim), r\cos(\theta\sim), 0] \qquad (4.5.4)$$

```
> ds:= simplify(norm(crossprod(drr,drt),2));
```

$$ds := \sqrt{2}\,|r| \qquad (4.5.5)$$

```
> Intv:=int(int(x^2*ds,r=0..1),theta=0..2*Pi);
```

$$Intv := \frac{1}{4}\sqrt{2}\,\pi \qquad (4.5.6)$$

```
> x:='x': y:='y': z:='z': r:='r':
```

Exercises

1. Find a parametrization of $z = 1 - x^2 - y^2$, $0 \le z$. Then draw its graph.
2. Find a parametrization of the hemisphere $x^2 + y^2 + z^2 = 4$, $0 \le z$. Then draw its graph.
3. Find the area of the paraboloid $x = y^2 + z^2$ that lies inside the cylinder $y^2 + z^2 = 9$.
4. Find the surface area of the portion of the cone $z = \sqrt{x^2 + y^2}$ between the planes $z = 1$ and $z = 2$.
5. Evaluate the surface integral of $g(x, y, z) = \sqrt{x^2 + y^2}$ over the surface $x = r\cos(\theta)$, $y = r\sin(\theta)$, $z = \theta$, $0 \le r \le 1$, $0 \le \theta \le 2\pi$.
6. Evaluate the surface integral of $g(x, y, z) = z\left(x^2 + y^2\right)$ over the surface $z = 4 - x^2 - y^2$, $0 \le z$.
7. Evaluate the surface integral of $g(x, y, z) = y$ over the hemisphere $x^2 + y^2 + z^2 = 4$, $0 \le z$.

17.5 Surface Integrals of Vector Fileds

(1) The vector surface integral defined on an **oriented surface** S is the following:

$$\iint_S \mathbf{F}\cdot d\mathbf{S} = \iint_S \mathbf{F}\cdot\mathbf{n}\,dS$$

where **n** is the unit normal vector of the oriented surface S.
(2) The **flux** of a three dimensional vector field **F** across an oriented surface S in the direction of **n** is

$$\text{Flux} = \iint_S \mathbf{F}\cdot\mathbf{n}\,dS$$

If the surface is given by the equation $g(x, y, z) = c$, then by $dS = \dfrac{|\nabla g|}{|\nabla g\cdot\mathbf{p}|}dA$, and $\mathbf{n} = \pm\dfrac{\nabla g}{|\nabla g|}$,

$$\text{Flux} = \iint_R \mathbf{F}\cdot\frac{\pm\nabla g}{|\nabla g\cdot\mathbf{p}|}dA$$

Particularly, if the surface is given by $z = z(x, y)$, and the orientation is the upward one, then $g(x, y, z) = z - z(x, y)$, $\nabla g = \left[-\dfrac{\partial z}{\partial x}, -\dfrac{\partial z}{\partial y}, 1\right]$, $|\nabla g\cdot\mathbf{p}| = 1$, and

$$\text{Flux} = \iint_R \mathbf{F}\cdot\nabla g\,dA$$

(3) The **flux** of a three dimensional vector field **F** across an oriented, parameterized surface S in the direction of **n** is

$$\text{Flux} = \iint_S \mathbf{F}\cdot\mathbf{n}\,dS = \int_c^d\int_a^b \mathbf{F}\cdot\left(\mathbf{r}_u\times\mathbf{r}_v\right)du\,dv$$

17.5.1 Vector Surface integral

Example 1. Evaluate the surface integral of the vector field **F**(x, y, z) = e^y**i** + ye^x**j** +x^2y**k** over the surface S, which is the part of the paraboloid $z = x^2 + y^2$ that lies above the square $0 \le x \le 1$, $0 \le y \le 1$, and has upward orientation.

Step 1. Plot the graph of the surface.

```
> plot3d(x^2+y^2, x=0..1, y=0..1, view=0..1, axes=normal,
  scaling=constrained);
```



Step 2. Calculate **F** • ∇g.

```
> x:='x': y:='y': z:=x^2+y^2:
> fieldf:=[exp(y),y*exp(x),x^2*y];
```

$$fieldf := \left[e^y, y\,e^x, x^2\,y\right]$$

```
> ds:=[-diff(z,x), -diff(z,y),1];
```

$$ds := [-2\,x, -2\,y, 1]$$

```
> ifunc:=linalg[innerprod](fieldf,ds);
```

$$ifunc := -2\,e^y\,x - 2\,y^2\,e^x + x^2\,y$$

Step 3. Evaluate the integral.

```
> int(int(ifunc,x=0..1),y=0..1);
```

$$\frac{11}{6} - \frac{5}{3}\,e$$

Example 2. Evaluate the surface integral of the vector field **F**(x, y, z) = y **i** + x **j** + z **k** over the surface S, which is the boundary of the solid region enclosed by the paraboloid $z = 1 - x^2 - y^2$ and the plane $z = 0$.

Step 1. Plot the graph of the surface.

```
> x:='x': y:='y': z:=1-x^2-y^2:
> plot3d(z, x=-1..1,y=-1..1,axes=normal, scaling=constrained,
  view=0..1);
```



Step2. Calculate $\mathbf{F} \cdot \nabla g$.

```
> fieldf:=[y,x,z];
```

$$fieldf := \left[y, x, 1 - x^2 - y^2\right]$$

```
> dg:=[-diff(z,x),-diff(z,y),1];
```

$$dg := [2\,x, 2\,y, 1]$$

```
> ifunc:=linalg[innerprod](fieldf,dg);
```

$$ifunc := 4\,y\,x + 1 - x^2 - y^2$$

Step3. Evaluate the integral in the polar coordinates.

```
> r:='r': theta:='theta':
> x:=r*cos(theta): y:=r*sin(theta):
> int(int(ifunc*r,r=0..1),theta=0..2*Pi);
```

$$\frac{1}{2}\,\pi$$

```
> x:='x': y:='y': z:='z':
```

17.5.2 Surface integral for flux

Example 3. Find the flux of the vector field $\mathbf{F}(x, y, z) = yz\mathbf{j} + z^2\mathbf{k}$ outward through the surface S cut from the cylinder $y^2 + z^2 = 1$, $z \geq 0$, by the planes $x = 0$ and $x = 1$.
Method 1. Directly apply the formula for calculating flux.
Step 1. Plot the graph of the surface.

```
> implicitplot3d(y^2+z^2=1,x=0..1,y=-1..1,z=0..1, axes=normal);
```



Step2. Calculate $(\mathbf{F}\cdot\nabla g) / |\nabla g\cdot\mathbf{p}|$.

```
> F:=[0, y*z, z^2]; g:=y^2+z^2; assume(z>0):
```

$$F := [0, y\,z, z^2]$$

$$g := y^2 + z^2 \tag{5.2.1}$$

```
> dg:=grad(g, [x,y,z]); k:=[0,0,1]:
```

$$dg := [0 \quad 2\,y \quad 2\,z\sim] \tag{5.2.2}$$

```
> ifunc:=simplify(innerprod(F, dg)/abs(inner(dg, k)));
```

$$ifunc := y^2 + z\sim^2 \tag{5.2.3}$$

By $y^2 + z^2 = 1$, we have

```
> ifunc:=1:
> flx:=int(int(ifunc, x=0..1), y=-1..1);
```

$$flx := 2 \tag{5.2.4}$$

Method 2. Solve for z as the function of y and x. Then calculate the surface integral for flux.

```
> z:= sqrt(1-y^2); F:=[0, y*z, z^2];
```

$$z := \sqrt{1 - y^2}$$

$$F := \left[0, y\sqrt{1-y^2}, 1 - y^2\right] \tag{5.2.5}$$

```
> dg:=[-diff(z,x),-diff(z,y), 1];
```

$$dg := \left[0, \frac{y}{\sqrt{1-y^2}}, 1\right] \tag{5.2.6}$$

```
> ifunc:=innerprod(F,dg);
```

$$ifunc := 1 \tag{5.2.7}$$

```
> flx:=int(int(ifunc, x=0..1), y=-1..1);
```

$$flx := 2 \tag{5.2.8}$$

```
> z:='z':
```

Example 4. Find the flux of the vector field $\mathbf{F}(x, y, z) = z\mathbf{i} + y\mathbf{j} + x\mathbf{k}$ outward across the unit sphere.
Step 1. Get the parametric equation.

```
> x:=cos(theta)*sin(phi):y:=sin(theta)*sin(phi):z:=cos(phi)
  :rvec:=[x,y,z]:
> thetallmt:=0: thetaulmt:=2*Pi: phillmt:=0: phiulmt:=Pi:
```

Step 2. Calculate $\mathbf{F} \cdot (\mathbf{r}_u \times \mathbf{r}_v) du dv$.

```
> dr1:=diff(rvec,phi);
```

$$dr1 := [\cos(\theta)\cos(\phi\sim), \sin(\theta)\cos(\phi\sim), -\sin(\phi\sim)]$$

```
> dr2:=diff(rvec,theta);
```

$$dr2 := [-\sin(\theta)\sin(\phi\sim), \cos(\theta)\sin(\phi\sim), 0]$$

```
> F:=[z,y,x];
```

$$F := [\cos(\phi\sim), \sin(\theta)\sin(\phi\sim), \cos(\theta)\sin(\phi\sim)]$$

```
> vds:=simplify(crossprod(dr1,dr2));
```

$$vds := \begin{bmatrix} \sin(\phi\sim)^2\cos(\theta) & \sin(\phi\sim)^2\sin(\theta) & \sin(\phi\sim)\cos(\phi\sim) \end{bmatrix}$$

```
> ifunc:=innerprod(F,vds);
```

$$ifunc := 2\cos(\phi\sim)\sin(\phi\sim)^2\cos(\theta) + \sin(\theta)^2\sin(\phi\sim)^3$$

```
> Flux:=int(int(ifunc,theta=thetallmt..thetaulmt),phi=phillmt..
  phiulmt);
```

$$Flux := \frac{4}{3}\pi$$

```
> x:='x': y:='y': z:='z': r:='r':
```

Example 5. Find the flux of the vector field $\mathbf{F}(x, y, z) = yz\mathbf{i} + x^2\mathbf{j} - z^2\mathbf{k}$ outward through the parabolic cylinder $y = x^2$, $0 < x \le 1$, $0 \le z \le 4$.

```
> y:=x^2; r:=[x,y,z];
```

$$y := x^2$$

$$r := [x, x^2, z] \quad (5.2.9)$$

```
> drx:=map(diff, r, x);
```

$$drx := [1, 2x, 0] \quad (5.2.10)$$

```
> drz:=map(diff, r, z);
```

$$drz := [0, 0, 1] \quad (5.2.11)$$

```
> nds:=crossprod(drx, drz);
```

$$nds := [2x \quad -1 \quad 0] \quad (5.2.12)$$

```
> F:=[y*z, x^2, z^2];
```

$$F := [x^2 z, x^2, z^2] \quad (5.2.13)$$

```
> func:=innerprod(F, nds);
```

$$func := 2x^3 z - x^2 \quad (5.2.14)$$

```
> Flux:=int(int(func, x=0..1),z=0..4);
```

$$Flux := \frac{8}{3} \tag{5.2.15}$$

Exercises

1. Let $\mathbf{F}(x, y, z) = y\mathbf{i} + z\mathbf{j} + x\mathbf{k}$. Let S be the oriented surface parameterized by $x = u^2 - v$, $y = u + v$, $z = v^2$ for $0 \le u \le 2$, $-1 \le v \le 1$. (a) Find $\mathbf{n}$ and $\mathbf{F}\cdot\mathbf{n}$. (b) $\iint_S \mathbf{F}\cdot d\mathbf{S}$.
2. Evaluate the surface integral of the vector field $\mathbf{F}(x, y, z) = x\mathbf{i} + y\mathbf{j} + z\mathbf{k}$ over the hemisphere $x^2 + y^2 + z^2 = 9$, $0 \le z$, with the upward-pointing normal.
3. Evaluate the surface integral of the vector field $\mathbf{F}(x, y, z) = x\mathbf{i} + y\mathbf{j} + e^z\mathbf{k}$ over the cylinder $x^2 + y^2 = 4$, $1 \le z \le 5$, with the upward-pointing normal.
4. Find the flux of the vector field $\mathbf{F}(x, y, z) = xz\mathbf{i} + yz\mathbf{j} + 1/z\mathbf{k}$ outward through the surface S cut from the part of the cylinder $x^2 + y^2 = 1$, $x \ge 0$, $y \ge 0$, and between the planes $z = 1$ and $z = 2$.
5. Find the flux of the vector field $\mathbf{F}(x, y, z) = xy\mathbf{i} + yz\mathbf{j} + zx\mathbf{k}$ outward across the unit sphere.

Chapter 18 FUNDAMENTAL THEOREMS OF VECTOR ANALYSIS

18.1 Green's Theorem

(1) **Green's Theorem:** Let R be a domain whose boundary C is a simple closed curve, oriented counterclockwise. If $M(x, y)$ and $N(x, y)$ are differentiable and have continuous first partial derivatives on R, then

$$\oint_C M\mathrm{d}x + N\mathrm{d}y = \iint_R \left(\frac{\partial N}{\partial x} - \frac{\partial M}{\partial y}\right)\mathrm{d}x\mathrm{d}y$$

(2) **The area formula of line integral:** The area enclosed in a closed curve C can be calculated be the line integral

$$A = \frac{1}{2}\oint_C x\mathrm{d}y - y\mathrm{d}x$$

(3) The **divergence (flux density)** of a two-dimensional vector field $\mathbf{F} = M\mathbf{i} + N\mathbf{j}$ at the point (x, y) is

$$\operatorname{div} \mathbf{F} = \frac{\partial M}{\partial x} + \frac{\partial N}{\partial y}$$

(4) The **k-component of the curl** (also called a **scalar curl** or **circulation density**) of a two-dimensional vector field $\mathbf{F} = M\mathbf{i} + N\mathbf{j}$ at the point (x, y) is

$$\operatorname{curl}_z(\mathbf{F}) = (\operatorname{curl} \mathbf{F})\cdot \mathbf{k} = \frac{\partial N}{\partial x} - \frac{\partial M}{\partial y}$$

More general forms of Green's Theorem are the following:

(5) **Green's Theorem (circulation-curl form** or **tangential form):**

$$\oint_C \mathbf{F}\cdot\mathrm{d}\mathbf{r} = \oint_C \mathbf{F}\cdot\mathbf{T}\mathrm{d}s = \iint_R \operatorname{curl}_z(\mathbf{F})\mathrm{d}x\mathrm{d}y$$

(6) **Green's Theorem (flux-divergence form** or **normal form):**

$$\oint_C \mathbf{F}\cdot\mathbf{n}\mathrm{d}s = \iint_R \operatorname{div}(\mathbf{F})\mathrm{d}x\mathrm{d}y$$

18.1.1 Find divergence

Example 1. Find the divergence of $\mathbf{F}(x, y) = (x^2 - y)\mathbf{i} + (xy - y^2)\mathbf{j}$.

```
> divF:=diff(x^2-y,x)+diff(x*y-y^2,y);
```

$$divF := 3\,x - 2\,y \qquad (1.1.1)$$

Example 2. Find the divergence of $\mathbf{F}(x, y) = (x + \mathrm{e}^x\sin(y))\mathbf{i} + (x + \mathrm{e}^x\cos(y))\mathbf{j}$.

```
> divF:=diff(x+exp(x)*sin(y),x)+diff(x+exp(x)*cos(y),y);
```

$$divF := 1 \qquad (1.1.2)$$

Note. You can use the command "**diverge**" in the "**linalg**" package to calculate the divergence of a vector field.

Example 3. Use the command "**diverge**" in the "**linalg**" package to find the divergence of $\mathbf{F}(x, y) = (x^2 - y)\mathbf{i} + (xy - y^2)\mathbf{j}$.

```
> with(linalg):
Warning, the protected names norm and trace have been redefined
and unprotected
> F:=[x^2-y,x*y-y^2];
```

$$F := \left[x^2 - y, x\,y - y^2\right] \qquad (1.1.3)$$

```
> diverge(F,[x,y]);
```

$$3\,x - 2\,y \qquad (1.1.4)$$

18.1.2 Use Green's Theorem to evaluate the line integrals

Example 4. Evaluate the counterclockwise circulation of the vector field $\mathbf{F}(x, y) = x^4\mathbf{i} + xy\mathbf{j}$ over C, where C consists the line segments from (0,0) to (1, 0), from (1, 0) to (0, 1), and from (0, 1) to (0, 0).

Step1. Get the graph of the curve C.

```
> plot({[0,0],[1,0],[0,1],[0,0]}, color='blue',style=LINE);
```



Step 2. Change the line integral to double integral using Green's Theorem.

```
> P:=x^4; Q:=x*y;
```

$$P := x^4$$

$$Q := x\,y$$

```
> func:=diff(Q,x)-diff(P,y);
```

$$func := y$$

```
> int(int(func, y=0..1-x),x=0..1);
```

$$\frac{1}{6}$$

Example 5. Evaluate the counterclockwise circulation of $\mathbf{F}(x, y) = (3\,y - e^{\sin(x)})\mathbf{i} + (7\,x + \sqrt{y^4+1}\,)\mathbf{j}$ over C, where C is the circle $x^2 + y^2 = 9$.

```
> M:=3*y-exp(sin(x)); N:=7*x+sqrt(y^4+1);
```

$$M := 3\,y - e^{\sin(x)}$$

$$N := 7\,x + \sqrt{y^4 + 1}$$

```
> func:=diff(N,x)-diff(M,y);
```
$$func := 4$$

```
> int(int(func,y=-sqrt(9-x^2)..sqrt(9-x^2)),x=-3..3);
```
$$36\,\pi$$

Example 6. Evaluate the counterclockwise circulation of the vector field $\mathbf{F}(x, y) = y^2\mathbf{i} + 3xy\mathbf{j}$ over C, where C is the boundary of the semi-annual region in the upper half-plane between the circles $x^2+y^2 = 1$ and the circle $x^2+y^2 = 4$.

```
> plot({sqrt(1-x^2), sqrt(4-x^2)}, x=-2..2, scaling=constrained)
  ;
```




```
> M:=y^2; N:=3*x*y;
```
$$M := y^2$$
$$N := 3\,x\,y$$

```
> func:=diff(N,x)-diff(M,y);
```
$$func := y$$

Change to the polar coordinates.

```
> x:=r*cos(theta): y:=r*sin(theta):
> int(int(func*r,r=1..2),theta=0..Pi);
```
$$\frac{14}{3}$$

```
> x:='x': y:='y':
```

Note: You can also use the rectangular coordinates to evaluate the integral above.

```
> int(int(y, y=0..sqrt(4-x^2)), x=-2..2)-int(int(y, y=0..sqrt(1-
  x^2)), x=-1..1);
```
$$\frac{14}{3} \qquad (1.2.1)$$

Example 7. Evaluate the line integral

$$\oint_C xy\mathrm{d}x - y^2\mathrm{d}y$$

where C is the square cut from the first quadrant by the lines $x = 1$ and $y = 1$.

```
> plot({[0,0],[1,0],[1,1],[0,1],[0,0]}, style=LINE);
```




```
> M:=y^2; N:=x*y;
```

$$M := y^2$$

$$N := x\,y \qquad (1.2.2)$$

```
> func:=diverge([M,N], [x,y]);
```

$$func := x \qquad (1.2.3)$$

```
> int(int(func, x=0..1),y=0..1);
```

$$\frac{1}{2} \qquad (1.2.4)$$

Example 8. Calculate the outward flux of the field $\mathbf{F}(x, y) = x\mathbf{i} + y^2\mathbf{j}$ across the square bounded by the lines $x = \pm 1$ and $y = \pm 1$.

```
> func:=diverge([x, y^2], [x,y]);
```

$$func := 1 + 2\,y \qquad (1.2.5)$$

```
> Flux:=int(int(func, x=-1..1),y=-1..1);
```

$$Flux := 4 \qquad (1.2.6)$$

18.1.3 Find the area of a region using the formula of line integral

Example 9. Find the area enclosed by the ellipse $\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1$ by the line integral.

Step1. Get the parametric equation of the ellipse.

```
> x:=a*cos(t); y:=b*sin(t);
```

$$x := a\cos(t)$$

$$y := b\sin(t)$$

Step 2. Evaluate the line integral.

```
> lowlmt:=0: uplmt:=2*Pi:
```

```
> ifunc:=1/2*(x*diff(y,t)-y*diff(x,t));
```

$$ifunc := \frac{1}{2}\, a \cos(t)^2\, b + \frac{1}{2}\, b \sin(t)^2\, a$$

```
> A:=int(ifunc,t=lowlmt..uplmt);
```

$$A := \pi\, a\, b$$

You can verify the result using the area formula in double integral.

```
> x:='x': y:='y': assume(a>0): assume(b>0):
> A:=int(int(1, y=-b*sqrt(1-x^2/a^2)..b*sqrt(1-x^2/a^2)), x=-a..
  a);
```

$$A := a\sim\, \pi\, b\sim \tag{1.3.1}$$

18.1.4 Use Green's Theorem to change the path of the line integrals

Example 10. If $\mathbf{F}(x, y) = (-y\mathbf{i} + x\mathbf{j})/(x^2+y^2)$ and C is the square bounded by the lines $x = \pm 1$ and $y = \pm 1$, find the integral of $\mathbf{F}$ over C.

```
> x:='x': y:='y':
> P:=-y/(x^2+y^2): Q:=x/(x^2+y^2):
> testf:=diff(Q,x)-diff(P,y);
```

$$testf := \frac{2}{x^2+y^2} - \frac{2\,x^2}{\left(x^2+y^2\right)^2} - \frac{2\,y^2}{\left(x^2+y^2\right)^2}$$

```
> simplify(testf);
```

$$0$$

Hence, $\frac{dQ}{dx} - \frac{dP}{dy} = 0$ except at the origin (0, 0), where the first partial derivatives do not exist. Then the path can be changed to any simple closed path containing the origin, for example, the unit circle. Hence, we can evaluate the integral using the new path $x = \cos(t)$, $y = \sin(t)$, $0 \le t \le 2\pi$, oriented counterclockwise.

```
> x:=cos(t); y:=sin(t); rvec:=[x,y];
```

$$x := \cos(t)$$

$$y := \sin(t)$$

$$rvec := [\cos(t), \sin(t)]$$

```
> fieldf:=simplify([-y/(x^2+y^2),x/(x^2+y^2)]);
```

$$fieldf := [-\sin(t), \cos(t)]$$

```
> dr:=diff(rvec,t);
```

$$dr := [-\sin(t), \cos(t)]$$

```
> assume(t>0): ifunc:=simplify(dotprod(fieldf,dr));
```

$$ifunc := 1$$

```
> int(ifunc,t=0..2*Pi);
```

$$2\,\pi$$

```
> x:='x': y:='y':
```

Exercises

1. Verify Green's Theorem for the line integral $\oint_C xy\,dx + y\,dy$, where C is the unit circle, oriented counterclockwise by (a) evaluating the integral directly, (b) applying Green's Theorem.
2. Use Green's Theorem to evaluate the line integral $\oint_C y^2 dx + x^2 dy$, where C is the boundary of the unit square $0 \le x \le 1$, $0 \le y \le 1$, oriented counterclockwise.
3. Use Green's Theorem to evaluate the line integral $\oint_C e^{2x+y} dx + e^{-y} dy$, where C is the triangle with vertices $(0, 0)$, $(1, 0)$, $(1, 1)$, oriented counterclockwise.
4. Use Green's Theorem to evaluate the work done by the force $\mathbf{F} = [x + y, x^2 - y]$, over the close loop C, where C is the boundary of the region enclosed by $y = x^2$, $y = \sqrt{x}$, $0 \le x \le 1$, oriented counterclockwise.
5. Use Green's Theorem to evaluate the line integral $\oint_C \mathbf{F}\cdot d\mathbf{r}$, where $\mathbf{F} = [x^2, x^2]$ and C consists of the arcs $y = x^2$, $y = x$, $0 \le x \le 1$, oriented counterclockwise.

18.2 Stokes' Theorem

(1) The **curl vector** of a vector field $\mathbf{F} = M\mathbf{i} + N\mathbf{j} + P\mathbf{k}$ is

$$\text{curl}\,(\mathbf{F}) = \nabla\times\mathbf{F} = \left(\frac{\partial P}{\partial y} - \frac{\partial N}{\partial z}\right)\mathbf{i} + \left(\frac{\partial M}{\partial z} - \frac{\partial P}{\partial x}\right)\mathbf{j} + \left(\frac{\partial N}{\partial x} - \frac{\partial M}{\partial y}\right)\mathbf{k}$$

(2) **Stokes' Theorem:** The circulation of a vector field $\mathbf{F} = M\mathbf{i} + N\mathbf{j} + P\mathbf{k}$ around the boundary C of an oriented surface S in the direction counterclockwise with respect to the surface's unit vector $\mathbf{n}$ is equal to the integral of curl $\mathbf{F}$ over S.

$$\oint_C \mathbf{F}\cdot d\mathbf{r} = \iint_S \text{curl}\,(\mathbf{F})\cdot d\mathbf{S} = \iint_S (\nabla\times\mathbf{F})\cdot\mathbf{n}\,d\sigma$$

(3) The following identity is important.

$$\text{curl grad} f = 0, \text{ i.e., } \nabla\times\nabla f = 0.$$

(4) If $\nabla\times\mathbf{F} = 0$ at every point of a simply connected open region D in space, then $\mathbf{F}$ is a conservative vector field on D.

Note. In MAPLE, the command "**curl**" in the "**linalg**" package can be used to calculate the curl vector of a vector field.

18.2.1 Find the curl and divergence of a vector field

Example 1. Find the curl of the vector field $\mathbf{F}(x, y, z) = x^2y\mathbf{i} + yz^2\mathbf{j} + zx^2\mathbf{k}$.

```
> F:=[x^2*y,y*z^2,z*x^2]; v:=[x,y,z];
```

$$F := [x^2 y, y z^2, z x^2]$$

$$v := [x, y, z]$$

```
> curl(F,v);
```

$$\begin{bmatrix} -2yz & -2zx & -x^2 \end{bmatrix}$$

Example 2. Find the curl of the vector field $\mathbf{F}(x, y, z) = \left(x^2 - y\right)\mathbf{i} + 4z\mathbf{j} + x^2\mathbf{k}$.

```
> F:=[x^2-y,4*z,x^2]; v:=[x,y,z];
```

$$F := \left[x^2 - y, 4\,z, x^2\right]$$

$$v := [x, y, z] \quad (2.1.1)$$

```
> curl(F,v);
```

$$[-4 \quad -2\,x \quad 1\,] \quad (2.1.2)$$

18.2.2 Determine by its curl whether or not a vector field is conservative

Example 3. Determine whether or not the vector field $\mathbf{F}(x, y, z) = yz\mathbf{i} - z^2\mathbf{j} + x^2\mathbf{k}$ is conservative.

```
> F:=[y*z,z^2,x^2]: v:=[x,y,z]:
> curl(F,v);
```

$$[-2\,z \quad y - 2\,x \quad -z]$$

Hence **F** is not a conservative field.

Example 4. Determine whether or not the vector field $\mathbf{F}(x, y, z) = yz\mathbf{i} + (y^2 + xz)\mathbf{j} + xy\mathbf{k}$ is conservative.

```
> F:=[y*z,y^2+x*z,x*y]:
> curl(F,v);
```

$$[0 \quad 0 \quad 0\,]$$

Hence **F** is a conservative field.

Example 5. Let $\mathbf{r} = [x, y, z]$, $r = \|\mathbf{r}\|$. To verify (1) grad $(r) = \mathbf{r}/r$; (2) curl($\mathbf{r}$) = $\mathbf{0}$; and (3) diverge $(r\mathbf{r}) = 4r$.

(1) Verify grad $(r) = \mathbf{r}/r$.

```
> r:=[x,y,z]: v=[x,y,z]:
> absr:=sqrt(x^2+y^2+z^2);
```

$$absr := \sqrt{x^2 + y^2 + z^2}$$

```
> grad(absr,v);
```

$$\left[\frac{x}{\sqrt{x^2 + y^2 + z^2}} \quad \frac{y}{\sqrt{x^2 + y^2 + z^2}} \quad \frac{z}{\sqrt{x^2 + y^2 + z^2}}\right] \quad (2.2.1)$$

Hence, grad $(r) = \mathbf{r}/r$.

(2) Verify curl($\mathbf{r}$) = $\mathbf{0}$.

```
> curl(r,v);
```

$$[0 \quad 0 \quad 0\,]$$

(3) Verify diverge $(r\mathbf{r}) = 4r$.

```
> func:=[absr*x,absr*y,absr*z];
```

$$func := \left[\sqrt{x^2+y^2+z^2}\,x, \sqrt{x^2+y^2+z^2}\,y, \sqrt{x^2+y^2+z^2}\,z\right]$$

```
> diverge(func,v);
```

$$\frac{x^2}{\sqrt{x^2+y^2+z^2}} + 3\sqrt{x^2+y^2+z^2} + \frac{y^2}{\sqrt{x^2+y^2+z^2}} + \frac{z^2}{\sqrt{x^2+y^2+z^2}}$$

```
> simplify(%);
```

$$4\sqrt{x^2+y^2+z^2}$$

Hence, diverge $(r\mathbf{r}) = 4r$.

18.2.3 Evaluate circulation using Stokes' Theorem

Example 6. Evaluate the circulation of vector field **F** around the curve C counterclockwise when viewed from above, where $\mathbf{F} = -y^2\mathbf{i} + x\mathbf{j} + z^2\mathbf{k}$ and C is the intersection of the plane $y+z=2$ and $x^2+y^2=1$.

Step 1. Calculate $(\nabla \times \mathbf{F})$

```
> F:=[-y^2, x, z^2];
```

$$F := \left[-y^2, x, z^2\right]$$

```
> cf:=curl(F,[x,y,z]);
```

$$cf := \begin{bmatrix} 0 & 0 & 1+2\,y \end{bmatrix}$$

Step 2. Establish the surface equation and calculate $\mathbf{n}d\sigma$.

The surface has equation $z = 2-y$, bounded by $x^2+y^2 = 1$, with upward orientation.

```
> z:=2-y;
```

$$z := 2-y$$

```
> nds:=[-diff(z,x),-diff(z,y),1];
```

$$nds := [0, 1, 1]$$

Step 3. Evaluate the integral.

```
> ifunc:=innerprod(cf,nds);
```

$$ifunc := 1+2\,y$$

```
>
> LineInt:=int(int(ifunc,y=-sqrt(1-x^2)..sqrt(1-x^2)),x=-1..1);
```

$$LineInt := \pi$$

```
> z:='z':
```

Example 7. Find the circulation of the field $\mathbf{F}(x, y, z) = (x^2 - y)\mathbf{i} + 4z\mathbf{j} + x^2\mathbf{k}$ around the curve C in which the plane $z=2$ meets the cone $z=\sqrt{x^2+y^2}$, counterclockwise as viewed from above.

Step1. Find curl of the vector field **F**.

```
> F:=[x^2-y, 4*z, x^2];
```

$$F := \left[x^2-y, 4\,z, x^2\right] \tag{2.3.1}$$

```
> cf:=curl(F, [x,y,z]);
```
$$cf := [\,-4 \quad -2\,x \quad 1\,] \tag{2.3.2}$$

Step 2. Find the parametric equation of the surface and calculate **n**dσ.

```
> assume(theta>0, r>0):x:=r*cos(theta);y:=r*sin(theta);z:=
  simplify(sqrt(x^2+y^2));
```
$$x := r\sim\cos(\theta\sim)$$
$$y := r\sim\sin(\theta\sim)$$
$$z := r\sim \tag{2.3.3}$$

```
> vecr:=[x,y,z];
```
$$vecr := [r\sim\cos(\theta\sim),\, r\sim\sin(\theta\sim),\, r\sim] \tag{2.3.4}$$

```
> drr:=diff(vecr,r);
```
$$drr := [\cos(\theta\sim),\, \sin(\theta\sim),\, 1] \tag{2.3.5}$$

```
> drt:=diff(vecr, theta);
```
$$drt := [-r\sim\sin(\theta\sim),\, r\sim\cos(\theta\sim),\, 0] \tag{2.3.6}$$

```
> nds:=crossprod(drr, drt);
```
$$nds := \left[\, -r\sim\cos(\theta\sim) \quad -r\sim\sin(\theta\sim) \quad \cos(\theta\sim)^2\, r\sim + \sin(\theta\sim)^2\, r\sim \,\right] \tag{2.3.7}$$

Step 3. Evaluate the line integral.

```
> ifunc:=innerprod(cf,nds);
```
$$ifunc := 4\, r\sim\cos(\theta\sim) + 2\, r\sim^2\cos(\theta\sim)\sin(\theta\sim) + \cos(\theta\sim)^2\, r\sim + \sin(\theta\sim)^2\, r\sim \tag{2.3.8}$$

```
> LineInt:=int(int(ifunc, r=0..2), theta=0..2*Pi);
```
$$LineInt := 4\,\pi \tag{2.3.9}$$

```
> x:='x': y:='y': z:='z':
```

Example 8. Use Stokes' Theorem to evaluate the line integral $\oint_C \mathbf{F}\cdot d\mathbf{r}$, if $\mathbf{F} = xz\mathbf{i} + xy\mathbf{j} + 3xz\mathbf{k}$ and C is the boundary of the portion of the plane $2x + y + z = 2$ in the first octant, traversed counterclockwise as viewed from above.

```
> plot3d(2-2*x-y,x=0..2,y=0..2,view=0..2, axes=normal);
```



```
> F:=[x*z, x*y, 3*x*z]; v:[x, y, z];
```

$$F := [x\,z, x\,y, 3\,x\,z]$$

$$[x, y, z] \quad (2.3.10)$$

```
> cf:=curl(F,v);
```

$$cf := [0 \quad x-3\,z \quad y\,] \quad (2.3.11)$$

```
> z:=2-2*x-y;
```

$$z := 2-2\,x-y \quad (2.3.12)$$

```
> nds:=[-diff(z,x), -diff(z,y),1];
```

$$nds := [2, 1, 1] \quad (2.3.13)$$

```
> ifunc:=innerprod(cf, nds);
```

$$ifunc := 7\,x-6+4\,y \quad (2.3.14)$$

The domain for the double integral is bounded by $2x+y=2$, x-axis, and y-axis.

```
> LineInt:=int(int(ifunc,y=0.. 2-2*x),x=0..1);
```

$$LineInt := -1 \quad (2.3.15)$$

```
> z:='z':
```

18.2.4 Evaluate the surface integral using Stokes' Theorem

Example 9. Compute the integral of vector field curl(**F**) outward across the surface S, where **F** = $yz\mathbf{i}+xz\mathbf{j}+xy\mathbf{k}$ and S is the part of the sphere $x^2+y^2+z^2 = 4$ that lies inside the cylinder $x^2+y^2=1$ and above the xy-plane.

Step 1. Find the equation of the boundary curve. The boundary curve is the intersection of $x^2+y^2+z^2 = 4$ and $x^2+y^2 = 1$ that is also the intersection of $z = \sqrt{3}$ and $x^2+y^2=1$. Hence, the parametric equation of the curve is the following:

```
> x:=cos(t): y:=sin(t): z:=sqrt(3): r:=[x,y,z]:
```

Step 2. Change to the line integral.

```
> lowlmt:=0: uplmt:=2*Pi:
> vr:=[x, y, z];
```

$$vr := \left[\cos(t\sim), \sin(t\sim), \sqrt{3}\,\right]$$

```
> F:=[y*z,x*z,x*y];
```

$$F := \left[\sin(t\sim)\sqrt{3}, \sqrt{3}\,\cos(t\sim), \cos(t\sim)\sin(t\sim)\right]$$

```
> dr:=diff(r,t);
```

$$dr := [-\sin(t\sim), \cos(t\sim), 0]$$

```
> ifunc:=simplify(innerprod(F,dr));
```

$$ifunc := \sqrt{3}\,\left(2\cos(t\sim)^2-1\right)$$

```
> int(ifunc,t=lowlmt..uplmt);
```

$$0$$

```
> x:='x': y:='y': z:='z': r:='r':
```

Exercises

1. Find the curl of the vector field $\mathbf{F}(x, y, z) = (z - y^2)\mathbf{i} + (x + z^3)\mathbf{j} + (y + x^2)\mathbf{k}$.
2. Find the curl of the vector field $\mathbf{F}(x, y, z) = e^{y}\mathbf{i} + \sin(x)\mathbf{j} + \cos(x)\mathbf{k}$.
3. Determine by its curl whether or not the vector field $\mathbf{F}(x, y, z)$ $= xyz\mathbf{i} - yz\mathbf{j} + xz\mathbf{k}$ is conservative.
4. Determine by its curl whether or not the vector field $\mathbf{F}(x, y, z) = y\sin(z)\mathbf{i} - x\sin(z)\mathbf{j} + xy\cos(z)\mathbf{k}$ is conservative.
5. Evaluate the circulation of vector field $\mathbf{F}$ around the curve C counterclockwise when viewed from above, where $\mathbf{F} = 2xy\mathbf{i} + x\mathbf{j} + (y + z)\mathbf{k}$ and C is the intersection of the plane $z = 1 - x^2 - y^2$ and $x^2 + y^2 = 1$.
6. Evaluate the circulation of vector field $\mathbf{F}$ around the curve C counterclockwise when viewed from above, where $\mathbf{F} = z\mathbf{i} + y\mathbf{j} + x\mathbf{k}$ and C is the boundary of the part of the plane $x + y + z = 4$ in the first octant.
7. Use Stokes' Theorem to evaluate the line integral $\oint_C \mathbf{F}\cdot d\mathbf{r}$, if $\mathbf{F} = (\sin(x))^2\mathbf{i} + (e^{y^2} + x^2)\mathbf{j}$ $+ (z^4 + 2x^2)\mathbf{k}$ and C is the triangle with the vertices $(3, 0, 0)$, $(0, 2, 0)$, $(0, 0, 1)$.

18.3 Divergence Theorem

(1) The **divergence** of a vector field $\mathbf{F} = M\mathbf{i} + N\mathbf{j} + P\mathbf{k}$ is

$$\operatorname{div}(\mathbf{F}) = \nabla\cdot\mathbf{F} = \frac{\partial M}{\partial x} + \frac{\partial N}{\partial y} + \frac{\partial P}{\partial z}$$

(2) **Divergence Theorem.** The flux of a vector field $\mathbf{F}$ across a closed oriented surface S in the direction of the surface's outward unit normal field $\mathbf{n}$ equals the integral of $\nabla\cdot\mathbf{F}$ over the region D enclosed by the surface:

$$\iint_S \mathbf{F}\cdot\mathbf{n}\,d\sigma = \iiint_D \nabla\cdot\mathbf{F}\,dV$$

Note. In MAPLE, the command "**diverge**" in the "**linalg**" package is used to calculate the divergence of a vector field.

18.3.1 Find divergences

Example 1. Find the divergence of the vector field $\mathbf{F}(x, y, z) = x^2y\mathbf{i} + yz^2\mathbf{j} + zx^2\mathbf{k}$.

```
> F:=[x^2*y,y*z^2,z*x^2]: v:=[x,y,z]:
> diverge(F,v);
```

$$2\,x\,y + z^2 + x^2$$

Example 2. Find the divergence of the vector field $\mathbf{F}(x, y, z) = (x^2 - y)\mathbf{i} + 4z\mathbf{j} + x^2\mathbf{k}$.

```
> F:=[x^2-y,4*z,x^2]; v:=[x,y,z]:
```

$$F := [x^2 - y, 4\,z, x^2] \tag{3.1.1}$$

```
> diverge(F,v);
```

$$2\,x \tag{3.1.2}$$

18.3.2 Evaluate flux by the Divergence Theorem

Example 3. Use the Divergence Theorem to evaluate the outward flux of the vector field $\mathbf{F} = -xz\mathbf{i} - yz\mathbf{j} + z^2\mathbf{k}$ across the surface of the ellipsoidal $\frac{x^2}{a^2} + \frac{y^2}{b^2} + \frac{z^2}{c^2} = 1$.

```
> F:=[-x*z,-y*z,z^2];
```

$$F := [-x\,z, -y\,z, z^2]$$

```
> ifunc:=diverge(F,[x,y,z]);
```

$$ifunc := 0$$

Then the flux is 0.

Example 4. Evaluate the outward flux of vector field **F** across the surface *S*, where $\mathbf{F} = xy\mathbf{i} + (y^2 + e^{xz^2})\mathbf{j} + \sin(xy)\mathbf{k}$ and *S* is the surface of the region bounded by the parabolic cylinder $z = 1 - x^2$ and the planes $z = 0$, $y = 0$, and $y + z = 2$.

```
> F:=[x*y, y^2+exp(x*z^2), sin(x*y)];
```

$$F := [x\,y, y^2 + e^{(x z^2)}, \sin(x\,y)]$$

```
> divf:=diverge(F,[x,y,z]);
```

$$divf := 3\,y$$

```
> plot3d({1-x^2,2-y}, x=-1..1,y=0..2,axes=normal, view=0..1);
```




```
> Flux:=int(int(int(divf,y=0..2-z),z=0..1-x^2),x=-1..1);
```

$$Flux := \frac{184}{35}$$

Example 5. Find the outward flux of vector field **F** across the surface *S*, where $\mathbf{F} = xy\mathbf{i} + yz\mathbf{j} + xz\mathbf{k}$ and *S* is the surface of the cube cut from the first octant by the plane $x = 1, y = 1$, and $z = 1$.

```
> F:=[x*y,y*z,x*z]; v:=[x,y,z]:
```

$$F := [x\,y, y\,z, x\,z] \tag{3.2.1}$$

```
> divf:=diverge(F,v);
```

$$divf := y + z + x \tag{3.2.2}$$

```
> Flux:=int(int(int(divf, x=0..1),y=0..1),z=0..1);
```

$$Flux := \frac{3}{2} \tag{3.2.3}$$

Example 6. Find the outward flux of vector field $\mathbf{F} = \dfrac{x\mathbf{i} + y\mathbf{j} + z\mathbf{k}}{\left(\sqrt{x^2 + y^2 + z^2}\right)^3}$ across the boundary of the region *D*: $0 < a^2 \le x^2 + y^2 + z^2 \le b^2$.

```
> v:=[x,y,z]: vnorm:=sqrt(innerprod(v, v));
```

$$vnorm := \sqrt{x^2 + y^2 + z^2} \tag{3.2.4}$$

```
> F:=map(x->x/vnorm^3, v);
```

$$F := \left[\frac{x}{\left(x^2 + y^2 + z^2\right)^{(3/2)}}, \frac{y}{\left(x^2 + y^2 + z^2\right)^{(3/2)}}, \frac{z}{\left(x^2 + y^2 + z^2\right)^{(3/2)}}\right] \tag{3.2.5}$$

```
> divf:=simplify(diverge(F,v));
```

$$divf := 0 \tag{3.2.6}$$

Note that div(**F**) is discontinuous at the origin, which is not included in the region *D*. Hence, the **flux is** 0.

Exercises

1. Find the divergence of the vector field $\mathbf{F}(x, y, z) = xy\mathbf{i} + yz\mathbf{j} + (y^2 - x^2)\mathbf{k}$.
2. Find the divergence of the vector field $\mathbf{F}(x, y, z) = (x - 2x^2z)\mathbf{i} + (z - xy)\mathbf{j} + z^2x^2\mathbf{k}$.
3. Use the Divergence Theorem to evaluate the outward flux of the vector field $\mathbf{F} = z\mathbf{i} + x\mathbf{j} + y\mathbf{k}$ across the surface of the box [0, 4] × [0, 2] × [0, 3].
4. Evaluate the outward flux of vector field **F** across the surface *S*, where $\mathbf{F} = y\mathbf{i} + z\mathbf{j} + x\mathbf{k}$ and *S* is the sphere $x^2 + y^2 + z^2 = 1$
5. Find the outward flux of vector field **F** across the surface *S*, where $\mathbf{F} = x\mathbf{i} + y^2\mathbf{j} + (y + z)\mathbf{k}$ and *S* is the boundary of the region contained in the cylinder $x^2 + y^2 = 4$ between the planes $x = z$ and $z = 8$.
6. Find the outward flux of vector field $\mathbf{F} = e^{z^2}\mathbf{i} + \sin\left(x^2 z\right)\mathbf{j} + \sqrt{x^2 + 9\,y^2}\,\mathbf{k}$ across the boundary of the region: $x^2 + y^2 \le z \le 8 - x^2 - y^2$.